



ObjectScript の使用法

Version 2024.1
2024-06-03

ObjectScript の使用法

InterSystems IRIS Data Platform Version 2024.1 2024-06-03

Copyright © 2024 InterSystems Corporation

All rights reserved.

InterSystems®, HealthShare Care Community®, HealthShare Unified Care Record®, IntegratedML®, InterSystems Caché®, InterSystems Ensemble®, InterSystems HealthShare®, InterSystems IRIS®, および TrakCare は、InterSystems Corporation の登録商標です。HealthShare® CMS Solution Pack™ HealthShare® Health Connect Cloud™, InterSystems IRIS for Health™, InterSystems Supply Chain Orchestrator™, および InterSystems TotalView™ For Asset Management は、InterSystems Corporation の商標です。TrakCare は、オーストラリアおよび EU における登録商標です。

ここで使われている他の全てのブランドまたは製品名は、各社および各組織の商標または登録商標です。

このドキュメントは、インターシステムズ社(住所: One Memorial Drive, Cambridge, MA 02142)あるいはその子会社が所有する企業秘密および秘密情報を含んでおり、インターシステムズ社の製品を稼動および維持するためにのみ提供される。この発行物のいかなる部分も他の目的のために使用してはならない。また、インターシステムズ社の書面による事前の同意がない限り、本発行物を、いかなる形式、いかなる手段で、その全てまたは一部を、再発行、複製、開示、送付、検索可能なシステムへの保存、あるいは人またはコンピュータ言語への翻訳はしてはならない。

かかるプログラムと関連ドキュメントについて書かれているインターシステムズ社の標準ライセンス契約に記載されている範囲を除き、ここに記載された本ドキュメントとソフトウェアプログラムの複製、使用、廃棄は禁じられている。インターシステムズ社は、ソフトウェアライセンス契約に記載されている事項以外にかかるソフトウェアプログラムに関する説明と保証をするものではない。さらに、かかるソフトウェアに関する、あるいはかかるソフトウェアの使用から起こるいかなる損失、損害に対するインターシステムズ社の責任は、ソフトウェアライセンス契約にある事項に制限される。

前述は、そのコンピュータソフトウェアの使用およびそれによって起こるインターシステムズ社の責任の範囲、制限に関する一般的な概略である。完全な参照情報は、インターシステムズ社の標準ライセンス契約に記載され、そのコピーは要望によって入手することができる。

インターシステムズ社は、本ドキュメントにある誤りに対する責任を放棄する。また、インターシステムズ社は、独自の裁量にて事前通知なしに、本ドキュメントに記載された製品および実行に対する代替と変更を行う権利を有する。

インターシステムズ社の製品に関するサポートやご質問は、以下にお問い合わせください:

InterSystems Worldwide Response Center (WRC)

Tel: +1-617-621-0700

Tel: +44 (0) 844 854 2917

Email: support@InterSystems.com

目次

1 ObjectScript の概要	1
1.1 機能	1
1.2 言語の概要	2
1.3 言語要素の概要	2
1.3.1 文とコマンド	3
1.3.2 関数	3
1.3.3 式	4
1.3.4 変数	4
1.3.5 演算子	5
2 構文規則	7
2.1 左から右の評価順序	7
2.2 大文字と小文字の区別	7
2.2.1 識別子	8
2.2.2 キーワード名	8
2.2.3 クラス名	8
2.2.4 ネームスペース名	8
2.3 Unicode	8
2.3.1 Unicode の文字	9
2.3.2 リストの圧縮	9
2.4 空白	9
2.5 コメント	10
2.5.1 INT コードのルーチンおよびメソッドに使用するコメント	10
2.5.2 MAC コードのルーチンおよびメソッドに使用するコメント	11
2.5.3 メソッド・コードの外部のクラス定義に使用するコメント	11
2.6 文字列リテラル	11
2.7 数値リテラル	13
2.8 識別子	14
2.8.1 識別子の句読点文字	14
2.9 ラベル	14
2.9.1 ラベルの使用法	15
2.9.2 ラベル付けされたコード・セクションの終了	15
2.10 ネームスペース	17
2.10.1 拡張参照	17
2.11 予約語	18
3 データ型とデータ値	19
3.1 文字列	19
3.1.1 NULL 文字列と \$CHAR(0)	19
3.1.2 引用符のエスケープ	20
3.1.3 連結文字列	20
3.1.4 文字列比較	20
3.1.5 ビット文字列	21
3.2 数値	22
3.2.1 数値の基本	22
3.2.2 数値のキャノニック形式	23
3.2.3 数値としての文字列	24
3.2.4 数字の連結	25

3.2.5 浮動小数点数	26
3.2.6 科学的記数法	27
3.2.7 極端に大きな数字	28
3.3 オブジェクト	30
3.4 永続多次元配列 (グローバル)	31
3.5 未定義の値	31
3.6 ブーリアン値	32
3.7 日付	32
4 変数	35
4.1 変数のカテゴリ	35
4.2 ローカル変数	35
4.2.1 名前付け規約	36
4.2.2 ローカル変数の範囲	36
4.3 グローバル	37
4.4 プロセス・プライベート・グローバル	38
4.4.1 名前付け規約	38
4.4.2 プロセス・プライベート・グローバルのリスト	39
4.5 添え字に関する規則	40
4.6 変数のタイプと変換	40
4.7 #dim (オプション)	41
4.8 グローバル変数とジャーナリング	42
4.9 特殊変数	42
5 演算子と式	43
5.1 演算子と式の概要	43
5.1.1 代入	43
5.2 演算子の優先順位	44
5.2.1 単項マイナス演算子	44
5.2.2 括弧と優先順位	44
5.2.3 関数と優先順位	45
5.3 文字列から数値への変換	46
5.3.1 数値文字列	46
5.3.2 非数値文字列	47
5.4 式	47
5.4.1 論理式	49
5.5 算術演算子	50
5.6 数値関係演算子	52
5.7 論理比較演算子	53
5.7.1 論理演算子の優先順位	53
5.7.2 論理演算子	53
5.8 文字列連結演算子 ()	55
5.9 文字列関係演算子	55
5.10 パターン・マッチング演算子 (?)	56
5.11 間接演算子 (@)	57
6 コマンド	59
6.1 コマンド・キーワード	59
6.2 コマンド引数	60
6.2.1 複数の引数	60
6.2.2 パラメータおよび後置条件付きの引数	61
6.2.3 引数なしコマンド	61

6.3 コマンド後置条件式	62
6.3.1 後置条件構文	63
6.3.2 後置条件の評価	63
6.4 単一行での複数コマンド	64
6.5 変数	64
6.6 エラー処理	65
6.7 トランザクション処理	65
6.8 ロックと並行処理の制御	65
6.9 コードの呼び出し	66
6.9.1 DO	66
6.9.2 JOB	67
6.9.3 XECUTE	67
6.9.4 QUIT および RETURN	67
6.10 制御フロー	68
6.10.1 条件付きの実行	68
6.10.2 FOR	69
6.10.3 WHILE と DO WHILE	71
6.11 入出力の制御	71
6.11.1 表示 (書き込み) コマンド	71
6.11.2 READ	74
6.11.3 OPEN、USE、および CLOSE	74
7 呼び出し可能なユーザ定義コードモジュール	75
7.1 プロシージャ、ルーチン、サブルーチン、関数、メソッドの概要	76
7.1.1 ルーチン	76
7.1.2 サブルーチン	77
7.1.3 関数	78
7.2 プロシージャの定義	79
7.2.1 プロシージャの呼び出し	79
7.2.2 プロシージャの構文	80
7.2.3 プロシージャ変数	81
7.2.4 パブリック・プロシージャとプライベート・プロシージャ	83
7.3 パラメータ渡し	84
7.3.1 値渡し	85
7.3.2 参照渡し	85
7.3.3 可変個数のパラメータ	86
7.4 プロシージャ・コード	88
7.5 プロシージャ範囲内の間接指定、XECUTE コマンド、および JOB コマンド	90
7.6 プロシージャ内でのエラー・トラップ	90
7.7 従来のユーザ定義コード	91
7.7.1 サブルーチン	91
7.7.2 関数	92
8 マクロとインクルード・ファイルの使用	97
8.1 マクロの基本情報	97
8.2 インクルード・ファイルの基本情報	98
8.3 マクロの定義	99
8.3.1 マクロを定義する場所	99
8.3.2 許可されるマクロ定義	99
8.3.3 マクロの名前付け規約	100
8.3.4 マクロの空白規約	100
8.3.5 マクロ・コメントとスタジオ・アシスト	101

8.4 インクルード・ファイルを含める方法	101
8.5 マクロが拡張される条件	102
8.6 関連項目	102
9 埋め込み SQL	105
9.1 埋め込み SQL	105
9.2 その他の形式のクエリ	105
10 多次元配列	107
10.1 多次元配列の概要	107
10.1.1 多次元ツリー構造	107
10.1.2 スパース多次元ストレージ	108
10.1.3 多次元配列の種類	108
10.2 多次元配列の操作	108
10.3 関連項目	109
11 文字列演算	111
11.1 基本的な文字列演算と関数	111
11.1.1 \$EXTRACT の高度な機能	112
11.2 区切り文字列演算	113
11.2.1 高度な \$PIECE 関数	114
11.3 リスト構造文字列演算	114
11.3.1 スパース・リストおよびサブリスト	116
11.4 リストと区切り文字列の比較	116
11.4.1 リストの利点	116
11.4.2 区切り文字列の利点	116
12 ロックと並行処理の制御	119
12.1 概要	119
12.2 ロック名	119
12.3 ロック・テーブル	120
12.4 ロックと配列	120
12.5 LOCK コマンドの使用法	121
12.5.1 増分ロックの追加	121
12.5.2 タイムアウト付き増分ロックの追加	122
12.5.3 ロックの削除	122
12.5.4 LOCK コマンドの他の基本的なバリエーション	123
12.6 ロック・タイプ	123
12.6.1 排他ロックと共有ロック	124
12.6.2 非エスカレート・ロックとエスカレート・ロック	124
12.6.3 ロック・タイプのまとめ	124
12.7 エスカレート・ロック	125
12.7.1 ロックのエスカレーションの例	125
12.7.2 エスカレート・ロックの削除	127
12.8 ロック、グローバル、およびネームスペース	127
12.8.1 シナリオ 1 : 同じグローバル・データベースを持つ複数のネームスペース	128
12.8.2 シナリオ 2 : ネームスペースがマップされたグローバルを使用	129
12.8.3 シナリオ 3 : ネームスペースがマップされたグローバル添え字を使用	129
12.8.4 シナリオ 4 : 拡張グローバル参照	130
12.9 デッドロックの回避	131
12.10 ロックの実際の使用方法	131
12.10.1 アプリケーション・データへのアクセスの制御	132
12.10.2 同時アクティビティの防止	132

12.11 SQL と永続クラスにおけるロックと並行処理	133
12.12 関連項目	133
13 ロック要求とデッドロックの詳細	135
13.1 待機ロック要求	135
13.2 配列ノードに対するロック要求のキュー	136
13.3 ECP ローカルおよびリモート・ロック要求	137
13.4 デッドロックの回避	137
13.5 関連項目	138
14 ロック・テーブルの管理	139
14.1 ロック・テーブルの管理のための使用可能ツール	139
14.2 管理ポータルでのロックの表示	139
14.3 管理ポータルでのロックの削除	142
14.4 ^LOCKTAB ユーティリティ	142
14.5 関連項目	143
15 トランザクション処理	145
15.1 InterSystems IRIS のトランザクションについて	145
15.2 アプリケーションでのトランザクション管理	145
15.2.1 トランザクション・コマンド	146
15.2.2 トランザクションでの LOCK の使用	146
15.2.3 トランザクションでの \$INCREMENT と \$SEQUENCE の使用法	147
15.2.4 アプリケーション内でのトランザクション・ロールバック	147
15.2.5 アプリケーション内のトランザクション処理の例	148
15.3 自動トランザクション・ロールバック	149
15.4 トランザクション処理に関するシステム全体の問題	149
15.4.1 トランザクション処理でのバックアップとジャーナリング	149
15.4.2 非同期エラーの通知	150
15.5 現在のすべてのトランザクションの一時停止	150
16 %Status 値の操作	153
16.1 ステータス値の操作の基本	153
16.2 例	154
16.3 バリエーション (%objlasterror)	154
16.4 1 つのステータス値で報告される複数のエラー	154
16.5 %Status を返す	155
16.6 %SYSTEM.Error	156
16.7 関連項目	156
17 TRY-CATCH の使用法	157
17.1 概要	157
17.2 TRY-CATCH と一緒に THROW を使用する	158
17.3 \$\$\$ThrowOnError および \$\$\$ThrowStatus マクロの使用法	159
17.4 %Exception.SystemException と %Exception.AbstractException クラスの使用	160
17.5 TRY-CATCH を使用する際の他の考慮事項	160
17.5.1 TRY-CATCH ブロック内での QUIT	160
17.5.2 TRY-CATCH と実行スタック	161
17.5.3 従来のエラー処理で TRY-CATCH を使用する	161
18 エラー・ログ	163
18.1 アプリケーション・エラーのログ作成	163
18.2 管理ポータルを使用したアプリケーション・エラー・ログの表示	163
18.3 ^%ERN を使用したアプリケーション・エラー・ログの表示	164

18.4 関連項目	165
19 コマンド行ルーチンのデバッグ	167
19.1 保護されたデバッグ・シェル	167
19.1.1 制限されるコマンドと関数	168
19.2 ObjectScript デバッガによるデバッグ	169
19.2.1 ブレークポイントとウォッチポイントの使用法	170
19.2.2 ブレークポイントとウォッチポイントの設定	170
19.2.3 ブレークポイントとウォッチポイントを無効にする	174
19.2.4 ブレークポイントとウォッチポイントの実行を遅らせる	175
19.2.5 ブレークポイントとウォッチポイントの削除	175
19.2.6 シングル・ステップ・ブレークポイントの動作	175
19.2.7 実行のトレース	176
19.2.8 割り込みキーと Break コマンド	177
19.2.9 現在のデバッグ環境情報の表示	178
19.2.10 デバッグ・デバイスの使用法	179
19.2.11 ObjectScript デバッガの例	180
19.2.12 ObjectScript デバッガ・エラーの理解	181
19.3 BREAK コマンドによるデバッグ	181
19.3.1 ルーチンの実行を中断する引数なし BREAK の使用法	182
19.3.2 ルーチンの実行を中断する引数付き BREAK の使用法	182
19.3.3 ターミナル・プロンプトのプログラム・スタック情報表示	183
19.3.4 FOR ループおよび WHILE ループ	184
19.3.5 BREAK あるいはエラー後の実行の再開	185
19.3.6 ターミナル・プロンプトで使用する NEW コマンド	186
19.3.7 ターミナル・プロンプトで使用する QUIT コマンド	186
19.3.8 InterSystems IRIS エラー・メッセージ	187
19.4 スタックを表示する %STACK の使用法	187
19.4.1 %STACK の実行	187
19.4.2 プロセス実行スタックの表示	188
19.4.3 スタックの表示の理解	189
19.5 その他のデバッグ・ツール	193
19.5.1 \$SYSTEM.OBJ.ShowReferences によるオブジェクトへの参照の表示	193
19.5.2 エラー・トラップ・ユーティリティ	193
付録A: (従来) ^%ETN を使用したエラー・ログ	197
付録B: (従来) 従来のエラー処理	199
B.1 従来のエラー処理の概要	199
B.1.1 内部的なエラー・トラップの振る舞い	199
B.1.2 現在のコンテキスト・レベル	200
B.1.3 エラー・コード	201
B.2 \$ZTRAP でのエラー処理	203
B.2.1 プロシージャでの \$ZTRAP の設定	203
B.2.2 ルーチンでの \$ZTRAP の設定	203
B.2.3 \$ZTRAP コードの記述	204
B.2.4 \$ZTRAP の使用法	204
B.2.5 エラー・トラップによる New コマンドの削除	205
B.2.6 \$ZTRAP フロー制御オプション	205
B.3 \$ETRAP でのエラー処理	206
B.3.1 \$ETRAP エラー・ハンドラ	207
B.3.2 コンテキスト固有の \$ETRAP エラー・ハンドラ	207

B.3.3 \$ETRAP フロー制御オプション	208
B.4 エラー・ハンドラによるエラー処理	209
B.4.1 \$ZTRAP エラー・ハンドラでのエラー	210
B.4.2 \$ETRAP エラー・ハンドラでのエラー	210
B.4.3 \$ZERROR 特殊変数と \$ECODE 特殊変数のエラー情報	210
B.5 強制エラー	210
B.5.1 \$ECODE の設定	211
B.5.2 アプリケーション固有エラーの生成	211
B.6 ターミナル・プロンプトでのエラー処理	211
B.6.1 エラー・メッセージ形式の理解	211
B.6.2 ターミナル・プロンプトの理解	212
B.6.3 エラーのリカバリ	212

図一覧

図 II-1: コール・スタックのフレーム	200
図 II-2: \$ZTRAP エラー・ハンドラ	206
図 II-3: \$ETRAP エラー・ハンドラ	208

テーブル一覧

テーブル 3-1: 日付形式	33
テーブル 4-1: ObjectScript の型の変換規則	41
テーブル 6-1: 表示形式	72
テーブル 6-2: 値の表示方法	73
テーブル 15-1: トランザクション・コマンド	146
テーブル 19-1: ターミナル・プロンプトのスタック・エラー・コード	184
テーブル 19-2: %STACK ユーティリティ情報	189
テーブル 19-3: フレーム・タイプと使用可能な値	189

1

ObjectScript の概要

ObjectScript は、InterSystems IRIS® データ・プラットフォームに組み込まれた、完全に汎用的なプログラミング言語です。ObjectScript のソース・コードは、InterSystems IRIS 仮想マシン内で実行されるオブジェクト・コードにコンパイルされます。このオブジェクト・コードは、文字列操作やデータベース・アクセスなど、ビジネス・アプリケーションで一般的な処理を実行するために、高度に最適化されています。ObjectScript プログラムは、InterSystems IRIS がサポートするすべてのプラットフォーム間で完全に移植可能です。

以下のいずれのコンテキストでも、ObjectScript を使用できます。

- ・ [InterSystems IRIS クラス](#)のメソッドの実装言語として使用 (クラス定義は正式には ObjectScript の一部ではありません。むしろ、クラス定義の特定の部分内で ObjectScript を使用できます)。
- ・ [InterSystems SQL](#) のストアド・プロシージャとトリガに対する実装言語として使用
- ・ InterSystems IRIS 内に含まれ実行される個別プログラムである ObjectScript [ルーチン](#)を生成するために使用
- ・ ターミナルのコマンド行からインタラクティブに使用

ObjectScript を詳細に学習するには、以下も参照してください。

- ・ [ObjectScript チュートリアル](#) – 大半の言語要素に対しインタラクティブな概要について
- ・ [ObjectScript リファレンス](#) – 個別のコマンドと関数について

重要 ObjectScript で演算子の評価順序は、必ず左から右です。したがって、式の演算は表示された順番で実行されます。式で明示的に小括弧を使用して、特定の演算子を先に処理させることができます。

1.1 機能

以下は、ObjectScript の主な機能です。

- ・ [文字列](#)と連動する強力な組み込み関数
- ・ メソッド、プロパティ、多態を含む[オブジェクト](#)のネイティブ・サポート
- ・ アプリケーションでコントロール・フローを管理する[コマンド](#)
- ・ 並行処理の制御のサポート
- ・ 入出力デバイスを処理するコマンド・セット
- ・ ローカルと[グローバル](#) (永続) の両方で多次元なスパース配列 (要素番号が連続しない配列) をサポート

- ・ 効率的な埋め込み SQL のサポート
- ・ 間接指定、実行時の評価とコマンド実行のサポート

1.2 言語の概要

ObjectScript は予約語が定義されていないため、任意の単語を識別子 (変数名など) として自由に使用できます。このために、ObjectScript は、組み込みのコマンド一式と特殊文字 (関数名の “\$” 接頭語など) を使用して、他の言語要素から識別子を区別します。

例えば、変数に値を割り当てる場合、SET コマンドを使用できます。

ObjectScript

```
SET x = 100
WRITE x
```

ObjectScript では、以下のプログラムに示すように、任意の有効な名前を識別子名として使用できます (推奨ではありません)。このプログラムと上記のプログラムは、機能的に同じです。

ObjectScript

```
SET SET = 100
WRITE SET
```

ObjectScript のコンポーネントには、コマンド名や関数名のように、大文字と小文字が区別されないものがあります。また、変数名、ラベル、クラス名、およびメソッド名のように、大文字と小文字が区別される ObjectScript のコンポーネントもあります。詳細は、“[大文字と小文字の区別](#)” を参照してください。

ObjectScript のほとんどの場所で、空白を挿入または省略できます。ただし、空白の使用法のうち 2 つには重要な意味があります。

1. コマンドとその引数は少なくとも 1 つのスペースで区切る必要があります。
2. 各コマンド行は少なくとも 1 つのスペースでインデントする必要があります。コマンドを行の最初の文字位置から開始したり、続行することはできません。

また、コメントもインデントする必要があります。ただし、[ラベル](#)は行の最初の文字位置で開始する必要があります。マクロ・プリプロセッサ文など、他の一部の構文は、行の最初の文字位置で開始できます。詳細は、“[空白](#)” を参照してください。

ObjectScript は、コマンド・ターミネータ文字や行ターミネータ文字を使用しません。

1.3 言語要素の概要

一番単純な ObjectScript 構文は、以下のような形式で、式でコマンドを呼び出します。

ObjectScript

```
WRITE x
```

これは、変数 x に WRITE コマンドを実行します (これにより、x の値を表示します)。上記の例で、x は式です。ObjectScript 式は、値を算出するために評価される 1 つ以上の “トークン” です。それぞれのトークンは、リテラル、変数、1 つ以上

の演算の結果 (2 つの数字の加算の合計など)、関数の評価からの戻り値、あるいはこれらのうちの組み合わせなどになります。文の有効な構文には、その[コマンド](#)、[関数](#)、[式](#)、および[演算子](#)が関係しています。

1.3.1 文とコマンド

ObjectScript プログラムは、多くの文から構成されています。それぞれの文が、プログラムで実行する特定の動作を定義します。文には、コマンドと引数があります。

以下に ObjectScript 文があります。

ObjectScript

```
SET x="World"
WRITE "Hello",!,x
```

WRITE はコマンドです。これは、その名前のとおり、引数として指定された内容を現在の主出力デバイスに書き込みます。この場合、WRITE は、リテラル文字列 “Hello”、“!” 文字 (改行またはキャリッジ・リターンを発行する WRITE コマンドに固有の記号演算子)、および実行時に最新の値に置き換えられるローカル変数 x の 3 つの引数を書き込みます。引数はコンマで区切られ、引数と引数の間に空白を挿入することもできます (いくつかの制限があります)。空白については、本ドキュメントの[構文規則](#)の章で説明しています。

ほとんどの ObjectScript のコマンド (および関数や特殊変数) には、長い形式と短い (省略された) 形式 (通常は 1、2 文字) があります。例えば、次のプログラムは前述のプログラムと同じものですが、コマンド名が省略された形で使用されています。

ObjectScript

```
S x="World"
W "Hello",!,x
```

短い形式のコマンド名は、長いコマンド名の入力を好まない開発者が使用する 1 つの手段です。長い形式とまったく同じコマンドを示します。このドキュメントでは、長い形式のコマンド名を使用します。完全なリストは、[ObjectScript で使用する省略形](#)を参照してください。

コマンドの詳細は、[コマンド](#)、あるいは[ObjectScript リファレンス](#)のそれぞれのリファレンス・ページを参照してください。

1.3.2 関数

関数とは、処理 (例えば、文字列をその等価の ASCII コード値に変換するなど) を実行して値を返すコードのことです。関数は、コマンド行内で呼び出されます。この呼び出しにより関数にパラメータ値が渡され、このパラメータ値を使用して処理を実行します。次に関数は単一の値 (結果) を呼び出し元コマンドに返します。式を使用できる場所ならどこでも関数を使用できます。

InterSystems IRIS では、大量のシステム指定の関数 (“内部” 関数とも呼ばれる) を提供しており、これらを変更することはできません。これらの関数は、1 つのドル記号 (“\$”) で始まり、パラメータが括弧で囲まれているため、識別可能です。パラメータが指定されない場合でも、括弧は必須です (特殊変数名も単一のドル記号で始まりますが、括弧は含まれません)。

多くのシステム指定の関数名には、省略形があります。このドキュメントでは、完全な関数名が使用されます。省略形は関数のリファレンス・ページに示され、[ObjectScript で使用する省略形](#)には完全なリストが用意されています。

関数は常に値を返します。通常、戻り値はコマンドに渡され、SET namelen=\$LENGTH("Fred Flintstone") または WRITE \$LENGTH("Fred Flintstone") のようになるか、あるいは別の関数に指定されて WRITE \$LENGTH(\$PIECE("Flintstone^Fred", "^", 1)) のようになります。戻り値の受取先を指定しないと、通常 <SYNTAX> エラーが発生します。ただし、一部の関数では、戻り値の受取先を指定することは必須ではありません。関数の実行によって行われる処理 (ポインタの移動など)、または関数のパラメータの 1 つを設定することは、関連する処理です。この

ような場合、DO または JOB コマンドを使用して、戻り値を受け取らずに関数を呼び出すことができます。例えば、DO \$CLASSMETHOD(clname,clmethodname,singlearg) のように指定します。

関数には、1 つのパラメータまたは複数のパラメータを指定することも、パラメータを指定しないこともできます。関数のパラメータは定位置にあり、コンマで区切られます。多くのパラメータはオプションです。パラメータを省略する場合、InterSystems IRIS はそのパラメータの既定値を使用します。パラメータは定位置のため、指定されたパラメータのリスト内でパラメータを通常省略することはできません。場合によっては (\$LISTTOSTRING など)、パラメータのリスト内でパラメータを省略し、プレースホルダのコンマを指定できます。オプションのパラメータについては、最後の指定されたパラメータの右側にプレースホルダのコンマを指定する必要はありません。

大半の関数は、同じパラメータを繰り返して指定できません。しかし、\$CASE、\$CHAR、\$SELECT は例外として使用できます。

一般的に、パラメータはリテラル、変数、または別の関数の戻り値として指定できます。パラメータをリテラルとして指定しなければいけない場合もあります。ほとんどの場合、関数パラメータとして指定するには、変数をまず定義する必要があります。そうしないと、〈UNDEFINED〉エラーが発生します。一部のケースでは (\$DATA など)、パラメータ変数を定義しなくてもよい場合があります。

一般的に、関数パラメータは関数に値を指定する入力パラメータです。関数は入力パラメータとして指定された変数の値を変更しません。関数は値を返すと共に出力パラメータを設定する場合もあります。例えば、\$LISTDATA は、指定された位置にリスト要素があるかどうかを示すブーリアン値を返します。また、(オプションで) リスト要素の値に対して 3 番目のパラメータを設定します。

すべての関数は SET コマンド (SET x=\$LENGTH(y) など) の右側で指定できます。一部の関数は SET コマンドの左側でも指定できます (SET \$LIST(list,position,end)=x など)。SET の左側で指定可能な関数は、参照ページの構文ブロックでそのように識別されます。

システム指定の関数は、InterSystems IRIS の一部として提供されます。“ObjectScript ランゲージ・リファレンス”では、システム指定の関数のそれぞれについて説明しています。クラスで提供される関数はメソッドと呼ばれます。InterSystems IRIS で提供されるメソッドについては、“インターシステムズ・クラス・リファレンス”で説明されています。

システム関数の他に、ObjectScript は ユーザ定義関数 (“外部” 関数ともいいます) もサポートします。また、ユーザ定義関数の定義と呼び出しの詳細は、“ユーザ定義コード”を参照してください。

1.3.3 式

式は、1 つの値を算出するために評価されるトークン式です。例えば、リテラル文字列 “hello” は式です。また 1 + 2 も式です。x などの変数、\$LENGTH などの関数、\$ZVERSION などの特殊変数も式として評価されます。

プログラムでは、式をコマンドや関数の引数として使用します。

ObjectScript

```
SET x = "Hello"
WRITE x,!
WRITE 1 + 2,!
WRITE $LENGTH(x),!
WRITE $ZVERSION
```

1.3.4 変数

ObjectScript で変数は、実行時の値が格納される場所の名前です。変数は、SET コマンドなどで定義する必要がありますが、決まった形式は必要ありません。ObjectScript の変数に決まった形式はありません。つまり、変数にはデータ型が割り当てられていないため、任意のデータ値を取ることができます (互換性を維持するために、\$DOUBLE 関数を使用して、形式の決まっていない浮動小数点数を特定の数値データ型形式に変換できます)。

ObjectScript は、さまざまな種類の変数をサポートします。

- ・ ローカル変数 – 変数を作成した InterSystems IRIS プロセスからのみアクセス可能でプロセス終了時に自動的に削除される変数。ローカル変数にはどのネームスペースからでもアクセスできます。
- ・ プロセス・プライベート・グローバル – InterSystems IRIS プロセスからのみアクセス可能でプロセス終了時に削除される変数。プロセス・プライベート・グローバルには、どのネームスペースからでもアクセスできます。プロセス・プライベート・グローバルは、サイズの大きいデータ値を一時的に格納する場合に特に便利です。
- ・ グローバル – InterSystems IRIS データベースに格納される永続変数。グローバルは、すべてのプロセスからアクセス可能であり、そのグローバルを作成したプロセスが終了しても存続します。グローバルは、各ネームスペースに固有です。
- ・ 配列変数 – 1 つあるいはそれ以上の添え字を持つ変数。すべてのユーザ定義変数は、ローカル変数、プロセス・プライベート・グローバル、グローバル、オブジェクト・プロパティを含め配列として使用されます。
- ・ 特殊変数 (システム変数とも呼ばれる) – InterSystems IRIS 操作環境における特定の状況の値を含む組み込み変数一式の 1 つ。すべての特殊変数は定義されます。InterSystems IRIS では、すべての特殊変数を初期値 (場合によっては NULL 文字列値) に設定します。特殊変数にはユーザが設定できるものと、InterSystems IRIS でのみ設定できるものがあります。特殊変数は配列変数ではありません。
- ・ オブジェクト・プロパティ – オブジェクトの特定のインスタンスに関連し、格納される値。

ObjectScript は、変数に対する、あるいは変数間でのさまざまな処理をサポートします。変数については、“[変数](#)”を参照してください。

1.3.5 演算子

ObjectScript は、多くの組み込み演算子を定義します。演算子には、算術演算子、論理演算子、パターン・マッチ演算子などがあります。詳細は“[演算子と式](#)”を参照してください。

2

構文規則

ここでは、ObjectScript 構文の基本的な規則について説明します。

2.1 左から右の評価順序

ObjectScript で演算子の評価順序は、必ず左から右です。したがって、式の演算は表示された順番で実行されます。これは、特定の演算子の優先順位が他の演算子よりも高くなることがある他の言語と異なります。詳細は、“[演算子の優先順位](#)”を参照してください。

2.2 大文字と小文字の区別

ObjectScript には、大文字と小文字を区別する場合とそうでない場合があります。通常、ObjectScript のユーザ定義部は、大文字と小文字を区別しますが、キーワードは区別しません。

- ・ 大文字と小文字の区別あり:変数名 (ローカル、グローバル、およびプロセス・プライベート・グローバル) および変数の添え字、クラス名、メソッド名、プロパティ名、プロパティのインスタンス変数の i% 序文、ルーチン名、マクロ名、マクロ・インクルード・ファイル (.inc ファイル) 名、ラベル名、ロック名、パスワード、埋め込みコード指示文マーカ文字列、埋め込み SQL ホスト変数名。
- ・ 大文字と小文字を区別しない:コマンド名、関数名、特殊変数名、ネームスペース名 (下記参照)、ユーザ名およびロール名、プリプロセッサ指示文 (#include など)、文字コード (LOCK、OPEN、または USE)、キーワード・コード (\$STACK)、パターン・マッチ・コード、埋め込みコード指示文 (&html、&js、&sql)。%ZLANG ルーチンをカスタマイズすることによって追加するカスタム言語要素は大文字と小文字を区別しませんが、作成するときは大文字を使用する必要があります。これらの言語要素を参照するときは、大文字でも小文字でもかまいません。テキスト分析のインデックス作成では、小文字に変換することでテキストを正規化しているので、ドメイン名も含め、大半の NLP 値は大文字と小文字を区別しません。
- ・ 通常は大文字と小文字を区別しない:デバイス名、ファイル名、ディレクトリ名、ディスク・ドライブ名で、大文字と小文字を区別するかどうかはプラットフォームによる。指数記号は、通常大文字と小文字を区別しません。大文字の “E” は、常に有効な指数記号です。小文字の “e” は、%SYSTEM.Process の ScientificNotation() メソッドを使用している現在のプロセス、または Config.Miscellaneous クラスの ScientificNotation プロパティを使用しているシステム全体に対して、有効または無効に構成できます。

2.2.1 識別子

ユーザ定義の **識別子** (変数、ルーチン名、ラベル名) は、大文字と小文字を区別します。String、string、STRING はすべて異なる変数を指しています。グローバル変数名は、ユーザが定義する場合も、あるいはシステムが提供する場合も大文字と小文字を区別します。

注釈 これに対して、**SQL 識別子**では大文字と小文字を区別しません。

2.2.2 キーワード名

コマンド、関数、システム変数キーワード (およびその省略形) は、大文字と小文字を区別しません。例えば、Write、write、WRITE、W、w は同じ Write コマンドを示しています。

2.2.3 クラス名

クラスに関連するすべての識別子 (クラス名、プロパティ名、メソッド名など) は大文字と小文字を区別します。しかし、一意性を確保するためには、このような名前では大文字と小文字は区別されないと見なします。つまり、2 つのクラス名は大文字小文字だけでは区別できません。

2.2.4 ネームスペース名

ネームスペース名は、大文字と小文字を区別しません。つまり、ネームスペース名は、大文字と小文字を好きなように組み合わせる入力することができます。ただし、InterSystems IRIS® データ・プラットフォームは常に、ネームスペース名を大文字で格納することに注意してください。したがって、InterSystems IRIS は、ユーザが指定した小文字ではなく、大文字でネームスペース名を返す場合があります。ネームスペース名の名前付け規約の詳細は、“**ネームスペース**”を参照してください。

2.3 Unicode

InterSystems IRIS は、Unicode 国際文字セットをサポートします。Unicode 文字は 16 ビット文字であり、ワイド文字と呼ばれることもあります。**\$ZVERSION** 特殊変数 (Build nnnU) および **\$SYSTEM.Version.IsUnicode()** メソッドは、InterSystems IRIS インストールで Unicode がサポートされていることを示すものです。

ほとんどの用途に、InterSystems IRIS は Unicode 基本多言語面 (16 進数 0000 から FFFF) のみをサポートします。これには、最も一般的に使用される国際文字が含まれています。内部的に、InterSystems IRIS では UCS-2 エンコーディングを使用します。これは基本多言語面の場合、UTF-16 と同じです。**\$WCHAR**、**\$WISWIDE**、および関連する関数を使用することで、Unicode 基本多言語面に存在しない文字で作業できます。

InterSystems IRIS は標準の UTF-16 エンコードと同様に、1 文字あたり 16 ビット (2 バイト) を割り当てることで、Unicode 文字列をメモリにエンコードします。ただし、Unicode 文字列をグローバルに保存する際に、すべての文字の数値が 255 以下の場合、InterSystems IRIS は 1 文字あたり 8 ビット (1 バイト) を使用して文字列を格納します。文字列に 255 より大きな数値の文字が含まれる場合、InterSystems IRIS は圧縮アルゴリズムを適用して、文字列がストレージ内で占有する領域を減らします。

Unicode と UTF-8 間の変換、および他の文字エンコーディングへの変換については、**\$ZCONVERT** 関数を参照してください。**ZZDUMP** を使用して、文字列の 16 進エンコーディングを表示できます。**\$CHAR** を使用して、文字 (または文字列) をその 10 進数 (10 進法) エンコーディングで指定することができます。**\$ZHEX** を使用して、16 進数を 10 進数に、または 10 進数を 16 進数に変換することができます。

2.3.1 Unicode の文字

InterSystems IRIS では、Unicode 文字を使用できる名前もありますが、Unicode 文字を含めることができない名前もあります。Unicode 文字は、255 より大きな 10 進文字コード値のアルファベット文字として定義されます。例えば、小文字のギリシャ文字ラムダは Unicode 文字で \$CHAR(955) になります。

InterSystems IRIS では、以下の例外を除き、どのような場合でも Unicode 文字を使用できます。

- ・ 変数名:ローカル変数名には Unicode 文字を使用できます。一方、[グローバル変数名とプロセス・プライベート・グローバル変数名](#)には、Unicode 文字を使用できません。あらゆるタイプの変数の添え字が Unicode 文字で指定できます。
- ・ データベースを暗号化する際に管理者が使用するユーザ名とパスワードには Unicode 文字を使用できません。

Unicode 文字を処理する際に、ローカル識別子は考慮されません。つまり、Unicode 文字で構成される識別子が、あるローカルで有効な場合、その識別子はどのローカルでも有効です。前述の例外は引き続き適用されます。

注釈 日本語ローカルでは、InterSystems IRIS の名前におけるアクセント記号付きラテン文字がサポートされていません。日本語の名前には、(日本語の文字の他に) ラテン文字 A-Z と a-z (65-90 と 97-122)、およびギリシャ語の大文字 (913-929 と 931-937) を使用できます。

2.3.2 リストの圧縮

ListFormat は、\$LIST のエンコードされた文字列に格納する際、Unicode 文字列を圧縮するかどうかを制御します。既定では、圧縮されません。圧縮形式は、InterSystems IRIS によって自動的に処理されます。圧縮形式がサポートされているかどうかを確認せずに、Java や C# などの外部クライアントに圧縮されたリストを渡さないでください。

%SYSTEM.Process クラスの ListFormat() メソッドを使用すると、プロセスごとの動作を制御できます。

Config.Miscellaneous クラスの ListFormat プロパティを設定するか、InterSystems IRIS の管理ポータルで **[システム管理]** から **[構成]**、**[追加の設定]**、**[互換性]** を選択して、システム全体の既定の動作を設定できます。

2.4 空白

特定の環境では、ObjectScript は構文的に意味のあるものとして空白を処理します。指定がない限り、空白はブランク、タブ、改行と同じ意味を示します。以下はその規則の概要です。

- ・ コードの各行または 1 行コメントの先頭には空白を記述する必要があります。以下の場合には先頭の空白が不要となります。
 - ラベル (タグまたはエントリ・ポイントとも呼ばれる): ラベルは直前の空白文字なしで列 1 にて記述する必要があります。行にラベルがある場合、そのラベルと同一行上の他のコードやコメントの間には空白が必要です。ラベルがパラメータ・リストを有する場合、ラベル名とパラメータ・リストの最初の括弧の間にある空白をなくすこともできます。パラメータ・リスト内のパラメータの前、間、または後には空白を入れることができます。
 - マクロ指示文: #define などのマクロ指示文は直前の空白文字なしで列 1 に記述できます。これは推奨の規約となりますが、マクロ指示文の前の空白は認められています。
 - 複数行コメント: 複数行コメントの最初の行の前には 1 つ以上のスペースを配置する必要があります。複数行コメントの 2 行目以降には、先頭に空白を置く必要はありません。
 - 空白行: 行に文字がない場合は、スペースを含める必要はありません。空白文字のみで構成される行も認められており、コメントとして扱われます。

- ・ コマンドと最初の引数の間には、必ずスペース文字を 1 つだけ記述する必要があります (タブ文字ではありません)。コマンドで後置条件を使用する場合、コマンドとその後置条件との間にはスペース文字を記述しません。
- ・ 後置条件式にスペースがある場合、式全体を括弧で囲む必要があります。
- ・ 組になったコマンド引数の間には、いくつもの空白を置くことができます。
- ・ 1 行にコードと 1 行のコメントが含まれる場合は、その間に空白を置く必要があります。
- ・ 通常、それぞれのコマンドは独自の行に記述されますが、同じ行に複数のコマンドを入力することができます。この場合、これらのコマンドの間には空白が必要です。引数がないコマンドの後には、空白を 2 つ (スペース文字を 2 つ、タブ文字を 2 つ、またはスペース文字とタブ文字をそれぞれ 1 つずつ) 記述する必要があります。これら必須の空白の後に、さらに空白を記述してもかまいません。

2.5 コメント

コードでコメントを使用してインラインのドキュメントを提供するのは適切な方法です。コメントは、コードの変更またはメンテナンス時に大切な情報源になります。ObjectScript では、いくつかの種類のコメントをサポートしており、さまざまな場所で使用できます。

- ・ [INT コードのルーチンおよびメソッドに使用するコメント](#)
- ・ [MAC コードのルーチンおよびメソッドに使用するコメント](#)
- ・ [メソッド・コード以外のクラス定義に使用するコメント](#)

2.5.1 INT コードのルーチンおよびメソッドに使用するコメント

ObjectScript コードは MAC コードとして記述され、その MAC コードから INT (中間) コードが生成されます。MAC コードに記述したコメントは通常、対応する INT コードで利用できます。[ZLOAD](#) コマンドを使用して INT コード・ルーチンをロードしてから、[ZPRINT](#) コマンドまたは [\\$TEXT](#) 関数を使用して、以下のコメントを含む INT コードを表示できます。以下のタイプのコメントを使用できます。これらのコメントはすべて 2 列目以降から開始する必要があります。

- ・ `/* */` 複数行コメントは、1 行または複数行にわたって記述できます。`/*` は、行の最初の要素として、または他の要素の後に配置できます。`*/` は、行の最後の要素として、または他の要素の前に配置できます。`/* */` の間のすべての行が INT コードに表示されます。これには、`/*` または `*/` のみで構成される行も含まれますが、完全に空白の行は含まれません。複数行コメント内の空白行は INT コードから省略されるため、行数に影響する可能性があります。
- ・ `//` コメントは、行の残りの部分がコメントであることを示します。このコメントは、行の最初の要素として、または他の要素の後に配置できます。
- ・ `;` コメントは、行の残りの部分がコメントであることを示します。このコメントは、行の最初の要素として、または他の要素の後に配置できます。
- ・ `;` コメントの特殊なケースである `;;` コメントは、ルーチンがオブジェクト・コードとして配信される場合にのみ、[\\$TEXT](#) 関数で使用できます。行内でこのコメントの前にコマンドがない場合に、[\\$TEXT](#) 関数でのみ使用可能です。

注釈 InterSystems IRIS は、`;;` コメントをオブジェクト・コード (実際に解釈され、実行されるコード) に保持するため、このコメントを挿入するとパフォーマンスに影響します。そのため、ループ内には配置しないでください。

複数行コメント (`/* comment */`) はコンマ区切り文字前後の、コマンドもしくは関数の引数の間に置くことができます。複数行コメントを引数内に置いたり、コマンドのキーワードと最初の引数の間、もしくは関数キーワードと最初の括弧の間

に置くことはできません。同一行で複数行コメントを 2 つのコマンドの間に置くことはできます。その場合、コマンドを区切るために必要な 1 つの空白として機能します。同一行で複数行コメント (*) の直後にコマンドを続けたり、同一行で単数行コメントを続けることもできます。行中に `/* comment */` を挿入する例を以下に示します。

ObjectScript

```
WRITE $PIECE("Fred&Ginger")/* WRITE "world" */, "&", 2), !
WRITE "hello", /* WRITE "world" */ " sailor", !
SET x="Fred"/* WRITE "world" */WRITE x, !
WRITE "hello"/* WRITE "world" */// WRITE " sailor"
```

2.5.2 MAC コードのルーチンおよびメソッドに使用するコメント

以下のコメント・タイプは MAC コードに記述できますが、対応する INT コードでは別の動作となります。

- ・ `#;` コメントは、任意の列から開始できますが、その行の最初の要素でなければなりません。`#;` コメントは、INT コードには表示されません。コメントもコメント・マーカ (`#;`) も INT コードには表示されず、空白行も保持されません。このため、`#;` コメントによって、INT コードの行番号が変わる可能性があります。
- ・ `##;` コメントは任意の列から開始できます。また、その行で最初の要素になることができ、他の要素に続けることもできます。`##;` コメントは、INT コードには表示されません。`##;` は、ObjectScript コード内や埋め込み SQL コード内、あるいは `#define`、`#deflarg`、または `##continue` の各マクロ・プリプロセッサ指示文と同一行に使用できます。
`##;` コメントが列 1 から開始される場合、コメントもコメント・マーカ (`##;`) も INT コードには表示されず、空白行も保持されません。ただし、`##;` コメントが列 2 以降から開始される場合、コメントもコメント・マーカ (`##;`) も INT コードには表示されませんが、空白行は保持されます。この使用法では、`##;` コメントによって INT コードの行番号が変わることはありません。
- ・ `///` コメントは、任意の列から開始できますが、その行の最初の要素でなければなりません。`///` が列 1 から開始される場合、INT コードには表示されず、空白行も保持されません。`///` が列 2 以降から開始される場合、このコメントは INT コードに表示され、`//` コメントであるかのように扱われます。

2.5.3 メソッド・コードの外部のクラス定義に使用するコメント

メソッド定義の外部のクラス定義で、いくつかの種類のコメントを使用できます。これらのコメントはすべて任意の列から開始できます。

- ・ `//` コメントおよび `/* */` コメントは、クラス定義で使用するコメントです。
- ・ `///` コメントには、その直後のクラスまたは [クラス・メンバ](#) のクラス・リファレンスのコンテンツを記述します。クラス自体に対しては、クラス定義の開始前に配置される `///` コメントには、クラスリファレンスのコンテンツとしてそのクラスの説明が記述されます。このコンテンツは、クラスに関する説明のキーワード値にもなります。クラス内では、[メンバ](#) の直前 (クラス定義の開始位置以降、または前のメンバの後) に配置される `///` コメントにはすべて、そのメンバのクラス・リファレンスのコンテンツが記述されます。複数行の記述は、HTML の 1 つのブロックとして扱われます。`///` コメントの規則の詳細およびクラス・リファレンスは、["クラス・ドキュメントの作成"](#) を参照してください。

2.6 文字列リテラル

文字列リテラルは、0 個以上の文字を引用符で区切って記述したリテラルです (対照的に、[数値リテラル](#) では、値を囲む区切り文字のペアを必要としません)。ObjectScript の文字列リテラルでは、区切り文字として二重引用符を使用し (例えば `"myliteral"`)、InterSystems SQL の文字列リテラルでは、区切り文字として一重引用符を使用します (例えば `'myliteral'`)。このような区切り文字としての引用符は、文字列の長さには算入されません。

文字列リテラルには、スペース文字や制御文字を初めとして、任意の文字を使用できます。許可される文字列の長さには上限があります(“[文字列長の制限](#)”を参照してください)。文字列に 0 から 255 のコードの文字のみが含まれる場合 (Latin-1 または ASCII 拡張文字とも呼ばれる)、各文字は 8 ビット (1 バイト) となります。文字列に 255 より大きなコードの文字が 1 文字以上含まれる場合 (Unicode またはワイド文字とも呼ばれる)、各文字は 16 ビット (2 バイト) となります。文字列の文字の格納に使用されるバイト数を表示するには、次の例に示すように、ZZDUMP コマンドを使用します。

以下の例は、8 ビット文字による文字列、16 ビット Unicode 文字 (ギリシャ文字) による文字列、および 8 ビット文字と 16 ビット Unicode 文字を組み合わせた文字列を示しています。

ObjectScript

```
DO AsciiLetters
DO GreekUnicodeLetters
DO CombinedAsciiUnicode
RETURN
AsciiLetters()
SET a="abc"
WRITE a
WRITE !,"the length of string a is ", $LENGTH(a)
ZZDUMP a
QUIT
GreekUnicodeLetters()
SET b=$CHAR(945)_ $CHAR(946)_ $CHAR(947)
WRITE !!,b
WRITE !,"the length of string b is ", $LENGTH(b)
ZZDUMP b
QUIT
CombinedAsciiUnicode()
SET c=a_b
WRITE !!,c
WRITE !,"the length of string c is ", $LENGTH(c)
ZZDUMP c
QUIT
```

入力不能文字を文字列に使用することもできます。以下の Unicode の例のように、\$CHAR 機能を使用すると、入力不能文字を指定できます。

ObjectScript

```
SET greekstr=$CHAR(952,945,955,945,963,945)
WRITE greekstr
```

表示不能文字を文字列に使用することもできます。このような文字として、出力不能文字 (制御文字) があります。WRITE コマンドでは、出力不能文字がボックス記号で表示されます。WRITE コマンドを使用すると、制御文字が実行されます。以下の例の文字列では、NULL 文字 (\$CHAR(0))、タブ文字 (\$CHAR(9))、キャリッジ・リターン文字 (\$CHAR(13)) に入れ替わる出力可能な文字を使用しています。

ObjectScript

```
SET a="a_"$CHAR(0)"_"b_"$CHAR(9)"_"c_"$CHAR(13)"_"d"
WRITE !,"the length of string a is ", $LENGTH(a)
ZZDUMP a
WRITE !,a
```

プログラムで実行している WRITE では印刷不能文字として表示される制御文字が、ターミナルのコマンドラインから実行した WRITE では実行されることがあります。ベル文字 (\$CHAR(7)) と垂直タブ文字 (\$CHAR(11)) はその例です。

引用符文字 (") を文字列に含めるには、次の例に示すようにその文字を重複して使用します。

ObjectScript

```
SET x="This quote"
SET y="This "" quote"
WRITE x,!," string length=", $LENGTH(x)
ZZDUMP x
WRITE !!,y,!," string length=", $LENGTH(y)
ZZDUMP y
```


値を含まない文字列は NULL 文字列と呼ばれ、2 つの引用符 (") で示されます。NULL 文字列は、定義済みの値と見なされます。その長さは 0 です。以下の例にあるように、NULL 文字列は、NULL 文字 (\$CHAR(0)) で構成した文字列とは異なります。

ObjectScript

```
SET x=""
WRITE "string=",x," length=", $LENGTH(x)," defined=", $DATA(x)
ZZDUMP x
SET y=$CHAR(0)
WRITE !!, "string=",y," length=", $LENGTH(y)," defined=", $DATA(y)
ZZDUMP y
```

文字列の詳細は、“[文字列](#)”を参照してください。

2.7 数値リテラル

数値リテラルとは、ObjectScript が数値として計算する値のことです。[文字列リテラル](#)とは対照的に、値を囲む区切り文字のペアは必要としません。InterSystems IRIS では、数値リテラルが[キャノニック形式](#) (数値リテラルの最も単純な数値形式) に変換されます。

ObjectScript

```
SET x = ++0007.00
WRITE "length:      ", $LENGTH(x), !
WRITE "value:       ", x, !
WRITE "equality:    ", x = 7, !
WRITE "arithmetic:  ", x + 1
```

引用符で区切った文字列リテラルとして数値を表現することもできます。数値の文字列リテラルはキャノニック形式に変換されませんが、演算操作では数値として使用できます。

ObjectScript

```
SET y = "++0007.00"
WRITE "length:      ", $LENGTH(y), !
WRITE "value:       ", y, !
WRITE "equality:    ", y = 7, !
WRITE "arithmetic:  ", y + 1
```

詳細は、“[数値としての文字列](#)”を参照してください。

ObjectScript では、以下の値を使用した値が数値として扱われます (ここにない値を使用すると数値扱いにはなりません)。

値	数量
0 から 9 までの数字。	1 個以上の任意の個数。
符号演算子、単項マイナス演算子 (-)、単項プラス演算子 (+)。	個数は任意ですが、他のすべての文字の前に記述する必要があります。
decimal_separator 文字 (既定ではピリオドまたは小数点文字、ヨーロッパのロケールではコンマ)。	最大 1 個。
文字 E (科学的記数法で使用)。	最大 1 個。2 つの数値の間に記述する必要があります。

これらの文字の使用と解釈の詳細は、“[数値の基本](#)”を参照してください。

ObjectScript は、以下の種類の数値を使用できます。

- ・ 整数 (100、0、-7 などの自然数)。
- ・ 小数：小数 (3.767、.0442 などの実数)。ObjectScript では、InterSystems IRIS 標準の浮動小数点数 (\$DECIMAL 数) または IEEE 倍精度浮動小数点数 (\$DOUBLE 数) のいずれかの内部表現で小数を記述できます。詳細は、“[\\$DOUBLE 関数](#)”を参照してください。
- ・ [科学的記数法](#)：指数表現の数字 (2.8E2 など)。

2.8 識別子

識別子とは、変数、ルーチン、またはラベルの名前です。一般に、正当な識別子は文字や数字で構成されています。まれな例外を除き、句読点文字は識別子に使用できません。識別子は小文字と大文字を区別します。

ユーザ定義のコマンド、関数、および特殊変数に対する名前付け規約は、識別子の名前付け規約よりも制限 (許可された文字のみ) を受けます。“[%ZLANG ルーチンによる言語の拡張](#)”を参照してください。

ローカル変数、プロセス・プライベート・グローバル、およびグローバルの命名規則の詳細は、“[変数](#)”を参照してください。

2.8.1 識別子の句読点文字

特定の識別子は、1 つ以上の句読点文字を持つことができます。以下はその概要です。

- ・ 識別子の最初の文字をパーセント (%) 記号にすることができます。% 文字で始まる InterSystems IRIS 名 (%Z または %z で始まるものは除く) は、システム要素として予約されています。詳細は、“[識別子のルールとガイドライン](#)”を参照してください。
- ・ (ローカル変数名ではなく) グローバルまたはプロセス・プライベート・グローバルの名前は、1 つ以上のピリオド (.) 文字を持つ場合があります。ルーチン名は、1 つ以上のピリオド (.) 文字を持つ場合があります。ピリオドを、識別子の最初または最後の文字にすることはできません。

グローバルおよびプロセス・プライベート・グローバルは、次に示すような 1 文字以上のキャレット (^) 文字で識別されることに注意してください。

Globals: [^] globname [^] " " globname [^] "myspace" globname [^] ["myspace"] globname	Process-Private Globals: [^] ppgname [^] " " ppgname [^] " " , " " ppgname [^] [" "] ppgname
---	---

これらの接頭文字は変数名を構成するものではなく、ストレージのタイプと (グローバルの場合は) このストレージで使用するネームスペースを識別します。実際の名前は、最後の垂直バーまたは閉じ角括弧の後から始まります。

2.9 ラベル

ObjectScript コードの行は、オプションでラベル (タグとして) を含むことができます。ラベルは、コード内の行の場所を参照するハンドルとして機能します。ラベルはインデントされない識別子で、列 1 で指定されます。すべての ObjectScript コマンドは、インデントする必要があります。

ラベルには以下の名前付け規約があります。

- 最初の文字は、英数字またはパーセント記号 (%) でなくてはなりません。ラベルは、先頭を数字とすることができる唯一の ObjectScript 名であることに注意してください。2 番目以降の文字はすべて英数字でなくてはなりません。ラベルは Unicode 文字を含めることができます。
- 最大長は 31 文字です。ラベルは、31 文字よりも長くすることができますが、最初の 31 文字は一意である必要があります。ラベル参照は、ラベルの最初の 31 文字のみと一致します。ただし、ラベルまたはラベル参照 (最初の 31 文字だけでなく) の文字はすべて、ラベル文字の名前付け規約に従う必要があります。
- ラベルは大文字と小文字を区別します。

注釈 CREATE PROCEDURE や CREATE TRIGGER などの SQL コマンドで指定した ObjectScript コードのブロックではラベルを使用できます。この場合は、ラベルの先頭文字の前の 1 カラムに、接頭文字としてコロン (:) を記述します。ラベルのそれ以降の部分は、ここで説明する命名と使用方法の要件に従います。

ラベルではパラメータの括弧を含めたり省略することができます。括弧を含めた場合、その中を空にしたり、またはコンマ区切りのパラメータ名を 1 つ以上含めることができます。括弧付きのラベルによりプロシージャ・ブロックが識別されます。

行は、ラベルだけで構成できます。ラベルの次に 1 以上のコマンド、またはラベルの次に 1 コメントなどです。コマンドまたはコメントが、同じ行のラベルの後に続く場合、スペースまたはタブ記号を使って両者を区別する必要があります。

以下は、すべて一意のラベルです。

ObjectScript

```
maximum
Max
MAX
86
agent86
86agent
%control
```

\$ZNAME 関数を使用すれば、ラベル名を検証することができます。ラベル名を検証する場合には、パラメータの括弧を含めないでください。

ZINSERT コマンドを使用すれば、ラベル名をソース・コードに挿入できます。

2.9.1 ラベルの使用法

ラベルは、コード・セクションの認識およびフロー・コントロールの管理に役立ちます。

DO および GOTO コマンドでは、ターゲットの場所をラベルとして指定することができます。\$ZTRAP 特殊変数では、エラー・ハンドラの場所をラベルとして指定することができます。JOB コマンドでは、実行されるルーチンをラベルとして指定することができます。

また、ソース・コード行を特定するには、PRINT、ZPRINT、ZZPRINT、ZINSERT、ZREMOVE、および ZBREAK コマンド、さらには \$TEXT 関数によりラベルを使用します。

ただし、CATCH コマンドと同一のコード行にて、または TRY ブロックと CATCH ブロックの間にてラベルを指定することはできません。

2.9.2 ラベル付けされたコード・セクションの終了

ラベルはエン트리・ポイントを提供しますが、カプセル化されたコード・ユニットを定義しません。これは、ラベル付けされたコードが実行されれば、実行が停止またはリダイレクトされない限り、次のラベル付けされたコード・ユニットへと実行が続くことを意味します。コード・ユニットの実行を停止するには、以下の 3 つの方法があります。

- コード実行が QUIT または RETURN に遭遇する。

- ・ コード実行が **TRY** の閉じ中括弧に (“}”) に遭遇する。これが発生すると、関連付けられた **CATCH** ブロックに続くコードの次の行へと実行を継続します。
- ・ コード実行が次のプロシージャ・ブロック (パラメータの括弧付きラベル) に遭遇する。括弧付きラベル行と遭遇すると、括弧内にパラメータがない場合においても、実行が停止します。

以下の例では、label0 以下から label1 以下までコードが連続して実行されます。

ObjectScript

```
SET x = $RANDOM(2)
IF x=0 {DO label0
    WRITE "Finished Routine0",! }
ELSE {DO label1
    WRITE "Finished Routine1",! }
QUIT
label0
WRITE "In Routine0",!
FOR i=1:1:5 {
    WRITE "x = ",x,!
    SET x = x+1 }
WRITE "At the end of Routine0",!
label1
WRITE "In Routine1",!
FOR i=1:1:5 {
    WRITE "x = ",x,!
    SET x = x+1 }
WRITE "At the end of Routine1",!
```

以下の例では、ラベル付けされたコード・セクションが **QUIT** または **RETURN** コマンドのいずれかで終了します。これによって実行が止まります。RETURN は常に実行を止めて、QUIT は現在のコンテキストの実行を止めることに注意してください。

ObjectScript

```
SET x = $RANDOM(2)
IF x=0 {DO label0
    WRITE "Finished Routine0",! }
ELSE {DO label1
    WRITE "Finished Routine1",! }
QUIT
label0
WRITE "In Routine0",!
FOR i=1:1:5 {
    WRITE "x = ",x,!
    SET x = x+1
    QUIT }
WRITE "Quit the FOR loop, not the routine",!
WRITE "At the end of Routine0",!
QUIT
WRITE "This should never print"
label1
WRITE "In Routine1",!
FOR i=1:1:5 {
    WRITE "x = ",x,!
    SET x = x+1 }
WRITE "At the end of Routine1",!
RETURN
WRITE "This should never print"
```

以下の例では、2 番目と 3 番目のラベルがプロシージャ・ブロック (パラメータの括弧にて指定されたラベル) を指定します。プロシージャ・ブロックのラベルと遭遇すると、実行が停止します。

ObjectScript

```
SET x = $RANDOM(2)
IF x=0 {DO label0
    WRITE "Finished Routine0",! }
ELSE {DO label1
    WRITE "Finished Routine1",! }
QUIT
label0
WRITE "In Routine0",!
FOR i=1:1:5 {
    WRITE "x = ",x,!
    SET x = x+1 }
WRITE "At the end of Routine0",!
label1()
WRITE "In Routine1",!
FOR i=1:1:5 {
    WRITE "x = ",x,!
    SET x = x+1 }
WRITE "At the end of Routine1",!
label2()
WRITE "This should never print"
```

2.10 ネームスペース

ネームスペース名は、明示的なネームスペース名または暗黙的なネームスペース名になります。明示的なネームスペース名は大文字と小文字を区別しません。入力時においての大文字と小文字に関係なく、常に大文字で格納と返送が行われます。

明示的なネームスペース名では、最初の文字は、英字またはパーセント記号 (%) に限られます。残りの文字は、英字、数字、ハイフン (-)、またはアンダースコア () にする必要があります。この名前は、255 文字以内にする必要があります。

InterSystems IRIS で、明示的なネームスペース名をルーチン名またはクラス名に変換するとき (キャッシュしたクエリ・クラス/ルーチン名の作成時など) は、句読文字が % = p、_ = u、- = d のように小文字に置き換えられます。暗黙のネームスペース名には、これら以外の句読文字が使用されていることがありますが、そのようなネームスペース名の変換では、それらの句読文字が小文字の "s" に置き換えられます。したがって、7 つの句読点文字は次のようになります。@ = s、: = s、/ = s、¥ = s、[= s、] = s、^ = s。

予約済みのネームスペース名は、%SYS、BIN、BROKER、および DOCUMATIC です。

InterSystems SQL CREATE DATABASE コマンドを使用した場合、SQL データベースの作成により、対応する InterSystems IRIS ネームスペースが作成されます。

InterSystems IRIS のインスタンスでは、ネームスペースはディレクトリとして存在します。現在のネームスペースのフル・パス名を返すには、以下の例に示すように、NormalizeDirectory() メソッドを呼び出します。

ObjectScript

```
WRITE ##class(%Library.File).NormalizeDirectory("")
```

ネームスペースの使用の詳細は、“[ネームスペースとデータベース](#)”を参照してください。ネームスペースの作成の詳細は、“[ネームスペースの構成](#)”を参照してください。

2.10.1 拡張参照

拡張参照とは他のネームスペースにあるエンティティへの参照のことです。ネームスペース名は引用符で囲んだ文字列リテラル、ネームスペース名に解決する変数、暗黙的なネームスペース名、または現在のネームスペースを指定するブレースホルダである NULL 文字列 (") として指定できます。拡張参照には、以下の 3 つのタイプがあります。

- ・ 拡張グローバル参照: 他のネームスペースのグローバル変数を参照します。次の構文形式をサポートしています。
^["namespace"]global および ^| "namespace" |global。詳細は、“[グローバル変数](#)”を参照してください。

- ・ 拡張ルーチン参照: 他のネームスペースのルーチンを参照します。
 - DO コマンド、[\\$TEXT](#) 関数、およびユーザ定義関数は次の構文形式をサポートしています。
| "namespace" | routine。
 - JOB コマンドは次の構文形式をサポートしています。routine | "namespace" |、routine["namespace"]、または routine: "namespace"。

これらのすべての場合において、拡張ルーチン参照の開始には ^ (キャレット) 文字を使用し、指定したエンティティが(ラベルやオフセットではなく) ルーチンであることを示します。このキャレットはルーチン名の一部ではありません。例えば、DO ^ | "SAMPLES" | fibonacci は、SAMPLES ネームスペースに置かれている fibonacci という名のルーチンを呼び出します。WRITE \$\$fun^ | "SAMPLES" | house コマンドは、SAMPLES ネームスペースに置かれている、ルーチン house のユーザ定義関数 fun() を呼び出します。

- ・ 拡張 SSVN 参照: 他のネームスペースの[構造化システム変数](#) (SSVN) を参照します。次の構文形式をサポートしています。^\$["namespace"]ssvn および ^\$ | "namespace" | ssvn。詳細は、[^\\$GLOBAL](#)、[^\\$LOCK](#)、および [^\\$ROUTINE](#) 構造化システム変数を参照してください。

もちろん、すべての拡張参照は名前によって明示的に、もしくは NULL 文字列のプレースホルダ指定により、現在のネームスペースを指定できます。

2.11 予約語

ObjectScript に予約語はありません。したがって、あらゆる有効な識別子を、変数名、関数名、ラベルとして使用できます。同時に、コマンド名、関数名、他の文字列である識別子の使用を避けるのが適切です。また、ObjectScript コードは埋め込み SQL をサポートするため、関数、オブジェクト、変数、[SQL 予約語](#)である他のエンティティの名前を避けることが賢明です。そうしなければ、障害が発生する原因となります。

3

データ型とデータ値

ObjectScript は、タイプのない言語です (変数タイプを宣言する必要はありません)。変数は、文字列、数値、オブジェクト値を持つことができます。それでも ObjectScript では、異種データをいつ使用するかを把握しておくことは重要な情報となります。

3.1 文字列

文字列とは、文字、数字、句読点などの文字の組み合わせを、1 組の引用符 (") で囲んだものです。

ObjectScript

```
SET string = "This is a string"
WRITE string
```

文字列についての項目は以下のとおりです。

- ・ [NULL 文字列と \\$CHAR\(0\)](#)
- ・ [引用符のエスケープ](#)
- ・ [連結文字列](#)
- ・ [文字列比較](#)
- ・ [ビット文字列](#)

“[文字列長の制限](#)” を参照してください。

3.1.1 NULL 文字列と \$CHAR(0)

- ・ SET mystr="" : NULL または空の文字列を設定します。文字列が定義され、ゼロの長さで、データは含まれません。

ObjectScript

```
SET mystr=""
WRITE "defined:", $DATA(mystr), !
WRITE "length: ", $LENGTH(mystr), !
ZZDUMP mystr
```

- ・ SET mystr=\$CHAR(0) : 文字列を NULL 文字に設定します。文字列が定義され、1 の長さで、16 進数値 00 の 1 文字が含まれます。

ObjectScript

```
SET mystr=$CHAR(0)
WRITE "defined:", $DATA(mystr), !
WRITE "length: ", $LENGTH(mystr), !
ZZDUMP mystr
```

これら 2 つの値は同じではないことに注意してください。ただし、[ビット文字列](#)ではこれらの値は同一のものとして処理されます。

InterSystems SQL では、これらの値が独自の方法で解釈されます。“[NULL および空文字列](#)”を参照してください。

3.1.2 引用符のエスケープ

“ (二重引用符) 文字の前にもう 1 つ ” 文字を置くと、文字列内のリテラルとして引用符を組み込むことができます。

ObjectScript

```
SET string = "This string has ""quotes"" in it."
WRITE string
```

ObjectScript 文字列リテラルに、他のエスケープ文字シーケンスはありません。

リテラルの引用符は、その他のインターシステムズのソフトウェアでは、別のエスケープ・シーケンスを使用して指定されます。これに該当するエスケープ・シーケンスについての表は、[\\$ZCONVERT](#) 関数を参照してください。

3.1.3 連結文字列

[連結演算子](#)を使用して、2 つの文字列を 1 つに結合できます。

ObjectScript

```
SET a = "Inter"
SET b = "Systems"
SET string = a_b
WRITE string
```

連結演算子を使用して、出力不能文字を文字列に組み込むことができます。以下の文字列には、改行 (\$CHAR(10)) 文字が含まれています。

ObjectScript

```
SET lf = $CHAR(10)
SET string = "This"_lf_"is"_lf_"a string"
WRITE string
```

注釈 出力不能文字の表示方法は、ディスプレイ・デバイスにより決定されます。例えば、ターミナルとブラウザでは、改行文字や他のポジショニング文字の表示方法が異なります。さらに、ブラウザが異なれば、ポジショニング文字 \$CHAR(11) および \$CHAR(12) の表示方法も異なります。

InterSystems IRIS エンコード文字列 (ビット文字列、リスト構造文字列、JSON 文字列) には、連結演算子の使用に関する制限があります。詳細は、“[連結エンコード文字列](#)”を参照してください。

数値を結合する際は、他の事項も考慮する必要があります。詳細は、“[数字の連結](#)”を参照してください。

3.1.4 文字列比較

等号 (=) および不等号 (≠) の演算子を使用して、2 つの文字列を比較できます。文字列の等値比較では大文字と小文字が区別されます。これらの演算子を使用して、文字列を数字と比較する際には注意が必要です。それは、この比較が

文字列比較であり、数値比較ではないからです。したがって、[キャノニック形式の数字](#)を含む文字列のみが、対応する数字と等価になります。("-0" はキャノニック形式の数ではありません。)詳細は、以下の例を参照してください。

ObjectScript

```
WRITE "Fred" = "Fred",! // TRUE
WRITE "Fred" = "FRED",! // FALSE
WRITE "-7" = -007.0,! // TRUE
WRITE "-007.0" = -7,! // FALSE
WRITE "0" = -0,! // TRUE
WRITE "-0" = 0,! // FALSE
WRITE "-0" = -0,! // FALSE
```

<、>、<=、>= などの演算子を使用して、文字列比較を実行することはできません。これらの演算子は[数値として文字列](#)を扱い、常に数値比較を実行します。非数値文字列は、これらの演算子を使用して比較された場合、いずれも数値 0 が割り当てられます。

3.1.4.1 大文字小文字の区別と文字列比較

文字列の等値比較では大文字と小文字が区別されます。[\\$ZCONVERT](#) 関数を使用すれば、比較する文字列の文字をすべて大文字に、またはすべて小文字に変換することができます。文字ではない符号や記号は不変となります。

小文字の文字形式しか有していない文字も多少あります。例えば、ドイツ語の eszett ([\\$CHAR\(223\)](#)) は小文字としてのみ定義されます。これを大文字に変換しても、結果は同じ小文字となります。これにより、英数字文字列を大文字または小文字のみの文字列に変換する場合は、常に小文字に変換することをお勧めします。

3.1.5 ビット文字列

ビット文字列は、番号付きビットとブーリアン値の論理セットを表します。文字列内のビットには、ビット番号 1 から始まる番号が付けられます。ブーリアン値 1 に明示的に設定されていない番号付きビットは、0 と評価されます。したがって、明示的に設定された番号付きビット以外のものを参照すると、ビット値 0 が返されます。

- ・ ビット値は、ビット文字列関数 [\\$BIT](#) および [\\$BITLOGIC](#) を使用することでのみ、設定できます。
- ・ ビット値は、ビット文字列関数 [\\$BIT](#)、[\\$BITLOGIC](#)、および [\\$BITCOUNT](#) を使用することでのみ、アクセスできます。

ビット文字列には論理長があります。これは、0 または 1 のいずれかに明示的に設定された最上位のビット位置です。この論理長には、[\\$BITCOUNT](#) 関数を使用することでのみアクセスできます。ただし、通常は、アプリケーション・ロジックでは論理長を使用しないでください。ビット文字列関数の場合、未定義のグローバル変数またはローカル変数は、ビット値 0 を返す指定された番号付きビットを持ち、[\\$BITCOUNT](#) 値 が 0 であるビット文字列と同等です。

ビット文字列は、内部形式を持つ通常の ObjectScript 文字列として保存されます。この内部文字列の表現には、ビット文字列関数ではアクセスできません。この内部形式のため、ビット文字列の[文字列長](#)は、文字列内のビット数に関する情報を判断する際に意味がありません。

ビット文字列の内部形式という理由により、[連結演算子](#)をビット文字列と共に使用することはできません。これを実行しようとすると、<INVALID BIT STRING> エラーが返されます。

同じ状態 (同じブーリアン値を持つ) の 2 つのビット文字列が、異なる内部文字列表現を持つ場合があります。したがって、文字列表現は、アプリケーション・ロジックで調査したり、比較したりしないでください。

ビット文字列関数の場合、未定義の変数として指定されたビット文字列は、すべてのビットが 0 で長さが 0 のビット文字列と同等です。

[通常の文字列とは異なり](#)、ビット文字列では、空の文字列と文字 [\\$CHAR\(0\)](#) は互いに等しく、0 ビットを表すものとして処理されます。これは、[\\$BIT](#) では非数値の文字列がすべて 0 として処理されるからです。したがって、以下ようになります。

ObjectScript

```
SET $BIT(bstr1,1)=" "
SET $BIT(bstr2,1)=$CHAR(0)
SET $BIT(bstr3,1)=0
IF $BIT(bstr1,1)=$BIT(bstr2,1) {WRITE "bitstrings are the same"} ELSE {WRITE "bitstrings different"}

WRITE $BITCOUNT(bstr1),$BITCOUNT(bstr2),$BITCOUNT(bstr3)
```

トランザクション中にグローバル変数で設定されたビットは前の値に戻され、その後にトランザクションのロールバックが行われます。ただし、グローバル変数のビット文字列は、ロールバック操作によって前の文字列長または前の内部文字列表現に戻りません。ローカル変数は、ロールバック操作によって元に戻りません。

論理ビットマップ構造は、ビット文字列の配列で表現できます。この配列の各要素が固定のビット数を持つ 1 つの「チャンク」を表しています。未定義はすべてのビットが 0 のチャンクと同等であるため、配列はスパースになることがあります。すべてのビットが 0 のチャンクを表す配列要素は、存在している必要は一切ありません。この理由および上記のロールバック動作のため、アプリケーション・ロジックでは、ビット文字列の長さや \$BITCOUNT(str) または \$BITCOUNT(str,0) を使用してアクセスできる 0 値のビットのカウントに依存しないようにする必要があります。

3.2 数値

数値に関連するトピックは、以下のとおりです。

- ・ [数値の基本](#)
- ・ [数値のキャノニック形式](#)
- ・ [数値としての文字列](#)
- ・ [数字の連結](#)
- ・ [浮動小数点数](#)
- ・ [科学的記数法](#)
- ・ [極端に大きな数字](#)

3.2.1 数値の基本

数値リテラルに句読点は必要ありません。任意の有効な数値文字を使用して数を指定できます。InterSystems IRIS では、数値が構文的に有効と評価されると、その数値がキャノニック形式に変換されます。

数値リテラルの構文要件は以下のようになります。

- ・ 0 ～ 9 の 10 進数を含むことができ、少なくともそれらの数値文字のいずれかを含んでいること。先頭または末尾にゼロを使用できます。ただし、InterSystems IRIS で数値をキャノニック形式に変換すると、先頭の整数ゼロが自動的に削除されます。このため、先頭の整数ゼロが重要な数値は、文字列として入力する必要があります。例えば、米国の郵便番号は、02142 のように先頭が整数ゼロの可能性もあるため、数値ではなく文字列として処理する必要があります。
- ・ 任意のシーケンスで、先頭に任意の数のプラス符号とマイナス符号を使用できます。ただし、プラス符号とマイナス符号は、科学的記数法の文字 “E” を除き、他の任意の文字の後に位置することはできません。数値式では、非符号文字の後の符号は加算または減算として評価されます。数値文字列では、非符号文字の後の符号は非数値文字として評価され、そこで文字列の数値部が終わります。

InterSystems IRIS では、PlusSign および MinusSign プロパティ値を現在のロケールに使用して、これらの符号文字（既定では “+” および “-”）を決定します。これらの符号文字はロケールに依存します。ユーザのロケールの PlusSign 文字および MinusSign 文字を決定するには、以下のように GetFormatItem() メソッドを呼び出します。

ObjectScript

```
WRITE ##class(%SYS.NLS.Format).GetFormatItem("PlusSign"),!
WRITE ##class(%SYS.NLS.Format).GetFormatItem("MinusSign")
```

- ・ 小数点区切り文字を 1 つ使用できます。数値式では、2 つ目の小数点区切りを用いると <SYNTAX> エラーとなります。数値文字列では、2 つ目の小数点区切りは最初の非数値文字として評価され、そこで文字列の数値部が終わります。小数点区切り文字は、数値式の最初の文字または最後の文字とすることができます。小数点区切り文字の選択はロケールに依存します。アメリカ形式では、小数点区切りとして既定のピリオド(.) が使用されます。ヨーロッパ形式では、コンマ(,) が使用されます。ユーザのロケールの DecimalSeparator 文字を決定するには、以下のように GetFormatItem() メソッドを呼び出します。

ObjectScript

```
WRITE ##class(%SYS.NLS.Format).GetFormatItem("DecimalSeparator")
```

- ・ **科学的記数法**では、“E” (または “e”) を最大で 1 つ使用して 10 を基数とする指数を指定できます。この科学的記数文字 (“E” または “e”) の前には整数または小数、その後には整数を記述する必要があります。

数値リテラル値は、以下をサポートしていません。

- ・ 数値グループ・セパレータは使用できません。これらはロケールに依存します。アメリカ形式ではコンマが使用され、ヨーロッパ形式ではピリオドが使用されます。**\$NUMBER** 関数を使用して数値グループ・セパレータを削除し、**\$FNUMBER** 関数を使用して数値グループ・セパレータを追加することができます。
- ・ 通貨記号、16 進数文字、その他の非数値文字は使用できません。演算子の前後を除き、空白スペースは使用できません。
- ・ 末尾にプラス符号やマイナス符号を使用できません。ただし、**\$FNUMBER** 関数は、末尾に符号の付いた文字列として数を表示でき、**\$NUMBER** 関数でこの形式の文字列を取得して、先頭に符号の付いた数に変換します。
- ・ 負の数 (借方) として数を表す場合、括弧に入れることはできません。ただし、**\$FNUMBER** 関数は、括弧で囲んだ文字列で負の数を表示でき、**\$NUMBER** 関数でこの形式の文字列を取得して、先頭に負符号の付いた数に変換します。

数または数値式には、1 組みの括弧を使用できます。これらの括弧は数の一部ではありませんが、処理の優先順位を決定します。既定では、InterSystems IRIS は、厳密に左から右の順序ですべての処理を実行します。

3.2.2 数値のキャノニック形式

ObjectScript では、数に対する数値演算のすべてをキャノニック形式で実行します。例えば、数値 +007.00 の長さは 1 になり、文字列 "+007.00" の長さは 7 になります。

InterSystems IRIS で数をキャノニック形式に変換するときには、以下の手順が実行されます。

1. 科学的記数法の指数が解決されます。例えば、3E4 は 30000 に変換され、3E-4 は .0003 に変換されます。
2. 先頭の符号が解決されます。まず、複数の符号が 1 つの符号に解決されます (例えば、2 つのマイナス符号が 1 つのプラス符号になります)。次に、先頭にあるプラス符号が削除されます。**\$FNUMBER** 関数を使用すると、InterSystems IRIS のキャノニック形式の正の数にプラス記号を明示的に指定 (追加) できます。

注釈 ObjectScript は、先頭のプラス記号とマイナス記号の任意の組み合わせを解決します。SQL では、2 つの連続したマイナス記号は、1 行コメント文字として解析されます。したがって、先頭に 2 つの連続したマイナス記号を付けて SQL で数値を指定すると、SQLCODE -12 エラーになります。

3. 先頭と末尾のゼロがすべて削除されます。これには、1 より小さい小数の先頭の整数ゼロの削除も含まれます。例えば、0.66 は .66 になります。

- ・ キヤノニック形式の小数に整数のゼロを追加するには、[\\$FNUMBER](#) 関数または [\\$JUSTIFY](#) 関数を使用します。`.66` は `0.66` になります。
- ・ キヤノニック形式以外的小数から整数ゼロを削除するには、[単項プラス](#) 演算子を使用して、数値文字列からキヤノニック形式の数値への変換を行います。例えば、`+$PIECE("65798,00000.66", ",", 2)` のタイムスタンプの秒の小数部 `00000.66` は `.66` になります。

この変換の一環として、ゼロの小数は 0 に単純化されます。表現方法 (`0.0`、`.0`、`.000`) にかかわらず、すべてのゼロ値は 0 に変換されます。

4. 末尾の小数点区切りが削除されます。
5. `-0` は 0 に変換されます。
6. 算術演算と数値の連結が実行されます。InterSystems IRIS は、こうした演算を厳密に左から右の順に実行します。このような演算は、キヤノニック形式の数で実行されます。詳細は、後述の [“数字の連結”](#) を参照してください。

InterSystems IRIS のキヤノニック形式の数は、その他のインターシステムズのソフトウェアで使用されているキヤノニック形式の数とは異なります。

- ・ ODBC: 整数ゼロの小数は、1 つのゼロ整数を含む ODBC に変換されます。そのため、`.66` と `000.66` は、どちらも `0.66` になります。[\\$FNUMBER](#) 関数または [\\$JUSTIFY](#) 関数を使用すると、InterSystems IRIS のキヤノニック形式の小数に整数のゼロを追加できます。
- ・ JSON: 先頭の 1 つのマイナス記号のみが許容されます。先頭のプラス記号や複数の記号は許容されません。指数は許容されますが、解決されません。`3E4` は、`3E4` として返されます。
先頭のゼロは許容されません。末尾のゼロは削除されません。
整数ゼロの小数には、1 つのゼロ整数が必要です。したがって、`.66` と `000.66` は無効な JSON の数値になりますが、`0.66` と `0.660000` は有効な JSON の数値となります。
末尾の小数点記号は許容されません。
ゼロの値は、変換されません。`0.0`、`-0`、および `-0.000` は、有効な JSON の数値として変更なしで返されます。

3.2.3 数値としての文字列

文字列を数値として処理する標準的な規則は以下のとおりです。詳細は、[“文字列から数値への変換”](#) を参照してください。特別な処理については、[“非常に大きな数値文字列”](#) を参照してください。

- ・ すべての数値演算では、キヤノニック形式の数字を含む文字列が、対応する数字と機能的に同一になります。例えば、`"3" = 3`、`"-2.5" = -2.5` となります。(0 はキヤノニック形式の数ではありません。)
- ・ 算術演算では、非キヤノニック形式の数値文字のみを含む文字列が、対応する数字と機能的に同一になります。例えば、`"003" + 3 = 6`、`"++-2.5000" + -2.5 = -5` となります。
- ・ 「より大きい/より小さい」などの比較演算では、非キヤノニック形式の数値文字のみを含む文字列が、対応する数字と機能的に同一になります。例えば、次の文は True となります。`"003" > 2`、`"++-2.5000" >= -2.5`
- ・ 等値演算 (`=`、`'=`) では、非キヤノニック形式の数値文字のみを含む文字列が、数値ではなく文字列として扱われます。例えば、次の文は True となります。`"003" = "003"`、`"003" '= 3`、`"003" '= "003"`

数値としての文字列の解析に関するその他のガイドラインを以下に述べます。

- ・ 混合数値文字列とは、数字で始まり、1 つまたは複数の数値でない文字が続く文字列です。例えば `"7 dwarves"` です。InterSystems IRIS の数値演算とブーリアン演算 (等値演算は除く) では共に、数値ではない文字を検出する

まで混合数値文字列を数値として解析します。数値でない文字を検出した時点で残りの文字列を無視します。以下に混合数値文字列の算術演算の例を示します。

ObjectScript

```
WRITE "7dwarves" + 2,!    // returns 9
WRITE "+24/7" + 2,!      // returns 26
WRITE "7,000" + 2,!      // returns 9
WRITE "7.0.99" + 2,!     // returns 9
WRITE "7.5.99" + 2,!     // returns 9.5
```

- 非数値文字列とは、数値文字の前に数値ではない文字がある任意の文字列です。空白は数値ではない文字と見なされます。InterSystems IRIS の数値演算とブーリアン演算（等値演算は除く）では共に、このような文字列を数値 0（ゼロ）を持つものとして解析します。以下に非数値文字列の算術演算の例を示します。

ObjectScript

```
WRITE "dwarves 7" + 2,!    // returns 2
WRITE "+ 24/7" + 2,!      // returns 2
WRITE "$7000" + 2,!       // returns 2
```

- 文字列の接頭語にプラス符号を置いて、等値演算の数値として評価させることができます。数値文字列はキャノニック形式の数値として解析され、非数値文字列は 0 として解析されます。（マイナス符号の接頭語でも等値演算の数値として文字列を評価されます。当然、マイナス符号は 0 以外の値の符号の反転となります。）以下では、プラス符号により等値演算での数値評価が強制される例を示します。

ObjectScript

```
WRITE +"7" = 7,!          // returns 1 (TRUE)
WRITE "+007" = 7,!        // returns 1 (TRUE)
WRITE +"7 dwarves" = 7,!  // returns 1 (TRUE)
WRITE +"dwarves" = 0,!    // returns 1 (TRUE)
WRITE +" " = 0,!          // returns 1 (TRUE)
```

“ObjectScript リファレンス” に示したとおり、個別のコマンドと関数に対する数値文字列の例外処理は共通となります。

3.2.3.1 非常に大きな数値文字列

通常、数値文字列は ObjectScript の 10 進数値に変換されます。ただし、非常に大きな数値の場合（9223372036854775807E127 より大きい）、常に数値文字列を 10 進数値に変換できるとは限りません。数値文字列を 10 進数値に変換すると <MAXNUMBER> エラーが発生する場合、InterSystems IRIS では代わりに IEEE バイナリ値に変換します。InterSystems IRIS は次の操作を実行して、数値文字列を数に変換します。

- 数値文字列を 10 進数の浮動小数点数に変換します。この操作で <MAXNUMBER> エラーが発生する場合は、手順 2 に進みます。それ以外の場合、10 進数値がキャノニック形式の数として返されます。
- \$SYSTEM.Process.TruncateOverflow() メソッド・ブーリアン値を確認します。0（既定）の場合、手順 3 に進みます。それ以外の場合、オーバーフローする 10 進数値が返されます（メソッドの説明を参照）。
- 数値文字列を IEEE バイナリ浮動小数点数に変換します。この操作で <MAXNUMBER> エラーが発生する場合は、手順 4 に進みます。それ以外の場合、IEEE バイナリ値がキャノニック形式の数として返されます。
- \$SYSTEM.Process.IEEEError() メソッド・ブーリアン値を確認します。この値に応じて、INF / -INF が返されるか、<MAXNUMBER> エラーが発生します。

3.2.4 数字の連結

連結演算子 () を使用して、ある数と別の数を連結できます。InterSystems IRIS は、まずそれぞれの数値をキャノニック形式に変換してから、その結果に対して文字列連結を実行します。したがって、12_34、12_+34、12_--34、12.0_34、

12.0034.0、12E0_34 はすべて 1234 になります。12._34 の連結は 1234 になりますが、12._34 は 12.34 になります。12_34 の連結は文字列 “12-34” になります。

InterSystems IRIS は、キャノニック形式に変換した後の数に対して、数値の連結と算術演算を実行します。こうした演算は厳密に左から右の順に実行されます。ただし、演算の優先順位を決める括弧が指定されている場合を除きます。以下の例では、この結果について説明しています。

ObjectScript

```
WRITE 7_-6+5 // returns 12
```

この例では、連結によって文字列 “7-6” が返されます。これが、キャノニック形式の数でないことは明白です。InterSystems IRIS は、最初の非数値文字 (埋め込みのマイナス記号) を切り捨てることで、この文字列をキャノニック形式の数に変換します。その後で、このキャノニック形式の数を使用して次の演算 $7 + 5 = 12$ を実行します。

3.2.5 浮動小数点数

InterSystems IRIS は、浮動小数点数を表すために使用できる 2 つの異なる数値タイプをサポートしています。

- 10 進数の浮動小数点：既定では、InterSystems IRIS は、固有の 10 進数の浮動小数点標準 (\$DECIMAL 数) を使用して小数を表します。これは、ほとんどの場合の優先形式であり、IEEE バイナリ浮動小数点よりも、高いレベルの精度が得られます。これは、InterSystems IRIS がサポートしているすべてのシステム・プラットフォームで一貫しています。10 進数の浮動小数点は、データ・ベースの値で優先的に使用されます。特に、0.1 などの小数は、10 進数の浮動小数記数法を使用して正確に表現できますが、小数 0.1 (およびほとんどの 10 進数の小数) は IEEE バイナリ浮動小数点では近似値で表現できるだけです。

内部的に、10 進算術は $M \times (10^N)$ 形式の数を使用して実行されます。M は -9223372036854775808 から 9223372036854775807 の整数値を含む整数の仮数で、N は -128 から 127 の整数値を含む 10 進指数です。仮数は 64 ビット符号付整数で表され、指数は 8 ビット符号付バイトで表されます。

10 進浮動小数点の平均精度は小数桁数 18.96 桁です。仮数が 10000000000000000000 から 9223372036854775807 の小数の精度は正確に 19 桁で、922337203685477581 から 9999999999999999 の有効な小数の精度は正確に 18 桁です。IEEE バイナリ浮動小数点は、精度は低いものの (精度は小数桁数約 15.95 桁)、小数文字列としての IEEE バイナリ表現の正確な無限精度の値は、1000 桁以上の有効小数桁数を持つことができます。

以下の例では、\$DECIMAL 関数は小数と 25 桁の整数を取り、19 桁の精度/有効桁数に丸められた小数を返します。

Terminal

```
USER>WRITE $DECIMAL(1234567890.123456781818181)
1234567890.123456782
USER>WRITE $DECIMAL(1234567890123456781818181)
1234567890123456782000000
```

- IEEE バイナリ浮動小数点：IEEE 倍精度バイナリ浮動小数点は、小数を表現する業界標準の方法です。IEEE 浮動小数点数は、バイナリ表現でエンコードされます。バイナリ浮動小数点表現は、ほとんどのコンピュータがバイナリ浮動小数点演算のための高速ハードウェアを搭載しているため、通常、高速計算を実行する場合に優先的に使用されます。

内部的に、IEEE バイナリ算術は $S \times M \times (2^N)$ 形式の数を使用して実行されます。S は -1 または +1 の値を含む符号で、M は 1 番目と 2 番目のバイナリ・ビット間に 2 進小数点がある 53 ビットのバイナリ小数値を含む仮数で、N は -1022 から 1023 の整数値を含む 2 進指数です。このため、表現は 64 ビットで構成されます。S は 1 つの符号ビット、指数 N は次の 11 ビットに格納され (2 つの追加の値が予約されます)、仮数 M は ≥ 1.0 かつ < 2.0 で最後の 52 ビットを含み、合計 53 バイナリ・ビットの精度となります。(M の最初のビットは常に 1 であるため、64 ビット表現に出現する必要はありません。)

倍精度バイナリ小数点の精度は 53 バイナリ・ビットで、小数点以下約 15.95 桁の精度に相当します。(対応する 10 進数の精度は 15.35 から 16.55 桁の間で変動します。)

バイナリ表現は、0.1 などの小数をバイナリ小数の有限シーケンスとして表現できないため、10 進数の小数と正確には一致しません。ほとんどの小数はこのバイナリ表現では正確に表すことができないため、IEEE 浮動小数点数は対応する InterSystems の 10 進浮動小数点数とは若干異なる場合があります。IEEE 浮動小数点数が小数として表示されるとき、バイナリ・ビットが18桁を大幅に超える小数桁数の小数に変換されることがよくあります。これは、IEEE 浮動小数点数が InterSystems の 10 進浮動小数点数より精度が高いことを意味するわけではありません。IEEE 浮動小数点数は、InterSystems の10 進数よりも大きい数および小さい数を表現できます。

以下の例では、\$DOUBLE 関数は 17 桁の整数のシーケンスを取り、小数点以下の有効桁数約 16 桁の値を返します。

Terminal

```
USER>FOR i=12345678901234558:1:12345678901234569 {W $DOUBLE(i),!}
12345678901234558
12345678901234560
12345678901234560
12345678901234560
12345678901234562
12345678901234564
12345678901234564
12345678901234564
12345678901234566
12345678901234568
12345678901234568
12345678901234568
```

IEEE バイナリ浮動小数点は、特殊な値 INF (無限大) と NAN (非数値) をサポートします。詳細は、“\$DOUBLE” 関数を参照してください。

INF および NAN 値を扱うための IEEEError 設定と、\$LIST 構造データ内の IEEE 浮動小数点数の圧縮を扱うための ListFormat 設定を使用して、IEEE 浮動小数点数の処理を構成できます。どちらも、%SYSTEM.Process クラス・メソッド \$SYSTEM.Process.IEEEError() を使用して、現在のプロセスに表示および設定できます。InterSystems IRIS の管理ポータルで [システム管理] から [構成]、[追加の設定]、[互換性] を選択して、システム全体の既定値を設定できます。

\$DOUBLE 関数を使用して、InterSystems IRIS 標準の浮動小数点数を IEEE 浮動小数点数に変換できます。また、\$DECIMAL 関数を使用して、IEEE 浮動小数点数を InterSystems IRIS 標準の浮動小数点数に変換できます。

既定では、InterSystems IRIS は小数を**キャノンニック形式**に変換して、先頭のゼロをすべて削除します。したがって、0.66 は .66 となります。**\$FNUMBER** (ほとんどの形式) および **\$JUSTIFY** (3 パラメータ形式) は常に、最低 1 桁の整数を伴う小数を返します。これらの関数のいずれかを使用した場合、.66 は 0.66 となります。

\$FNUMBER および **\$JUSTIFY** を使用すれば、指定した小数桁数に対して丸めや埋め込みを行うことができます。InterSystems IRIS では、5 以上を切り上げて、4 以下を切り捨てます。埋め込みでは、ゼロを必要な小数桁だけ追加します。少数を整数に丸めた場合、小数点区切り文字は削除されます。整数をゼロ・パディングして小数にした場合、小数点区切り文字が追加されます。

3.2.6 科学的記数法

ObjectScript で科学的記数法 (指数表現) を指定するには、以下の形式を使用します。

```
[-]mantissaE[-]exponent
```

各項目の内容は次のとおりです。

要素	説明
-	オプションー1つ以上の単項マイナス演算子または単項プラス演算子。これらのマイナス記号文字とプラス記号文字は構成可能です。キャノニック形式への変換では、科学的記数法を解決した後にこれらの演算子が解決されます。
mantissa	整数または小数。先頭と末尾のゼロおよび末尾の小数点区切り文字が含まれる場合があります。
E	指数表現を区切る演算子。大文字の“E”は標準の指数演算子です。小文字の“e”は、%SYSTEM.Process クラスの ScientificNotation() メソッドを使用して構成できる指数演算子です。
-	オプションー単一の単項マイナス演算子または単項プラス演算子。負数の指数を指定するために使用できます。これらのマイナス記号文字とプラス記号文字は構成可能です。
exponent	指数 (10 のべき乗値) を指定する整数。先頭にゼロを使用できます。小数点区切り文字は記述できません。

例えば、10 を表現するには 1E1、2800 を表現するには 2.8E3、0.05 を表現するには 5E-2 と記述します。

mantissa、E、およびexponentの間には、スペースは許可されません。この構文内では、括弧、連結、およびその他の演算子は許可されません。

科学的記数法を解決することが、数値を**キャノニック形式**に変換する第一歩であるため、一部の変換処理は利用できません。mantissaとexponentは数値リテラルである必要があり、変数または算術式にすることはできません。exponentは、(最大でも) 1つのプラスまたはマイナス記号の付いた整数である必要があります。

詳細は、%SYSTEM.Process クラスの ScientificNotation() メソッドの説明を参照してください。

3.2.7 極端に大きな数字

正確に表現できる最大の整数は、19桁の整数の -9223372036854775808 と 9223372036854775807 です。なぜなら、これらの数値は 64ビット符号付き整数で表現できる最大数だからです。これを超える整数は、この 64ビット制限に収まるように自動的に丸められます。以下に例を示します。

ObjectScript

```
SET x=9223372036854775807
WRITE x,!
SET y=x+1
WRITE y
```

同様に、128を超える指数も、64ビット符号付き整数の表現になるように自動的に丸められます。以下に例を示します。

ObjectScript

```
WRITE 9223372036854775807e-128,!
WRITE 9223372036854775807e-129
```

この丸めにより、19桁の整数を超える数になる算術演算は、下位の桁がゼロに置き換えられます。これにより、以下のような状況となる可能性があります。

ObjectScript

```
SET longnum=9223372036854775790
WRITE longnum,!
SET add17=longnum+17
SET add21=longnum+21
SET add24=longnum+24
WRITE add17,! ,add24,! ,add21,!
IF add24=add21 {WRITE "adding 21 same as adding 24"}
```

InterSystems IRIS でサポートされる最大の 10 進数の浮動小数点数は 9.223372036854775807E145 です。サポートされる最大の \$DOUBLE 値 (INFINITY への IEEE オーバーフローが無効であると想定) は 1.7976931348623157081E308 です。\$DOUBLE 型は、InterSystems IRIS の 10 進数型よりも広い範囲の値をサポートしますが、InterSystems IRIS の 10 進数型の方が高精度をサポートします。InterSystems IRIS の 10 進数型の精度は約 18.96 桁 (普通は 19 桁ですが、時折 18 桁の精度) で、\$DOUBLE 型の通常の精度は約 15.95 桁 (53 バイナリ桁) です。既定では、InterSystems IRIS は数値リテラルを 10 進数の浮動小数点数として表現します。ただし、数値リテラルが InterSystems IRIS の 10 進数で表現できる数値より大きい場合 (9.223372036854775807E145 より大きい場合)、InterSystems IRIS はその数値を \$DOUBLE 表現に自動的に変換します。

1.7976931348623157081E308 (308 桁または 309 桁) より大きな数値の場合、<MAXNUMBER> エラーが発生します。

10 進数の浮動小数点からバイナリ浮動小数点への自動変換のため、丸めの動作は、9.223372036854775807E145 (整数に応じて 146 桁または 147 桁) で変更されます。詳細は、以下の例を参照してください。

ObjectScript

```
TRY {
    SET a=1
    FOR i=1:1:310 {SET a=a_1 WRITE i+1," digits = ",+a,! }
}
CATCH exp { WRITE "In the CATCH block",!
    IF 1=exp.%IsA("%Exception.SystemException") {
        WRITE "System exception",!
        WRITE "Name: ", $ZCVT(exp.Name,"O","HTML"),!
        WRITE "Location: ",exp.Location,!
        WRITE "Code: "
    }
    ELSE { WRITE "Some other type of exception",! RETURN }
    WRITE exp.Code,!
    WRITE "Data: ",exp.Data,!
    RETURN
}
```

ObjectScript

```
TRY {
    SET a=9
    FOR i=1:1:310 {SET a=a_9 WRITE i+1," digits = ",+a,! }
}
CATCH exp { WRITE "In the CATCH block",!
    IF 1=exp.%IsA("%Exception.SystemException") {
        WRITE "System exception",!
        WRITE "Name: ", $ZCVT(exp.Name,"O","HTML"),!
        WRITE "Location: ",exp.Location,!
        WRITE "Code: "
    }
    ELSE { WRITE "Some other type of exception",! RETURN }
    WRITE exp.Code,!
    WRITE "Data: ",exp.Data,!
    RETURN
}
```

309 桁より大きな数値は数値文字列として表現できます。この値は数値ではなく文字列として保存されるため、丸めも <MAXNUMBER> エラーも適用されません。

ObjectScript

```
SET a="1"
FOR i=1:1:360 {SET a=a_"1" WRITE i+1," characters = ",a,! }
```

許可される最大桁数を超える数値になる指数は、〈MAXNUMBER〉エラーを生成します。指数の最大許容数は、その指数を受け入れる数のサイズによって異なります。1 桁の仮数の場合、最大指数は 307 または 308 です。

InterSystems IRIS の小数または IEEE 倍精度数を使用する際の大きな数字に関する考慮事項の詳細は、["インターシステムズ・アプリケーションでの数値の計算"](#) を参照してください。

3.3 オブジェクト

オブジェクト値は、メモリ内のオブジェクトのインスタンスを参照します。オブジェクト参照 (OREF) をローカル変数に割り当てることができます。

ObjectScript

```
SET myperson = ##class(Sample.Person).%New()
WRITE myperson
```

オブジェクトのインスタンスのメソッドやプロパティを参照するには、ドット構文を使用します。

ObjectScript

```
SET myperson.Name = "El Vez"
```

変数にオブジェクトが使用されているかどうかを判断するには、[\\$ISOBJECT](#) 関数を使用します。

ObjectScript

```
SET str = "A string"
SET myperson = ##class(Sample.Person).%New()

IF $ISOBJECT(myperson) {
    WRITE "myperson is an object.",!
} ELSE {
    WRITE "myperson is not an object."
}

IF $ISOBJECT(str) {
    WRITE "str is an object."
} ELSE {
    WRITE "str is not an object."
}
```

グローバルにはオブジェクト値を代入できません。代入すると、実行時エラーが発生します。

オブジェクト値を変数 (またはオブジェクトのプロパティ) に割り当てると、以下の例にあるように、オブジェクトの内部参照カウントが増分される悪影響があります。

ObjectScript

```
SET x = ##class(Sample.Person).%New()
WRITE x,!
SET y = ##class(Sample.Person).%New()
WRITE y,!
SET z = ##class(Sample.Person).%New()
WRITE z,!
```

オブジェクトの参照番号が 0 に到達した場合、システムは自動的にそのオブジェクトを破壊します ([%OnClose\(\)](#) コールバック・メソッドを呼び出し、そのオブジェクトをメモリから削除します)。

3.4 永続多次元配列 (グローバル)

グローバルは、多次元データベースのスパース配列です。グローバルは、グローバル変数名がキャレット記号 (^) で開始する以外は、他の配列タイプと同一です。データは、任意の数の添え字を付けてグローバルに格納できます。InterSystems IRIS の添え字にはタイプはありません。

以下は、グローバル使用の例です。グローバル ^x を設定すると、その値を検証できます。

ObjectScript

```
SET ^x = 10
WRITE "The value of ^x is: ", ^x,!
SET ^x(2,3,5) = 17
WRITE "The value of ^x(2,3,5) is: ", ^x(2,3,5)
```

グローバルの詳細は、“[多次元配列](#)” および “[グローバルの使用法](#)” を参照してください。

3.5 未定義の値

ObjectScript 変数を明示的に宣言または定義する必要はありません。変数に値を代入するとすぐ、この変数が定義されます。この最初の代入が行われるまで、この変数へのすべての参照は未定義となります。`$DATA` 関数を使用して、変数が定義済みであるか未定義であるかを判断できます。

`$DATA` は、1 つあるいは 2 つの引数を取ります。1 つの引数を取る場合、変数が値を持つかどうかをテストします。

ObjectScript

```
WRITE "Does ""MyVar"" exist?",!
IF $DATA(MyVar) {
    WRITE "It sure does!"
} ELSE {
    WRITE "It sure doesn't!"
}

SET MyVar = 10
WRITE !,!, "How about now?",!
IF $DATA(MyVar) {
    WRITE "It sure does!"
} ELSE {
    WRITE "It sure doesn't!"
}
```

`$DATA` は、変数が値を持つ場合は True (1) を、値を持たない場合 (データがない場合) は False (0) のブーリアン値を返します。2 つの引数を取る場合、テストを実行し、2 番目の引数の変数を、テストされた変数の値と同じ値に設定します。

ObjectScript

```
IF $DATA(Var1,Var2) {
    WRITE "Var1 has a value of ",Var2,".",!
} ELSE {
    WRITE "Var1 is undefined.",!
}

SET Var1 = 3
IF $DATA(Var1,Var2) {
    WRITE "Var1 has a value of ",Var2,".",!
} ELSE {
    WRITE "Var1 is undefined.",!
}
```

3.6 ブーリアン値

論理コマンドあるいは演算子と共に使用するなどの特定の場、値はブーリアン値 (真偽値) に解釈されます。このような場合、式がゼロ以外の数値に評価される場合は 1 (True)、ゼロの数値に評価される場合は 0 (False) と解釈されます。数値文字列はそれ自身の数値として評価されますが、非数値文字列は 0 (False) に評価されます。

例えば、以下の値は True になります。

ObjectScript

```
IF 1 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF 8.5 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF "1 banana" { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF 1+1 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF -7 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF +"007"=7 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
```

以下の値は False になります。

ObjectScript

```
IF 0 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF 3-3 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF "one banana" { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF "" { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF -0 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
IF "007"=7 { WRITE "evaluates as true",! }
    ELSE { WRITE "evaluates as false",! }
```

数値としての文字列の評価の詳細は、[“文字列から数値への変換”](#)を参照してください。

3.7 日付

ObjectScript には、組み込みの日付タイプはありません。その代わりに、文字列として表される日付値を操作または形式設定するための関数が多数あります。日付形式は以下のようになります。

テーブル 3-1: 日付形式

形式	説明
\$HOROLOG	これは、 \$HOROLOG (\$H) 特殊変数で返される形式です。これはコンマで区切られた 2 つの整数からなる文字列です。最初の整数は、1840 年 12 月 31 日以降の日数、2 番目は今日の日付の午前 0 時以降の秒数です。 \$HOROLOG では秒の小数部はサポートされません。 \$NOW 関数は、秒の小数部を使用した \$HOROLOG 形式の日付を提供します。InterSystems IRIS は、日付を \$HOROLOG 形式に設定したり、この形式の日付を検証したりするための多数の関数を用意しています。
ODBC Date	これは、ODBC と他の多くの外部表現で使用される形式です。文字列の形式は、“YYYY-MM-DD HH:MM:SS” となります。ODBC の日付値は順番に並べられます。つまり、ODBC 日付形式でデータをソートした場合、このデータは自動的に年代順に並べられます。
ロケールの日付	これは現在のロケールで使用される形式です。ロケールによって、日付の形式は以下のように異なります。 “アメリカ” の日付は mm/dd/yyyy (dateformat 1) の形式に設定されます。“ヨーロッパ” の日付は dd/mm/yyyy (dateformat 4) の形式に設定されます。csyw、deuw、engw、espw、eurw、fraw、itaw、mitw、ptbw、rusw、skyw、svnw、turw、ukrw を除くすべてのロケールでは、dateformat 1 を使用します。これらのロケールでは dateformat 4 を使用します。 アメリカの日付では、秒の小数部を表すためにピリオド (.) を小数点区切り文字として使用します。ヨーロッパの日付では、秒の小数部を表すためにコンマ (,) を小数点区切り文字として使用します。ただし、engw、eurw、skyw ではピリオドを使用します。 すべてのロケールでは、スラッシュ (/) を日付区切り文字として使用しますが、例外としてチェコ語 (csyw)、ロシア語 (rusw)、スロバキア語 (skyw)、スロベニア語 (svnw)、およびウクライナ語 (ukrw) ではピリオド (.) を使用します。
System Time	これは、 \$ZHOROLOG (\$ZH) 特殊変数で返される形式です。システムが実行されている秒数 (とその一部) を含む浮動小数点数です。InterSystems IRIS を停止して再起動すると、この数値がリセットされます。一般的にこの形式は、処理のタイミングを計り検証するために使用します。

以下は、異なる日付形式の使用方法についての例です。

ObjectScript

```

SET now = $HOROLOG
WRITE "Current time and date ($H): ",now,!

SET odbc = $ZDATETIME(now,3)
WRITE "Current time and date (ODBC): ",odbc,!

SET ldate = $ZDATETIME(now,-1)
WRITE "Current time and date in current locale format: ",ldate,!

SET time = $ZHOROLOG
WRITE "Current system time ($ZH): ",time,!

```


4

変数

変数は、値が格納される場所の名前です。ObjectScript では、変数にデータ型がなく、宣言の必要はありません。

一般的には、[SET](#) コマンドを使用して、変数に値を代入することによって変数を定義します。null 文字列 (") の値も変数に割り当てられます。コマンドおよび関数の多くは、変数を定義してから、その変数を参照する必要があります。変数が未定義である場合、既定においては、変数を参照すると `<UNDEFINED>` エラーが生成されます。`%SYSTEM.Process.Undefined()` メソッドを設定することで、未定義の変数を参照する際に `<UNDEFINED>` エラーを生成しないように InterSystems IRIS® Data Platform の動作を変更できます。

[READ](#) コマンド、[\\$INCREMENT](#) 関数、[\\$BIT](#) 関数、引数を 2 つ使用した形式の [\\$GET](#) 関数など、演算によっては、未定義の変数を使用できます。これらの関数は値を変数に代入します。[\\$DATA](#) 関数の場合、未定義または定義済みの変数を使用して、その状態を返します。

多くのコンピュータ言語とは異なり、ObjectScript では変数を宣言する必要はありません。変数は、値が割り当てられたときに作成されます。ObjectScript は "タイプのない" 言語です。変数はあらゆるタイプのデータを受け取ることができます。["#dim \(オプション\)"](#) および ["変数のタイプと変換"](#) を参照してください。

4.1 変数のカテゴリ

ObjectScript には、いくつかの変数の種類があります。

- ・ [ローカル変数](#)
- ・ [グローバル変数](#) (グローバル)
- ・ [プロセス・プライベート・グローバル変数](#) (PPG)
- ・ [i%property インスタンス変数](#) (別のページで説明)
- ・ [特殊変数](#) (システム変数)

4.2 ローカル変数

ローカル変数は、現在の InterSystems IRIS プロセス内のメモリで利用可能な変数です。作成したプロセスだけにアクセス可能となります。すべてのネームスペースからアクセスできます。つまり、プロセスがネームスペースを変更しても、変数は引き続き利用可能です。プロセスの終了時、そのプロセスのローカル変数はすべて削除されます。

4.2.1 名前付け規約

“識別子のルールとガイドライン”では詳細を説明していますが、簡潔に説明すると、以下のようになります。

- ローカル変数名の最初の文字は、必ず、文字、もしくはパーセント記号 (%) のいずれかです。
% 記号で始まる変数名は、“パーセント変数”と呼ばれ、その有効範囲の規則は異なります。コードでは、ローカル・パーセント変数の名前は、%Z または %z で初めます。他の名前はシステムで使用するために予約されています。
- ローカル変数名を含め、すべての変数名では、大文字と小文字が区別されます。例えば、MYVAR、MyVar および myvar は、3 つの異なるローカル変数となります。
- ローカル変数名は、現在のプロセスに一意であることが必要です。その他のプロセスは、同じ名前のローカル変数を持つことができます。
- プロセス・プライベート・グローバルまたはグローバルは、ローカル変数と同じ (見かけ上の) 名前を持つことができます。例えば、myvar、^|myvar、^myvar は、3 つの異なる変数となります。
- ローカル変数には、添え字を使用できます。“添え字に関する規則”を参照してください。

規則に準拠しないローカル変数名の場合、<SYNTAX> エラーが生成されます。ただし例外が 1 つあり、先頭がアンダースコア文字でその後に文字が続く変数名を作成しようとすると、<CALLBACK SYNTAX> エラーが生成されます。例えば、SET x=_a です。

4.2.2 ローカル変数の範囲

ObjectScript コードでは、ローカル変数はすべてパブリックです。このため、現在のコンテキスト内で ObjectScript コードのプロセスによって実行されるいずれの処理からもローカル変数にアクセスできます。ローカル変数値へのアクセスには、以下のような制約があります。

- NEW コマンドは、新しいローカル変数コンテキストを作成します。引数なしの NEW は、既存のローカル変数に何も定義されていない新しいコンテキストを作成します。NEW var は、ローカル変数 var が定義されていない新しいコンテキストを作成します。QUIT コマンドは、前のローカル変数コンテキストに戻します。
- プロシージャ・ブロック内では、ローカル変数は既定でプライベートです。プライベート・ローカル変数は、そのプロシージャ・ブロック内でのみ定義されます。

プロシージャ・ブロック内のローカル変数は以下のように動作します。

- プライベート変数。プロシージャ・ブロック内で使用されるローカル変数は、プライベート変数です。パブリック変数として宣言されるか % 変数でない限り、そのプロシージャ・ブロック内でのみ定義されます。既定では、スタジオで生成されるすべてのオブジェクト・メソッドがプロシージャ・ブロックを使用するため (プロシージャ・ブロック・クラス・キーワードは、クラス定義内で設定されます)、メソッドで生成されるすべての変数がプライベート変数です。プロシージャ・ブロック内のプライベート変数に NEW コマンドを使用することはできません。
- パブリック変数。プロシージャ・ブロックでは、ローカル変数のリストをパブリック変数として明示的に宣言できます。パブリック変数として宣言された変数の値にはプロシージャ・ブロックの外部からアクセスできます。パブリック変数のこのコンマ区切りリストには、存在しない変数や % 変数を含めることができます。プロシージャ・ブロック内のパブリック変数には NEW コマンドを使用できます。

2 つのローカル変数 var1 と var2 のパブリック変数リストは、MyProc(x,y) [var1,var2] PUBLIC { code body } のように指定します (PUBLIC キーワードはプロシージャがパブリックであることを指定するもので、パブリック変数リストとは関係ないことに注意してください)。パブリック変数は、コンマ区切りのリストで指定します。添え字なしのローカル変数のみを指定できます。パブリック変数リストに添え字なしの変数を指定すると、その添え字レベルもすべてパブリックになります。単純なオブジェクト参照 (OREF) のみを指定できます。パブリック変数リストに OREF を指定すると、そのオブジェクト・プロパティもすべてパブリックになります。パブリック変数リストには、未定義の変数を含めることができます。

- % 変数。名前が“%”で始まるローカル変数は、自動的にパブリック変数として宣言されます。これにより、プロセス内のすべてのコードから参照できる変数を、パブリックとして明示的にリストすることなく定義できます。“%Z”または“%z”で始まる変数のみをアプリケーション・コードで使用できます。その他すべての%変数は、規則に従ってシステムで使用するために予約されています。この規則の詳細は、“[識別子のルールとガイドライン](#)”を参照してください。プロシージャ・ブロック内の%変数には **NEW** コマンドを使用できます。

これらのメカニズムの詳細は、“[プロシージャ変数](#)”を参照してください。

- ・ **XECUTE** コマンドは、既定でローカル変数をパブリックとして定義します。XECUTE コマンド内で、明示的にローカル変数をプライベートとして定義できます。ローカル変数をプライベートまたはパブリックのいずれかとして明示的に定義する方法の詳細は、“[XECUTE](#)”を参照してください。

引数の指定なしで WRITE または ZWRITE コマンドを使用すると、現在定義されているローカル変数をすべてリストできます。また、KILL コマンドを使用して、ローカル変数を削除できます。

4.3 グローバル

グローバルは、InterSystems IRIS データベースに自動的に格納される特殊な変数です。これは固有のネームスペースにマッピングされるので、拡張参照を使用しない限り、そのネームスペース内でのみアクセスできます。グローバルは任意のプロセスからアクセスできます。グローバルは、そのグローバルを作成したプロセスの終了後も存続します。グローバルは、明示的に削除されるまで存続します。

ObjectScript コードでは、他の変数と同様にグローバルを使用できます。構文的には、グローバル名は先頭のキャレット (^) 文字により識別されます。

ObjectScript

```
SET mylocal = "This is a local variable"
SET ^myglobal = "This is a global stored in the current namespace"
```

名前の詳細は、“[識別子のルールとガイドライン](#)”を参照してください。概要は以下のようになります。

- ・ キャレットの後の最初の文字は必ず、文字、もしくはパーセント記号 (%) のいずれかです。
^% で始まるグローバル名は、“パーセント変数”と呼ばれ、すべてのネームスペースで使用できます。コードでは、これらの変数の名前は、^%Z または ^%z で始めます。他の名前はシステムで使用するために予約されています。
- ・ グローバル名の 2 番目およびその後に続く文字は、文字、数字、ピリオド記号 (.) になります。ピリオドを、名前の最初または最後の文字にすることはできません。
- ・ ローカル変数とは異なり、グローバル名には、Unicode 文字 (ASCII 255 よりもコード値が大きい文字) を使用できません。グローバル名に Unicode 文字を使用すると <WIDE CHAR> エラーが発生します。
- ・ グローバル名は、大文字と小文字を区別します。
- ・ グローバル名は、そのネームスペース内で重複しないようにします。
- ・ グローバル名は、接頭語を除き最大 31 文字です。31 文字より長い名前を指定することもできますが、最初の 31 文字のみが使用されます。このため、グローバル名は最初の 31 文字内で一意である必要があります。
- ・ その他の変数と同様に、グローバルには[添え字](#)を使用できます。“[添え字に関する規則](#)”を参照してください。
- ・ インターシステムズで使用する目的で保持されているグローバル名もあります。“[回避する必要があるグローバル変数名](#)”を参照してください。

別のネームスペースにあるグローバルを参照するには、[拡張参照](#)を使用できます。この構文には、キャレット文字のすぐ後に垂直バーまたは角括弧のペアを使用します (^|"samples"|myglobal または ^|" "|myglobal など)。これらの構文をプロセス・プライベート・グローバルと混同しないでください。

グローバルの詳細は、“[グローバルの使用法](#)”を参照してください。

4.4 プロセス・プライベート・グローバル

プロセス・プライベート・グローバルは、これを作成したプロセスによってのみアクセス可能な変数です。プロセスの終了時、プロセス・プライベート・グローバルはすべて削除されます。

- ・ プロセス固有：プロセス・プライベート・グローバルは、作成元のプロセスからのみアクセス可能で、そのプロセスが完了すると存在しなくなります。これは、ローカル変数と同様です。
- ・ 常にパブリック：プロセス・プライベート・グローバルは常にパブリック変数です。これは、グローバル変数と同様です。
- ・ ネームスペース非依存：プロセス・プライベート・グローバルは、すべてのネームスペースからアクセスできます。
- ・ 引数なしの KILL、NEW、WRITE、または ZWRITE の影響を受けない：プロセス・プライベート・グローバルを KILL、WRITE、または ZWRITE の引数として指定できます。これは、グローバル変数と同様です。

プロセス・プライベート・グローバルは、サイズの大きなデータ値に使用することができます。多くの場合、Mgr/Temp ディレクトリで使用するための置換値の役割を果たし、プロセスの終了時に自動クリーンアップを行います。

4.4.1 名前付け規約

プロセス・プライベート・グローバル名は、次のいずれかの形式となります。

```
^|name
^|"^"|name
^{ "|^"|name
^[ "|^", "|^"|name
```

これらの 4 つの接頭語は同等で、4 つはいずれも同じプロセス・プライベート・グローバルを参照します。最初の形式 (^|name) がもっとも一般的で、新しいコードに推奨されるものです。2 番目 ~ 4 番目の形式は、グローバルを定義する既存のコードとの互換性のために提供されています。

接頭語を除き、プロセス・プライベート・グローバルは、“[識別子のルールとガイドライン](#)”で説明するように、通常のグローバルと同じ名前付け規約を使用します。概要は以下のようになります。

- ・ 最初の文字 (2 番目の垂直バーの後) は、文字、もしくはパーセント記号 (%) のいずれかにする必要があります。
% で始まるプロセス・プライベート変数名は、“パーセント変数”と呼ばれ、その[有効範囲の規則](#)は異なります。コードでは、これらの変数の名前は、%z または %z で始めます。他の名前はシステムで使用するために予約されています。例えば、^| |%zmyvar のようになります。
- ・ ローカル変数とは異なり、グローバル名 (プロセス・プライベート・グローバルを含む) には、Unicode 文字 (ASCII 255 よりもコード値が大きい文字) を使用できません。プロセス・プライベート・グローバル名に Unicode 文字を使用すると <WIDE CHAR> エラーが発生します。
- ・ このプロセス・プライベート・グローバル名を含め、すべての変数名では、大文字と小文字が区別されます。
- ・ プロセス・プライベート・グローバル名は、プロセス内で一意である必要があります。

- ・ ローカル変数とは異なり、プロセス・プライベート・グローバル名は、接頭語を除き最大 31 文字です。31 文字より長い名前を指定することもできますが、最初の 31 文字のみが使用されます。このため、プロセス・プライベート・グローバル名は最初の 31 文字内で一意である必要があります。
- ・ その他の変数と同様に、プロセス・プライベート・グローバルには添え字を使用できます。“添え字に関する規則”を参照してください。

4.4.2 プロセス・プライベート・グローバルのリスト

`^$GLOBAL()` の `^$||GLOBAL()` 構文形式を使用して、現在のプロセスに属するプロセス・プライベート・グローバルに関する情報を返すことができます。

`^GETPPGINFO` ルーチンを使用して、現在のすべてのプロセス・プライベート・グローバルの名前およびそれらのスペース割り当てをブロック単位で表示できます。`^GETPPGINFO` はプロセス・プライベート・グローバルの添え字も値もリストしません。プロセス ID (pid) を指定することで、特定プロセスのプロセス・プライベート・グローバルを表示できます。また、ワイルドカード文字列 (*) を指定することで、すべてのプロセスのプロセス・プライベート・グローバルを表示できます。`^GETPPGINFO` を呼び出すには `%SYS` ネームスペースで作業する必要があります。

以下の例では、`^GETPPGINFO` を使用して、現在のすべてのプロセスのプロセス・プライベート・グローバルをリストします。

ObjectScript

```
SET ^||flintstones(1)="Fred"
SET ^||flintstones(2)="Wilma"
NEW $NAMESPACE
SET $NAMESPACE="%SYS"
DO ^GETPPGINFO(" *")
```

`^GETPPGINFO` ルーチンは、以下のように引数を取ります。

ObjectScript

```
do ^GETPPGINFO("pdf","options","outfile")
```

これらの引数は以下のとおりです。

- ・ pdf はプロセス ID または * ワイルドカードにすることができます。
- ・ options は次の文字列のあらゆる組み合わせにすることができます。
 - b (バイト数で値を返す)
 - Mnn (nn 個以上のブロックを使用するプロセス・プライベート・グローバルがあるプロセスのみをリストする)

プロセス・プライベート・グローバルがないプロセスをリストに含めるには `M0` を使用します。

プロセス・プライベート・グローバルがないプロセスはリストから除外し、グローバル・ディレクトリ・ブロックのみがあるプロセスを含めるには `M1` を使用します (これが既定です)。

プロセス・プライベート・グローバルがないプロセスをリストから除外し、グローバル・ディレクトリ・ブロックのみがあるプロセスも除外するには `M2` を使用します。

- s (outfile と共に使用して画面表示を抑制)
- T (プロセスの合計数のみを表示)
- ・ outfile は、`^GETPPGINFO` 出力を受け取る CSV (コンマ区切り値) 形式ファイルのファイル・パスです。

次の例では、プロセス・プライベート・グローバルを `ppgout` という名前の出力ファイルに書き込みます。S オプションで画面表示を抑制し、出力を 500 ブロック以上使用するプロセス・プライベート・グローバルがあるプロセスのみを M500 オプションで表示します。

ObjectScript

```
NEW $NAMESPACE
SET $NAMESPACE="%SYS"
DO ^GETPPGINFO("","SM500","/home/myspace/ppgout")
```

4.5 添え字に関する規則

ローカル変数、プロセス・プライベート変数、およびグローバル変数はすべて、添え字を使用できます。いずれの場合も、以下の点に注意してください。

- ・ 添え字は数値または文字列にすることができます。添え字には Unicode 文字をはじめ、あらゆる文字を含めることができます。有効な数値の添え字は、正と負の数、ゼロ、および小数を含みます。
- ・ 空の文字列 (" ") は有効な添え字ではありません。
- ・ 添え字の値は、大文字と小文字で区別されます。
- ・ 数値の添え字はキャノニック形式に変換されます。そのため、例えば、グローバル・ノード `^a(7)`、`^a(007)`、`^a(7.000)`、および `^a(7.)` はすべて同じです。これらすべてにおいて添え字は実際には同一であるためです。
- ・ 文字列の添え字はキャノニック形式に変換されません。そのため、例えば `^a("7")`、`^a("007")`、`^a("7.000")`、および `^a("7.")` では、添え字はすべて異なるため、これらはすべて異なるグローバル・ノードになります。また、`^a("7")` および `^a(7)` はどちらも同じグローバル・ノードを参照します。これらの添え字は同じであるためです。
- ・ 添え字の長さ、添え字レベルの数には制限があります。“添え字の制限”を参照してください。

また、“一般的なシステム制限”も参照してください。

ロック名の添え字は変数の添え字と同一の規約に従います。

`$QSUBSCRIPT` 関数を使って、指定した変数のコンポーネント(名前と添え字)を返したり、`$QLENGTH` 関数を使って、添え字レベルの数を返すことができます。

4.6 変数のタイプと変換

ObjectScript の変数には決まった形式はありません。つまり、特定のデータ型はありません(これは、JavaScript、VBScript、および Document Data Base/JSON にも当てはまります)。したがって、文字列値を変数に割り当て、その後同じ変数に数値を割り当てることができます。最適化として、InterSystems IRIS は文字列、整数、数字、オブジェクトに異なる内部表現を使用しますが、アプリケーション・プログラマからは見えません。InterSystems IRIS は、使用されるコンテキストを基にして、変数の値を自動的に変換(または解釈)します。

以下はその例です。

ObjectScript

```
// set some variables
SET a = "This is a string"
SET b = "3 little pigs"
SET int = 22
SET num = 2.2
SET obj = ##class(Sample.Person).%New()
```

```
// Display them
WRITE "Here are the variables themselves: ",!
WRITE "a: ",a,!
WRITE "b: ",b,!
WRITE "int: ",int,!
WRITE "num: ",num,!
WRITE "obj: ",obj,!

// Now use them as other "types"
WRITE "Here are the numeric interpretation of",!
WRITE "a, b, and obj: ",!
WRITE "+a: ",+a,!
WRITE "+b: ",+b,!
WRITE "+obj: ",+obj,!

WRITE "Here are concatenations of int and num:",!
WRITE "Concatenating int: ", "I found " _ int _ " apples.",!
WRITE "Concatenating num: ", "There are " _ num _ " pounds per kilogram.",!
```

InterSystems IRIS は、以下のように値を変換します。

テーブル 4-1: ObjectScript の型の変換規則

変換元	変換先	規則
番号	文字列	数値を表す文字列が使用され、前の例では変数 num は 2.2 のようになります。
文字列	数値	文字列の先頭文字は数値リテラルとして解釈されます。これについては、“ 文字列から数値への変換 ”で説明しています。例えば、“-1.20abc”は -1.2 と解釈され、“abc123”は 0 と解釈されます。
オブジェクト	数値	指定されたオブジェクト参照の内部オブジェクト・インスタンス番号が使用されます。その値は整数です。
オブジェクト	文字列	n@cls 形式の文字列が使用されます。n は内部オブジェクトのインスタンス番号で、cls は指定されたオブジェクトのクラス名です。
数値	オブジェクト	許可されていません。
文字列	オブジェクト	許可されていません。

4.7 #dim (オプション)

他の言語と異なり、ObjectScript で変数を宣言する必要はありません。ただし、[#dim](#) プリプロセッサ指示文をコードの記述の補助として使用できます。IDE ではこれを利用してコード入力サポートを提供します。

#dim の構文形式は、以下のとおりです。

ObjectScript

```
#dim VariableName As DataTypeName
#dim VariableName As List Of DataTypeName
#dim VariableName As Array Of DataTypeName
```

ここで、VariableName は、データ型を指定する変数で、DataTypeName は、そのデータ型を指定します。

4.8 グローバル変数とジャーナリング

InterSystems IRIS では、グローバルの SET と KILL は、ジャーナル化されたトランザクション・イベントとして扱われます。このようなイベントでは、トランザクションをロールバックすると、これらの処理が取り消されて元の状態に戻ります。他のプロセスからアクセスできないように、変更したトランザクションがコミットされるまでロックを使用できます。詳細は、“[トランザクション処理](#)”を参照してください。

対照的に、ローカル変数またはプロセス・プライベート・グローバル変数の SET または KILL は、ジャーナル化されたトランザクション・イベントとして扱われません。このようなイベントでは、トランザクションをロールバックしても、これらの処理には何の影響もありません。

4.9 特殊変数

ObjectScript には、特定のシステム情報をアプリケーションで有効にするために使用する特殊変数（システム変数ともいいます）があります。特殊変数はすべて InterSystems IRIS で供給される変数で、\$ 記号の接頭辞を使用した名前を使用します。ユーザがその他の特殊変数を定義することはできません。特殊変数はすべてのネームスペースで使用できます。

その値は、ユーザの現在の動作環境に合わせて設定されます。特殊変数の一部は、はじめ NULL 文字列 (“”) に設定されているため、特殊変数を参照しても <UNDEFINED> エラーは生成されません。特殊変数の値は現在のプロセスに固有のもので、別のプロセスからアクセスできません。

ユーザは、SET コマンドを使用して一部の特殊変数を設定できますが、それ以外の特殊変数をユーザが変更することはできません。詳細は、個々の特殊変数を参照してください。

次の例では、特殊変数 \$HOROLOGY を使用します。

ObjectScript

```
SET starttime = $HOROLOGY
HANG 5
WRITE !,$ZDATETIME(starttime)
WRITE !,$ZDATETIME($HOROLOGY)
```

特殊変数 \$HOROLOGY は、現在のシステムの日付と日時を格納します。SET コマンドはこの特殊変数を使用して、ユーザ定義のローカル変数 starttime にこの値を設定します。HANG コマンドは、プログラムを 5 秒間停止します。最後に、2 つの \$ZDATETIME 関数は、starttime と現在のシステムの日付と時間をユーザが読み取れる形式で返します。

以下はその他の特殊変数の例です。

ObjectScript

```
WRITE !,"$JOB = ",$JOB // Current process ID
WRITE !,"$ZVERSION = ",$ZVERSION // Version info
```

多くの特殊変数は読み取り専用で、SET コマンドを使用して設定することができません。\$DEVICE などのその他の特殊変数は、読み取り/書き込み可能で、SET コマンドで設定できます。

特殊変数には、添え字を使用できません。特殊変数は、\$INCREMENT 関数を使用してインクリメントすることや KILL コマンドを使用して削除することができません。特殊変数は、“[表示 \(書き込み\) コマンド](#)”で説明するように、WRITE、ZWRITE、ZZWRITE、または ZZDUMP コマンドを使用して表示できます。

特殊変数のリストと詳細説明は、“[ObjectScript リファレンス](#)”を参照してください。

5

演算子と式

ObjectScript は、多様な演算子をサポートします。数理演算動作、論理比較などさまざまな動作を実行します。演算子は、最終的に 1 つの値に評価される変数や他のエンティティである式に対して作用します。ここでは、式と演算子について説明します。

5.1 演算子と式の概要

演算子とはシンボル文字で、対応するオペランドで実行する処理を指定するものです。各オペランドは、1 つ以上の式または式アトムから構成されます。演算子とその演算子に対応するオペランドと一緒に使用する場合、以下の形式になります。

[operand] 演算子 operand

演算子の中には、1 つのオペランドのみを使用する単項演算子があります。2 つのオペランドを使用するものは、二項演算子といいます。

式は、演算子とそこで使用されるオペランドで構成されます。

5.1.1 代入

ObjectScript の [SET](#) コマンドは、代入演算子(=)を併用して変数に値を代入します。代入コマンドの右側に式が置かれます。

ObjectScript

```
SET value = 0
SET value = a + b
```

ObjectScript では、代入コマンドの左側にも特定の関数を使用できます。

ObjectScript

```
SET pies = "apple,banana,cherry"
WRITE "Before: ",pies,!

// set the 3rd comma-delimited piece of pies to coconut
SET $Piece(pies,"",3) = "coconut"
WRITE "After: ",pies
```

5.2 演算子の優先順位

ObjectScript で演算子の評価順序は、必ず左から右です。したがって、式の演算は表示された順番で実行されます。これは、特定の演算子の優先順位が他の演算子よりも高くなることもある他の言語と異なります。式で明示的に小括弧を使用して、特定の演算子を先に処理させることができます。

ObjectScript

```
WRITE "1 + 2 * 3 = ", 1 + 2 * 3,! // returns 9
WRITE "2 * 3 + 1 = ", 2 * 3 + 1,! // returns 7
WRITE "1 + (2 * 3) = ", 1 + (2 * 3),! // returns 7
WRITE "2 * (3 + 1) = ", 2 * (3 + 1),! // returns 8
```

InterSystems SQL では、[演算子の優先順位](#)を構成でき、ObjectScript の演算子の優先順位と一致させる（または一致しないようにする）ことができます。

5.2.1 単項マイナス演算子

ObjectScript は、二項算術演算子より単項マイナス演算子を優先します。ObjectScript はまず数値式を検査し、単項マイナス演算子を実行します。その後、式を評価して結果を算出します。

ObjectScript

```
WRITE -123 - 3,! // returns -126
WRITE -123 + -3,! // returns -126
WRITE -(123 - 3),! // returns -120
```

5.2.2 括弧と優先順位

式の評価の順序は、それぞれの式を対の小括弧で入れ子にして変更できます。小括弧は、囲んだ式(算術式と関係式の両方)をグループ化し、ObjectScript が式で実行する演算の順序を制御します。以下の例を考えてみます。

ObjectScript

```
SET TorF = ((4 + 7) > (6 + 6)) // False (0)
WRITE TorF
```

上記では、小括弧で 4 と 7、および 6 と 6 を加算しているため、論理式は $11 > 12$ となり、結果は False になります。以下のコードと比較します。

ObjectScript

```
SET Value = (4 + 7 > 6 + 6) // 7
WRITE Value
```

この場合、演算の優先順位は左から右になります。したがって、最初に 4 と 7 を加算します。その合計の 11 と 6 を比較し、11 は 6 より大きくなるため、論理演算の結果は 1 (True) になります。その後、1 に 6 を加算するため、結果は 7 になります。

優先順位により結果のタイプが異なることに注意してください。上記の例で、最初の式の演算は最終的にブーリアン値を返し、2 番目の式は数値を返します。

以下の例では、複数レベルの入れ子を示します。

ObjectScript

```
WRITE 1+2*3-4*5,! // returns 25
WRITE 1+(2*3)-4*5,! // returns 15
WRITE 1+(2*(3-4))*5,! // returns -5
WRITE 1+(((2*3)-4)*5),! // returns 11
```

内側の入れ子の式から 1 レベルずつ外側に進み、各レベルで左から右へと式が評価されます。

Tip ヒン 極めて単純な ObjectScript 式を除いて、すべての式全体を括弧で囲むことをお勧めします。これにより、評価のト 順序のあいまいさが解消され、コードの本来の意図について今後疑問が出ることもなくなります。

例えば、すべての演算子と同様に && 演算子は、左から右の順に実行されるため、以下のコード例の最後の文は 0 に評価されます。

ObjectScript

```
SET x = 3
SET y = 2
IF x && y = 2 {
    WRITE "True",! }
ELSE {
    WRITE "False",! }
```

これは評価が次のように実行されるからです。

1. 最初に、x が定義されていて、0 以外の値であるかどうかを確認されます。x は 3 なので評価は続行されます。
2. 次に、y が定義されていて、0 以外の値であるかどうかを確認されます。y は 2 なので評価は続行されます。
3. 次に、3 && 2 の値が評価されます。3 も 2 も 0 ではないので、この式は True で、1 に評価されます。
4. 次に、返された値を 2 と比較します。1 は 2 ではないので、この評価は 0 を返します。

多くのプログラミング言語に精通している人にとって、これは予想外の結果です。x に 0 でない値が定義されていて、y が 2 のときに True を返すことを意図している場合は、次のように括弧が必要です。

ObjectScript

```
SET x = 3
SET y = 2
IF x && (y = 2) {
    WRITE "True",! }
ELSE {
    WRITE "False",! }
```

5.2.3 関数と優先順位

関数など、式のタイプによって副次的作用が発生する場合があります。以下の論理式を考えてみます。

ObjectScript

```
IF var1 = ($$ONE + (var2 * 5)) {
    DO ^Test
}
```

ObjectScript は、最初に var1、次に関数 \$\$ONE、その次に var2 を評価します。その後、var2 を 5 倍し、最後に ObjectScript は、加算の結果が var1 の値に等しいかどうかをテストします。等しい場合、DO コマンドを実行して Test ルーチン呼び出します。

別の例として、以下の論理式を考えてみます。

ObjectScript

```
SET var8=25,var7=23
IF var8 = 25 * (var7 < 24) {
    WRITE !,"True" }
ELSE {
    WRITE !,"False" }
```

ObjectScript は、厳密に左から右の順に式を評価します。プログラマは、括弧を使用して優先順位を確立する必要があります。この場合、ObjectScript はまず var8=25 を評価し、結果が 1 になります。次に、この 1 に括弧内の式の結果を乗算します。var7 は、24 より小さいので、括弧内の式は、1 と評価されます。したがって、ObjectScript は、1 * 1 の乗算を行い、結果は 1 (True) となります。

5.3 文字列から数値への変換

文字列は数値、部分的数値、または非数値となり得ます。

- ・ 数値文字列はすべて数字で構成されます。例えば、"123"、"+123"、".123"、"++0007"、"-0" となります。
- ・ 部分的数値の文字列とは、数字で始まり、非数値の文字が続く文字列です。例えば、"3 blind mice"、"-12 degrees" となります。
- ・ 非数値の文字列は数値ではない文字で始まります。例えば、" 123"、"the 3 blind mice"、"three blind mice" となります。

5.3.1 数値文字列

数値文字列または部分的文字列を算術式で使用する場合、文字列は数値として解釈されます。文字列で左から右に数値文字をスキャンしてこの数値を取得し、[数値リテラル](#)と解釈できる開始文字の最長シーケンスを見つけます。以下の文字が可能です。

- ・ 0 から 9 までの数字。
- ・ PlusSign プロパティ値および MinusSign プロパティ値。既定では、それらは + および - の文字となりますが、ロケールに依存します。%SYS.NLS.Format.GetFormatItem() メソッドを使用して、現在の設定を返します。
- ・ DecimalSeparator プロパティ値。既定では、これは . の文字となりますが、ロケールに依存します。%SYS.NLS.Format.GetFormatItem() メソッドを使用して、現在の設定を返します。
- ・ 文字の e および E は、シーケンス内で 4E3 などの[科学的記数法](#)を表す場合に、数値文字列の一部を含むことができます。

NumericGroupSeparator プロパティ値 (既定では、, の文字) は、数値文字とは見なされないことに注意してください。したがって、文字列 "123,456" は部分的数値文字列となり、数 "123" に分解されます。

数値文字列および部分的数値文字列は、算術演算 (加算および減算) および「より大きい/より小さい」などの比較演算 (<, >, <=, >=) に優先して、[キャノニック形式](#)に変換されます。数値文字列は等値比較 (=, '=) に優先して、キャノニック形式へ変換されません。この理由は、それらの演算子が文字列比較にも使用されるためです。

以下の例は、数値文字列の算術比較を示しています。

ObjectScript

```
WRITE "3" + 4,!           // returns 7
WRITE "003.0" + 4,!       // returns 7
WRITE "++--3" + 4,!       // returns 7
WRITE "3 blind mice" + 4,! // returns 7
```

以下の例は、数値文字列の < (より小さい) 比較を示しています。

ObjectScript

```
WRITE "3" < 4,!           // returns 1
WRITE "003.0" < 4,!       // returns 1
WRITE "++--3" < 4,!       // returns 1
WRITE "3 blind mice" < 4,! // returns 1
```

以下の例は、数値文字列の <= 比較を示しています。

ObjectScript

```
WRITE "4" <= 4,!           // returns 1
WRITE "004.0" <= 4,!       // returns 1
WRITE "++--4" <= 4,!       // returns 1
WRITE "4 horsemen" <= 4,!   // returns 1
```

以下の例は、数値文字列の等値比較を示しています。非キャノニック形式の数値文字列は、数値ではなく、文字列として比較されます。-0 は非キャノニック形式の数値文字列です。したがって、数値ではなく、文字列として比較されます。

ObjectScript

```
WRITE "4" = 4.00,!         // returns 1
WRITE "004.0" = 4,!        // returns 0
WRITE "++--4" = 4,!        // returns 0
WRITE "4 horsemen" = 4,!    // returns 0
WRITE "-4" = -4,!          // returns 1
WRITE "0" = 0,!            // returns 1
WRITE "-0" = 0,!           // returns 0
WRITE "-0" = -0,!          // returns 0
```

5.3.2 非数値文字列

文字列の先頭の文字が数値文字ではない場合、すべての算術演算に対する文字列の数値は 0 となります。<, >, >=, <=, <', >', および >= による比較演算でも、数値ではない文字列は 0 として扱われます。等号は数値的等値演算子および文字列比較演算子の両方として使用されるので、= および ' の演算では文字列比較が優先されます。PlusSign プロパティ値 (既定では +) を追加することで、文字列の数値的評価を強制することができます (例: "+123")。x および y が異なる非数値文字列 (x="Fred", y="Wilma" など) である場合、結果は以下の論理値となります。

x, y	x, x	+x, y	+x, +y	+x, +x
x=y は False	x=x は True	+x=y は False	+x=+y は True	+x=+x は True
x'=y は True	x=y は False	+x'=y は True	+x'='+y は False	+x'='+x は False
x<y は FALSE	x<x は FALSE	+x<y は FALSE	+x<+y は FALSE	+x<+x は FALSE
x<=y は TRUE	x<=x は TRUE	+x<=y は TRUE	+x<=+y は TRUE	+x<=+x は TRUE

5.4 式

ObjectScript 式は、値を算出するために評価される 1 つ以上のトークン です。最も単純な式は、リテラルあるいは変数です。

ObjectScript

```
SET expr = 22
SET expr = "hello"
SET expr = x
```

配列、演算子、多くの ObjectScript 関数の 1 つを使用して、さらに複雑な式を記述できます。

ObjectScript

```
SET expr = +x
SET expr = x + 22
SET expr = array(1)
SET expr = ^data("x",1)
SET expr = $Length(x)
```

式は、オブジェクト・プロパティ、インスタンス・メソッド呼び出し、クラス・メソッド呼び出しから構成されます。

ObjectScript

```
SET expr = person.Name
SET expr = obj.Add(1,2)
SET expr = ##class(MyApp.MyClass).Method()
```

ルーチン呼び出しの前に \$\$ を置き、ObjectScript ルーチン呼び出しを式で直接実行できます。

ObjectScript

```
SET expr = $$MyFunc^MyRoutine(1)
```

式は、返す値の種類で分類できます。

- ・ 算術式は算術演算子を含み、オペランドを数値として解釈し、数値結果を算出します。

ObjectScript

```
SET expr = 1 + 2
SET expr = +x
SET expr = a + b
```

算術式で使用される文字列は、数値として評価されます (有効な数値がない場合は 0 となります)。また、単項加算演算子 (+) を使用すると、文字列値を数値に暗黙に変換します。

- ・ 文字列式は文字列演算子を含み、オペランドを文字列として解釈し、文字列の結果を返します。

ObjectScript

```
SET expr = "hello"
SET expr = "hello" _ x
```

- ・ 論理式は関係演算子と論理演算子を含み、オペランドを論理解釈し、True (1) あるいは False (0) のブーリアン値を返します。

ObjectScript

```
SET expr = 1 && 0
SET expr = a && b
SET expr = a > b
```

- ・ オブジェクト式 は、結果としてオブジェクト参照を生成します。

ObjectScript

```
SET expr = object
SET expr = employee.Company
SET expr = ##class(Person).%New()
```

5.4.1 論理式

論理式は、[論理演算子](#)、[数値関係演算子](#)、[文字列関係演算子](#) を使用します。式を評価し、1 (True) または 0 (False) のブーリアン値を返します。論理式は、通常、以下を使用します。

- ・ [IF コマンド](#)
- ・ [\\$SELECT 関数](#)
- ・ [後置条件式](#)

ブーリアン・テストでは、ゼロ以外の数値に評価される式はいずれもブーリアン 1 (True) の値を返します。ゼロの数値に評価される式はいずれもブーリアン 0 (False) の値を返します。InterSystems IRIS® データ・プラットフォームでは、非数値文字列がゼロの数値を有しているものとして評価します。詳細は、“[文字列から数値への変換](#)” を参照してください。

論理演算子を使用することにより、複数のブーリアン論理式を組み合わせることができます。すべての InterSystems IRIS の式と同様に、これらは厳密に左から順に評価されます。論理演算子のタイプには、正規論理演算子 (& および !) と簡易論理演算子 (&& および ||) の 2 つがあります。

正規論理演算子を使用して論理式を組み合わせる場合、InterSystems IRIS では、すべての式の評価が完了する前にブーリアン値の結果が分かっていたとしても、指定された式をすべて評価します。これにより、すべての式の妥当性が保証されます。

簡易論理演算子を使用して論理式を組み合わせる場合、InterSystems IRIS ではブーリアン値結果の割り出しに必要な式のみを評価します。例えば、複数の AND テストをする場合、0 を返す最初の式によって全体のブーリアン値結果が決まります。この式の右側にある論理式は評価されません。これにより、不必要な時間的浪費となる式の評価を回避できます。

[コンマ区切りリスト](#)を引数値として指定できるコマンドもあります。この場合、InterSystems IRIS ではリストの各引数を独立コマンド文のように扱います。したがって、IF x=7,y=4,z=2 は IF x=7 THEN IF y=4 THEN IF z=2 と解析され、簡易論理演算子の文の IF (x=7)&&(y=4)&&(z=2) と機能的に同一となります。

以下の例では、IF テストにて正規論理演算子 (&) を使用しています。したがって、最初の関数が 0 (False) を返すため、自動的に式全体の結果が False になりますが、すべての関数が実行されます。

ObjectScript

```
LogExp
  IF $$One() & $$Two() {
    WRITE !,"Expression is TRUE." }
  ELSE {
    WRITE !,"Expression is FALSE." }
One()
  WRITE !,"one"
  QUIT 0
Two()
  WRITE !,"two"
  QUIT 1
```

以下の例では、IF テストにて簡易論理演算子 (&&) を使用しています。したがって、最初の関数の実行により 0 (False) を返すため、自動的に式全体の結果が False になります。2 番目の関数は実行されません。

ObjectScript

```
LogExp
  IF $$One() && $$Two() {
    WRITE !,"Expression is TRUE." }
  ELSE {
    WRITE !,"Expression is FALSE." }
One()
  WRITE !,"one"
  QUIT 0
Two()
  WRITE !,"two"
  QUIT 1
```

以下の例では、IF テストにてコンマ区切り引数を指定しています。コンマは論理演算子ではありませんが、簡易 && 論理演算子の指定と同じ効果があります。最初の関数の実行により 0 (False) を返すため、自動的に式全体の結果が False になります。2 番目の関数は実行されません。

ObjectScript

```
LogExp
  IF $$One(),$$Two() {
    WRITE !,"Expression is TRUE." }
  ELSE {
    WRITE !,"Expression is FALSE." }
One()
  WRITE !,"one"
  QUIT 0
Two()
  WRITE !,"two"
  QUIT 1
```

5.5 算術演算子

算術演算子は、オペランドを数値として解釈し、数値結果を生成します。文字列について演算する場合、算術演算子は、“[文字列から数値への変換](#)” で説明したルールに従って、文字列をその数値として扱います。ObjectScript は、以下の算術演算子を提供します。

単項プラス (+)

単項プラス演算子 (+) は、単一のオペランドを数値として解釈します。数値として解釈できない文字に遭遇するまで、文字列の文字を数値として順番に解析することで、これを実行します。そして、適格な数値に変換した文字列の先頭部分を返します (そのような解釈が可能でない場合は 0 を返します)。以下に例を示します。

ObjectScript

```
WRITE + "32 dollars and 64 cents" // 32
```

詳細は、“[単項プラス \(+\)](#)” のリファレンス・ページを参照してください。

単項マイナス (-)

単項マイナス演算子 (-) は、数値として解釈されるオペランドの符号を反転します。以下に例を示します。

ObjectScript

```
SET x = -60
WRITE " x: ", x,! // -60
WRITE "-x: ", -x,! // 60
```

ObjectScript は、二項 (2 オペランド) 算術演算子より単項マイナス演算子を優先します。

詳細は、“[単項マイナス \(-\)](#)” のリファレンス・ページを参照してください。

数値式の絶対値を返すには、[\\$ZABS](#) 関数を使用します。

加算 (+)

加算演算子は、2 つの数値を加算します。以下に例を示します。

ObjectScript

```
WRITE 2936.22 + 301.45 // 3237.67
```

詳細は、“[加算 \(+\)](#)” のリファレンス・ページを参照してください。

減算 (-)

減算演算子は、ある数値を別の数値から減算します。以下に例を示します。

ObjectScript

```
WRITE 2936.22 - 301.45 // 2634.77
```

詳細は、“[減算 \(-\)](#)” のリファレンス・ページを参照してください。

乗算 (*)

乗算演算子は、2 つの数値を乗算します。以下に例を示します。

ObjectScript

```
WRITE 9 * 5.5 // 49.5
```

詳細は、“[乗算 \(*\)](#)” のリファレンス・ページを参照してください。

除算 (/)

除算演算子は、ある数値を別の数値で除算します。以下に例を示します。

ObjectScript

```
WRITE 9 / 5.5 // 1.6363636363636364
```

詳細は、“[除算 \(/\)](#)” のリファレンス・ページを参照してください。

整数除算 (¥)

整数除算演算子は、ある数値を別の数値で除算し、小数値を破棄します。以下に例を示します。

ObjectScript

```
WRITE "355 \ 113 = ", 355 \ 113 // 3
```

詳細は、“[整数除算 \(¥\)](#)” のリファレンス・ページを参照してください。

モジュロ (#)

2 つのオペランドが正の場合、モジュロ演算の結果は、右のオペランドで左のオペランドを整数除算した剰余です。以下に例を示します。

ObjectScript

```
WRITE "37 # 10 = ", 37 # 10,! // 7  
WRITE "12.5 # 3.2 = ", 12.5 # 3.2,! // 2.9
```

詳細は、“[モジュロ \(#\)](#)” のリファレンス・ページを参照してください。

べき乗 (**)

べき乗演算子は、ある数値を、別の数値を指数としてべき乗します。以下に例を示します。

ObjectScript

```
WRITE "9 ** 2 = ", 9 ** 2,! // 81
```


詳細は、“[べき乗 \(**\)](#)” のリファレンス・ページを参照してください。べき乗計算は、[\\$ZPOWER](#) 関数を使用しても可能です。

注釈 InterSystems IRIS では、ObjectScript の 10 進数浮動小数点数 (10 進形式と呼ばれる) と IEEE 倍精度バイナリ浮動小数点数 (\$DOUBLE と呼ばれ、通常特殊な目的に使用される) の 2 つの数の表現がサポートされています。ObjectScript は、以下の状況では、10 進数値を対応する \$DOUBLE 値に自動的に変換します。

- ・ 算術演算に \$DOUBLE 値が含まれている場合、ObjectScript は演算内のすべての数値を \$DOUBLE に変換します。
- ・ 演算の結果の数値が 10 進形式で表すには大きすぎる場合、ObjectScript は、`<MAXNUMBER>` エラーを発行するのではなく、この数値を \$DOUBLE に自動的に変換します。

これらの形式の詳細は、“[インターシステムズ・アプリケーションでの数値の計算](#)” を参照してください。

5.6 数値関係演算子

数値関係演算子は、オペランドの数値を使用してブーリアン値の結果を返します。文字列に対する演算では、“[文字列から数値への変換](#)” に説明があるルールに従い、文字列は数値関係演算子によってそれぞれが該当する数値として扱われます。

数値関係演算子を非数値文字列の比較で使用しないでください。

ObjectScript は、以下の数値関係演算子を提供します。

より小さい関係演算子 (<)

より小さい関係演算子は、左のオペランドが右のオペランドより小さいかどうかをテストします。以下に例を示します。

ObjectScript

```
WRITE 9 < 6      // 0
```

詳細は、“[より小さい \(<\)](#)” のリファレンス・ページを参照してください。

より大きい関係演算子 (>)

より大きい関係演算子は、左のオペランドが右のオペランドより大きいかどうかをテストします。以下に例を示します。

ObjectScript

```
WRITE 15 > 15    // 0
```

詳細は、“[より大きい \(>\)](#)” のリファレンス・ページを参照してください。

以下演算子 (<= または >=)

以下演算子は、左のオペランドが右のオペランド以下であるかどうかをテストします。以下に例を示します。

ObjectScript

```
WRITE 9 <= 6     // 0
```

“[以下 \(<= または >=\)](#)” のリファレンス・ページを参照してください。

以上演算子 (>= または '>=')

以上演算子は、左のオペランドが右のオペランド以上であるかどうかをテストします。以下に例を示します。

ObjectScript

```
WRITE 15 >= 15 // 1
```

“以上 (>= または '>=')” のリファレンス・ページを参照してください。

注釈 InterSystems IRIS では、ObjectScript の 10 進数浮動小数点数 (10 進形式と呼ばれる) と IEEE 倍精度バイナリ浮動小数点数 (\$DOUBLE 形式と呼ばれ、通常特殊な目的に使用される) の 2 つの数の表現がサポートされています。

これらの形式の数値間では、以下/以上の比較は丸めを行わず正確に行われます。ただし、10 進数と \$DOUBLE の数値の等値比較は、予期しない結果が発生することが多いので避ける必要があります。詳細は、“[インターシステムズ・アプリケーションでの数値の計算](#)” を参照してください。

5.7 論理比較演算子

論理比較演算子は、オペランド値を比較し、True (1) か False (0) のブーリアン値を返します。

5.7.1 論理演算子の優先順位

ObjectScript は、厳格に演算子を左から右へ評価するため、その他の演算子が関与する論理比較では、必要な優先順位を実現するために、演算をグループ化する括弧を使用する必要があります。例えば、以下のプログラムで True (1) を返すために、論理 OR (!) テストが要求されます。

ObjectScript

```
SET x=1,y=0
IF x=1 ! y=0 {WRITE "TRUE"}
ELSE {WRITE "FALSE"}
// Returns 0 (FALSE), due to evaluation order
```

しかし、この論理比較を適切に実行するには、その他の演算を入れ子にする括弧を使用する必要があります。以下の例では、期待される結果が得られます。

ObjectScript

```
SET x=1,y=0
IF (x=1) ! (y=0) {WRITE "TRUE"}
ELSE {WRITE "FALSE"}
// Returns 1 (TRUE)
```

5.7.2 論理演算子

ObjectScript は、以下の論理演算子を提供します。

否定 (!)

否定演算子は、ブーリアン型オペランドの真偽値を反転します。オペランドが True (1) の場合、否定演算子は False (0) になります。オペランドが False (0) の場合、否定演算子は True (1) になります。

例えば、以下の文は False (0) の結果を返します。

ObjectScript

```
SET x=0
WRITE x
```

[“否定 \(!\)”](#) のリファレンス・ページを参照してください。

論理積 (& または &&)

論理積演算子は、オペランドの両方の値が True (1) であるかどうかを判断します。オペランドが両方とも True の場合 (つまり、数値として計算した場合、ゼロ以外の値となる)、ObjectScript は、True (1) を返します。それ以外の場合、False (0) を返します。

論理積演算子には、以下の 2 つの形式があります。

- ・ & 演算子は、両方のオペランドを評価し、いずれかのオペランドの値がゼロの場合、False (0) を返します。それ以外は True (1) を返します。
- ・ && 演算子は、左のオペランドを評価し、そのオペランドの値がゼロの場合、False (0) を返します。左のオペランドがゼロ以外の場合にのみ、&& 演算子は右のオペランドを評価します。右のオペランドの評価がゼロの場合、False (0) を返します。それ以外は True (1) を返します。

以下の例は、2 つのゼロ以外のオペランドを True と評価して True (1) を返します。

ObjectScript

```
SET A=-4,B=1
WRITE A&B // TRUE (1)
```

[“論理積 \(& または &&\)”](#) のリファレンス・ページを参照してください。また、[“論理積否定 \(NAND\) \(!&\)”](#) のリファレンス・ページも参照してください。

論理和 (! または ||)

論理和演算子は、いずれか一方のオペランドが True の値を持つ場合、あるいは両方のオペランドが True (1) の値を持つ場合、True (1) を返します。論理和演算子は、両方のオペランドが False (0) の場合にのみ False (0) を返します。

論理和演算子には、以下の 2 つの形式があります。

- ・ ! 演算子は、両方のオペランドを評価し、両方のオペランドの値がゼロの場合、False (0) を返します。それ以外は True (1) を返します。
- ・ || 演算子は、左のオペランドを評価します。左のオペランドがゼロ以外の値に評価された場合、|| 演算子は、右のオペランドを評価せずに True (1) を返します。左のオペランドがゼロの場合にのみ、|| 演算子は、右のオペランドを評価します。右のオペランドがゼロの場合、False (0) を返します。それ以外は True (1) を返します。

以下の例は、2 つの True の (ゼロではない) オペランドに論理和演算を実行し、True の結果を返します。

ObjectScript

```
SET A=5,B=7
WRITE "A!B = ",A!B,!
SET A=5,B=7
WRITE "A||B = ",A||B,!
```

[“論理和 \(! または ||\)”](#) のリファレンス・ページを参照してください。また、[“論理和否定 \(NOR\) \(!|\)”](#) のリファレンス・ページも参照してください。

5.8 文字列連結演算子 ()

文字列連結演算子 () は、その 2 つのオペランドを文字列と解釈し、1 つ目の文字列に 2 つ目の文字列を追加した文字列値を返します。

以下の例では、現在のデバイスに文字列 Highchair を書き込みます。

ObjectScript

```
WRITE "High_" "chair"
```

詳細は、“[文字列の連結 \(\)](#)” のリファレンス・ページを参照してください。

5.9 文字列関係演算子

文字列関係演算子は、オペランドを文字列として解釈してブーリアン値の結果を返します。文字列関係演算子は、否定論理演算子 (!) を先頭に付けて、論理結果を反転することができます。ObjectScript は、以下の文字列関係演算子を提供します。

等価 (=)

等価演算子は、2 つのオペランドが文字列として等しいかを判断します。等価演算子を 2 つの文字列で実行すると、ObjectScript は、2 つのオペランドの文字順序が同一で、スペースを含め他の異なる文字を持たない同一の文字列の場合に True (1) を返します。それ以外の場合は False (0) を返します。以下に例を示します。

ObjectScript

```
WRITE "SEVEN"="SEVEN"      // 1
WRITE "SEVEN"="seven"       // 0
WRITE "SEVEN"=" SEVEN "     // 0
```

詳細は、“[等価 \(= \)](#)” のリファレンス・ページを参照してください。また、“[不等価 \(!= \)](#)” のリファレンス・ページも参照してください。

包含 (⊇)

包含演算子は、右のオペランドの一連の文字が、左の文字の部分文字列であるかを判断します。左のオペランドが、右のオペランドの文字列を含んでいる場合、結果は True (1) となります。左のオペランドが、右のオペランドの文字列を含んでいない場合、結果は False (0) となります。右のオペランドが NULL 文字列の場合、結果は常に True となります。

以下に例を示します。

ObjectScript

```
SET L="Steam Locomotive"
SET S="Steam"
WRITE L⊇S  /// 1
```

詳細は、“[包含 \(⊇ \)](#)” のリファレンス・ページを参照してください。また、“[非包含 \(!⊇ \)](#)” のリファレンス・ページも参照してください。

後続 ([])

後続関係演算子は、左のオペランドの文字が、ASCII 文字順で右のオペランドの文字の後に来るかどうかを判断します。後続関係演算子は、両方の文字列が、それぞれの最左端の文字から始まるものとします。

以下に例を示します。

ObjectScript

```
WRITE "LAMPPOON" ] "LAMP" // 1
```

詳細は、“[後続 \(\[\]\)](#)” のリファレンス・ページを参照してください。また、“[非後続 \('\]'\)](#)” のリファレンス・ページも参照してください。

前後関係 ([])

前後関係演算子は、左のオペランドが数値添え字の照合順序で右のオペランドの後に順番に並んでいるかを判定します。数値照合順序では、NULL 文字列が最初に置かれます。次に負数から数値順に並べた[キャノニック形式の数値](#)、ゼロ、整数と続き、最後に数値以外の値が置かれます。

以下に例を示します。

ObjectScript

```
WRITE 122[]2 // 1
```

詳細は、“[前後関係 \(\[\]\)](#)” のリファレンス・ページを参照してください。また、“[非前後関係 \('\]'\)](#)” のリファレンス・ページも参照してください。

5.10 パターン・マッチング演算子 (?)

ObjectScript のパターン・マッチング演算子は、左オペランドの文字が、その右オペランドのパターンと正確に一致しているかどうかを判断します。

以下の例は、文字列 ssn に有効な米国社会保障番号 (数字 3 桁、ハイフン、数字 2 桁、ハイフン、数字 4 桁) が含まれているかどうかをテストします。

ObjectScript

```
SET ssn="123-45-6789"  
SET match = ssn ? 3N1 "-" 2N1 "-" 4N  
WRITE match
```

パターン構文およびその他の詳細は、“[パターン・マッチング \(?\)](#)” のリファレンス・ページを参照してください。

注釈 ObjectScript では、多くのソフトウェア・ベンダよりサポートされる (バリエーションを用いた) パターン・マッチング構文である[正規表現](#)もサポートします。正規表現は、[\\$LOCATE](#) および [\\$MATCH](#) 関数と [%Regex.Matcher](#) クラスのメソッドで使用できます。詳細は、“[正規表現](#)” のリファレンス・ページを参照してください。

これらのパターン・マッチング・システムは完全に別個のものであり、異なるパターンとフラグを持つ異なる構文を使用します。

5.11 間接演算子 (@)

間接演算とは、コマンド行、コマンド、コマンド引数の一部あるいはすべてを、データ・フィールドの内容で、実行時にダイナミックに置換する手段です。

間接演算は、添え字間接演算の場合を除いて、間接演算子 (@) で指定され、以下の形式になります。

@variable

variable は、置換する値を取得する変数を識別します。置換する値で参照されるすべての変数は、プロシージャで使用されているとしてもパブリック変数です。変数は配列ノードにできます。

以下のルーチンは、間接演算がその右側にある変数全体の値を参照することを示します。

ObjectScript

```
IndirectionExample
SET x = "ProcA"
SET x(3) = "ProcB"
; The next line will do ProcB, NOT ProcA(3)
DO @x(3)
QUIT
ProcA(var)
WRITE !,"At ProcA"
QUIT
ProcB(var)
WRITE !,"At ProcB"
QUIT
```

詳細は、“[間接 \(@\)](#)” のリファレンス・ページを参照してください。

注釈 間接演算は、他の方法より能率的かつ汎用的にコーディングできますが、必ずしも使用する必要はありません。XECUTE コマンドなど他の方法を使用して、間接演算と同様の機能を常に行えます。

6

コマンド

コマンドは、ObjectScript プログラミングの基本的なコード・ユニットです。ObjectScript の実行タスクは、すべてコマンドで実行されます。すべてのコマンドは、コマンド・キーワードと (ほとんどの場合) これに続く 1 つまたは複数のコマンド引数から構成されます。

ここでは、最もよく使用されるコマンドの概要を提供します。“[ObjectScript リファレンス](#)” も参照してください。

6.1 コマンド・キーワード

ObjectScript では、すべてのコマンド文は明示的です。ObjectScript コードの実行可能行は、すべてコマンド・キーワードで始まる必要があります。例えば、変数に値を代入するには、SET キーワードを指定し、続いて変数の引数および代入する値を指定する必要があります。

コマンド・キーワードは、常にキーワードで始まります。以下の例を考えてみます。

ObjectScript

```
WRITE "Hello"
```

WRITE という単語がコマンド・キーワードです。コマンドは、実行するアクションを指定します。WRITE コマンドは、その名前から分かるように、引数として指定された内容を主デバイスに書き込みます。この場合、WRITE コマンドは文字列 Hello を書き込みます。

ObjectScript のコマンド名では、大文字と小文字は区別されません。ほとんどのコマンド名は省略形で表現できます。したがって、WRITE、Write、write、W、および w はいずれも、WRITE コマンドの有効な形式です。コマンドの省略形のリストは、“[省略形テーブル](#)” を参照してください。

コマンド・キーワードは予約語ではありません。したがって、コマンド・キーワードを変数のユーザ割り当て名、ラベル、またはその他の識別子として使用できます。

ObjectScript プログラムでは、コード行の最初のコマンドをインデントにする必要があります。コマンド・キーワードは列 1 では使用できません。ターミナル・コマンド行プロンプトから、または EXECUTE コマンドからコマンドを発行する場合、インデントは必要ありません (インデントにしてもかまいません)。

コードの実行可能行は、それぞれが独自のコマンド・キーワードを持つ 1 つまたは複数のコマンドを含むことができます。単一行における複数コマンドは 1 つ以上のスペースで区切ります。1 つ以上のコマンドは同一行のラベルの後に続きます。ラベルとコマンドは 1 つ以上のスペースで区切ります。

ObjectScript では、コマンド末端や行末の区切り文字が必要なく、または不許可となります。コマンドの後には、[行内コメント](#)を指定して、コマンド行の残りの部分がコメントであることを示すことができます。コマンドの末端とコメント構文の間に

は空白スペースが必要になりますが、`##;` および `/* comment */` の構文については例外となります。`/* comment */` の複数行コメントは、1 つのコマンド内かつコマンド末端にて指定可能です。

6.2 コマンド引数

コマンド・キーワードの後ろには、オブジェクト(複数も可)またはコマンドの範囲を指定する1つ以上の引数を置くことができます。また、引数を置かない場合もあります。コマンドが1つ以上の引数を取得する場合、コマンド・キーワードと最初の引数の間にスペースを1つだけ置く必要があります。例えば以下のようになります。

ObjectScript

```
SET x = 2
```

最初の引数の最初の文字が(上記のように)正確に1つのスペースでコマンド自体から区切られている限り、引数内または引数間でスペースを使用できます。したがって、以下はすべて有効です。

ObjectScript

```
SET a = 1
SET b=2
SET c=3,d=4
SET e= 5      ,    f =6
SET g
      =          7
WRITE a,b,c,d,e,f,g
```

コマンドが **後置条件** を取る場合、コマンド・キーワードと後置条件間にスペースを置くことはできません。また、後置条件と1番目の引数の最初に必ず1つのスペースを置きます。以下の形式の QUIT コマンドはすべて有効です。

ObjectScript

```
QUIT x+y
QUIT x + y
QUIT:x<0
QUIT:x<0 x+y
QUIT:x<0 x + y
```

引数間にスペースは必要ありませんが、複数の空白スペースを引数間で使用する場合があります。これらの空白スペースは、コマンドの実行に影響を与えません。また、改行、タブ、コメントもコマンド引数内や引数間に含まれますが、コマンドの実行に影響はありません。詳細は、“[空白](#)”を参照してください。

6.2.1 複数の引数

多くのコマンドでは、複数の独立引数を指定することができます。コマンド引数の区切り記号はコンマ(,)です。つまり、1つのコマンドの後ろに、コンマで区切られたリストとして複数の引数を指定します。例えば以下のようになります。

ObjectScript

```
SET x=2,y=4,z=6
```

Set コマンドは、3つの引数を使用して、指定した3つの変数に値を割り当てます。ここでは、複数の引数が繰り返されます。つまり、コマンドは、指定された順番で各引数に個別に割り当てられます。内部的には、InterSystems IRIS データ・プラットフォームがこれを3つの別個の SET コマンドとして解析します。**デバッグ**時には、これらの複数の引数それぞれが別個の手順となります。

コマンド・リファレンス・ページで紹介しているコマンド構文では、繰り返して記述できる引数の後にコンマと省略記号 `,...` が続きます。このコンマは引数に必要な区切り文字です。また、省略記号 (...) は、不特定の数の繰り返し引数を指定できることを示しています。

繰り返し引数は、厳密に左から右の順で実行されます。したがって、以下のコマンドが有効となります。

ObjectScript

```
SET x=2,y=x+1,z=y+x
```

ただし、以下のコマンドは無効です。

ObjectScript

```
SET y=x+1,x=2,z=y+x
```

各繰り返し引数は、指定した順序で独立して実行されるため、無効な引数が検出されるまで有効な引数が実行されます。以下の例では、`SET x` では値が `x` に代入され、`SET y` では `<UNDEFINED>` エラーが発生し、`SET z` が評価されていないため、`<DIVIDE>` (0 による除算) エラーは検出されません。

ObjectScript

```
KILL x,y,z
SET x=2,y=z,z=5/0
WRITE "x is:",x
```

6.2.2 パラメータおよび後置条件付きの引数

いくつかのコマンド引数は、パラメータも受け取ります（“関数”で説明する関数パラメータと混同しないでください）。指定する引数がパラメータを取得できる場合、パラメータの区切り記号はコロンの (:) です。

以下のコマンド例では、引数の区切り記号として使用されているコンマと、パラメータ区切り記号として使用されているコロンを示しています。ここでは、2 つの引数があり、各引数には 3 つのパラメータがあります。

ObjectScript

```
VIEW X:y:z:a,B:a:y:z
```

少数のコマンド (DO, XECUTE、および GOTO) では、引数に続くコロンは、その引数を実行するかどうかを決める [後置条件式](#) を指定します。

6.2.3 引数なしコマンド

引数を持たないコマンドは、引数なしコマンドといいます。キーワードに追加された [後置条件式](#) は引数とは見なされません。

常に引数を持たないコマンドが少数あります。例えば、HALT、CONTINUE、TRY、TSTART および TCOMMIT は、引数なしコマンドです。

いくつかのコマンドは必要に応じて引数なしになります。例えば、BREAK、CATCH、FOR、GOTO、KILL、LOCK、NEW、QUIT、RETURN、TROLLBACK、WRITE、および ZWRITE には、すべて引数なしの構文形式があります。このような場合、同じコマンドでも引数ありと引数なしでは、意味が多少異なる場合があります。

引数なしコマンドを行の最後で使用する場合、末尾のスペースは必要ありません。引数なしコマンドを他のコマンドと同一のコード行で使用する場合、引数なしコマンドと後続のコマンドの間には 2 つ (あるいはそれ以上) のスペースを置く必要があります。例えば以下ようになります。

ObjectScript

```
QUIT:x=10 WRITE "not 10 yet"
```

この場合、QUIT は後置条件式を持つ引数なしのコマンドです。このコマンドと次のコマンドの間に、少なくとも 2 つのスペースが必要です。

6.2.3.1 引数なしコマンドと中括弧

中括弧で区切られたコマンド・ブロック内で引数なしコマンドを使用する場合、空白の制限はありません。

- 引数なしコマンドの直後に左中括弧が続く場合、コマンド名と中括弧の間に空白は必要ありません。何も指定しないか、1 つあるいは複数のスペース、タブ、または改行を指定します。これは、FOR などのように引数を使用できる引数なしコマンド、および ELSE などのように引数を使用できない引数なしコマンドのどちらにも当てはまります。

ObjectScript

```
FOR {
    WRITE !,"Quit out of 1st endless loop"
    QUIT
}
FOR{
    WRITE !,"Quit out of 2nd endless loop"
    QUIT
}
FOR
{
    WRITE !,"Quit out of 3rd endless loop"
    QUIT
}
```

- 右中括弧は区切り文字として機能するため、引数なしコマンドの直後に右中括弧が続く場合、後続のスペースは必要ありません。例えば以下の方法で、引数なしの QUIT を使用します。

ObjectScript

```
IF 1=2 {
    WRITE "Math error"}
ELSE {
    WRITE "Arithmetic OK"
    QUIT}
WRITE !,"Done"
```

6.3 コマンド後置条件式

ほとんどの場合、ObjectScript のコマンドを指定する際、後置条件 を追加することができます。

後置条件とは、コマンドや(時には) コマンド引数に追加されるオプションの式で、InterSystems IRIS がそのコマンドやコマンド引数を実行するかどうかを制御します。後置条件式の評価が True (ゼロ以外の値) の場合、InterSystems IRIS はそのコマンドあるいはコマンド引数を実行します。また、後置条件式の評価が False (ゼロ) の場合、InterSystems IRIS はそのコマンドあるいはコマンド引数を実行せず、次のコマンドあるいはコマンド引数に実行を移します。

すべての ObjectScript コマンドでは、フロー制御コマンド (IF、ELSEIF、ELSE、FOR、WHILE、および DO WHILE) 以外、およびブロック構造エラー処理コマンド (TRY、THROW、CATCH) 以外で、後置条件式を使用できます。

ObjectScript コマンド DO と XECUTE は、コマンド・キーワードとコマンド引数の両方に後置条件式を追加できます。後置条件式は常にオプションです。例えば、コマンドの引数の中でも、後置条件式が追加されている引数と追加されていない引数があります。

コマンド・キーワードとコマンドの引数の両方が後置条件を持つ場合、キーワード後置条件が最初に評価されます。このキーワード後置条件が True に評価された場合にのみ、コマンド引数の後置条件が評価されます。コマンド・キーワード

の後置条件が False の場合、コマンドは実行されず、プログラムは次のコマンドに移って実行を続けます。コマンド引数の後置条件が False の場合、引数は実行されず、コマンドは次の引数を左から右の順で実行を続けます。

6.3.1 後置条件構文

コマンドに後置条件を追加するには、コマンド・キーワードの直後にコロン(:)と式を置きます。後置条件式を持つコマンドの構文は以下のようになります。

Command:pc

Command はコマンド・キーワード、コロンは必須のリテラル文字で、pc は任意の有効な式にできます。

コマンド後置条件には、以下の構文規則があります。

- ・ スペース、タブ、改行、コメントは、コマンド・キーワードと後置条件式の間、あるいはコマンド引数と後置条件式の間には使用できません。コロン文字の前後にスペースを使用することはできません。
- ・ スペース、タブ、改行、コメントは、後置条件式内では使用できません。ただし、後置条件式全体が括弧で囲まれている場合や、括弧で囲まれた引数リストが後置条件式にある場合は使用できます。スペース、タブ、改行、コメントは括弧内で使用できます。
- ・ 後置条件式の後に置く必要があるスペースは、コマンド・キーワードの場合と同様です。キーワード後置条件式の最後の文字と 1 番目の引数の最初の文字の間には、スペースを正確に 1 つ置きます。引数なしコマンドでは、後置条件の直後に右中括弧が続かない限り、後置条件式の最後の文字と同じ行にある次のコマンドの間に、2 つ以上のスペースを置く必要があります。(括弧を使用する場合、閉じ括弧は後置条件式の最後の文字として処理されます。)

後置条件式は、厳密に言えばコマンド引数ではありません(しかし、ObjectScript リファレンス・ページで、後置条件式は、引数セクションの一部として示されています)。後置条件は常にオプションです。

6.3.2 後置条件の評価

InterSystems IRIS は、後置条件式を True あるいは False として評価します。通常、これらは 1 あるいは 0 の値で評価することをお勧めします。しかし、InterSystems IRIS はあらゆる値の後置条件式を実行し、式の値が 0 の場合は False、ゼロ以外の場合は True として評価します。

- ・ InterSystems IRIS は、有効なゼロ以外の数値を True として評価します。算術演算のように、有効な数値に対し同じ評価基準を使用します。したがって、1、“1”、007、3.5、-.007、7.0、3 little pigs、\$CHAR(49)、0_“1” はすべて、True と評価されます。
- ・ InterSystems IRIS は、ゼロ (0) 値およびすべての非数値 (空白文字 (“ ”) が含まれる文字列や NULL 文字列 (“”) など) を False として評価します。したがって、0、-0.0、A、-、\$、The 3 little pigs、\$CHAR(0)、\$CHAR(48)、“0_1” はすべて、False と評価されます。
- ・ 一般的な等式の規則が適用されます。したがって、0=0、0=“0”、“a”=\$CHAR(97)、0=\$CHAR(48)、“”=\$CHAR(32)) は、True と評価されます。一方、0=“”、0=\$CHAR(0)、“”=\$CHAR(32)) は False と評価されます。

次の例では、どの WRITE コマンドが実行されるかは、変数 count の値によって異なります。

ObjectScript

```
FOR count=1:1:10 {
    WRITE:count<5 count," is less than 5",!
    WRITE:count=5 count," is 5",!
    WRITE:count>5 count," is greater than 5",!
}
```

6.4 単一行での複数コマンド

ObjectScript のソース・コードの単一行には、複数のコマンドおよびそれらの引数を含めることができます。これらは厳密に左から右の順序で実行され、別々の行で記述したコマンドと機能的に同一となります。引数付きのコマンドは、単一の空白文字によって後続コマンドと区切る必要があります。引数なしのコマンドは、2 つの空白文字によって後続コマンドと区切る必要があります。[ラベル](#)には、同一行にて 1 つ以上のコマンドを後続させることができます。[コメント](#)は、同一行にて 1 つ以上のコマンドの後に続けることができます。

ソース・コードの行の最大長に関する詳細は、“[一般的なシステム制限](#)”を参照してください。[スタジオ](#)を使用してソース・コードを記述/編集する場合、この制限は異なることがあります。

6.5 変数

変数に値を割り当てるには、SET コマンドを使用します。

SET コマンドは、変数に値を割り当てます。単一の変数または複数の変数に、一度に値を割り当てることができます。

以下は SET の最も基本的な構文です。

ObjectScript

```
SET variable = expression
```

これにより、1 つの変数の値が設定されます。また、以下のようにいくつかの手順があります。

- ObjectScript は、value 式を評価し、値を決定します (可能な場合)。この手順では、式に未定義変数、無効な構文 (例えばゼロによる除算)、他のエラーが含まれている場合、エラーを生成します。
- 変数が存在しない場合は、ObjectScript が変数を生成します。
- 一度変数が生成されると、あるいは変数が既に存在している場合、ObjectScript はその式の変数に値を設定します。

以下の構文を使用して、複数の変数それぞれに値を設定します。

ObjectScript

```
SET variable1 = expression1, variable2 = expression2, variable3 = expression3
```

以下の構文を使用して、1 つの式と等しい複数の変数を設定します。

ObjectScript

```
SET (variable1,variable2,variable3)= expression
```

例えば、以下のコードを使用して、**Person** クラス・インスタンスの **Gender** プロパティ値を設定します。

ObjectScript

```
SET person.Gender = "Female"
```

ここで、person は、**Person** クラスの関連インスタンスへのオブジェクト参照です。

複数の **Person** オブジェクトの **Gender** プロパティを同時に設定することもできます。

ObjectScript

```
SET (per1.Gender, per2.Gender, per3.Gender) = "Male"
```

ここで、per1、per2、per3 は、**Person** クラスの 3 つの異なるインスタンスへのオブジェクト参照です。

SET を使用して、値を返すメソッドを呼び出すことができます。メソッドを呼び出す場合、SET では、メソッドの戻り値と同一の変数、グローバル参照、またはプロパティを設定できます。引数の形式は、メソッドが**インスタンス**か**クラス・メソッド**かどうかにより異なります。クラス・メソッドを呼び出すには、以下の構文を使用します。

ObjectScript

```
SET retval = ##class(PackageName.ClassName).ClassMethodName()
```

ClassMethodName() は呼び出すクラス・メソッド名、**ClassName** はメソッドを含むクラス名、**PackageName** はクラスを含むパッケージ名です。メソッドの戻り値は retval ローカル変数に割り当てられます。##class() 構文は、コードに必要なリテラル部です。

インスタンス・メソッドを呼び出すには、ローカルでインスタンスを生成されたオブジェクトへのハンドルのみが必要です。

ObjectScript

```
SET retval = InstanceName.InstanceMethodName()
```

InstanceMethodName() は呼び出すインスタンス・メソッド名、**InstanceName** はメソッドを含むインスタンス名です。メソッドの戻り値は retval ローカル変数に割り当てられます。

詳細は ["SET"](#) を参照してください。

注釈 プロシージャ・ブロックを使用しない古いコードでは、**KILL** コマンドと **NEW** コマンドが変数の範囲を制御するのに役立ちます。詳細は、これらのコマンドのリファレンスを参照してください。

6.6 エラー処理

エラー処理には TRY / CATCH ブロック構造を使用します。エラー処理のためにブロック構造を作成するには、TRY コマンドおよび CATCH コマンドを使用することをお勧めします。

["TRY-CATCH メカニズム"](#)、および ["TRY"](#)、["THROW"](#)、["CATCH"](#) を参照してください。

6.7 トランザクション処理

トランザクション処理には TSTART、TCOMMIT、および TROLLBACK コマンドを使用します。["トランザクション処理"](#)、および ["TSTART"](#)、["TCOMMIT"](#)、["TROLLBACK"](#) を参照してください。

6.8 ロックと並行処理の制御

リソースをロックまたはロック解除するには LOCK コマンドを使用します。["ロックと並行処理の制御"](#) および ["LOCK"](#) を参照してください。

ロックはトランザクション処理とも関連があります。["トランザクション処理"](#) を参照してください。

6.9 コードの呼び出し

この節では、1 つまたは複数のコマンドの実行を呼び出すために使用するコマンドについて説明します。

- [DO](#)
- [JOB](#)
- [XECUTE](#)
- [QUIT](#) および [RETURN](#)

6.9.1 DO

ObjectScript で任意のルーチン、プロシージャ、またはメソッドを呼び出すには、DO コマンドを使用します。以下は DO の基本的な構文です。

ObjectScript

```
DO ^CodeToInvoke
```

CodeToInvoke は、InterSystems IRIS システム・ルーチンあるいはユーザ定義ルーチンです。キャレット文字 ^ は、ルーチン名の直前に付ける必要があります。

また、ルーチン内でプロシージャの開始場所を示すラベル (タグともいいます) を参照して、プロシージャを実行することもできます。ラベルは、キャレットの直前に置かれます。以下はその例です。

ObjectScript

```
SET %X = 484
DO INT^%SQROOT
WRITE %Y
```

このコードは、%X システム変数値を 484 に設定します。その後 DO コマンドを使用して、InterSystems IRIS システム・ルーチン %SQROOT の INT プロシージャを呼び出し、%X の平方根を算出して、%Y に格納します。その後、WRITE コマンドを使用して、%Y の値を表示します。

メソッドの呼び出し時、DO は 1 つの引数として、メソッドを指定する式全体を取得します。引数の形式は、メソッドが [インスタンスかクラス・メソッド](#) かどうかにより異なります。クラス・メソッドを呼び出すには、以下の構文を使用します。

ObjectScript

```
DO ##class(PackageName.ClassName).ClassMethodName()
```

ClassMethodName() は呼び出すクラス・メソッド名、**ClassName** はメソッドを含むクラス名、**PackageName** はクラスを含むパッケージ名です。##class() 構文は、コードに必要なリテラル部です。

インスタンス・メソッドを呼び出すには、ローカルでインスタンスを生成されたオブジェクトへのハンドルのみが必要です。

ObjectScript

```
DO InstanceName.InstanceMethodName()
```

InstanceMethodName() は呼び出すインスタンス・メソッド名、**InstanceName** はメソッドを含むインスタンス名です。

詳細は ["DO"](#) を参照してください。

6.9.2 JOB

DO はフォアグラウンドでコードを実行し、JOB はバックグラウンドで実行します。これは、通常ユーザとの対話なしに、現在のプロセスで単独に実行されます。ジョブ起動プロセスは、明示的に指定された場合を除き、すべてのシステムの既定値を継承します。

詳細は “JOB” を参照してください。

6.9.3 XECUTE

XECUTE コマンドは、1 つ以上の ObjectScript コマンドを実行し、引数として取得する式を評価します（その引数は、1 つ以上の ObjectScript コマンドを含む文字列に評価される必要があります）。実際、各 XECUTE 引数は、DO コマンドによって呼び出される一行サブルーチンに類似しており、引数の最後に達するか、QUIT コマンドに遭遇すると終了します。InterSystems IRIS が引数を実行した後、制御は XECUTE 引数の直後に返されます。

詳細は “XECUTE” を参照してください。

6.9.4 QUIT および RETURN

QUIT コマンドと RETURN コマンドは、両方ともメソッドを含むコード・ブロックの実行を終了させます。引数がない場合、呼び出し元のコードを単に終了します。引数がある場合、戻り値としてその引数を使用します。QUIT は、現在のコンテキストを終了し、囲んでいるコンテキストに移動します。RETURN は、現在のプログラムを終了し、プログラムが呼び出された場所に移動します。

以下のテーブルは、QUIT を使用するか RETURN を使用するかを選択する方法を示しています。

Location	QUIT	RETURN
ルーチン・コード(ブロック構造になっていない)	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。
TRY または CATCH ブロック	TRY / CATCH ブロック構造のペアを終了し、ルーチンの次のコードに移動します。入れ子にされた TRY または CATCH ブロックから発行された場合、1 レベルを終了し、囲んでいる TRY または CATCH ブロックに移動します。	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。
DO または XECUTE	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。
IF	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。ただし、FOR、WHILE、または DO WHILE ループで入れ子にされている場合、そのブロック構造を終了し、コード・ブロックの後の次の行で続行します。	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。
FOR、WHILE、DO WHILE	ブロック構造を終了し、コード・ブロックの後の次の行で続行します。入れ子にされたブロックから発行された場合、1 レベルを終了し、囲んでいるブロックに移動します。	ルーチンを終了し、呼び出し元のルーチン(存在する場合)に戻ります。

詳細は、“[QUIT](#)” および “[RETURN](#)” を参照してください。

6.10 制御フロー

コードでロジックを構築するためには、フロー制御が必要です。コードのブロックを条件付きで実行するかまたはバイパスするか、またはコードのブロックを繰り返し実行します。そのために、ObjectScript は以下のコマンドをサポートします。

- ・ [IF](#)、[ELSEIF](#)、および [ELSE](#)
- ・ [FOR](#)
- ・ [WHILE](#) と [DO WHILE](#)

6.10.1 条件付きの実行

コードのブロックを条件付きで実行するには、ブーリアン (True/False) テストに基づいて、IF コマンドを使用できます。(後置条件式を使用して、個別の ObjectScript コマンドの条件付きの実行を行うことができます。)

IF は、式を引数として取得し、式が True か False かを判断します。True の場合、式の後ろに続くコードのブロックが実行され、False の場合は実行されません。通常、これらは 1 あるいは 0 の値で評価することをお勧めします。しかし、

InterSystems IRIS はあらゆる値の条件付きの実行を行い、式の値が 0 の場合は False、ゼロ以外の場合は True として評価します。詳細は、“[演算子と式](#)”を参照してください。

複数の IF ブーリアン・テスト式を、コンマ区切りのリストで指定することができます。これらのテストは、一連の論理 AND テストとして左から順に評価されます。したがって、IF は、すべてのテスト式が True に評価されると、True に評価されます。IF は、テスト式の 1 つが False と評価されると False に評価されます。残りのテスト式は評価されません。

コードは、常に多数のコマンドを含む コード・ブロック で表示されます。コード・ブロックとは、単に {} 括弧内の 1 行以上のコード行のことで、コード・ブロックの前と中には改行を置くことができます。以下の例を考えてみます。

6.10.1.1 IF、ELSEIF、および ELSE

IF 文は、複数の条件を評価し、条件によってどのコードを実行するのかを指定できます。単純なコマンドと異なり、文とは 1 つ以上のコマンド・キーワード、条件式、コード・ブロックを含みます。IF 文は、以下のものから構成されます。

- ・ 1 つまたは複数の条件式を含む 1 つの IF 節。
- ・ それぞれが 1 つまたは複数の条件式を含む任意の数の ELSEIF 節。ELSEIF 節 (オプション)。ELSEIF 節は複数でもかまいません。
- ・ 条件式のない、1 つ以下の ELSE 節。ELSE 節は、オプションです。

以下は、IF 文の例です。

ObjectScript

```
READ "Enter the number of equal-length sides in the polygon: ",x
IF x=1 {WRITE !,"It's so far away that it looks like a point"}
ELSEIF x=2 {WRITE !,"I think that's a line, not a polygon"}
ELSEIF x=3 {WRITE !,"It's an equilateral triangle"}
ELSEIF x=4 {WRITE !,"It's a square"}
ELSE {WRITE !,"It's a polygon with ",x," number of sides" }
WRITE !,"Finished the IF test"
```

詳細は、“[IF](#)” のリファレンスを参照してください。

6.10.2 FOR

FOR 文を使用すると、コードのセクションを繰り返し実行できます。数値や文字列値を基に FOR ループを生成できます。

通常、FOR は、コードのそれぞれのループの開始でインクリメントあるいはデクリメントされる数値制御変数の値を基に、コード・ブロックを 0 回以上実行します。制御変数が最後の値に達したとき、制御は FOR ループを終了します。終了値がない場合、ループは QUIT コマンドに遭遇するまで実行を続けます。制御がループを終了すると、制御変数は、最後のループの実行による値を保持します。

以下は、数値 FOR ループの形式です。

ObjectScript

```
FOR ControlVariable = StartValue:IncrementAmount:EndValue {
    // code block content
}
```

すべての値は正数あるいは負数で、スペースは許可されていますが、等符号とコロンの前後には不要です。FOR に続くコード・ブロックは、変数に割り当てられた各値を繰り返します。

例えば、以下の FOR ループは 5 回実行します。

ObjectScript

```
WRITE "The first five multiples of 3 are:",!
FOR multiple = 3:3:15 {
    WRITE multiple,!
}
```

また、変数を使用して、最後の値を決めることもできます。以下の例は、ループが繰り返される回数を指定します。

ObjectScript

```
SET howmany = 4
WRITE "The first ",howmany," multiples of 3 are "
FOR multiple = 1:1:howmany {
    WRITE (multiple*3),",", "
    IF multiple = (howmany - 1) {
        WRITE "and "
    }
    IF multiple = howmany {
        WRITE "and that's it!"
    }
}
QUIT
```

この例は、制御変数 multiple を使用して 3 倍しているため、式は multiple*3 となります。また、IF コマンドを使用して、最後の計算の前に and を挿入しています。

注釈 この例の IF コマンドは、ObjectScript の優先順位を含む分かりやすい例です（優先順位は、階層に関係なく常に左から右です）。IF 式 multiple = howmany - 1 の括弧がない場合、あるいは全体が括弧で囲まれている場合、式の最初の部分 multiple = howmany は、False (0) となります。したがって、式全体は 0 - 1 となり、結果は -1 であるため、この式は True となります（ループで、最後の繰り返し以外、それぞれの文字列に and を挿入します）。

FOR の引数は、値のリストに設定された変数にすることもできます。この場合、コード・ブロックは、変数に割り当てられたリストで、各項目に対して繰り返します。

ObjectScript

```
FOR item = "A", "B", "C", "D" {
    WRITE !, "Now examining item: "_item
}
```

QUIT コマンドを特定の状況で実行されるようコード・ブロック内に配置し、FOR を終了させるように指定すると、最終値のない FOR の数値形式を記述できます。これにより、繰り返し回数を示すカウンタを提供し、カウンタの値に基づかない条件を使用して、FOR を制御できます。例えば、以下のループはカウンタを使用し、推測された回数をユーザに知らせます。

ObjectScript

```
FOR i = 1:1 {
    READ !, "Capital of MA? ", a
    IF a = "Boston" {
        WRITE "...did it in ", i, " tries"
        QUIT
    }
}
```

カウンタが不要な場合、引数なしの FOR を使用できます。

ObjectScript

```
FOR {
    READ !, "Know what? ", wh
    QUIT:(wh = "No!")
    WRITE "    That's what!"
}
```

詳細は“FOR”を参照してください。

6.10.3 WHILE と DO WHILE

2つの関連するフロー制御コマンド、WHILE と DO WHILE は、それぞれがコード・ブロックでループを実行し、条件に応じて終了します。2つのコマンドは、条件の評価方法が異なります。WHILE はコード・ブロック全体の前に条件を評価し、DO WHILE はブロックの後に条件を評価します。FOR と同様、コード・ブロック内に QUIT コマンドを記述してループを終了します。

2つのコマンドの構文は、以下のとおりです。

```
DO {code} WHILE condition
WHILE condition {code}
```

以下の例は、ユーザが指定した値までフィボナッチの数列で値を2回表示します（最初は DO WHILE 次はWHILE を使用）。

ObjectScript

```
fibonacci() PUBLIC { // generate Fibonacci sequences
    READ !, "Generate Fibonacci sequence up to where? ", upto
    SET t1 = 1, t2 = 1, fib = 1
    WRITE !
    DO {
        WRITE fib," " set fib = t1 + t2, t1 = t2, t2 = fib
    }
    WHILE ( fib '> upto )

    SET t1 = 1, t2 = 1, fib = 1
    WRITE !
    WHILE ( fib '> upto ) {
        WRITE fib," "
        SET fib = t1 + t2, t1 = t2, t2 = fib
    }
}
```

WHILE、DO WHILE、FOR の明確な違いは、WHILE はループの実行前、DO WHILE はループの実行後に制御式の値をテストする必要があり、一方 FOR はループ内ならどこでもテストできる点です。つまり、コード・ブロックに2つの部分があり、式の評価によって2番目を実行する場合、FOR 文が一番適切です。それ以外の場合、式をコード・ブロックの前か後のいずれに評価するかによって選択します。

詳細は、“WHILE” および “DO WHILE” を参照してください。

6.11 入出力の制御

ObjectScript の入出力コマンドは、InterSystems IRIS の内部あるいは外部でデータを取得する基本的な機能を提供します。以下のとおりです。

- ・ Write コマンド
- ・ READ
- ・ OPEN、USE、および CLOSE

6.11.1 表示（書き込み）コマンド

ObjectScript は、現在の出力デバイスにリテラルと変数値を表示（書き込み）する、以下の4つのコマンドをサポートしています。

- ・ WRITE コマンド

- ・ [ZWRITE](#) コマンド
- ・ [ZZDUMP](#) コマンド
- ・ [ZZWRITE](#) コマンド

6.11.1.1 引数なしの表示コマンド

- ・ 引数なしの WRITE は、定義済みの各ローカル変数の名前と値を表示します (行ごとに 1 つの変数を表示します)。パブリック変数とプライベート変数の両方がリスト表示されます。グローバル変数、プロセス・プライベート・グローバルまたは特殊変数はリスト表示されません。変数は照合シーケンス順でリスト表示されます。添え字ツリー順で添え字付き変数をリストします。

すべてのデータ値は、二重引用符文字で区切られ、引用符で囲まれた文字列として表示されます。ただし、キャノニック形式の数とオブジェクト参照を除きます。オブジェクト参照 (OREF) 値が割り当てられた変数は、`variable=<OBJECT REFERENCE>[oref]` として表示されます。%List 形式の値またはビット文字列の値は、引用符で囲まれた文字列として、それらの値がエンコードされた形式で表示されます。このようなエンコードされた形式には非表示文字が含まれている可能性があるため、%List やビット文字列の表示は空の文字列になることがあります。

WRITE は、特定の非表示文字を表示しません。そうした非表示文字を表現するための、プレースホルダやスペースは表示されません。WRITE は、制御文字を実行します (改行やバックスペースなど)。

- ・ 引数なしの ZWRITE は、引数なしの WRITE と同様に機能します。
- ・ 引数なしの ZZDUMP は、無効なコマンドであり、<SYNTAX> エラーを生成します。
- ・ 引数なしの ZZWRITE は、空の文字列を返す空命令です。

6.11.1.2 引数ありの表示コマンド

以下の表に、4 つのコマンドの引数付きの形式の機能をリストします。4 つのすべてのコマンドは、1 つの引数または引数のコンマ区切りリストを取ります。4 つのすべてのコマンドは、引数として、ローカル変数、グローバル変数またはプロセス・プライベート変数、リテラル、式、または特殊変数を取ります。

以下の表には、%Library.Utility.FormatString() メソッドの既定の戻り値も示されています。FormatString() メソッドは、ZZWRITE とほとんど同じですが、戻り値の一部として %val= をリスト表示しないことと、オブジェクト参照 (OREF) 識別子のみを返す点が異なります。FormatString() を使用すると、ZWRITE/ZZWRITE 形式の戻り値に変数を設定できるようになります。

テーブル 6-1: 表示形式

	WRITE	ZWRITE	ZZDUMP	ZZWRITE	FormatString()
各値を別個の行に表示	なし	あり	あり (1 行に 16 文字)	あり	1 つの入力値のみ
識別された変数名	なし	あり	なし	%val= で表される	なし
未定義の変数を <UNDEFINED> エラーと表示	あり	なし (スキップされ、変数名は返されない)	あり	あり	あり

4 つのすべてのコマンドは、式を評価して、キャノニック形式で数値を返します。

テーブル 6-2: 値の表示方法

	WRITE	ZWRITE	ZZDUMP	ZZWRITE	FormatString()
16 進表現	なし	なし	あり	なし	なし
文字列を囲む引用符 (数値と区別するため)	なし	あり	なし	あり (文字列リテラルは %val="value" として返される)	あり
添え字ノードの表示	なし	あり	なし	なし	なし
別のネームスペースのグローバル変数 (拡張グローバル参照) の表示	あり	あり (拡張グローバル参照構文が表示される)	あり	あり	あり
非表示文字の表示	なし、表示されない (制御文字は実行される)	あり、\$c(n) として表示される	あり、16 進数として表示される	あり、\$c(n) として表示される	あり、\$c(n) として表示される
リスト値の形式	エンコードされた文字列	\$lb(val) 形式	エンコードされた文字列	\$lb(val) 形式	\$lb(val) 形式
%Status 形式	エンコードされたリストを含んでいる文字列	\$lb(val) 形式のリストを含んでいる文字列、エラーとメッセージを明示する /*... */ コメントが付加される。	エンコードされたリストを含んでいる文字列	\$lb(val) 形式のリストを含んでいる文字列、エラーとメッセージを明示する /*... */ コメントが付加される。	\$lb(val) 形式のリストを含んでいる文字列、エラーとメッセージを明示する /*... */ コメントが付加される (既定の場合)。
ビット文字列形式	エンコードされた文字列	\$zwc 形式、1 つのビットをリストする /* \$bit() */ コメントが付加される。例: %\$zwc(0,1,2,3) \$1,2,4%	エンコードされた文字列	\$zwc 形式、1 つのビットをリストする /* \$bit() */ コメントが付加される。例: %\$zwc(0,1,2,3) \$1,2,4%	\$zwc 形式、1 つのビットをリストする /* \$bit() */ コメントが付加される (既定の場合)。例: %\$zwc(0,1,2,3) \$1,2,4%
オブジェクト参照 (OREF) 形式	OREF のみ	<OBJECT REFERENCE>[oref] 形式の OREF。一般情報や属性値などの詳細がリストされる。すべてのサブノードがリストされる	OREF のみ	<OBJECT REFERENCE>[oref] 形式の OREF。一般情報や属性値などの詳細がリストされる。	OREF のみ (引用符で囲まれた文字列形式)

JSON の動的配列と JSON のダイナミック・オブジェクトは、該当するコマンドのすべてで OREF 値として返されます。JSON のコンテンツを返すには、以下の例に示すように、%ToJSON() を使用する必要があります。

```
SET jobj={"name":"Fred","city":"Bedrock"}
WRITE "JSON object reference:",!
ZWRITE jobj
WRITE !!, "JSON object value:",!
ZWRITE jobj.%ToJSON()
```

詳細は、“[WRITE](#)”、“[ZWRITE](#)”、“[ZZDUMP](#)”、および “[ZZWRITE](#)” を参照してください。

6.11.2 READ

READ コマンドを使用すると、現在の入力装置からエンド・ユーザが入力した内容を受け取り、格納することができます。READ コマンドは、以下の引数を持つことができます。

ObjectScript

```
READ format, string, variable
```

format は、ユーザの入力エリアが画面に表示される場所を制御し、string は入力プロンプトの前に画面に表示され、variable は入力データを格納します。

以下の形式コードを使用して、ユーザ入力エリアを制御します。

形式コード	結果
!	新しい行を開始します。
#	新しいページを開始します。ターミナルでは、現在の画面をクリアして、新規画面の一番上から開始します。
?n	n 番目の列に移動します。n は正の整数です。

詳細は “[READ](#)” を参照してください。

6.11.3 OPEN、USE、および CLOSE

さらに高度なデバイス処理を実行するために、InterSystems IRIS は多彩なオプションを提供します。つまり、OPEN コマンドでデバイスを開き、USE コマンドで現在のデバイスを指定して、CLOSE コマンドで実行中のデバイスを閉じることができます。この処理の詳細は、“[入出力デバイス・ガイド](#)” で説明されています。

詳細は、“[OPEN](#)”、“[USE](#)”、および “[CLOSE](#)” を参照してください。

7

呼び出し可能なユーザ定義コードモジュール

このトピックでは、InterSystems IRIS® データ・プラットフォームで ObjectScript コードのユーザ定義モジュール（メソッド、関数、プロシージャ、ルーチン、またはサブルーチン）を作成する方法および呼び出す方法について説明します。最も一般的なコード形式はメソッドであり、クラスに定義できます。このトピックでは、メソッドについて具体的には説明しませんが、既定ではメソッドはプロシージャです。このため、プロシージャに関する内容はすべてメソッドにも適用されます。

他の言語と同様に、ObjectScript は直接呼び出すことができる名前付きのコード・ブロックを生成できます。このようなブロックをプロシージャともいいます。ObjectScript 用語で厳密に言うと、プロシージャであるコード・ブロックは、特定の構文と構造を持ちます。

プロシージャ定義の構文は、以下のようになります。

ObjectScript

```
ProcedureName(Parameters) [PublicVariables]
{
    /* code goes here */
    RETURN ReturnValue
}
```

ここで ProcedureName としているプロシージャの要素は以下のとおりです。

- ・ パラメータ (0 か 1) – あらゆるタイプが可能です。一般的な ObjectScript 構文と同様、プロシージャの定義時にタイプを宣言する必要はありません。既定で、パラメータは(参照渡しではなく)値によって渡されます。指定がない限り、その範囲はプロシージャに対しローカルです。パラメータの一般的な詳細は、“[プロシージャ・パラメータ](#)”を参照してください。
- ・ パブリック変数への参照 (0 かそれ以上) – あらゆるタイプが可能です。プロシージャは、変数の値を参照することも、設定することもできます。パブリック変数の詳細は、“[プロシージャ変数](#)”を参照してください。
- ・ プロシージャの Public 宣言 (オプション) – 既定では、プロシージャはプライベートであるため、同じルーチン内からのみ、そのプロシージャを呼び出すことができます (ObjectScript 用語で、ルーチンとは 2 つ以上のプロシージャ、あるいはユーザ定義のコード・ブロックを含むファイルのことです)。また、プロシージャ名の後に PUBLIC キーワードを使用して、パブリック・プロシージャを生成できます。パブリック・プロシージャは他のルーチンやメソッドからも呼び出せます。パブリック・プロシージャとプライベート・プロシージャの詳細は、“[パブリック・プロシージャとプライベート・プロシージャ](#)”を参照してください。
- ・ コード – プロシージャ内のコードは、ObjectScript で使用できるすべての機能を備えています。プロシージャ・コードには、Java も使用できます。コードは中括弧で区切られ、プロシージャ・ブロックとも呼ばれます。
- ・ 返り値 (オプション) – プロシージャが返す値です。標準の ObjectScript 式である必要があります。プロシージャ内のフロー制御では、計算式の値、RETURN 文、またはその両方を使用して、さまざまな返り値を指定できます。

注釈 通常は、サブルーチンやユーザ定義関数を記述するよりも、プロシージャを記述する方が望ましいです。プロシージャ・パラメータは、プロシージャでのスコープで自動的にローカルになります。これらのパラメータはプロシージャにとってプライベートであり、シンボル・テーブルとやり取りしないので、他の値を上書きしないようにする NEW コマンドは不要です。また、パブリック変数を明示的に宣言することで、例えば銀行全体の金利といったアプリケーション内のグローバル変数を参照できます。あるいは、他のアプリケーションに有効なプロシージャ内で、変数の値を生成、設定することもできます。

プロシージャは、ObjectScript ルーチンの一種です。

InterSystems IRIS は、多くのシステム関数を備えています。内部関数とも呼ばれるこの関数の詳細は、“ObjectScript ランゲージ・リファレンス” で説明しています。システム関数の呼び出しは、接頭語 \$ で識別されます。

7.1 プロシージャ、ルーチン、サブルーチン、関数、メソッドの概要

このトピックでは、プロシージャを使用して、独自のコードを実装する方法について説明します。ユーザ定義の機能実装では、この方法をお勧めします。InterSystems のドキュメントでは、プロシージャ、ルーチン、サブルーチン、関数、およびメソッドについて説明しています。すべての項目が機能を共有していますが、それぞれに独自の特徴もあります。

名前付きのユーザ定義のコード・ブロックを記述するのに、最も柔軟で強力な推奨される形式は、プロシージャです。以下はプロシージャの特徴です。

- ・ プライベートあるいはパブリックです。
- ・ ゼロ個以上のパラメータを受け取ることができます。
- ・ その中で生成された変数を、ローカルの範囲として自動的に維持します。
- ・ 範囲外の変数を参照し変更できます。
- ・ あらゆるタイプの値を返す、あるいは値を返しません。

以下はプロシージャとの比較です。

- ・ サブルーチンは常にパブリックで、値を返しません。
- ・ 関数は常にパブリックで、ローカル変数を明示的に宣言する必要があります (そうしなければ、外部変数を上書きします)。また、必ず返り値があります。
- ・ 既定で、メソッドはクラス定義の一部として指定されるプロシージャで、1 つ以上のオブジェクトあるいはクラスで呼び出すことができます。メソッドを関数として明示的に宣言すると、付随するすべての特徴も関数に属するため、これはお勧めできません。
- ・ ルーチンは、ObjectScript プログラムです。1 つ以上のプロシージャ、サブルーチン、関数、あるいはこれらの組み合わせになります。

注釈 ObjectScript は、マクロ機能を通して、ユーザ定義コードに関連する形式もサポートしています。

7.1.1 ルーチン

ルーチンは、ユーザ記述コードの呼び出し可能なブロックで、ObjectScript プログラムです。ルーチンは汎用的な処理を実行します。ルーチン名は、保存時に選択する .MAC ファイル名から付けられます。ルーチンが値を返すかどうかによって、以下の構文の 1 つあるいは両方を使用して、ルーチンを呼び出すことができます。

ObjectScript

```
DO ^RoutineName
SET x = $$^RoutineName
```

ルーチンはネームスペース内で定義されます。[拡張ルーチン参照](#)を使用し、以下のようにして現在のネームスペース以外のネームスペースで定義されたユーザ定義ルーチンを実行できます。

ObjectScript

```
DO ^|"USER"|RoutineName
```

一般的に、ルーチンは、サブルーチン、メソッド、プロシージャのコンテナとして機能します。

ルーチンは、コード・ブロックの最初にあるラベル (タグともいいます) で識別されます。このラベルがルーチン名です。(通常)ラベルの後ろに続く括弧内には、呼び出し元のプログラムからルーチンに渡すパラメータ・リストが記述されます。

ルーチンをファイルに保存する場合、ファイル名には、アンダースコア (_)、ハイフン (-)、セミコロン (;) を使用できません。このような文字を含む名前は無効です。

7.1.2 サブルーチン

サブルーチンは、ルーチン内で名前の付いたコード・ブロックです。通常、サブルーチンはラベルから始まり、QUIT 文で終了します。パラメータを受け取ることはできますが、値は返しません。サブルーチンを呼び出すには、以下の構文を使用します。

```
DO Subroutine^Routine
```

Subroutine は、Routine ファイル (**Routine.MAC**) 内のコード・ブロックです。

以下は、サブルーチンの形式です。

ObjectScript

```
Label(parameters) // comment
// code
QUIT // note that QUIT has no arguments
```

サブルーチンの詳細は、従来のコードを説明している以下の [“サブルーチン”](#) セクションを参照してください。

{ } 括弧でコードと QUIT 文を囲む場合、サブルーチンはプロシージャとなるため、プロシージャとして処理されます。そのような場合、QUIT 文が重複するので省略してもかまいません。例えば、以下の 2 つのサブルーチンの定義は同等です。

ObjectScript

```
Label(parameters) PUBLIC {
// code
QUIT
}
```

および、

ObjectScript

```
Label(parameters) PUBLIC {
// code
}
```

7.1.3 関数

InterSystems IRIS は、“[ObjectScript ランゲージ・リファレンス](#)” で説明しているように、多くの[システム関数](#) (内部関数とも呼ばれる) を備えています。このセクションでは、ユーザ定義 (extrinsic) 関数について説明します。

関数は、ルーチン内で名前の付いたコード・ブロックです。通常、関数はラベルから始まり、RETURN 文で終了します。パラメータを受け取り、値も返します。関数の呼び出しにおいて、以下の 2 つが有効な形式の構文となります。

ObjectScript

```
SET rval=$$Function() /* returning a value */
DO Function^Routine /* ignoring the return value */
```

Function は、Routine ファイル (**Routine.MAC**) 内のコード・ブロックです。両方の構文形式で、[拡張ルーチン参照](#)を使用し、異なるネームスペースにある関数を実行できます。

以下は、関数の形式です。

ObjectScript

```
Label(parameters)
// code
RETURN ReturnValue
```

{ } 括弧でコードと RETURN 文を囲むと、関数はプロシージャとなるため、プロシージャとして処理されます。プロシージャは既定によりプライベートであるため、以下のように PUBLIC キーワードを指定することが必要になる場合もあります。

ObjectScript

```
Label(parameters) PUBLIC {
// code
RETURN ReturnValue }
```

以下の例では、簡単な関数 (MyFunc) を指定したうえで、それを呼び出し、2 つのパラメータを渡して、さらに返される値を受け取ります。

ObjectScript

```
Main ;
TRY {
    KILL x
    SET x=$$MyFunc(7,10)
    WRITE "returned value is ",x,!
    RETURN
}
CATCH { WRITE $ZERROR,! }
MyFunc(a,b)
SET c=a+b
RETURN c
```

関数を呼び出すコードは返される値を無視できますが、関数の RETURN コマンドでは返される値を指定することが必要となります。引数なしの RETURN にて関数の終了を試みると、<COMMAND> エラーが生成されます。<COMMAND> エラーによって関数を呼び出したコールの場所が指定され、呼び出された関数内の引数なし RETURN コマンドのオフセット位置を指定するメッセージが続きます。詳細は“[\\$ZERROR](#)”を参照してください。

関数の詳細は、従来のコードを説明している以下の“[関数](#)”セクションを参照してください。

7.2 プロシージャの定義

他のプログラミング言語と同様、プロシージャは特定のタスクを実行する一連の ObjectScript コマンド (大規模なルーチン) です。If 文の構造のように、プロシージャのコードは中括弧 ({}) で囲まれています。

プロシージャは、パブリックかプライベートのいずれかで変数を定義できます。例えば、以下は MyProc というプロシージャの例です。

ObjectScript

```
MyProc(x,y) [a,b] PUBLIC {
  Write "x + y = ", x + y
}
```

これは、2 つのパラメータ x と y を取る MyProc というパブリック・プロシージャを定義しています。2 つのパブリック変数 a と b も定義しています。このプロシージャで使用する他のすべての変数 (この場合、x と y) はプライベート変数です。

既定では、プロシージャはプライベートであるため、同じルーチン内からのみ呼び出すことができます。また、プロシージャ名の後に PUBLIC キーワードを使用して、パブリック・プロシージャを生成できます。パブリック・プロシージャは他のルーチンからも呼び出せます。

プロシージャでは、定義済みのパラメータは必要ありません。パラメータでプロシージャを生成する場合、ラベルの直後に括弧で囲んだ変数リストを置きます。

7.2.1 プロシージャの呼び出し

プロシージャを呼び出すには、DO コマンドを発行してプロシージャを指定するか、\$\$ 構文を使用して関数として呼び出します。プロシージャが他のプログラムから実行できるかどうか (パブリック)、あるいは存在するプログラム内でのみ実行できるかどうか (プライベート) を制御できます。DO コマンドで呼び出す場合、プロシージャは値を返しません。関数呼び出しとして呼び出す場合は、値を返します。\$\$ 形式は最も機能的であり、一般的に好まれる形式です。

7.2.1.1 \$\$ 接頭語の使用法

式を使用できるコンテキスト内で、ユーザ定義関数を呼び出すことができます。以下の形式で、ユーザ定義関数を呼び出します。

```
$$name([param[ ,...]])
```

以下を示します。

- ・ name 関数の名前を指定します。関数の定義場所によって、以下のように名前を指定できます。
 - label は、現在のルーチンにある行ラベルです。
 - label^routine は、ディスクにある名前付きルーチンの行ラベルです。
 - ^routine は、ディスクに保存されているルーチンです。ルーチンは、実行する関数のコードのみを含んでいる必要があります。
- ・ param は関数に渡される値を指定します。提供されるパラメータは、実パラメータ・リストと呼ばれます。これらは、その関数に関して定義された仮パラメータ・リストと一致する必要があります。例えば、関数コードには 2 つのパラメータがあり、1 番目は数値、2 番目は文字リテラルとします。この場合、1 番目のパラメータに文字リテラル値、2 番目に数値を指定すると、関数は不正確な値を返すか、あるいはエラーを生成する可能性があります。仮パラメータ・リストのパラメータでは、関数から NEW コマンドが必ず呼び出されます。詳細は、[NEW コマンドの説明](#)を参照してください。パラメータは、値または参照によって渡されます。詳細は、["パラメータ渡し"](#)を参照してください。関数の仮

パラメータ・リストに記載されている数より少数のパラメータを関数に渡すと、パラメータの既定値が使用されます(定義されている場合)。既定値がない場合、パラメータは未定義のままとなります。

7.2.1.2 DO コマンドの使用法

DO コマンドを使用して、ユーザ定義関数を呼び出すことができます (DO コマンドを使用して、システム関数を呼び出すことはできません)。DO コマンドで呼び出した関数は、値を返しません。つまり、関数は必ず返り値を生成しますが、DO コマンドはこの返り値を無視します。したがって、ユーザ定義関数を呼び出す DO コマンドの用途は大幅に制限されます。

DO コマンドを使用してユーザ定義関数を呼び出すには、以下の構文でコマンドを発行します。

```
DO label(param[,...])
```

DO コマンドは、label という関数を呼び出し、param で指定されたパラメータ (存在する場合) を関数に渡します。\$\$ 接頭語を使用しないことと、パラメータの括弧が必須であることに注意してください。label と param を指定する際には、\$\$ 接頭語を使用するユーザ定義関数を呼び出すときと同じ規則が適用されます。

関数は、常に値を返す必要があります。しかし、DO コマンドで関数を呼び出す場合、返された値は呼び出し元のプログラムで無視されます。

7.2.2 プロシージャの構文

以下はプロシージャの構文です。

```
label([param [= default]][, ...]) [[pubvar[, ...]]] [access]
{
  code
}
```

以下の構文で呼び出します。

```
DO label([param[, ...]])
```

あるいは以下ようになります。

```
command $$label([param[, ...]])
```

以下はその説明です。

引数	説明
label	プロシージャ名です。標準ラベルです。必ず 1 列目から始まります。label に続くパラメータの括弧は必須です。
param	プロシージャに必要な各パラメータの変数です。必要なパラメータは、仮パラメータ・リストと呼ばれます。パラメータ自体はオプションですが (何もない、1 つある、あるいは 1 つ以上の param がある)、括弧は必須です。複数の param 値がある場合、コンマによって区切られます。パラメータは、値または 参照 で仮パラメータ・リストに渡すことができます。ルーチンであるプロシージャにはパラメータのタイプ情報が記述されていませんが、メソッドであるプロシージャにはこの情報が記述されています。仮パラメータの最大数は 255 個です。

引数	説明
default	直前の param に対するオプションの既定値です。各パラメータに対して既定値を提供しても、省略してもかまいません。仮パラメータに対応する実パラメータが指定されていない場合、または実パラメータが参照渡しされたときに当該のローカル変数に値がない場合は、既定値が適用されます。この既定値はリテラル、つまり引用符で囲まれた数字あるいは文字列です。null 文字列 () も既定値として指定できます。null 文字列は変数を定義するため、これは既定値を指定しないこととは異なります。一方でパラメータ値が未指定の場合や既定値がない場合、変数は未定義となります。リテラル以外の既定値を指定すると、InterSystems IRIS は <PARAMETER> エラーを発行します。
pubvar	パブリック変数です。プロシージャによって使用され、他のルーチンとプロシージャで使用できる、パブリック変数の任意のリストです。これは、このプロシージャ内で定義され、他のルーチンでも使用可能な変数、あるいは他のルーチンで定義され、このプロシージャでも使用可能な変数のリストです。pubvar を指定する場合、それを角括弧で囲みます。pubvar を指定しない場合、角括弧は省略できます。複数の pubvar 値がある場合、コンマによって区切ります。Public として宣言されていない変数は、すべてプライベート変数です。プライベート変数は、プロシージャの現在の呼び出しに対してのみ有効です。その変数は、プロシージャの呼び出し時には未定義です。また、プロシージャが終了する際に破棄されます。プロシージャがそのプロシージャ外のコードを呼び出すと、プライベート変数は保存されますが、制御がプロシージャに戻るまでは使用できません。また、すべての % 変数は、ここにリストされているかどうかにかかわらず、常にパブリックです。パブリック変数のリストは、このルーチンに対して指定された 1 つ以上の param を含むことができます。
access	プロシージャが Public か Private を宣言する任意のキーワードです。利用できる値は 2 つあります。PUBLIC キーワードは、このプロシージャがどのルーチンからでも呼び出しできることを宣言します。PRIVATE キーワードは、このプロシージャが定義されたルーチン内でのみ呼び出されることを宣言します。既定は PRIVATE です。
code	中括弧で囲まれたコード・ブロックです。左中括弧 ({) は、その後ろに続く先頭の文字から少なくとも 1 つのスペースを置く、あるいは 1 行の改行が必要です。同じ行の右中括弧 (}) の後ろにはコードを記述できず、ブランクあるいはコメントのみ置くことができます。通常、右中括弧は 1 列目に記述します。このコード・ブロックは、ラベルによってのみ入力されます。

コマンドと引数の間には改行を挿入できません。

それぞれのプロシージャはルーチンの一部として実装され、各ルーチンは複数のプロシージャを持つことができます。

標準的な ObjectScript 構文の他に、ルーチンを管理する特殊な規則があります。ルーチン行はオプションとして、ObjectScript コードの最初にラベル (タグとも呼びます) を、最後にコメントを置くことができます。

インターシステムズでは、ルーチンの 1 行目にルーチン名を表すラベル、その後タブまたはスペースを置いた後、ルーチンの目的を簡単に説明する短いコメントを入れることをお勧めします。行にラベルを置く場合、タブやスペースを含む残りの行からラベルを離す必要があります。つまり、スタジオを使用してルーチンに行を追加し、ラベルとタブ/スペースに続けて ObjectScript コードを入力するか、ラベルを置かずにタブかスペースに続けて ObjectScript を入力します。したがって、いずれの場合でも、それぞれの行には、最初のコマンドの前にタブかスペースが必要です。

1 行コメントを示す場合、二重スラッシュ (/) かセミコロン (;) のいずれかを使用します。コメントがコードの後ろに続く場合、スラッシュやセミコロンの前にスペースが必要です。その行にコメントしか記述しない場合、スラッシュやセミコロンの前に、タブかスペースが必要です。当然のことながら、1 行コメントは改行できません。複数行コメントの場合、コメントの先頭に /* を、末尾には */ を付けます。

7.2.3 プロシージャ変数

プロシージャとメソッドは両方共、プライベート変数とパブリック変数をサポートします。以下の文はすべてプロシージャとメソッドに同等に適用されます。

プロシージャ内で使用する変数は、自動的にプライベートになります。したがって、これらは宣言する必要がなく、NEW コマンドは不要です。このプロシージャが呼び出すプロシージャの変数を共有したい場合、変数をパラメータとして他のプロシージャに渡します。

パブリック変数の宣言も可能です。パブリック変数は、このプロシージャまたはメソッドの呼び出し元と呼び出し先のすべてのプロシージャおよびメソッドで使用できます。アプリケーションで環境変数として動作するよう、比較的少数のパブリック変数をこの方法で定義します。パブリック変数を定義するには、プロシージャ名とパラメータの後に角括弧内 ([]) で変数をリストします。

以下の例では、宣言された 2 つのパブリック変数 [a, b] と 2 つのプライベート変数 (c, d) が含まれるプロシージャが定義されます。

ObjectScript

```
publicvarsexample
; examples of public variables
;
DO procl() ; call a procedure
QUIT ; end of the main routine
;
procl() [a, b]
; a private procedure
; "c" and "d" are private variables
{
WRITE !, "setting a" SET a = 1
WRITE !, "setting b" SET b = 2
WRITE !, "setting c" SET c = 3
SET d = a + b + c
WRITE !, "The sum is: ", d
}
```

Terminal

```
USER>WRITE

USER>DO ^publicvarsexample

setting a
setting b
setting c
The sum is: 6
USER>WRITE

a=1
b=2
USER>
```

7.2.3.1 パブリック変数とプライベート変数

プロシージャ内では、ローカル変数はパブリックあるいはプライベートのいずれでも定義できます。パブリック・リスト (pubvar) は、プロシージャ内のどの変数参照がパブリック変数のセットに追加されるかを宣言します。プロシージャ内の他の変数参照はすべて、プロシージャの現在の呼び出しによってのみ参照される、プライベート・セットに追加されます。

プライベート変数は、プロシージャの呼び出し時には未定義です。また、プロシージャが終了する際に破棄されます。

プロシージャ内のコードがプロシージャ外のコードを呼び出す場合、プライベート変数は、プロシージャに戻ったときに復元されます。呼び出されたプロシージャやルーチンは、パブリック変数にアクセスできます (それ自身のプライベート変数も同様に使用できます)。したがって、pubvar はこのプロシージャで参照されるパブリック変数と、プロシージャが呼び出すルーチンによって参照できるこのプロシージャで使用する変数の両方を指定します。

パブリック・リストが空の場合、すべての変数はプライベートです。この場合、角括弧は任意となります。

名前の先頭に % 文字が付く変数は、通常、システムや特定用途のために使用する変数です。InterSystems IRIS は、システムで使用するためにすべての % 変数 (%z 変数および %Z 変数を除く) を予約しています。ユーザ・コードでは、%z または %Z で始まる % 変数のみを使用してください。すべての % 変数は、暗黙的にパブリックです。これらの変数は、(ドキュメント用に) パブリック・リストに記載することもできますが、必須ではありません。

7.2.3.2 プライベート変数と NEW コマンドで生成される変数

プライベート変数は、NEW で生成される変数とは異なることに注意してください。プロシージャの変数を、他のプロシージャやサブルーチンで使いたい場合、その変数は必ず Public に宣言し、パブリック・リストに記述する必要があります。このプロシージャによってパブリック変数が導入されていれば、それに対して NEW を実行する意味があります。この場合、パブリック変数は、プロシージャが終了するときに自動的に破棄されます。しかし、パブリック変数が保持していた以前の値は保護されます。以下はコードの例です。

ObjectScript

```
MyProc(x,y)[name]{
  NEW name
  SET name="John"
  DO xyz^abc
}
```

name 変数はパブリックであるため、このコードによって、この変数の値 John を abc ルーチンの xyz プロシージャで参照できます。name に対して NEW コマンドを実行すると、MyProc プロシージャを呼び出したときに既に存在しているすべてのパブリック変数 name が保護されます。

NEW コマンドは、プライベート変数ではなくパブリック変数にのみ作用します。プロシージャ内で x がパブリック・リストになく、% 変数でない場合、NEW x または NEW (x) を指定すると違反になります。

7.2.3.3 仮リスト・パラメータをパブリックにする

プロシージャに仮リスト・パラメータ (例えば MyProc(x,y) の x あるいは y) が存在し、それを他のプロシージャ (あるいはこのプロシージャ外のサブルーチン) で使用する場合は、それをパブリック・リストに置く必要があります。

以下はコードの例です。

ObjectScript

```
MyProc(x,y)[x] {
  DO abc^rou
}
```

y ではなく x の値が、abc^rou ルーチンで使用可能となります。

7.2.4 パブリック・プロシージャとプライベート・プロシージャ

プロシージャは、パブリックまたはプライベートです。プライベート・プロシージャは、プロシージャが定義されたルーチン内でのみ呼び出すことができます。一方パブリック・プロシージャは、どのルーチンからも呼び出すことができます。PUBLIC キーワードと PRIVATE キーワードが省略されている場合、既定はプライベートです。

以下はその例です。

ObjectScript

```
MyProc(x,y) PUBLIC { }
```

上記はパブリック・プロシージャを定義しています。

ObjectScript

```
MyProc(x,y) PRIVATE { }
```

あるいは以下のコードを入力します。

ObjectScript

```
MyProc(x,y) { }
```

上記 2 つのコードは、プライベート・プロシージャを定義しています。

7.3 パラメータ渡し

プロシージャの重要な特徴は、パラメータ渡しに対するサポートです。これは、プロシージャに値または変数をパラメータとして渡すための機能です。当然のことながら、パラメータ渡しは必須ではありません。例えば、パラメータ渡しとしていないプロシージャを使用して、乱数の生成や、既定以外の形式でシステム日付を返す操作などが可能です。それでも、多くの場合、プロシージャではパラメータ渡しを使用します。

パラメータ渡しを設定するには、以下を指定します。

- ・ プロシージャ呼び出し時の実パラメータ・リスト
- ・ プロシージャ定義時の仮パラメータ・リスト

InterSystems IRIS がユーザ定義プロシージャを実行するとき、実パラメータ・リストのパラメータが、それに対応する仮リストのパラメータに位置を基準としてマップされます。したがって、実リストの先頭にあるパラメータの値は仮リストの先頭にある変数に配置され、2 番目の値は 2 番目の変数に配置され、以降の値も同様に配置されます。これらのパラメータの照合は、名前ではなく、パラメータの位置で行われます。したがって、実パラメータに使用する変数と仮パラメータに対して使用する変数は、同じ名前を持つ必要はありません（また、通常持つべきではありません）。プロシージャでは、渡された値に該当する変数を仮リストの中で参照することにより、渡された値にアクセスします。

実パラメータ・リストと仮パラメータ・リストの間で、収録されているパラメータの数が異なっていることがあります。

- ・ 実パラメータ・リストにあるパラメータが、仮パラメータ・リストにあるパラメータより少ない場合、仮パラメータ・リストと一致しない要素は未定義となります。以下の例のように、未定義の仮パラメータに使用するデフォルト値を指定できます。

ObjectScript

```
Main
/* Passes 2 parameters to a procedure that takes 3 parameters */
SET a="apple",b="banana",c="carrot",d="dill"
DO ListGroceries(a,b)
WRITE !,"all done"
ListGroceries(x="?",y="?",z="?") {
    WRITE x," ",y," ",z,! }
```

- ・ 実パラメータ・リストにあるパラメータが、仮パラメータ・リストにあるパラメータより多い場合は、以下の例のように、`<PARAMETER>` エラーが発生します。

ObjectScript

```
Main
/* Passes 4 parameters to a procedure that takes 3 parameters.
   This results in a <PARAMETER> error */
SET a="apple",b="banana",c="carrot",d="dill"
DO ListGroceries(a,b,c,d)
WRITE !,"all done"
ListGroceries(x="?",y="?",z="?") {
    WRITE x," ",y," ",z,! }
```

仮リストにある変数が実リストにあるパラメータよりも多く、既定値が各パラメータに指定されていない場合、余った変数は未定義のままとなります。プロシージャのコードに適切な If テストを記述し、プロシージャの各参照で、使用可能な値が

得られることを確認する必要があります。実パラメータ数と仮パラメータ数の一致を簡素化するには、[可変個数のパラメータ](#)を指定します。

実パラメータの最大数は 254 個です。

ユーザ定義プロシージャにパラメータを渡す場合は、[値渡し](#)または[参照渡し](#)が可能です。同じプロシージャの呼び出しの中で、値渡しと参照渡しを混用できます。

- ・ プロシージャには、パラメータを値で渡すことも参照で渡すこともできます。
- ・ サブルーチンには、パラメータを値で渡すことも参照で渡すこともできます。
- ・ ユーザ定義関数には、パラメータを値で渡すことも参照で渡すこともできます。
- ・ システム関数には、パラメータを値でのみ渡すことができます。

7.3.1 値渡し

値渡しをする場合は、実パラメータ・リストでリテラル値、式、またはローカル変数（添え字付きでも添え字なしでも可）を指定します。式を指定した場合、InterSystems IRIS は最初にその式を評価し、次に結果の値を渡します。ローカル変数を指定した場合、InterSystems IRIS はその変数の現在の値を渡します。指定したすべての変数名が必ず存在し、値が設定されている必要があります。

プロシージャの仮パラメータ・リストには、添え字なしのローカル変数名を記述します。このリストで明示的な添え字付き変数を指定することはできません。一方、[可変個数のパラメータ](#)を指定すると、添え字付き変数が暗黙的に作成されます。

InterSystems IRIS は、プロシージャ内で使用される非パブリック変数をすべて暗黙に生成し宣言します。したがって、呼び出し元コードで同じ名前を持つ既存の変数は上書きされません。これらの変数の既存の値（存在する場合）をプログラム・スタックに置きます。QUIT コマンドまたは RETURN コマンドを呼び出すとき、InterSystems IRIS は各仮変数に対して暗黙の KILL コマンドを実行し、スタックから前の値を復元します。

以下の例で、SET コマンドは 3 つの異なる形式を使用して、参照先の Cube プロシージャに同じ値を渡します。

ObjectScript

```
DO Start()
WRITE "all done"
Start() PUBLIC {
  SET var1=6
  SET a=$$Cube(6)
  SET b=$$Cube(2*3)
  SET c=$$Cube(var1)
  WRITE !,"a: ",a," b: ",b," c: ",c,!
  RETURN 1
}
Cube(num) PUBLIC {
  SET result = num*num*num
  RETURN result
}
```

7.3.2 参照渡し

参照渡しの場合、以下の形式を使用して、実パラメータ・リストのローカル変数名あるいは添え字なしの配列名を指定します。

```
.name
```

参照渡しでは、プロシージャでまだ参照していない段階であれば、指定された変数名または配列名が実際に存在している必要はありません。通常、配列は参照渡しのみを使用します。添え字付き変数は参照渡しできません。

- ・ 実パラメータ・リスト：実パラメータ・リストのローカルな変数名または配列名の前にはピリオドを記述する必要があります。これは、変数が値ではなく、参照で渡されることを指定するものです。
- ・ 仮パラメータ・リスト：仮パラメータ・リストでは特殊な構文を必要とせずに、参照渡しで変数を受け取ります。仮パラメータ・リストでは、ピリオドを接頭語として使用することはできません。ただし、仮パラメータ・リストの変数名の前にアンパサンド (&) を接頭語として使用することはできます。規約により、この & 接頭語の使用は、変数が参照渡しされることを示します。& 接頭語はオプションであり、処理を実行するものではありません。ソース・コードを読みやすく、かつ保守しやすくするために役立つ規約です。

“[引数の渡し方の指示](#)” で説明しているように、メソッド定義では参照渡しを指定できます。

参照渡しでは、実リストの各変数および配列名が、関数の仮リストの対応する変数名に結合されます。参照渡しは、参照するルーチンと関数の間を双方向で通信する効果的なメカニズムを提供します。関数が仮リストにある変数を変更すると、対応する実リスト上の参照渡しの変数も変更されます。これは、KILL コマンドにも適用されます。仮リストにある参照渡しの変数が関数によって削除されると、対応する実リスト上の変数も削除されます。

実リストで指定した変数あるいは配列名が存在しない場合、関数参照により未定義として処理されます。関数が対応する仮リストの値を変数に代入すると、実の変数あるいは配列もこの値で定義されます。

以下の例では、参照渡しを値渡しと比較しています。変数 a を値で渡し、変数 b を参照で渡します。

ObjectScript

```
Main
  SET a="6",b="7"
  WRITE "Initial values:",!
  WRITE "a=",a," b=",b,!
  DO DoubleNums(a,.b)
  WRITE "Returned to Main:",!
  WRITE "a=",a," b=",b
DoubleNums(foo,&bar) {
  WRITE "Doubled Numbers:",!
  SET foo=foo*2
  SET bar=bar*2
  WRITE "foo=",foo," bar=",bar,!
}
```

以下の例では参照渡しを使用して、変数 result を通じて参照元のルーチンと関数の間で双方向通信します。ピリオド接頭語は、result が参照渡しされることを指定します。この関数を実行すると、実パラメータ・リストに result が作成され、関数の仮パラメータ・リストにある z にバインドされます。計算された値は z に代入され、result の参照元のルーチンに渡されます。仮パラメータ・リストの z に対する & 接頭語はオプションであり、機能ありませんが、ソース・コードの明確化に役立ちます。num と powr は、参照渡しではなく値渡しであることに注意してください。以下は、値渡しと参照渡しを組み合わせた例です。

ObjectScript

```
Start ; Raise an integer to a power.
  READ !,"Integer= ",num RETURN:num=""
  READ !,"Power= ",powr RETURN:powr=""
  SET output=$$Expo(num,powr,.result)
  WRITE !,"Result= ",output
  GOTO Start
Expo(x,y,&z)
  SET z=x
  FOR i=1:1:y {SET z=z*x}
  RETURN z
```

7.3.3 可変個数のパラメータ

プロシージャでは、可変個数のパラメータが使用可能であることを指定できます。そのように指定するには、vals... のように最後のパラメータ名に 3 つのドットを付加します。このドットを付加するパラメータは、パラメータのリストの最後にあるものとする必要があります。パラメータのリストに記述した唯一のパラメータであってもかまいません。この ... 構文を記述することで、1 つ以上のパラメータを渡すことができるほか、パラメータをまったく渡さないこともできます。

リストのパラメータ間、ならびにリストの最初のパラメータの前と最後のパラメータの後には、スペースと新しい行が認められています。3 つのドットの間の空白は認められていません。

この構文を使用するには、最後のパラメータ名の後に ... が続くシグニチャを指定します。この手法でメソッドに渡す複数のパラメータには、データ型による値またはオブジェクトによる値を設定できるほか、これらの設定によるパラメータを混在して使用することもできます。可変個数のパラメータの使用を指定するパラメータには、任意の有効な識別子名を設定できます。

ObjectScript では、添え字付き変数を作成することで、可変個数のパラメータを渡す処理を扱います。この場合は、渡す変数ごとに 1 つの添え字が作成されます。最上位の変数には、渡すパラメータの個数を記述し、変数の添え字には、渡す値を記述します。

これを、以下の例に示します。ここでは、invals... を仮パラメータ・リストにある唯一のパラメータとしています。ListGroceries(invals...) は、可変個数の値を値渡しで受け取ります。

ObjectScript

```
Main
    SET a="apple",b="banana",c="carrot",d="dill",e="endive"
    DO ListGroceries(a,b,c,d,e)
    WRITE !,"all done"
ListGroceries(invals...) {
    WRITE invals," parameters passed",!
    FOR i=1:1:invals {
        WRITE invals(i),!
    }
}
```

以下の例では、2 つの定義済みパラメータの後に最後のパラメータとして args... を使用しています。このプロシージャは、3 つ目のパラメータから、可変個数の追加パラメータを受け取ることができます。最初の 2 つのパラメータは必須で、定義済みパラメータ DO AddNumbers(a,b,c,d,e) またはプレースホルダのコンマ DO AddNumbers(, ,c,d,e) のいずれかを指定します。

ObjectScript

```
Main
    SET a=7,b=8,c=9,d=100,e=2000
    DO AddNumbers(a,b,c,d,e)
    WRITE "all done"
AddNumbers(x,y,morenums...) {
    SET sum = x+y
    FOR i=1:1:$GET(morenums, 0) {
        SET sum = sum + $GET(morenums(i))
    }
    WRITE "The sum is ",sum,!
}
```

以下の例では、2 つの定義済みパラメータの後に最後のパラメータとして args... を使用しています。このプロシージャが受け取るパラメータの個数は 2 つのみで、可変個数の追加パラメータ args... は 0 です。

ObjectScript

```
Main
    SET a=7,b=8,c=9,d=100,e=2000
    DO AddNumbers(a,b)
    WRITE "all done"
AddNumbers(x,y,morenums...) {
    SET sum = x+y
    FOR i=1:1:$GET(morenums, 0) {
        SET sum = sum + $GET(morenums(i))
    }
    WRITE "The sum is ",sum,!
}
```

指定に応じて、AddNumbers(x,y,morenums...) は 2 個以上、255 個以下の範囲でパラメータを受け取ることができます。定義済みのパラメータに既定値を指定して AddNumbers(x=0,y=0,morenums...) とすると、このプロシージャは 0 個以上、255 個以下の範囲でパラメータを受け取ることができます。

以下の例では、唯一のパラメータとして args... を使用しています。これは参照渡しで可変個数の値を受け取ります。

ObjectScript

```

Main
  SET a=7,b=8,c=9,d=100,e=2000
  DO AddNumbers(.a,.b,.c,.d,.e)
  WRITE "all done"
AddNumbers(&nums...) {
  SET sum = 0
  FOR i=1:1:$GET(nums, 0) {
    SET sum = sum + $GET(nums(i)) }
  WRITE "The sum is ",sum,!
  RETURN sum
}

```

変数パラメータ・リスト `params...` が参照によって渡されたパラメータを受け取り、`params...` をルーチンに渡すと、その中間ルーチンは、参照によっても渡される追加パラメータ (`params` 配列の追加ノード) を追加できます。以下に例を示します。

ObjectScript

```

Main
  SET a(1)=10,a(2)=20,a(3)=30
  DO MoreNumbers(.a)
  WRITE !,"all done"
MoreNumbers(&params...) {
  SET params(1,6)=60
  SET params(1,8)=80
  DO ShowNumbers(.params) }
ShowNumbers(&tens...) {
  SET key=$ORDER(tens(1,1,""),1,targ)
  WHILE key="" {
    WRITE key," = ",targ,!
    SET key=$ORDER(tens(1,1,key),1,targ)
  }
}

```

以下の例では、この `args...` 構文を仮パラメータ・リストと実パラメータ・リストの両方で使用できることを示しています。この例の可変個数のパラメータ (`invals...`) は、`ListNums` に値で渡され、そこで 2 倍の値にされたうえで `invals...` として `ListDoubleNums` に渡されます。

ObjectScript

```

Main
  SET a="1",b="2",c="3",d="4"
  DO ListNums(a,b,c,d)
  WRITE !,"back to Main, all done"
ListNums(invals...) {
  FOR i=1:1:invals {
    WRITE invals(i),!
    SET invals(i)=invals(i)*2 }
  DO ListDoubleNums(invals...)
  WRITE "back to ListNums",!
}
ListDoubleNums(twicevals...) {
  WRITE "Doubled Numbers:",!
  FOR i=1:1:twicevals {
    WRITE twicevals(i),! }
}
QUIT

```

“[可変個数の引数の指定](#)” も参照してください。

7.4 プロシージャ・コード

中括弧内のコードがプロシージャのコードですが、従来の ObjectScript コードとは異なります。以下はその詳細です。

- ・ プロシージャに入ることができるのは、プロシージャ・ラベルの場所のみです。 `label+offset` 構文を使ってプロシージャにアクセスすることはできません。

- ・ プロシージャのラベルは、プロシージャに対しプライベートであり、プロシージャ内からのみアクセスできます。PRIVATE キーワードは不要ですが、プロシージャ内のラベルで PRIVATE キーワードを使用できます。PUBLIC キーワードは、プロシージャ内のラベルで使用すると構文エラーが発生します。\$TEXT システム関数は、プロシージャ・ラベル名を使用している label+offset をサポートしていますが、名前プライベート・ラベルにアクセスできません。
- ・ ラベルの重複はプロシージャ内では許可されていませんが、特定の状況ではルーチン内で許可されています。特に、異なるプロシージャ内でラベルは重複して使用できます。また、同じラベルはプロシージャ内や、プロシージャが定義されたルーチン内の他の場所で表示できます。例えば、以下の 3 つの Label1 は使用できます。

ObjectScript

```
Roul // Roul routine
Proc1(x,y) {
Label1 // Label1 within the proc1 procedure within the Roul routine
}

Proc2(a,b,c) {
Label1 // Label1 within the Proc2 procedure (local, as with previous Label1)
}

Label1 // Label1 that is part of Roul and neither procedure
```

- ・ プロシージャにルーチン名なしの DO コマンドやユーザ定義関数がある場合、ラベルが存在しているとプロシージャ内のラベルを参照します。そうでない場合、ルーチン内ではあるがプロシージャの外部にあるラベルを参照します。
- ・ ルーチン名を持つ DO またはユーザ定義関数がプロシージャにある場合、そのプロシージャでは必ずその外部にある行を特定します。その名前がプロシージャを含むルーチンを識別する場合も同様です。例えば以下のようになります。

ObjectScript

```
ROU1 ;
PROC1(x,y) {
DO Label1^ROU1
Label1 ;
}
Label1 ; The DO calls this label
```

- ・ プロシージャに GOTO が含まれる場合、プロシージャ内のプライベート・ラベルに移動する必要があります。GOTO では、プロシージャを終了できません。
- ・ label+offset 構文は、いくつかの例外はありますが、プロシージャ内ではサポートされていません。
 - \$TEXT は、プロシージャ・ラベルから label+offset をサポートします。
 - Break またはエラー発生後にプロシージャに戻る方法として、GOTO label+offset を、プロシージャ・ラベルから始まるダイレクト・モード行で使用できます。
 - ZBREAK コマンドでは、プロシージャ・ラベルから始まる label+offset を指定できます。
 - プロシージャが呼び出されたときに有効になっていた \$TEST 状態は、そのプロシージャが終了すると復元されます。
 - プロシージャの終了を意味する } は、行の最初の文字位置を含め、どの文字位置にでも置くことができます。コードは、行の } の前に記述します。その後ろには記述できません。
 - 右中括弧の直前に、暗黙的な QUIT が置かれます。
 - 間接指定と XECUTE コマンドは、プロシージャの範囲外に存在するかのように動作します。

7.5 プロシージャ範囲内の間接指定、XECUTE コマンド、および JOB コマンド

プロシージャ内の名前間接指定、引数間接指定、XECUTE コマンドは、プロシージャの範囲内では実行されません。したがって、XECUTE コマンドは、プロシージャ外にあるサブルーチンの暗黙的な DO のように動作します。

間接演算子と XECUTE は、パブリック変数にのみアクセスできます。その結果、間接演算子または XECUTE コマンドが変数 *x* を参照する場合、プライベート変数である *x* がプロシージャに存在していても、パブリック変数である *x* を参照します。例えば以下ようになります。

ObjectScript

```
SET x="set a=3" XECUTE x ; sets the public variable a to 3
SET x="label1" DO @x ; accesses the public subroutine label1
```

同様に、間接指定または XECUTE で参照するラベルは、プロシージャ外のラベルです。プロシージャ内での GOTO は、その内部のラベルに移動する必要があるため、GOTO @A はプロシージャ内でサポートされていません。

間接演算と XECUTE コマンドの詳細は、このドキュメントの他の個所で説明されています。

同様に、プロシージャの範囲内で JOB コマンドを発行する場合、メソッド外の子プロセスを開始します。つまり、次のようなコードになります。

ObjectScript

```
KILL ^MyVar
JOB MyLabel
QUIT $$$OK
MyLabel
SET ^MyVar=1
QUIT
```

子プロセスでラベルが見えるようにするには、プロシージャ・ブロックにメソッドまたはクラスを含めることはできません。

7.6 プロシージャ内でのエラー・トラップ

エラー・トラップがプロシージャ内から設定された場合、プロシージャのプライベート・ラベルに直接設定する必要があります。(これは、+offset やルーチン名を含む従来のコードとは異なります。この規則は、エラー・トラップの実行とは、基本的にエラー・トラップまでスタックを巻き戻し、GOTO を実行することである、という考えに一致しています。)

プロシージャ内でエラーが発生した場合、\$ZERROR はプライベートの label+offset ではなく、プロシージャの label+offset に設定されます。

エラー・トラップを設定するには、通常の \$ZTRAP を使用します。ただし、値はリテラルにする必要があります。以下はその例です。

ObjectScript

```
SET $ZTRAP = "abc"
// sets the error trap to the private label "abc" within this block
```

エラー・トラップの詳細は、“[TRY-CATCH の使用法](#)”を参照してください。

7.7 従来のユーザ定義コード

InterSystems IRIS にプロシージャが追加される前、ユーザ定義コードはサブルーチンと関数の形式でサポートされていました(これらは現在、プロシージャとして実装できます)。ここでは、これらの従来のユーザ定義コードについて説明していますが、これは既存のコードを説明することが主な目的です。これらを継続して使用することはお勧めしません。

7.7.1 サブルーチン

7.7.1.1 構文

以下は、ルーチンの構文です。

```
label [ ( param [ = default ] [ , ... ] ) ]
      code
      QUIT
```

以下の構文で呼び出します。

```
DO label [ ( param [ , ... ] ) ]
```

あるいは以下ようになります。

```
GOTO label
```

引数	説明
label	サブルーチンの名前標準ラベルです。必ず 1 列目から始まります。label の後に続くパラメータの括弧はオプションです。パラメータが指定されたサブルーチンは、GOTO では呼び出されません。パラメータの括弧は、直前に実行されるコードがサブルーチン内に“フォールスルーする”のを防ぎます。InterSystems IRIS は、パラメータの括弧付きラベルに遭遇した場合(中身が空でも)、暗黙の QUIT を実行し、ルーチンの次の行に実行が移らないように実行を終了します。
param	呼び出し元のプログラムからサブルーチンに渡されるパラメータの値。GOTO コマンドを使用して呼び出されるサブルーチンは、param 値を持つことはできず、パラメータの括弧を使うこともできません。DO コマンドを使用して呼び出されるサブルーチンには、param 値があってもなくてもかまいません。param 値がない場合、空のパラメータの括弧は記述しても省略してもかまいません。サブルーチンに必要な各パラメータに対し、param 変数を指定します。必要なパラメータは、仮パラメータ・リストと呼ばれます。このリストには、何も指定しなくても、1 つ以上の param を定義してもかまいません。複数の param 値がある場合、コンマによって区切られます。InterSystems IRIS は、参照された param 変数に対して、自動的に NEW を呼び出します。パラメータは、値または 参照 で仮パラメータ・リストに渡すことができます。
default	直前の param に対するオプションの既定値です。各パラメータに対して既定値を提供しても、省略してもかまいません。仮パラメータに対応する実パラメータが指定されていない場合、または実パラメータが 参照 渡されたときに当該のローカル変数に値がない場合は、既定値が適用されます。この既定値はリテラル、つまり引用符で囲まれた数字あるいは文字列です。null 文字列(" ")も既定値として指定できます。null 文字列は変数を定義するため、これは既定値を指定しないこととは異なります。一方でパラメータ値が未指定の場合や既定値がない場合、変数は未定義となります。リテラル以外の既定値を指定すると、InterSystems IRIS は <PARAMETER> エラーを発行します。

引数	説明
code	コード・ブロックです。コード・ブロックは、通常ラベルを呼び出してアクセスします。しかし、コード・ブロック内で別のラベルを呼び出したり、label + offset の GOTO コマンドを使用しても入力（再入力）できます。コード・ブロックでは、他のサブルーチン、関数、プロシージャを入れ子にして呼び出すことができます。入れ子になった呼び出しは、リンクされた一連の GOTO コマンドではなく、DO コマンドまたは関数呼び出しを使用して実行することをお勧めします。コード・ブロックは、通常、明示的な QUIT コマンドで終了します。必須ではありませんが、QUIT コマンドの記述をお勧めします。外部ラベルに GOTO を使用して、サブルーチンを終了することもできます。

7.7.1.2 概要

サブルーチンは、最初の行の 1 列目にあるラベルで識別されるコード・ブロックです。サブルーチンの実行は、通常、明示的な QUIT 文によって終了します。

サブルーチンは、DO コマンドまたは GOTO コマンドで呼び出します。

- DO コマンドはサブルーチンを実行し、その後、呼び出し元のルーチンの実行を再開します。したがって、InterSystems IRIS がサブルーチンで QUIT コマンドに遭遇した場合、呼び出し元のルーチンに戻り、DO コマンドに続く次の行を実行します。
- GOTO コマンドは、サブルーチンを実行しますが、呼び出し元のプログラムに制御を戻しません。InterSystems IRIS は、サブルーチンの中で QUIT コマンドに遭遇したときに実行を終了します。

DO コマンドで呼び出されたサブルーチンにパラメータを渡すことができますが、GOTO コマンドで呼び出されたサブルーチンには渡せません。値、もしくは参照によってパラメータを渡せます。詳細は、“[パラメータ渡し](#)”を参照してください。

サブルーチンと呼び出し元のルーチンは、同じ変数を使用できます。

サブルーチンは値を返しません。

7.7.2 関数

関数は既定でプロシージャ（推奨）となります。ただし、プロシージャではない関数を定義することも可能です。このセクションでは、このような関数について説明しています。

7.7.2.1 構文

以下はプロシージャではない関数の構文です。

```
label ( [param [ = default ] ] [ , ... ] )
  code
  QUIT expression
```

以下の構文で呼び出します。

```
command $$label([param[ ,...]])
```

あるいは以下ようになります。

```
DO label([param[ ,...]])
```


引数	説明
label	関数名。標準ラベルです。必ず 1 列目から始まります。label に続くパラメータの括弧は必須です。
param	関数に必要な各パラメータの変数です。必要なパラメータは、仮パラメータ・リストと呼ばれます。このリストには、何も指定しなくても、1 つ以上の param を定義してもかまいません。複数の param 値がある場合、コンマによって区切られます。InterSystems IRIS は、参照された param 変数に対して、自動的に NEW を呼び出します。パラメータは、値または参照で仮パラメータ・リストに渡すことができます。
default	直前の param に対するオプションの既定値です。各パラメータに対して既定値を提供しても、省略してもかまいません。仮パラメータに対応する実パラメータが指定されていない場合、または実パラメータが参照渡しされたときに当該のローカル変数に値がない場合は、既定値が適用されます。この既定値はリテラル、つまり引用符で囲まれた数字あるいは文字列です。null 文字列 () も既定値として指定できます。null 文字列は変数を定義するため、これは既定値を指定しないこととは異なります。一方でパラメータ値が未指定の場合や既定値がない場合、変数は未定義となります。リテラル以外の既定値を指定すると、InterSystems IRIS は <PARAMETER> エラーを発行します。
code	コード・ブロックです。コード・ブロックでは、他のサブルーチン、関数、プロシージャを入れ子にして呼び出すことができます。入れ子になった呼び出しは、DO コマンドまたは関数呼び出しで実行する必要があります。GOTO コマンドでは、関数のコード・ブロックを終了できません。コード・ブロックは、式を持つ明示的な QUIT コマンドを使用してのみ終了できます。
expression	有効な ObjectScript 式で指定される関数戻り値です。式を持つ QUIT コマンドは、ユーザ定義関数では必須です。式から生じる値は、関数の結果として呼び出し元に返されます。

7.7.2.2 概要

このセクションでは、ユーザ定義関数について説明します。ユーザ定義関数の呼び出しは、接頭語 \$\$ で識別されます。(ユーザ定義関数は、外部関数 ともいいます。)

ユーザ定義関数により、InterSystems IRIS が提供する関数に新たな関数を追加できます。通常、どのプログラムからでも呼び出せる汎用的な処理を実装するために関数を使用します。

関数は、常に ObjectScript コマンドから呼び出されます。式として評価され、呼び出し元のコマンドに単一の値を返します。例えば以下ようになります。

ObjectScript

```
SET x=$$myfunc()
```

7.7.2.3 関数パラメータ

原則として、ユーザ定義関数はパラメータ渡しを使用します。ただし、外部に提供する値を持たない場合も、関数は機能します。例えば、乱数を生成する関数の定義や、システム日付を既定以外の形式で返す関数の定義ができます。このような場合、パラメータの括弧は、パラメータ・リストが空でも関数定義と関数呼び出しの両方に必要です。

パラメータ渡しでは、以下が必要です。

- 関数呼び出しの実パラメータ・リスト
- 関数定義の仮パラメータ・リスト

InterSystems IRIS がユーザ定義関数を実行するとき、実パラメータ・リストのパラメータが、それに対応する仮リストのパラメータに位置を基準としてマップされます。例えば、実リストにある最初のパラメータの値は、仮リストにある最初の変数に配置し、2 番目の値は 2 番目の変数に配置し、以降の値も同様に配置します。これらのパラメータの照合は、名前ではなく、パラメータの位置で行われます。したがって、実パラメータに使用する変数と仮パラメータに対して使用する変数は、同じ名前を持つ必要はありません（また、通常持つべきではありません）。関数は、仮リスト内の該当する変数を参照することにより、渡された値にアクセスします。

仮リストにある変数が実リストにあるパラメータよりも多く、既定値が各パラメータに指定されていない場合、余った変数は未定義のままとなります。したがって、関数のコードには適切な If テストを記述して、各関数の参照において使用可能な値が必ず渡されることを確認する必要があります。

ユーザ定義関数にパラメータを渡す場合、値または[参照](#)で渡すことができます。同じ関数呼び出し内で、値または参照渡しの両方を混合して指定することができます。詳細は、“[パラメータ渡し](#)”を参照してください。

7.7.2.4 返り値

ユーザ定義関数を定義するための構文は以下のとおりです。

```
label ( parameters )
  code
  QUIT expression
```

関数は、QUIT コマンドの後に式を含む必要があります。InterSystems IRIS は、QUIT に遭遇すると関数の実行を終了し、関連式が生成する単一値を呼び出し元プログラムに返します。

式を含めずに QUIT コマンドを指定すると、InterSystems IRIS はエラーを生成します。

7.7.2.5 変数

呼び出し元のプログラムと呼び出される関数は、同じ変数一式を使用します。以下は特に考慮すべき点です。

- InterSystems IRIS は仮リストの各パラメータに対し、暗黙の NEW コマンドを実行します。以下の例では、myfunc が呼び出されると x が再度初期化されます。

ObjectScript

```
mainprog
  SET x=7
  SET y=$$myfunc(99)
myfunc(x)
  WRITE x
  QUIT 66
```

- システムは、関数を開始した時点でシステム変数 \$TEST の現在値を保存し、関数の終了時にその値を復元します。他の方法を使用して値を保存するコードを明示的に記述しない限り、関数実行中に \$TEST 値を変更しても、関数の終了時にその値は破棄されます。

7.7.2.6 関数の位置

ユーザ定義関数は、それを参照するルーチン内に定義する、あるいは複数のプログラムから参照できるよう別々のルーチン内に定義できます。しかし、1 つのルーチンにすべてのユーザ定義関数を定義することをお勧めします。これにより、関数の検証や更新の際に、関数定義の検索が簡単にできます。

7.7.2.7 ユーザ定義関数の呼び出し

\$\$ 接頭語または DO コマンドを使用して、ユーザ定義関数を呼び出すことができます。\$\$ 形式は最も機能的であり、一般的に好まれる形式です。

\$\$ 接頭語の使用法

式を使用できるコンテキスト内で、ユーザ定義関数を呼び出すことができます。以下の形式で、ユーザ定義関数を呼び出します。

```
$$name([param [, ...]])
```

以下はその説明です。

- ・ name 関数の名前を指定します。関数が定義される場所に応じて、以下のように名前を指定できます。
 - label – 現在のルーチン内の行ラベル
 - label^routine – ディスクにある名前付きルーチン内の行ラベル。
 - ^routine – ディスクに保存されているルーチン。ルーチンは、実行する関数のコードのみを含んでいる必要があります。

ルーチンはネームスペース内で定義されます。[拡張ルーチン参照](#)を使用し、以下のようにして現在のネームスペース以外のネームスペースで定義されたルーチン内にあるユーザ定義関数を実行できます。

ObjectScript

```
WRITE $$myfunc^|"USER"|routine
```

- ・ param は関数に渡される値を指定します。提供されるパラメータは、実パラメータ・リストと呼ばれます。これらは、その関数に関して定義された仮パラメータ・リストと一致する必要があります。例えば、関数コードには 2 つのパラメータがあり、1 番目は数値、2 番目は文字リテラルとします。この場合、1 番目のパラメータに文字リテラル値、2 番目に数値を指定すると、関数は不正確な値を返すか、あるいはエラーを生成する可能性があります。仮パラメータ・リストのパラメータでは、関数から NEW コマンドが必ず呼び出されます。詳細は、[NEW コマンドの説明](#)を参照してください。パラメータは、値または[参照によって](#)渡されます。詳細は、“[パラメータ渡し](#)”を参照してください。関数の仮パラメータ・リストに記載されている数より少数のパラメータを関数に渡すと、パラメータの既定値が使用されます(定義されている場合)。既定値がない場合、パラメータは未定義のままとなります。

DO コマンドの使用法

DO コマンドを使用して、ユーザ定義関数を呼び出すことができます (DO コマンドを使用して、システム関数を呼び出すことはできません)。DO コマンドで呼び出した関数は、値を返しません。つまり、関数は必ず返り値を生成しますが、DO コマンドはこの返り値を無視します。したがって、ユーザ定義関数を呼び出す DO コマンドの用途は大幅に制限されます。

DO コマンドを使用してユーザ定義関数を呼び出すには、以下の構文でコマンドを発行します。

```
DO label(param[ , ...])
```

DO コマンドは、label という関数を呼び出し、param で指定されたパラメータ (存在する場合) を関数に渡します。\$\$ 接頭語を使用しないことと、パラメータの括弧が必須であることに注意してください。label と param を指定する際には、\$\$ 接頭語を使用するユーザ定義関数を呼び出すときと同じ規則が適用されます。

関数は、常に値を返す必要があります。しかし、DO コマンドで関数を呼び出す場合、返された値は呼び出し元のプログラムで無視されます。

8

マクロとインクルード・ファイルの使用

ここでは、マクロと(マクロが含まれる)インクルード・ファイルを定義して使用方法について説明します。InterSystems IRIS® データ・プラットフォームには、[システム・マクロ](#)が用意されており、こちらも使用できます。

重要

インクルード・ファイルというフレーズは、これまでの経緯から使用されていますが、残念ながらいくらかの混乱も生じさせています。InterSystems IRIS では、インクルード・ファイルは、実際にはファイル (オペレーティング・システム内の個別のスタンドアロンのファイル) ではありません。クラスやルーチンと同じように、インクルード・ファイルは、InterSystems IRIS データベース内に保存されるコード・ユニットです。

その他のコード要素の場合と同様に、適切な IDE により、インクルード・ファイルを作成するオプションが提供され、コードがデータベースに正しく保存されます。同様に、IDE がソース・コントロール・システムに接続されている場合、各コード要素は、ソース・コントロールを介して管理される外部ファイルに投影されます。

8.1 マクロの基本情報

マクロは、便利な置換表現として以下のように定義して使用できます。

1. マクロは特殊な構文、通常は `#define` 指示文を介して定義します。以下に例を示します。

ObjectScript

```
#define StringMacro "Hello, World!"
```

この構文は、`StringMacro` と呼ばれるマクロを定義します。このマクロ名の大文字と小文字は区別されます。

2. 後で、このマクロを構文 `$$$macroname` を使用して呼び出せます。

ObjectScript

```
write $$$StringMacro
```

前述した構文は、以下と等価です。

ObjectScript

```
write "Hello, World!"
```

この置換は、コード（クラスまたはルーチン）のコンパイル時に行われます。特に、クラスまたはルーチン自体が変更されていなくても .INT コードが生成されている場合は、置換が行われたことを示しています（コードがどのようにコンパイルされるかについての詳細は、“[これらのコード要素がどのように連携しているか](#)”を参照してください）。

マクロはテキスト置換であることに注意してください。置換が実行されたら、結果として生成される文は、構文的に正しいものである必要があります。したがって、式を定義するマクロは、式を必要とするコンテキスト内で呼び出される必要があります。また、コマンドとその引数に対するマクロは、ObjectScript の独立した行として置くことができます。

8.2 インクルード・ファイルの基本情報

通常、マクロはインクルード・ファイル内で定義します。その後、このインクルード・ファイルをその他のコード内に含めて、そのコードがマクロを参照できるようにします。これは、以下のように機能します。

1. インクルード・ファイルは、データベース内に格納される特殊な種類のコード・ユニットです。以下は、部分的な例です。

ObjectScript

```
#; Optional comment lines
#define RELEASEID $GET(^MyGlobal("ReleaseID"), "")
#define RELEASENUMBER $GET(^MyGlobal("ReleaseNumber"), "")
#define PRODUCT $GET(^MyGlobal("Product"), "")
#define LOCALE $GET(^MyGlobal("Locale"), "en-us")
```

各行はコメント行であるか、`#define` 指示文で開始することに注意してください。空白行も許可されます。より複雑なマクロを定義できる `#define` の代替手段もありますが、それらについては[別の場所](#)でより詳しく説明します。

一般的なシナリオでは、IDE でインクルード・ファイルを作成し、それを `MyMacros` のような特定の名前で保存します。

2. マクロを使用する必要があるクラスまたはルーチン内に、インクルード・ファイルを含めます。以下に例を示します。

Class Definition

```
include MyMacros

Class
MyPackage.MyClass {

    //
}
```

この例では、インクルード・ファイルの名前は `MyMacros` です。

この手順により、`MyMacros` のマクロがクラスまたはルーチン内で使用できるようになります。

ルーチンごとに異なる構文のすべてのバリエーションについては、“[インクルード・ファイルを含める方法](#)”を参照してください。

3. 同じクラスまたはルーチン内で、構文 `$$$macroname` を使用してマクロを参照します。以下に例を示します。

ObjectScript

```
set title=$$$PRODUCT_ " _$$$RELEASENUMBER
```

注釈 実行されるテキストでは、インクルード・ファイル名に通常 `.inc` を追加します。例えば、一連の有用な[システム・マクロ](#)は `%occStatus.inc` および `%occMessages.inc` インクルード・ファイルで定義されています。

8.3 マクロの定義

最も基本的な形式では、“[マクロの基本情報](#)”で示したように、マクロは `#define` 指示文を使用して作成されます。

引数を受け入れ、より複雑なシナリオをサポートするマクロを定義できるようにする、追加の指示文もあります。また、`##continue` を使用して、`#define` 指示文を次の行に継続することもできます。詳細は、“[システム・プリプロセッサ・コマンド・リファレンス](#)”を参照してください。

ここでは、[マクロを定義できる場所](#)、[マクロ定義](#)に含めることができる内容、[マクロ名](#)のルール、[マクロ内での空白](#)の使用、および[マクロのコメント](#)に関する情報を提供します。

8.3.1 マクロを定義する場所

マクロは以下の場所で定義できます。各場所が、マクロの可用性に影響を及ぼします。

- ・ [インクルード・ファイル](#)内でマクロを定義できます。この場合、マクロは必要なインクルード・ファイルを含む任意のコード内で利用できます。
クラスにインクルード・ファイルが含まれる場合、そのクラスのサブクラスには自動的に同じインクルード・ファイルが含まれます。
- ・ メソッド内でマクロを定義できます。この場合、マクロはそのメソッド内で利用できます。
- ・ ルーチン内でマクロを定義できます。この場合、マクロはそのルーチン内で利用できます。

8.3.2 許可されるマクロ定義

以下はサポート対象の機能です。

- ・ 上記の例のような文字列置換
- ・ 数値置換

ObjectScript

```
#define NumberMacro 22
```

ObjectScript

```
#define 25M ##expression(25*1000*1000)
```

通常、ObjectScript のマクロ定義では、文字列には引用符が必要ですが、数値には必要ありません。

- ・ 変数置換

ObjectScript

```
#define VariableMacro Variable
```

マクロ名は、既に定義済みの変数名に置換されます。変数が未定義の場合、`<UNDEFINED>` エラーが発生します。

- ・ コマンドと引数の呼び出し

ObjectScript

```
#define CommandArgumentMacro(%Arg) WRITE %Arg,!
```

マクロ引数名は、上記の %Arg 引数のように、% 文字で始まる必要があります。このマクロは、%Arg 引数を使用する WRITE コマンドを呼び出します。

- 関数、式、演算子の使用

ObjectScript

```
#define FunctionExpressionOperatorMacro ($ZDate(+$Horolog))
```

ここでは、マクロ全体が式で、その値は、\$ZDate 関数の返り値になります。\$ZDate は、システム変数 \$Horolog に格納されているシステム時刻に + 演算子を付けた演算の結果として生じる式を処理します。上記のように、式を括弧で囲むことをお勧めします。これにより、式が使用されている文とのやり取りが最小限に抑えられます。

- 他のマクロの参照

ObjectScript

```
#define ReferenceOtherMacroMacro WRITE $$$ReferencedMacro
```

このマクロは、WRITE コマンドへの引数として、他のマクロの式の値を使用します。

注釈 1 つのマクロが別のマクロを参照する場合、参照先のマクロは、参照元のマクロの前にコンパイル済みのコード行に表示されている必要があります。

8.3.3 マクロの名前付け規約

- 最初の文字は、英数字またはパーセント記号 (%) でなくてはならない。
- 2 番目以降の文字は英数字でなくてはならない。マクロ名には、スペース、アンダースコア、ハイフンなどの記号文字を含めることはできません。
- マクロ名では大文字と小文字が区別される。
- マクロ名は最長 500 文字。
- マクロ名には日本語の全角文字および半角かな文字を使用可能。詳細は、“[パターン・マッチ演算子 \(?\)](#)”にある“パターン・コード”のテーブルを参照してください。
- ISCname.inc ファイルがシステム用に予約されているため、マクロ名を ISC で開始することはできません。

8.3.4 マクロの空白規約

- 規約では、マクロ指示文をインデントさせずに、列 1 にて記述することになっています。しかし、マクロ指示文はインデントさせても構いません。
- 1 つ以上のスペースをマクロ指示文に続けることができます。マクロ内では、任意数のスペースをマクロ指示文、マクロ名、マクロ値の間に入れることができます。
- マクロ指示文は単一行の文となります。マクロ指示文、マクロ名、およびマクロ値はすべて同一行で記述する必要があります。[##continue](#) を使用すると、マクロ指示文を次の行に継続できます。
- [#if](#) 指示文と [#elseif](#) 指示文はテスト式の形をとります。テスト式にはスペースを含めることはできません。
- [#if](#) 式、[#elseif](#) 式、[#else](#) 指示文、および [#endif](#) 指示文はすべて、それぞれの独立した行に記述します。同一行にてこれらの指示文のいずれかに続くものはコメントと見なされるので、解析はされません。

8.3.5 マクロ・コメントとスタジオ・アシスト

マクロには、その定義の一部として渡すコメントを記述できます。`/*`と`*/`、`/**`、`*/`、`#`、`;`、`;`、および`;`で区切ったコメントは、すべて通常どおりに機能します。“[コメント](#)”を参照してください。

`///`で始まるコメントには特別な機能があります。インクルード・ファイルにあるマクロでスタジオ・アシストを使用する場合、マクロの定義の直前の行に`///`コメントを記述します。これにより、そのマクロの名前がスタジオ・アシストのポップアップに表示されます（現在のファイルにあるすべてのマクロがスタジオ・アシストのポップアップに表示されます）。例えば、以下のコードを`#include`指示文で参照する場合、最初のマクロの名前はスタジオ・アシストのポップアップに表示されますが、2番目のマクロの名前は表示されません。

ObjectScript

```
/// A macro that is visible with Studio Assist
#define MyAssistMacro 100
//
// ...
//
// A macro that is not visible with Studio Assist
#define MyOtherMacro -100
```

インクルード・ファイルでマクロを使用可能にする方法については、“[インクルード・ファイルを含める方法](#)”を参照してください。

8.4 インクルード・ファイルを含める方法

ここでは、コードに[インクルード・ファイル](#)を含める方法について説明します。

- ・ クラス内またはルーチンの開始部にインクルード・ファイルを含めるには、以下の形式で指示文を使用します。

ObjectScript

```
#include MacroIncFile
```

ここで、`MacroIncFile`は、**MacroIncFile.inc**と呼ばれるマクロを含むインクルード・ファイルを参照しています。参照先ファイルが`#include`の引数となっている場合には、`.inc`接尾語が参照先ファイルの名前に含まれないことに注意してください。`#include`指示文では大文字と小文字は区別されません。

クラスにインクルード・ファイルが含まれる場合、そのクラスのサブクラスには自動的に同じインクルード・ファイルが含まれます。

例えば、**MyMacros.inc**というファイルに1つ以上のマクロがある場合、以下の呼び出しに組み込むことができます。

ObjectScript

```
#include MyMacros
```

- ・ 1つのルーチン内に複数のインクルード・ファイルを含めるには、同じ形式の複数の指示文を使用します。以下に例を示します。

ObjectScript

```
#include MyMacros
#include YourMacros
```

- ・ クラス定義の開始部に複数のインクルード・ファイルを組み合わせるには、構文を以下の形式とします。

```
include (MyMacros, YourMacros)
```

この `include` 構文には、先頭にシャープ記号がないことに注意してください。この構文は `#include` には使用できません。

[#include](#) のリファレンス・セクションを参照してください。

クラス定義をコンパイルすると、そのプロセスによってさまざまな方法で(空白の削除など)クラス定義が正規化されることに注意してください。このような正規化の一環として、`include` 指示文の大文字と小文字が変換されます。

ObjectScript コンパイラは、外部マクロを含めることを許可する `/defines` 修飾子を提供します。詳細は、“\$SYSTEM” リファレンス・ページの “[コンパイラ修飾子](#)” を参照してください。

8.5 マクロが拡張される条件

前述のように、クラスやルーチンをコンパイルする際は、システムにより INT コード (中間 ObjectScript) コードが生成されます。この INT コードを表示し、読み取ることで、ある種のトラブルシューティングの実行に役立ちます。

注釈 プリプロセッサは、ObjectScript パーサによって埋め込み SQL が処理される前にマクロを拡張します。プリプロセッサは、埋め込みまたは遅延いずれかのコンパイル・モードの[埋め込み SQL](#)をサポートしています。プリプロセッサは、[ダイナミック SQL](#)でのマクロ拡張は行いません。

ObjectScript パーサは、プリプロセッサ指示文を解析する前に、複数の行から成るコメントを削除します。そのため、`/* */` で囲んで記述した複数行コメントの中で指定したマクロ・プリプロセッサ指示文は、実行されません。

また以下のグローバルには MAC コード (元のソース・コード) が含まれます。これらのグローバルとその添え字を表示するには、`ZWRITE` を使用します。

- ・ `^rINDEX(routinename,"MAC")` には、MAC コードが変更後に最後に保存された日時を表すタイムスタンプと、この MAC コード・ファイルの文字数が含まれます。文字数には、コメントおよび空白行が含まれます。MAC コードが最後に保存された日時を表すタイムスタンプ、コンパイルされた日時を表すタイムスタンプ、および使用された `#include` ファイルに関する情報は、.INT コードの [^ROUTINE グローバル](#)に記録されます。.INT コードの詳細は、“[ZLOAD](#)” コマンドを参照してください。
- ・ `^rMAC(routinename)` には、MAC ルーチンの各コード行の添え字ノードが含まれます。また、`^rMAC(routinename,0,0)` には行数、`^rMAC(routinename,0)` には MAC ルーチンが最後に保存された日時のタイムスタンプ、`^rMAC(routinename,0,"SIZE")` には文字数がそれぞれ含まれます。
- ・ `^rMACSAVE(routinename)` には、MAC ルーチンの履歴が含まれます。ここには、保存された過去の MAC ルーチン・バージョン 5 つ分の `^rMAC(routinename)` と同じ情報が含まれます。現在の MAC バージョンに関する情報は含まれません。

8.6 関連項目

- ・ [#define](#)
- ・ [#include](#)
- ・ [システム・プリプロセッサ・コマンド・リファレンス](#)

- ・ [システム・マクロ](#)

9

埋め込み SQL

ObjectScript 内に SQL を埋め込むことができます。

9.1 埋め込み SQL

埋め込み SQL を使用すると、ObjectScript プログラム内に SQL コードを含めることができます。構文は `&sql()` です。例えば、以下のようになります。

ObjectScript

```
&sql( SELECT Name INTO :n FROM Sample.Person )  
WRITE "name is: ",n
```

埋め込み SQL は、これを含むルーチンのコンパイル時にはコンパイルされません。埋め込み SQL のコンパイルは、SQL コードの最初の実行時 (ランタイム) に行われます。

詳細は、“[埋め込み SQL の使用法](#)” を参照してください。

9.2 その他の形式のクエリ

%SQL クラスで提供される API を使用することで、その他の方法で ObjectScript 内に SQL クエリを含めることができます。“[ダイナミック SQL の使用法](#)” を参照してください。

10

多次元配列

InterSystems IRIS® Data Platform は、多次元配列をサポートします。多次元配列は、1 つ以上の要素を含む永続変数で、それぞれが一意的な添え字を持ちます。ユーザは異なる種類の添え字を組み合わせで使用できます。例は、以下の MyVar 配列です。

- ・ MyVar
- ・ MyVar(22)
- ・ MyVar(-3)
- ・ MyVar("MyString")
- ・ MyVar(-123409, "MyString")
- ・ MyVar("MyString", 2398)
- ・ MyVar(1.2, 3, 4, "Five", "Six", 7)

配列ノード MyVar は [ObjectScript 変数](#)であり、この変数タイプの規約に従います。

MyVar の添え字は、正と負の数、文字列、あるいはこれらの組み合わせになります。添え字には Unicode 文字をはじめ、あらゆる文字を含めることができます。数値添え字は[キャノニック形式の数](#)として格納および参照されます。文字列の添え字は大文字と小文字を区別するリテラルとして格納および参照されます。キャノニック形式の数（または、キャノニック形式の数へと変わる数）およびキャノニック形式の数を含む文字列は同等の添え字となります。

10.1 多次元配列の概要

多次元配列とは、添え字で示される n 次元の永続配列です。個別ノードは“グローバル”ともいい、InterSystems IRIS データ・ストレージの構成要素でもあります。以下のような他の特性もあります。

- ・ ツリー構造で存在します。
- ・ スパース（まばらな）配列
- ・ 3 つの基本的な種類の 1 つです。

10.1.1 多次元ツリー構造

多次元配列の全体的な構造は ツリー と呼ばれ、上から下に拡張します。ルート、上記の MyVar が頂点です。ルートと他の添え字が付いた形式を ノード と呼びます。下にノードを持たないノードは、葉と呼ばれます。下にノードを持つノード

ドは、親や先祖と呼ばれます。親を持つノードは、子や子孫と呼ばれます。また、同じ親を持つ子は兄弟と呼ばれます。すべての兄弟がツリーに追加されると、数値あるいはアルファベット順で自動的に並べ替えられます。

10.1.2 スパース多次元ストレージ

多次元配列はスパース (まばらな) 配列です。つまり、上記の例では、それぞれ定義されたノードに対し、7 つのメモリ位置のみを使用しています。また、配列を宣言したり、ディメンジョンを指定したりする必要がないため、以下のようなメモリに関する利点があります。すなわち、多次元配列に対しては事前に予約される領域がない、多次元配列は必要になるまで領域を使用しない、および多次元配列が使用する領域はすべて動的に割り当てられる、の 3 点です。例えば、配列を使用して、チェッカーと呼ばれるゲームでのプレーヤの区画を追跡するとします。チェッカーボードは、 8×8 です。 8×8 のチェッカーボード・サイズの配列を必要とする言語では、チェッカーによって占められる位置が 24 個を超えることがないにもかかわらず、64 個のメモリ位置が使用されます。一方で、ObjectScript では、配列は最初に必要な位置は 24 個となり、必要な位置はゲームの進行と共にさらに少なくなります。

10.1.3 多次元配列の種類

多次元配列は、3 つの基本的な種類のいずれかになります。いずれの場合も、配列は、1 つ以上の添え字を持つという点でスカラ式と区別できます。

- ・ 添え字を持つローカル変数はすべて配列です。例えば、`x(1)` を作成すると、`x` が配列として定義されます。
- ・ 添え字を持つグローバルはすべて配列です。例えば、`^y(1)` を作成すると、`^y` が配列として定義されます。
- ・ クラス内のプロパティは、その定義に `MultiDimensional` キーワードがあれば多次元配列にすることができます。例えば、以下のようになります。

```
Property MyProp as %String [ MultiDimensional ];
```

つづいて、以下のような文で値を設定できます。

```
myObj.MyProp(1) = "hello world"
```

永続クラスまたはシリアル・クラスの **多次元プロパティ** は、プロパティを保存するカスタム・コードを記述しない限り、オブジェクトを保存してもディスクに保存されません。

10.2 多次元配列の操作

Read コマンドと Write コマンドをそれぞれ使用して、多次元配列からデータの読み取りと書き込みができます。

InterSystems IRIS は、多次元配列と連動する包括的なコマンドと関数一式を提供します。

- ・ **Set** コマンドは、配列に値を設定します。
- ・ **Kill** コマンドは、配列構造のすべて、または一部を削除します。
- ・ **Merge** コマンドは、配列構造のすべてまたは一部を 2 番目の配列構造にコピーします。
- ・ **\$Order** および **\$Query** コマンドは、配列のコンテンツ全体に対して繰り返すことができます。
- ・ **\$Data** コマンドは、配列内にノードが存在するかどうかをテストできます。

このコマンドと関数一式は、多次元グローバルと多次元ローカル配列で生成できます。グローバルは “^” (キャレット文字) で始まるため、簡単に識別できます。

10.3 関連項目

多次元配列の詳細は、“[グローバルの使用法](#)”を参照してください。

11

文字列演算

ObjectScript は、それぞれ独自の目的と機能に合わせて文字列に関連する演算グループを提供します。これらは、基本的な文字列演算と関数、区切り文字列演算、およびリスト構造文字列演算です。

11.1 基本的な文字列演算と関数

ObjectScript の基本的な文字列演算では、文字列でさまざまな処理を実行できます。その処理は、以下のとおりです。

- ・ `$LENGTH` 関数は、文字列の文字数を返します。以下はそのコードの例です。

ObjectScript

```
WRITE $LENGTH("How long is this?")
```

これは、文字列の長さ、17 を返します。詳細は、“[\\$LENGTH](#)”を参照してください。

- ・ `$JUSTIFY` は、文字列の左側をスペース文字で埋め、右揃えした文字列を返します (数値演算も実行できます)。以下はコードの例です。

ObjectScript

```
WRITE "one", !, $JUSTIFY("two", 8), !, "three"
```

これは、文字列 `two` の位置を 8 文字に調整し、返します。

```
one      two
three
```

詳細は、“[\\$JUSTIFY](#)”を参照してください。

- ・ `$ZCONVERT` は、文字列の形式を変換します。大文字と小文字の変換 (大文字変換、小文字変換、タイトル文字変換など) とコード変換 (さまざまな形式でコード化された文字間の変換) をサポートします。以下はコードの例です。

ObjectScript

```
WRITE $ZCONVERT("cRAZy cAPs", "t")
```

これは、以下を返します。

```
CRAZY CAPS
```

詳細は、“\$ZCONVERT”を参照してください。

- ・ \$FIND 関数は、文字列を部分文字列で検索し、その部分文字列の次にある文字の位置を返します。以下はコードの例です。

ObjectScript

```
WRITE $FIND("Once upon a time...", "upon")
```

これは、upon の直後の文字位置である 10 を返します。詳細は、“\$FIND”を参照してください。

- ・ \$TRANSLATE 関数は、文字列内で文字単位の置換を行います。以下はコードの例です。

ObjectScript

```
SET text = "11/04/2008"
WRITE $TRANSLATE(text, "/", "-")
```

これは、日付のスラッシュをハイフンに置き換えます。詳細は、“\$TRANSLATE”を参照してください。

- ・ \$REPLACE 関数は、文字列内で文字列単位の置換を行います。演算の対象とする文字列の値は変更しません。例えば、次のコードは 2 つの異なる処理を行います。

ObjectScript

```
SET text = "green leaves, brown leaves"
WRITE text,!
WRITE $REPLACE(text,"leaves","eyes"),!
WRITE $REPLACE(text,"leaves","hair",15),!
WRITE text,!
```

最初の呼び出しの \$REPLACE では、文字列 leaves を文字列 eyes に置き換えます。2 番目の呼び出しの \$REPLACE では、(4 番目の引数で指定された位置である) 15 番目の文字の前にあるすべての文字を破棄し、文字列 leaves を文字列 hair に置き換えます。text 文字列の値は、どの \$REPLACE 呼び出しでも変更されません。詳細は、“\$REPLACE”を参照してください。

- ・ \$EXTRACT 関数は、文字列の中で指定の位置から部分文字列を返します。以下はコードの例です。

ObjectScript

```
WRITE $EXTRACT("Nevermore"),$EXTRACT("prediction",5),$EXTRACT("xon/xoff",1,3)
```

これは、3 つの文字列を返します。1-引数形式は文字列の最初の文字、2-引数形式は文字列から指定された文字、3-引数形式は指定の文字から始まり指定の文字で終わる部分文字列をそれぞれ返します。ここでは改行がないため、以下の値を返します。

```
Nixon
```

詳細は、次のセクションと“\$EXTRACT”を参照してください。

11.1.1 \$EXTRACT の高度な機能

\$EXTRACT 関数と SET コマンドを組み合わせることで、文字列の左側をスペースで埋めることができます。

ObjectScript

```
SET x = "abc"
WRITE x,!
SET $EXTRACT(y, 3) = x
SET x = y
WRITE x
```

このコードは、文字列 abc を取得し、文字列 y の 3 番目に置きます。y には値が指定されていないので、\$EXTRACT では、文字列に対する埋め込み文字として機能する空白と見なされます。

\$EXTRACT を使用して、変数の特定位置に新たな文字列を挿入することもできます。新、旧文字列の長さが同じかどうかにかかわらず、指定の文字を抜き出し、そこに提供された部分文字列を置きます。例えば以下ようになります。

ObjectScript

```
SET x = "1234"
WRITE x,!
SET $EXTRACT(x, 3) = "abc"
WRITE x,!
SET $EXTRACT(y, 3) = "abc"
WRITE y
```

このコードは、x の値として 1234 を設定します。次に \$EXTRACT を使用して、x の 3 番目の文字を抜き出し、その位置に abc を挿入します。その結果、文字列 12abc4 が作成されます。

11.2 区切り文字列演算

ObjectScript には、データの関連する部分を 1 つの文字列として渡すことができるよう、文字列を一連の部分文字列として処理できる関数があります。これらの関数は以下のとおりです。

- ・ \$PIECE – 指定の区切り文字に基づいて文字列内の特定の部分を返します。その範囲や、複数の区切り文字を元に、1 つの文字列から複数の個所を返すこともできます。
- ・ \$LENGTH – 指定の区切り文字に基づいて 1 つの文字列にある構成要素の数を返します。

\$PIECE 関数独自の重要な機能として、専用の区切り文字 (例えば ^) で区切って記述した複数の部分文字列を単一の文字列として使用できます。長い文字列はレコードとして、部分文字列はそのフィールドとして動作します。

\$PIECE の構文は以下のとおりです。

ObjectScript

```
WRITE $PIECE("ListString","QuotedDelimiter",ItemNumber)
```

ListString は使用されている完全なレコードを含む、引用符で囲まれた文字列です。QuotedDelimiter は指定の区切り文字で、引用符内に記述する必要があります。ItemNumber は、返されるように指定された部分文字列です。例えば以下の構文で、スペースで区切られた以下のリストで 2 番目のアイテムを表示します。

ObjectScript

```
WRITE $PIECE("Kennedy Johnson Nixon"," ",2)
```

これは、Johnson を返します。

また、リストから複数の値を返すこともできます。以下はその構文です。

ObjectScript

```
WRITE $PIECE("Nixon***Ford***Carter***Reagan","***",1,3)
```

これは、Nixon***Ford***Carter を返します。この値は、実際の部分文字列を参照する必要があります。また、3 番目の引数 (ここでは 1) は、4 番目の引数 (ここでは 3) よりも小さい値になる必要があることに注意してください。

以下のリストのように、区切り文字は任意に決めることができます。

ObjectScript

```
SET x = $PIECE("Reagan,Bush,Clinton,Bush,Obama",",",3)
SET y = $PIECE("Reagan,Bush,Clinton,Bush,Obama","Bush",2)
WRITE x,! ,y
```

これは、以下を返します。

```
Clinton
,Clinton,
```

最初のコードの区切り文字はコンマです。2 番目は、文字列 Bush が区切り文字であるため、返される文字列はコンマを含みます。区切り文字を正確に理解するために、次のセクションではリストに関連する関数を説明しています。

11.2.1 高度な \$PIECE 関数

要素値を区切ってリストに記述する \$PIECE を呼び出すことで、リストの中で固有の項目に部分文字列を配置し、他の部分は空として、十分な数の項目をリストに追加できます。例えば、あるコードで、リストの 1 番目、4 番目、そして 20 番目に項目を設定するとします。

ObjectScript

```
SET $PIECE(Alphalist, "^", 1) = "a"
WRITE "First, the length of the list is ", $LENGTH(Alphalist, "^"), ". ", !
SET $PIECE(Alphalist, "^", 4) = "d"
WRITE "Then, the length of the list is ", $LENGTH(Alphalist, "^"), ". ", !
SET $PIECE(Alphalist, "^", 20) = "t"
WRITE "Finally, the length of the list is ", $LENGTH(Alphalist, "^"), ". ", !
```

この \$LENGTH 関数は、1、4、20 の値を順番に返します。その長さに相当する必要な数のアイテムが、このコードで作成されているからです。ただし、アイテム 2、3、そして 5 から 19 には、値が設定されていません。そのため、これらの値を表示させようとしても、何も表示されません。

区切り文字列アイテムは、区切り文字列も含むことができます。このようなサブリストから値を取得するには、以下のように \$PIECE 関数の呼び出しを入れ子にします。

ObjectScript

```
SET $PIECE(Powers, "^", 1) = "1::1::1::1::1"
SET $PIECE(Powers, "^", 2) = "2::4::8::16::32"
SET $PIECE(Powers, "^", 3) = "3::9::27::81::243"
WRITE Powers,!
WRITE $PIECE($PIECE(Powers, "^", 2), "::", 3)
```

このコードは 2 行の出力を返します。1 行目は文字列 Powers で、すべての区切り文字を含みます。2 行目は 8 で、これは Powers の 2 番目の要素に含まれるサブリストの 3 番目の要素の値です。(Powers リストで、n 番目のアイテムは、2 を 1 乗から 5 乗までしたサブリストです。したがって、例えばサブリストの最初のアイテムは、n の 1 乗となります。)

詳細は、“\$PIECE”を参照してください。

11.3 リスト構造文字列演算

ObjectScript は、リストという特別な文字列の種類を定義します。これは、要素と呼ぶ部分文字列のエンコード・リストから構成されます。これらの InterSystems IRIS リストは、以下のリスト関数のみを使用して処理できます。

- ・ リスト作成：
 - － **\$LISTBUILD** は、パラメータ値として各要素を指定することでリストを作成します。

- `$LISTFROMSTRING` は、デリミタを含む文字列を指定することでリストを作成します。この関数はデリミタを使用して、文字列を要素に分割します。
- `$LIST` は、既存のリストからその一部を抽出してリストを作成します。
- ・ リスト・データ取得：
 - `$LIST` は、リスト要素の値をその位置別に返します。リストの先頭または末尾から位置をカウントできます。
 - `$LISTNEXT` は、リストの先頭から順番にリスト要素の値を返します。`$LIST` および `$LISTNEXT` はともにリストから要素を順番に返すために使用されますが、大量のリスト要素を返す場合は、`$LISTNEXT` の方が大幅に高速な処理が可能です。
 - `$LISTGET` は、リスト要素の値を位置別に返すか、または既定値を返します。
 - `$LISTTOSTRING` は、リストの要素の値すべてを区切り文字列として返します。
- ・ リストの操作：
 - `SET $LIST` は、リストの要素を挿入、更新、または削除します。`SET $LIST` は、リストの要素または要素の範囲を1つ以上の値で置換します。`SET $LIST` はリストの要素を複数の要素で置換できるため、これを使用してリストに要素を挿入できます。`SET $LIST` はリストの要素を `NULL` 文字列で置換できるため、これを使用してリストの要素または要素の範囲を削除できます。
- ・ リストの評価：
 - `$LISTVALID` は、文字列がリストに有効かどうかを判断します。
 - `$LISTLENGTH` は、リスト内の要素の数を判断します。
 - `$LISTDATA` は、指定のリスト要素がデータを含んでいるかどうか判断します。
 - `$LISTFIND` は、指定した値がリストにあるかどうかを判断して、リストの位置を返します。
 - `$LISTSAME` は、2つのリストが同一かどうか判断します。

リストはエンコードされた文字列であるため、InterSystems IRIS ではリストの扱いが標準の文字列の場合とは多少異なります。したがって、リストでは標準の文字列関数を使用しないようにします。さらに、ほとんどのリスト関数は、標準の文字列で使用する <LIST> エラーを発生します。

以下のプロシージャは、さまざまなリスト関数の使用法を示しています。

ObjectScript

```
ListTest() PUBLIC {
    // set values for list elements
    SET Addr="One Memorial Drive"
    SET City="Cambridge"
    SET State="MA"
    SET Zip="02142"

    // create list
    SET Mail = $LISTBUILD(Addr, City, State, Zip)

    // get user input
    READ "Enter a string: ", input, !, !

    // if user input is part of the list, print the list's content
    IF $LISTFIND(Mail, input) {
        FOR i=1:1:$LISTLENGTH(Mail) {
            WRITE $LIST(Mail, i), !
        }
    }
}
```

このプロシージャは、リストの重要な機能を示しています。

- ・ \$LISTFIND は、テストされた値がリスト項目に一致する場合にのみ 1 (True) を返します。
- ・ \$LISTFIND と \$LISTLENGTH は式で使います。

リスト関数の詳細は、“[ObjectScript リファレンス](#)” の関連リファレンス・ページを参照してください。

11.3.1 スパース・リストおよびサブリスト

リストの指定された位置に要素値を追加する関数では、適正位置に値を配置するために十分な数のリスト要素を追加します。以下はその例です。

ObjectScript

```
SET $LIST(Alphalist,1)="a"
SET $LIST(Alphalist,20)="t"
WRITE $LISTLENGTH(Alphalist)
```

この例にある 2 番目の \$LIST で 20 番目のリスト要素を作成しているので、\$LISTLENGTH は値 20 を返しますが、2 ～ 19 番目の要素には値が設定されていません。そのため、これらの値を表示させようとすると、〈NULL VALUE〉エラーが発生します。\$LISTGET を使用して、このエラーを回避できます。

List の要素自体をリストとすることもできます。このようなサブリストから値を取得するには、以下のように \$LIST 関数呼び出しを入れ子にします。

ObjectScript

```
SET $LIST(Powers,2)=$LISTBUILD(2,4,8,16,32)
WRITE $LIST($LIST(Powers,2),5)
```

このコードは、Powers リストの 2 番目の要素に含まれるサブリストの 5 番目の要素値、32 を返します。(Powers リストで 2 番目のアイテムは、2 を 1 乗から 5 乗までのサブリストです。したがって、例えばサブリストの最初のアイテムは、2 の 1 乗になります。)

11.4 リストと区切り文字列の比較

11.4.1 リストの利点

- ・ リストは特定の区切り文字を必要としません。\$PIECE 関数では複数のデータ項目を含む文字列を管理できますが、これは専用のデリミタとしての文字(または文字列)の設定にもよります。区切り文字を使用する場合は、データ項目の中にデータとして区切り文字が存在すると、データの区切り位置が混乱する可能性が常に存在しています。リストは区切り文字による混乱を避けるうえで効果的です。したがって、あらゆる文字または文字の組み合わせをデータとして入力できます。
- ・ データの要素は、\$PIECE を使用して区切り文字列から取得するよりも、\$LIST または \$LISTNEXT を使用してリストから取得する方が高速です。データを順番に取得する場合は、\$LIST よりも \$LISTNEXT の方が大幅に高速ですが、双方とも \$PIECE よりはきわめて高速です。

11.4.2 区切り文字列の利点

- ・ 区切り文字列では、\$FIND 関数を使用して、より柔軟なデータ内容の検索ができます。\$LISTFIND は、完全一致を必要とするので、リストの部分一致文字列を検索できません。したがって、上記の例では、アドレス One Memorial

Drive が文字 One で始まっているにもかかわらず、\$LISTFIND を使用してメール・リストの文字列 One を検索しても 0 (失敗を示す) が返されます。

- ・ 区切り文字列は標準文字列であるので、標準文字列関数をすべて使用できます。InterSystems IRIS リストはエンコードされた文字列なので、InterSystems IRIS リストでは \$List 関数のみが使用できます。

12

ロックと並行処理の制御

マルチプロセス・システムでの重要な機能の 1 つに並行処理の制御があります。これは、異なるプロセスがデータの特定の要素を同時に変更し、破損に至ることを防止する機能です。そのために、InterSystems IRIS® データ・プラットフォームにはロック管理システムが用意されています。このページでは、概要について説明します。

12.1 概要

基本のロック・メカニズムは、LOCK コマンドです。このコマンドの目的は、あるプロセス内のアクティビティを、別のプロセスが処理継続 OK の合図を送るまで遅延することです。

InterSystems IRIS では、ロックはそれ自体ではアクティビティを防止しません。規約では、ロックが機能するには、互いに競合するプロセスがすべて同じロック名でロックを実装していることが要求されています。例えば、以下に一般的なシナリオについて説明します。

1. プロセス A が LOCK コマンドを発行し、InterSystems IRIS がロックを作成します (既定では、排他ロック)。通常、プロセス A はグローバルのノードを変更します。詳細はアプリケーション固有です。
2. プロセス B が同じロック名で LOCK コマンドを発行します。排他ロックが存在するため、プロセス B は一時停止します。具体的には、この LOCK コマンドは復帰しなくなり、コードの後続行が実行できなくなります。
3. プロセス A がロックを解放すると、プロセス B の LOCK コマンドが復帰してプロセス B を継続します。通常、プロセス B は同じグローバルのノードを変更します。

12.2 ロック名

LOCK コマンドの引数の 1 つにロック名があります。ロック名は、任意の名前ですが、共通の規則として、プログラマはロックする項目の名前と同じロック名を使用します。通常、ロックする項目は、グローバルまたはグローバルのノードになります。したがって、通常ロック名は、グローバル名の名前またはグローバルのノードの名前のようになります。(このページでは、最も一般的であるキャレットで始まるロック名のみについて説明します。キャレットで始まらない名前のロックの詳細は、“[LOCK](#)” を参照してください。)

形式的には、ロック名は、“[変数](#)” で説明したローカル変数およびグローバル変数と同じ名前付け規約に従います。変数と同様に、ロック名は大文字と小文字を区別し、添え字を付けることができます。プロセス・プライベート・グローバル名をロック名として使用しないでください (定義によってそのようなグローバルにアクセスできるプロセスは 1 つしかいないため、そのようなロックは必要ありません)。

Tip ヒン 規約によってロックが機能し、ロック名が任意であるため、同じ名前のロックを作成する前に任意の変数を定義する必要はありません。

InterSystems IRIS によるメモリの割り当ておよび管理の方法により、ロック名の形式はパフォーマンスに影響します。ロックは、添え字を使用するロック名に対して最適化されます。例えば、`^sample.person(id)` です。

一方、InterSystems IRIS は、`^name_concatenated_identifier` のようなロック名に対しては最適化されません。添え字のないロック名を使用すると、ECP に関するパフォーマンスの問題の原因になる場合もあります。

12.3 ロック・テーブル

InterSystems IRIS は、現在のすべてのロックと、そのロックを所有しているプロセスを記録するために、システム規模のテーブルをメモリ内に保持しています。このテーブル (ロック・テーブル) は、管理ポータルからアクセスできます。管理ポータルでは、ロックの表示と、まれに必要なロックの削除を実行できます。特定のプロセスが、ロック名が異なる複数のロックを所有している可能性がある点に注意してください (複数のロックが同じ名前の場合もあります)。

プロセスが終了すると、システムは、そのプロセスが所有するすべてのロックを自動的に解放します。そのため、通常は、管理ポータルからロックを削除する必要はありません。ただし、アプリケーション・エラーが発生した場合を除きます。

ロック・テーブルは、固定サイズを超過することはありません (このサイズは `locksiz` 設定を使用して指定できます)。詳細は、“[ロックの監視](#)”を参照してください。そのため、ロック・テーブルが一杯になり、それ以上のロックが不可能になることがあります。この場合、InterSystems IRIS は、`messages.log` ファイルに以下のエラー・メッセージを書き込みます。


```
LOCK TABLE FULL
```

通常は、ロック・テーブルが一杯になっても、アプリケーション・エラーとは見なされません。InterSystems IRIS には、ロック・キューも用意されているため、プロセスはロック・テーブルにロックを追加する領域ができるまで待機します。(ただし、デッドロックは、アプリケーションのプログラミング・エラーです。“[デッドロックの回避](#)”を参照してください。)

12.4 ロックと配列

配列をロックする場合は、配列全体をロックすることも、配列内の 1 つ以上のノードをロックすることもできます。配列ノードをロックすると、他のプロセスは、そのノードに従属するどのノードもロックできなくなります。また、他のプロセスは、ロックされたノードの直接の祖先もロックできなくなります。

以下の図に例を示します。



暗黙的なロックは、ロック・テーブルに含まれていないため、[ロック・テーブル](#)のサイズに影響しません。

InterSystems IRIS のロック・キュー・アルゴリズムでは、同じロック名のロックは、リソースが直接的に競合しない場合でも受信した順にキューに置かれます。詳細と使用例は、[配列ノードに対するロックのキュー](#)を参照してください。

12.5 LOCK コマンドの使用法

ここでは、LOCK コマンドを使用してロックを追加および削除する方法について説明します。

12.5.1 増分ロックの追加

ロックを追加するには、以下のように LOCK コマンドを使用します。

```
LOCK +lockname
```

lockname はリテラルのロック名です。プラス記号(+)によって増分ロックが作成されます。これは一般的なシナリオです。一般的ではない代替手段は、[単純ロックの作成](#)を参照してください。

このコマンドは、以下を実行します。

1. 指定のロックをロック・テーブルに追加しようとします。つまり、このエントリはロック・キューに追加されます。

2. ロックを取得できるまで実行を一時停止します。

異なる動作をするさまざまなロック・タイプがあります。既定でないロック・タイプのロックを追加するには、以下のバリエーションを使用します。

```
LOCK +lockname#locktype
```

locktype は、二重引用符で囲んだロック・タイプ・コードの文字列です。“[ロック・タイプ](#)”を参照してください。

特定のプロセスが同じ名前の複数の増分ロックを追加できる点に注意してください。これらのロックは異なるタイプにもすべて同じタイプにもできます。

12.5.2 タイムアウト付き増分ロックの追加

正しくない方法で増分ロックを使用すると、デッドロックと呼ばれる好ましくない状態になる可能性があります。デッドロックについては、後述する“[デッドロックの回避](#)”で説明します。デッドロックを回避する 1 つの方法は、ロック作成時にタイムアウト時間を指定することです。これを実行するには、以下のように LOCK コマンドを使用します。

```
LOCK +lockname#locktype :timeout
```

timeout は、タイムアウト時間の秒数です。コロンの前のスペースはオプションです。timeout に 0 を指定すると、InterSystems IRIS はロックの追加を 1 回試行します（ただし、以下の[注意](#)を参照してください）。

このコマンドは、以下を実行します。

1. 指定のロックをロック・テーブルに追加しようとします。つまり、このエントリはロック・キューに追加されます。
2. ロックを取得できるまでまたはタイムアウト時間が経過するまで、実行を一時停止します。
3. \$TEST 特殊変数の値を設定します。ロックが取得されると、InterSystems IRIS は \$TEST を 1 に設定します。それ以外の場合、InterSystems IRIS は \$TEST を 0 に設定します。

つまり、タイムアウト引数を使用している場合、コードで \$TEST 特殊変数の値を調べ、その値を使用して続行するかどうかを選択する必要があります。以下に例を示します。

ObjectScript

```
Lock +^ROUTINE(routinename):0
If '$TEST { Return $$ERROR("Cannot lock the routine: ",routinename)}
```

12.5.2.1 ゼロ・タイムアウトに関する注意

前述のように、timeout に 0 を指定すると、InterSystems IRIS はロックの追加を 1 回試行します。ただし、ゼロのタイムアウトを使用して親ノードをロックしようとしたときに、既に子ノードをロックしている場合、ゼロのタイムアウトは無視され、代わりに内部の 1 秒のタイムアウトが使用されます。

12.5.3 ロックの削除

既定のタイプのロックを削除するには、以下のように LOCK コマンドを使用します。

```
LOCK -lockname
```

このコマンドを実行するプロセスが特定の名前の（既定のタイプの）ロックを所有している場合、このコマンドによってそのロックが削除されます。または、そのプロセスが複数の（既定のタイプの）ロックを所有している場合、このコマンドによってそれらのロックの 1 つが削除されます。

別のタイプのロックを削除するには、以下のようにします。

```
LOCK -lockname#locktype
```

locktype は、ロック・タイプ・コードの文字列です。“[ロック・タイプ](#)”を参照してください。ロック・タイプ・コードは、ロック作成時と同じ順序でなくてもかまいません。

12.5.4 LOCK コマンドの他の基本的なバリエーション

完全を期すために、ここではLOCK コマンドの他の基本的なバリエーションについて説明します。LOCK コマンドを使用した[単純ロック](#)の作成と LOCK コマンドを使用した[すべてのロックの削除](#)です。これらのバリエーションは実際には一般的ではありません。

12.5.4.1 単純ロックの作成

LOCK コマンドでは、+ 演算子を省略すると、LOCK コマンドは、まずこのプロセスが保持している既存のロックをすべて削除し、新しいロックを追加しようとします。この場合のロックは、増分ロックではなく単純ロックと呼ばれます。プロセスが以下のような構文で複数の単純ロックをすべて同時に作成する場合、プロセスで複数の単純ロックを所有することが可能です。

```
LOCK (^MyVar1, ^MyVar2, ^MyVar3)
```

複数のロックを保持し、コード内の異なるステップでそれらを取得する必要があることが多いため、単純ロックは実際には一般的ではありません。したがって、増分ロックを使用するほうが実用的です。

ただし、単純ロックが適切な場合、単純ロックの作成時に locktype 引数と timeout 引数を指定できます。また、単純ロックを削除するには、マイナス記号 (-) を付けて LOCK コマンドを使用できます。

12.5.4.2 すべてのロックの削除

現在のプロセスが保持しているすべてのロックを削除するには、引数を付けずに LOCK コマンドを使用します。実際には、この方法でこのコマンドを使用することは、以下の 2 つの理由により一般的ではありません。

- ・ できるだけ早く特定のロックを解放することが最良である。
- ・ プロセスが終了すると、すべてのロックが自動的に解放される。

12.6 ロック・タイプ

locktype 引数は、追加または削除するロックのタイプを指定します。ロックを追加する場合、この引数を以下のように使います。

```
LOCK +lockname#locktype
```

ロックを削除する場合には以下のようにします。

```
LOCK -lockname#locktype
```

どちらの場合でも、locktype は、二重引用符で囲まれた (任意の順序の) 1 つ以上のロック・タイプ・コードです。locktype 引数を指定する場合、シャープ文字 (#) を使用してロック名とロック・タイプを区切る必要があります。

以下のように 4 つのロック・タイプ・コードがあります。これらは大文字と小文字を区別しません。

- ・ S – 共有ロックを追加します。“[排他ロックと共有ロック](#)”を参照してください。
- ・ E – エスカレート・ロックを追加します。“[非エスカレート・ロックとエスカレート・ロック](#)”を参照してください。
- ・ I – 即時アンロック・タイプのロックを追加します。

- ・ D – 遅延アンロック・タイプのロックを追加します。

ロック・タイプ・コード D と E はトランザクションにおいて独特の動作をします。詳細は“[LOCK](#)”を参照してください。
同じロック名に対してこれら 2 つのロック・タイプ・コードを同時に使用することはできません。

以下、数セクションにわたって、最も一般的なバリエーションについて説明し、[最後のサブセクション](#)ですべてのロック・タイプについてまとめます。

12.6.1 排他ロックと共有ロック

あらゆるロックは、排他（既定）または共有のどちらかになります。これらのタイプは、次に示す重要な意味を持ちます。

- ・ あるプロセスが特定のロック名の付いた排他ロックを所有しているときに、その他のプロセスは、そのロック名の付いたロックを取得できません。
- ・ あるプロセスが特定のロック名の付いた共有ロックを所有しているときに、その他のプロセスは、そのロック名の付いた共有ロックを取得できます。ただし、その他のプロセスはそのロック名の付いた排他ロックを取得できません。

排他ロックの一般的な目的は、値を変更する予定があることを示し、その他のプロセスがその値の読み取りまたは変更を試行してはいけなことを示すことです。共有ロックの一般的な目的は、値を読み取る予定があることを示し、その他のプロセスがその値の変更を試行してはいけなことを示すことです。ただし、その他のプロセスがその値を読み取ることは可能です。“[ロックの実際の使用方法](#)”も参照してください。

12.6.2 非エスカレート・ロックとエスカレート・ロック

さらに、あらゆるロックは、非エスカレート（既定）またはエスカレートのどちらかになります。エスカレート・ロックの目的は、メモリを消費し、[ロック・テーブル](#)が埋め尽くされる可能性が大きくなる大量のロックの管理を簡単にすることです。

エスカレート・ロックは、同じ配列に含まれる複数のノードをロックする場合に使用します。エスカレート・ロックの場合、あるプロセスがある配列の並列ノードに特定の数（既定では 1000）よりも多くのロックを作成すると、InterSystems IRIS は、個別のロック名を置き換えて、それらを新しい 1 つのロック（ロック・カウントを含むロック）に置き換えます（一方、InterSystems IRIS は非エスカレート・ロックに対してはこの動作を行いません）。詳細と使用例は、“[エスカレート・ロック](#)”を参照してください。

注釈 添え字を含むロック名に対してのみエスカレート・ロックを作成できます。添え字のないロック名でエスカレート・ロックを作成しようとすると、InterSystems IRIS は <COMMAND> エラーを発行します。

12.6.3 ロック・タイプのまとめ

以下の表に可能なすべてのロック・タイプとその説明を示します。

	排他ロック	共有ロック (＃"S" ロック)
非エスカレート・ロック	<ul style="list-style-type: none"> ・ locktype 省略 – 既定のロック・タイプ ・ ＃"I" – 即時アンロック・タイプの排他ロック ・ ＃"D" – 遅延アンロック・タイプの排他ロック 	<ul style="list-style-type: none"> ・ ＃"S" – 共有ロック ・ ＃"SI" – 即時アンロック・タイプの共有ロック ・ ＃"SD" – 遅延アンロック・タイプの共有ロック
エスカレート・ロック (＃"E" ロック)	<ul style="list-style-type: none"> ・ ＃"E" – 排他エスカレート・ロック ・ ＃"EI" – 即時アンロック・タイプの排他エスカレート・ロック ・ ＃"ED" – 遅延アンロック・タイプの排他エスカレート・ロック 	<ul style="list-style-type: none"> ・ ＃"SE" – 共有エスカレート・ロック ・ ＃"SEI" – 即時アンロック・タイプの共有エスカレート・ロック ・ ＃"SED" – 遅延アンロック・タイプの共有エスカレート・ロック

複数のロック・コードを使用するロック・タイプの場合、ロック・コードの順序は任意にできます。例えば、ロック・タイプ ＃"SI" と ＃"IS" は同じです。

即時アンロックと遅延アンロックの詳細は、“[LOCK](#)”を参照してください。同じロック名に対してこれら 2 つのロック・タイプ・コードを同時に使用することはできません。

12.7 エスカレート・ロック

大量のロックを管理するためにエスカレート・ロックを使用します。これは配列のノードをロックする場合に関係します。特に同じ添え字レベルで複数のノードをロックする場合です。

あるプロセスが同じ配列の特定の添え字レベルで特定の数 (既定では 1000) よりも多くのエスカレート・ロックを作成すると、InterSystems IRIS は、個別のロック名をすべて削除し、それらを新しいロックに置き換えます。この新しいロックは親レベルになります。つまり、この配列の分岐全体が暗黙的にロックされます。(次に示す) 例でこのことを示します。

アプリケーションは、適切なタイミングが来たらすぐに特定の子ノードのロックを解放する必要があります (非エスカレート・ロックと同様)。ロックを解放すると、InterSystems IRIS は対応するロック・カウントをデクリメントします。アプリケーションが十分なロックを削除すると、InterSystems IRIS は親ノードのロックを削除します。[2 番目のサブセクション](#)に例を示します。

ロックしきい値 (既定値は 1000) の指定の詳細は、“[LockThreshold](#)”を参照してください。

12.7.1 ロックのエスカレーションの例

^MyGlobal ("sales", EU, salesdate) という形式で salesdate が日付を表しているロックが 1000 個あるとします。ロック・テーブルは以下のようになります。

	Owner	ModeCount	Reference	Directory
	1284	Exclusive	^%SYS("CSP","Daemon")	c:\intersystems\iris\mgr\
	26324	Exclusive	^ISC.LFMON("License Monitor")	c:\intersystems\iris\mgr\
	23400	Exclusive	^ISC.Monitor.System	c:\intersystems\iris\mgr\
	23180	Exclusive	^TASKMGR	c:\intersystems\iris\mgr\
	23948	Exclusive	^%cspSession("vgMJ4iLMCL")	c:\intersystems\iris\mgr\irislocaldata\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-03")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-04")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-05")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-06")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-07")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-08")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-09")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-10")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-11")	c:\intersystems\iris\mgr\user\
	19776	Exclusive_e	^MyGlobal("sales","EU","2015-07-12")	c:\intersystems\iris\mgr\user\

所有者 19776 のエントリに注目してください (これはロックを所有するプロセスです)。[ModeCount] 列は、これらが排他のエスカレート・ロックであることを示しています。

同じプロセスが同じ形式の別のロックを作成しようとする、InterSystems IRIS はこれらをエスカレートします。これらのロックを削除し、^MyGlobal("sales","EU") という名前の単一のロックで置き換えます。ロック・テーブルは以下のようになります。

	Owner	ModeCount	Reference	Directory
	1284	Exclusive	^%SYS("CSP","Daemon")	c:\intersystems\iris\mgr\
	26324	Exclusive	^ISC.LFMON("License Monitor")	c:\intersystems\iris\mgr\
	23400	Exclusive	^ISC.Monitor.System	c:\intersystems\iris\mgr\
	23180	Exclusive	^TASKMGR	c:\intersystems\iris\mgr\
	23948	Exclusive	^%cspSession("vgMJ4iLMCL")	c:\intersystems\iris\mgr\irislocaldata\
	19776	Exclusive/1001E	^MyGlobal("sales","EU")	c:\intersystems\iris\mgr\user\

[ModeCount] 列は、これが共有のエスカレート・ロックで、そのカウントが 1001 であることを示しています。

以下の重要な点に注意してください。

- ・ ^MyGlobal("sales","EU") のすべての子ノードは、この時点で[配列ロック](#)の基本的な規則に従って暗黙的にロックされます。
- ・ ロック・テーブルには、^MyGlobal("sales","EU") のどの子ノードが具体的にロックされたかに関する情報が含まれなくなります。これは、ロックを削除する際に重要な意味を持ちます。[次のサブセクション](#)を参照してください。

同じプロセスが ^MyGlobal("sales","EU",salesdate) の形式のロック名をさらに追加すると、ロック・テーブルは、ロック名 ^MyGlobal("sales","EU") のロック・カウントをインクリメントします。ロック・テーブルは以下のようになります。

	Owner	ModeCount	Reference	Directory
	1284	Exclusive	^%SYS("CSP","Daemon")	c:\intersystems\iris\mgr\
	26324	Exclusive	^ISC.LFMON("License Monitor")	c:\intersystems\iris\mgr\
	23400	Exclusive	^ISC.Monitor.System	c:\intersystems\iris\mgr\
	23180	Exclusive	^TASKMGR	c:\intersystems\iris\mgr\
	23948	Exclusive	^%cspSession("vgMJ4iLMCL")	c:\intersystems\iris\mgr\irislocaldata\
	19776	Exclusive/1026E	^MyGlobal("sales","EU")	c:\intersystems\iris\mgr\user\

[ModeCount] 列は、このロックのロック・カウントが 1026 になったことを示しています。

12.7.2 エスカレート・ロックの削除

非エスカレート・ロックと正確に同じ方法で、アプリケーションは、特定の子ノードのロックをできるだけ早く解放する必要があります。これを行うと、InterSystems IRIS は、エスカレートされたロックのロック・カウントをデクリメントします。例えば、コードが `^MyGlobal("sales","EU",salesdate)` のロックを削除したとします (salesdate は 2011 年の任意の日付に対応し、したがって 365 個のロックが削除されます)。ロック・テーブルは以下ようになります。

Owner	ModeCount	Reference	Directory
1284	Exclusive	^%SYS("CSP","Daemon")	c:\intersystems\iris\mgr\
26324	Exclusive	^ISC.LMFMON("License Monitor")	c:\intersystems\iris\mgr\
23400	Exclusive	^ISC.Monitor.System	c:\intersystems\iris\mgr\
23180	Exclusive	^TASKMGR	c:\intersystems\iris\mgr\
23948	Exclusive	^%cspSession("vgMJ4iLMCL")	c:\intersystems\iris\mgr\irislocaldata\
19776	Exclusive/660E	^MyGlobal("sales","EU")	c:\intersystems\iris\mgr\user\

ロックの数がしきい値 (1000) 未満になったにもかかわらず、ロック・テーブルには `^MyGlobal("sales","EU",salesdate)` のロックの個別のエントリが含まれていないことがわかります。

ノード `^MyGlobal("sales")` は、プロセスが `^MyGlobal("sales","EU",salesdate)` の形式のロックをさらに 661 個削除するまで明示的にロックされたままです。

重要 前記の説明に関して考慮する必要がある点があります。そもそもロックされていない配列ノードのロックをアプリケーションが解放し、エスカレートされたロックのロック・カウントが不正確になり、エスカレートされたロックの解放が適切になる前にこれが“解放”されてしまう可能性があります。

例えば、2010 年から現在までの `^MyGlobal("sales","EU",salesdate)` のノードをプロセスがロックしたとします。1000 個を超えるロックが作成され、計画どおりにこのロックはエスカレートされます。アプリケーションのバグによって 1970 年のノードのロックが削除されたとします。InterSystems IRIS は、これらのノードが以前ロックされなかったにもかかわらずこの操作を許可します。InterSystems IRIS はロック・カウントを 365 デクリメントします。結果として得られるロック・カウントは、目的のロックのカウントとして正確ではなくなります。その後アプリケーションが別の年のロックを削除すると、エスカレートされたロックは、予期しない早さで削除される可能性があります。

12.8 ロック、グローバル、およびネームスペース

ロックは通常、グローバルへのアクセスを制御するために使用されます。グローバルは複数のネームスペースからアクセスできるため、InterSystems IRIS は、グローバルのロック・メカニズム用に、自動クロスネームスペース・サポートを提供しています。この動作は自動で、操作する必要はありませんが、参考のためにここで説明します。以下のような考慮すべきシナリオがいくつかあります。

- 永続クラスと追加グローバルのデータを格納する既定データベースがどのネームスペースにもあります。これが、そのネームスペースのグローバル・データベースです。ユーザがデータに (任意の方法で) アクセスする場合、他の考慮事項がない限り、InterSystems IRIS によってそのデータベースからそれが取得されます。特定のデータベースを複数のネームスペースのグローバル・データベースにすることも可能です。[シナリオ 1](#) を参照してください。
- ネームスペースには、他のデータベースに格納されているグローバルへのアクセスを提供するマッピングを含めることができます。[シナリオ 2](#) を参照してください。
- ネームスペースには、他のデータベースに一部格納されているグローバルへのアクセスを提供する添え字レベルのグローバル・マッピングを含めることができます。[シナリオ 3](#) を参照してください。

- 1 つのネームスペースで実行されているコードは、拡張参照を使用して、拡張参照を使用しない場合にはこのネームスペースでは使用できないグローバルにアクセスできます。[シナリオ 4](#) を参照してください。


ロック名は本来任意ですが、キャレット (^) で始まるロック名を使用する場合、InterSystems IRIS は、これらのシナリオに適切な特別な動作を提供します。以降の項で詳しく説明します。わかりやすくするために、排他ロックのみを説明します。論理は共有ロックでも同様です。

12.8.1 シナリオ 1 : 同じグローバル・データベースを持つ複数のネームスペース

前述のとおり、プロセス A が特定のロック名の付いた排他ロックを所有しているときに、その他のプロセスは、同じロック名の付いたロックを取得できません。

ロック名がキャレットで始まる場合、この規則は、同じグローバル・データベースを使用するすべてのネームスペースに適用されます。

例えば、ネームスペース ALPHA と BETA の両方がデータベース GAMMA をグローバル・データベースとして使用するように構成されているとします。以下は、部分的な例です。



次に以下のシナリオを考えてみます。

- ネームスペース ALPHA では、プロセス A が ^MyGlobal(15) という名前の排他ロックを取得します。
- ネームスペース BETA では、プロセス B が ^MyGlobal(15) という名前のロックを取得しようとします。この LOCK コマンドは復帰しません。このプロセスは、プロセス A がロックを解放するまでブロックされます。


このシナリオでは、プロセス A が所有するロックのエントリのみがロック・テーブルに含まれています。ロック・テーブルを調べると、このロックの適用先のデータベースを示していることがわかります。[\[ディレクトリ\]](#) 列を参照してください。以下に例を示します。

Owner	ModeCount	Reference	Directory
1284	Exclusive	^%SYS("CSP","Daemon")	c:\intersystems\iris\mgr\
26324	Exclusive	^ISC.LFMON("License Monitor")	c:\intersystems\iris\mgr\
23400	Exclusive	^ISC.Monitor.System	c:\intersystems\iris\mgr\
23180	Exclusive	^TASKMGR	c:\intersystems\iris\mgr\
19776	Exclusive_e	^MyGlobal(15)	c:\intersystems\iris\mgr\gammadb\
23948	Exclusive	^%cspSession("vgMJ4iLMCL")	c:\intersystems\iris\mgr\irislocaldata\

12.8.2 シナリオ 2 : ネームスペースがマップされたグローバルを使用

1 つまたは複数のネームスペースにグローバル・マッピングが含まれている場合、システムは、該当のネームスペースにわたってロック・メカニズムを自動的に強制適用します。InterSystems IRIS は、既定以外のネームスペースでロックが取得されると、追加のロック・テーブル・エントリを自動的に作成します。

例えば、ネームスペース ALPHA がデータベース ALPHADB をグローバル・データベースとして使用するように構成されているとします。また、ネームスペース BETA が異なるデータベース (BETADB) をグローバル・データベースとして使用するように構成されているとします。ネームスペース BETA には、`^MyGlobal` が ALPHADB データベースに格納されることを指定するグローバル・マッピングも含まれています。以下は、部分的な例です。



次に以下のシナリオを考えてみます。

1. ネームスペース ALPHA では、プロセス A が `^MyGlobal(15)` という名前の排他ロックを取得します。

前述のシナリオと同様に、プロセス A が所有するロックのエントリのみがロック・テーブルに含まれています。このロックは ALPHADB データベースに適用されます。

19776	Exclusive_e	<code>^MyGlobal(15)</code>	<code>c:\intersystems\iris\mgr\alphadb\</code>
-------	-------------	----------------------------	--


2. ネームスペース BETA では、プロセス B が `^MyGlobal(15)` という名前のロックを取得しようとし、この LOCK コマンドは復帰しません。このプロセスは、プロセス A がロックを解放するまでブロックされます。

12.8.3 シナリオ 3 : ネームスペースがマップされたグローバル添え字を使用

添え字レベル・マッピングを使用するグローバル・マッピングが 1 つまたは複数のネームスペースに含まれている場合、システムは、該当のネームスペースにわたってロック・メカニズムを自動的に強制適用します。この場合、InterSystems IRIS は、既定以外のネームスペースでロックが取得されると、追加のロック・テーブル・エントリを自動的に作成します。

例えば、ネームスペース ALPHA がデータベース ALPHADB をグローバル・データベースとして使用するように構成されているとします。ネームスペース BETA は BETADB データベースをグローバル・データベースとして使用します。

また、ネームスペース BETA には、`^MyGlobal(15)` が ALPHADB データベースに格納される (一方、このグローバルの残りはネームスペースの既定の場所に格納される) ように添え字レベルのグローバル・マッピングも含まれているとします。以下は、部分的な例です。



次に以下のシナリオを考えてみます。

1. ネームスペース ALPHA では、プロセス A が ^MyGlobal(15) という名前の排他ロックを取得します。
 前述のシナリオと同様に、プロセス A が所有するロックのエントリのみがロック・テーブルに含まれています。このロックは ALPHADB データベース (c:\InterSystems\IRIS\mgr\alphadb など) に適用されます。
2. ネームスペース BETA では、プロセス B が ^MyGlobal(15) という名前のロックを取得しようとします。この LOCK コマンドは復帰しません。このプロセスは、プロセス A がロックを解放するまでブロックされます。

既定以外のネームスペースがロックを取得しても全体的な動作は同じですが、InterSystems IRIS による詳細の処理は若干異なります。ネームスペース BETA で、あるプロセスが ^MyGlobal(15) という名前のロックを取得するとします。この場合、ロック・テーブルには2つのエントリが含まれます。1つは ALPHADB データベース用、もう1つは BETADB データベース用です。これらのロックは、両方ともネームスペース BETA のプロセスによって所有されます。

19776	Exclusive_e	^MyGlobal(15)	c:\intersystems\iris\mgr\alphadb\
19776	Exclusive_e	^MyGlobal(15)	c:\intersystems\iris\mgr\betadb\

このプロセスがロック名 ^MyGlobal(15) を解放すると、システムは両方のロックを自動的に削除します。

12.8.4 シナリオ 4 : 拡張グローバル参照

1 つのネームスペースで実行されているコードは、拡張参照を使用して、拡張参照を使用しない場合にはこのネームスペースでは使用できないグローバルにアクセスできます。この場合、InterSystems IRIS は、関連するデータベースに影響するロック・テーブルにエントリを追加します。このロックは、これを作成したプロセスに所有されます。例えば、以下のシナリオを考えてみます。わかりやすくするために、このシナリオにはグローバル・マッピングはありません。

1. プロセス A は ALPHA ネームスペースで実行されていて、このプロセスは以下のコマンドを使用して、BETA ネームスペースで利用できるグローバルでロックを取得します。

ObjectScript

```
lock ^["beta"]MyGlobal(15)
```

2. ここで、ロック・テーブルには以下のエントリが含まれています。

19776	Exclusive_e	^MyGlobal(15)	c:\intersystems\iris\mgr\betadb\
-------	-------------	---------------	----------------------------------

これはグローバル名のみを示していることに注意してください(これにアクセスするために使用する参照ではありません)。また、このシナリオでは、BETADB は BETA ネームスペースの既定のデータベースです。

3. ネームスペース BETA では、プロセス B が ^MyGlobal(15) という名前のロックを取得しようとします。この LOCK コマンドは復帰しません。このプロセスは、プロセス A がロックを解放するまでブロックされます。

プロセス・プライベート・グローバルは技術的には拡張参照の一種ですが、InterSystems IRIS は、プロセス・プライベート・グローバル名をロック名として使用することをサポートしていません。定義によってそのようなグローバルにアクセスできるプロセスは 1 つしかないため、そのようなロックは必要ありません。

12.9 デッドロックの回避

増分ロックの使用は、潜在的な危険を伴います。これは、デッドロック と呼ばれる状況を引き起こす可能性があるからです。この状況は、既に他のプロセスによってロックされている変数に対して、2 つのプロセスがそれぞれ増分ロックをアサートするときに発生します。試行されているロックが増分ロックなので、既存のロックは解除されません。その結果、各プロセスは、他のプロセスが既存のロックを解除するまで待機する間、停止します。

以下に例を示します。

1. プロセス A が次のコマンドを発行します : lock + ^MyGlobal(15)
2. プロセス B が次のコマンドを発行します : lock + ^MyOtherGlobal(15)
3. プロセス A が次のコマンドを発行します : lock + ^MyOtherGlobal(15)

この LOCK コマンドは復帰しません。このプロセスは、プロセス B がロックを解放するまでブロックされます。

4. プロセス B が次のコマンドを発行します : lock + ^MyGlobal(15)

この LOCK コマンドは復帰しません。このプロセスは、プロセス A がこのロックを解放するまでブロックされます。しかし、プロセス A はブロックされ、ロックを解放できません。ここで、これらのプロセスは両方ともお互いの処理を待ちます。

デッドロックを防止するには以下のような方法があります。

- ・ 常に `timeout` 引数を使用する。
- ・ 増分 LOCK コマンドを発行する際にその順序に関して厳格なプロトコルに従う。すべてのプロセスが、ロック名に関して同じ順序に従っている限り、デッドロックが発生することはありません。単純なプロトコルは、照合順序でロックを追加するものです。
- ・ 増分ロックではなく単純ロックを使用する (つまり、+ 演算子を使用しない)。前述のとおり、単純ロックでは、LOCK コマンドは最初に、プロセスによって以前から保持されていたすべてのロックを解放します (ただし、実際には単純ロックはあまり使用されません)。

デッドロックが発生した場合、管理ポータルまたは ^LOCKTAB ルーチンを使用して解決できます。“監視ガイド”の“[ロックの監視](#)”を参照してください。

12.10 ロックの実際の使用方法

ここでは、実際にロックが使用される基本的な方法について説明します。

12.10.1 アプリケーション・データへのアクセスの制御

ロックは、グローバルに格納されているアプリケーション・データへのアクセスを制御するために使用されることが非常に多いです。アプリケーションは、このデータの特定の一部分を読み取りまたは変更する必要がある、これを行う前に以下のように 1 つまたは複数のロックを作成する場合があります。

- ・ アプリケーションが 1 つまたは複数のグローバル・ノードを読み取る必要がある、読み取り操作中に他のプロセスがこの値を変更しないようにする場合、これらのノードの共有ロックを作成します。
- ・ アプリケーションが 1 つまたは複数のグローバル・ノードを変更する必要がある、変更中に他のプロセスがこれらのノードを読み取らないようにする場合、これらのノードの排他ロックを作成します。

その後、計画どおりに読み取りまたは変更を行います。完了したら、ロックを削除します。

ロック・メカニズムは純粋に規約によって動作することに注意してください。これらのノードの読み取りまたは変更を行う他のコードも、これらの操作を実行する前にロックを取得しようとする必要があります。

12.10.2 同時アクティビティの防止

ロックは、複数のプロセスが同じアクティビティを実行することを防止するためにも使用されます。このシナリオでは、グローバルも使用しますが、そのグローバルには、純粋なアプリケーション・データではなくアプリケーションの内部用データが含まれています。単純な例として、どのような場合でも複数のプロセスで実行されてはいけないルーチン (^NightlyBatch) があるとします。このルーチンは、処理のごく初期段階に以下のことを実行できます。

1. 特定のグローバル・ノードに排他ロックを作成します (例えば、^AppStateData("NightlyBatch"))。この操作にタイムアウトを指定します。
2. ロックが取得された場合、グローバルにノードを設定し、ルーチンが開始されたこと (および他の関連情報) を記録します。以下に例を示します。

ObjectScript

```
set ^AppStateData("NightlyBatch")=1
set ^AppStateData("NightlyBatch","user")=$USERNAME
```

タイムアウト時間内にロックが取得されなかった場合、このルーチンが既に開始されていることを示すエラー・メッセージで終了します。

処理の最後に、同じルーチンが該当のグローバル・ノードをクリアし、ロックを解放します。

以下の部分的な例はこの方法を示しています。これは、InterSystems IRIS が内部的に使用するコードから採用されています。

ObjectScript

```
lock ^AppStateData("NightlyBatch"):0
if '$TEST {
    write "You cannot run this routine right now."
    write !, "This routine is currently being run by user: " _ ^AppStateData("NightlyBatch","user")
    quit
}
set ^AppStateData("NightlyBatch")=1
set ^AppStateData("NightlyBatch","user")=$USERNAME
set ^AppStateData("NightlyBatch","starttime")=$h

//main routine activity omitted from example

kill ^AppStateData("NightlyBatch")
lock -^AppStateData("NightlyBatch")
```

12.11 SQL と永続クラスにおけるロックと並行処理

InterSystems SQL または永続クラスを操作する場合、ObjectScript LOCK コマンドを直接使用する必要はありません。ユース・ケースに適した、これに代わる手段があるためです (内部的にはこれらの代替手段はすべて LOCK を使用します)。

- ・ InterSystems SQL では、ロックを操作するコマンドが提供されています。詳細は、“[InterSystems SQL リファレンス](#)”を参照してください。同様に、[INSERT](#)、[UPDATE](#)、および [DELETE](#) の各操作のロックが自動的に実行されます (%NOLOCK キーワードを指定していない場合)。
- ・ **%Persistent** クラスでは、オブジェクトへの同時アクセスを制御する手段も提供されています。すなわち、このクラスの `%OpenId()` およびその他のメソッドの同時処理引数です。すべての永続オブジェクトは、それらのメソッドを継承します。“[オブジェクト同時処理のオプション](#)”を参照してください。

このクラス **%Persistent** では、`%GetLock()`、`%ReleaseLock()`、`%LockId()`、`%UnlockId()`、`%LockExtent()`、および `%UnlockExtent()` メソッドも提供されています。詳細は、“**%Persistent**” のクラス・リファレンスを参照してください。

12.12 関連項目

- ・ [LOCK コマンドのリファレンス](#)
- ・ [~\\$LOCK](#) (`$LOCK` はロックに関する情報を含む構造化システム変数です。)
- ・ [トランザクション処理](#)
- ・ [ロック要求とデッドロックの詳細](#)
- ・ [ロック・テーブルの管理](#)
- ・ [ロックの監視](#)

13

ロック要求とデッドロックの詳細

ここでは、InterSystems IRIS® データ・プラットフォームにおけるロック要求の処理方法と、デッドロックのシナリオに関する詳細情報を提供します。

13.1 待機ロック要求

あるプロセスが排他ロックを保持する場合、別のプロセスが同じロック、または保持されたロックの上層や下層レベルのノード上のロックを取得しようとする、待機状態が発生します。添え字付きグローバル (配列ノード) をロックする場合、以下のように、ユーザがロックするものと他のプロセスでロック可能なものとの区別が重要になります。

- ・ ユーザがロックするもの: ユーザは指定したノードに対してのみ明示的なロックを有します。その上層や下層のレベルのノードに対してはロックを有しません。例えば、`^student(1,2)` をロックする場合、ユーザは `^student(1,2)` に対してのみ明示的なロックを有します。上層レベルのノード (`^student(1)` など) の解放によってこのノードを解放することはできません。なぜなら、上層レベルのノードに対して明示的なロックを有していないためです。もちろん、ユーザは任意の順序で上層または下層のノードを明示的にロックすることができます。
- ・ 他プロセスが何をロックできるか: ユーザがノードをロックすることにより、他のプロセスがその同一ノード、または上層や下層レベルのノード (該当ノードの親や子) をロックできません。親の `^student(1)` をロックすることはできません。なぜなら、これをロックすると、子の `^student(1,2)` (プロセスによって既に明示的にロックされている) も暗示的にロックされるためです。子の `^student(1,2,3)` をロックすることはできません。なぜなら、プロセスが既に親の `^student(1,2)` をロックしているためです。それらの他のプロセスは、指定の順序によりロック・キューにおいて待機となります。これらはロック・テーブルにリストされ、キュー内において他に優先することを指定された最上層レベルのノードで待機となります。これはロックされたノード、またはロックを待つノードとなる場合があります。

以下に例を示します。

1. プロセス A は `^student(1,2)` をロックします。
2. プロセス B は `^student(1)` のロックを試行しますが、禁じられます。これは、プロセス B が `^student(1)` をロックすると、`^student(1,2)` も (暗示的に) ロックされるためです。ただし、プロセス A は `^student(1,2)` に対するロックを保持しています。ロック・テーブルには、これが `WaitExclusiveParent ^student(1,2)` としてリストされます。
3. プロセス C は `^student(1,2,3)` のロックを試行しますが、禁じられます。ロック・テーブルには、これが `WaitExclusiveParent ^student(1,2)` としてリストされます。プロセス A は `^student(1,2)` に対するロックを保持しており、そのため `^student(1,2,3)` に対する暗示的なロックを保持しています。ただし、プロセス C はキュー内でプロセス B よりも下層となるため、プロセス B が `^student(1)` をロックして解放するのをプロセス C は待つ必要があります。

4. プロセス A は `^student(1,2,3)` をロックします。待機ロックはそのまま変更ありません。
5. プロセス A は `^student(1)` をロックします。待機ロックは以下のように変化します。
 - ・ プロセス B は `WaitExclusiveExact ^student(1)` としてリストされます。プロセス B は、プロセス A が保持するものと同一のロック (`^student(1)`) をロックすることを待ちます。
 - ・ プロセス C は `WaitExclusiveChild ^student(1)` としてリストされます。プロセス C はプロセス B よりもキュー内で下層となるため、要求されたロックをプロセス B がロックして解放するのを待ちます。そのときには、プロセス C がプロセス B ロックの子をロックできるようになります。同様に、プロセス A が `^student(1)` を解放するのをプロセス B は待ちます。
6. プロセス A は `^student(1)` をアンロックします。待機ロックは `WaitExclusiveParent ^student(1,2)` に戻ります。(手順 2 および 3 と同一。)
7. プロセス A は `^student(1,2)` をアンロックします。待機ロックは `WaitExclusiveParent ^student(1,2,3)` に変化します。プロセス B は `^student(1)` (現在のプロセス A ロック `^student(1,2,3)` の親) のロックを待ちます。プロセス C はプロセス B が `^student(1)` (プロセス C により要求された `^student(1,2,3)` ロックの親) をロックしてからアンロックするのを待ちます。
8. プロセス A は `^student(1,2,3)` をアンロックします。プロセス B は `^student(1)` をロックします。プロセス C は現在、プロセス B によって禁じられています。プロセス C は `WaitExclusiveChild ^student(1)` としてリストされます。プロセス C は `^student(1,2,3)` (現在のプロセス B ロックの子) のロックを待ちます。

13.2 配列ノードに対するロック要求のキュー

配列ロックに対する基本的なキュー・アルゴリズムでは、リソースが直接競合していない場合でも、同一リソースに対するロック要求は受信した順で厳密にキューに置かれます。この様子を以下の例に示します。ここでは、同じグローバル配列上の 3 つのロックが、以下に示す順序で別々のプロセスによって要求されます。

```
Process A: LOCK ^x(1,1)
Process B: LOCK ^x(1)
Process C: LOCK ^x(1,2)
```

これらの要求のステータスは以下のとおりです。

- ・ プロセス A は `^x(1,1)` に対するロックを保持しています。
- ・ プロセス B は、プロセス A が `^x(1,1)` に対するロックを解除するまで `^x(1)` をロックできません。
- ・ プロセス C もブロックされていますが、プロセス A のロックに伴うブロックではありません。プロセス B が `^x(1)` の明示的なロックと、`^x(1,2)` の暗示的なロックを待機しており、これによりプロセス C がブロックされます。

シーケンスでロックを保持しているプロセスの後にある次のジョブを高速化するために、このような手法が採用されています。キューの中でプロセス C がプロセス B を飛び越えることができるようにすると、プロセス C は速く実行できますが、プロセス B に重大な遅延が発生する可能性があります。プロセス C のようなジョブが多数存在する場合は、特に顕著になります。

親ノードに対するロックを保持しているプロセスには、そのノードの子に対して要求されたロックが直ちに付与されます。これは、要求は受信した順で処理されるという一般的なルールの例外です。例えば、前の例の拡張を以下で考えます。

```
Process A: LOCK ^x(1,1)
Process B: LOCK ^x(1)
Process C: LOCK ^x(1,2)
Process A: LOCK ^x(1,2)
```


この例では、プロセス A には、プロセス B とプロセス C よりも前に、 $\wedge x(1,2)$ に対して要求されたロックが直ちに付与されています。プロセス A が既に $\wedge x(1,1)$ に対するロックを保持しているからです。

注釈 このプロセスのキューイング・アルゴリズムは、添え字を持つすべてのロック要求に適用されます。ただし、添え字のないロック要求 (LOCK $\wedge x$) と添え字のあるロック要求 (LOCK $\wedge x(1,1)$) の両方が待機中である場合、添え字のないロック (LOCK $\wedge x$ など) の解除は特殊なケースです。この場合、どのロック要求が付与されるかは予測できず、プロセスのキューイングに従わない場合があります。

13.3 ECP ローカルおよびリモート・ロック要求

ロックを解除する際、ECP クライアントは、パフォーマンスを向上させるためにその他のシステム上の待機に優先してローカルの待機にロックを付与できます。リモート・ロック待機で許容できないほどの遅延が発生しないように、これが発生できる回数は制限されています。

13.4 デッドロックの回避

既存の共有ロックを保持している際の (+) 排他ロックの要求は、潜在的な危険を伴います。これは、“デッドロック”として知られる状況を引き起こす可能性があるからです。この状況は、既に他のプロセスによって共有ロックとしてロックされているロック名に対して、2 つのプロセスがそれぞれ排他ロックを要求するときに発生します。その結果、各プロセスは、他のプロセスが既存の共有ロックを解除するまで待機する間、停止します。

次の例では、これがどのように行われるかについて説明します (数字は処理の順番を示します)。

プロセス A	プロセス B
1.LOCK $\wedge a(1)\# "S"$ プロセス A は共有ロックを取得します。	
	2.LOCK $\wedge a(1)\# "S"$ プロセス B は共有ロックを取得します。
3.LOCK $\wedge a(1)$ プロセス A は排他ロックを要求し、プロセス B がその共有ロックを解除するまで待機します。	
	4.LOCK $\wedge a(1)$ プロセス B は排他ロックを要求し、プロセス A がその共有ロックを解除するまで待機します。これによってデッドロックが発生します。

これは最も単純な形式のデッドロックです。デッドロックは、保持されているロックの親ノードまたは子ノード上のロックをプロセスが要求する際にも発生する可能性があります。

デッドロックを防止するには、プラス記号なしで排他ロックを要求する (そうすることで共有ロックをロック解除する) 必要があります。以下の例では、両方のプロセスが排他ロックを要求する際に、それぞれが有している以前のロックを解除する

ことでデッドロックを防止しています (数字は操作の順序を示しています)。どのプロセスが排他ロックを取得するかに注意してください。

プロセス A	プロセス B
1.LOCK ^a(1)#"S" プロセス A は共有ロックを取得します。	
	2.LOCK ^a(1)#"S" プロセス B は共有ロックを取得します。
3.LOCK ^a(1) プロセス A は共有ロックを解除してから排他ロックを要求し、プロセス B がその共有ロックを解除するまで待機します。	
	4.LOCK ^a(1) プロセス B は共有ロックを解除してから排他ロックを要求します。プロセス A は直ちに、要求した共有ロックを取得します。プロセス B は、プロセス A がその共有ロックを解除するまで待機します。

他のデッドロック回避方法には、LOCK + および LOCK - コマンドを実行する順序に関する、厳密なプロトコルに従う方法もあります。すべてのプロセスが、この同じ順序に従っている限り、デッドロックが発生することはありません。単純なプロトコルは、すべてのプロセスにおいて照合順序でロックを適用および解除するものです。

デッドロック状況の影響を最小限に抑えるために、プラス記号のロックを使用する際に timeout 引数を指定する必要があります。例えば、LOCK + ^a(1):10 オペレーションは 10 秒後にタイムアウトします。

デッドロックが発生した場合は、管理ポータルまたは **LOCKTAB** を使用して、問題になっているロックのいずれかを削除できます。管理ポータルから、**[ロックを管理]** ウィンドウを開き、デッドロック・プロセスに対して **[削除]** オプションを選択します。

13.5 関連項目

- ・ [ロックと並行処理の制御](#)
- ・ [ロック・テーブルの管理](#)

14

ロック・テーブルの管理

ここでは、インターシステムズ製品でロック・テーブルを表示および管理するためのツールについて説明します（“[ロックの監視](#)”も参照してください）。

14.1 ロック・テーブルの管理のための使用可能ツール

ロックを表示し、(場合によっては) 削除することが必要なこともあります。このために、インターシステムズでは、以下のツールが用意されています。

- ・ 管理ポータル の [ロック] ページ。ここでは、[ロックの表示](#)および[ロックの削除](#)が可能です。
- ・ `_LOCKTAB` ユーティリティ。
- ・ `%SYS.LockQuery` クラス。これにより、ロック・テーブルの情報を読み取ることができます。
- ・ `SYS.Lock` クラス。これは `%SYS` ネームスペース内で利用可能です。

最後の 2 つのクラスの詳細は、クラス・リファレンスを参照してください。

14.2 管理ポータルでのロックの表示

管理ポータルを使用すれば、システム全体で現在保持または要求（待機）されているすべてのロックを表示できます。管理ポータルで、[システムオペレーション]→[ロック]→[ロック表示] の順に選択します。[ロック表示] ウィンドウには、ディレクトリごとのアルファベット順で（[ディレクトリ]）、なおかつ各ディレクトリ内にはロック名による照合順で（[リファレンス]）、ロック（およびロック要求）の一覧が表示されます。各ロックはプロセス ID（[所有者]）で特定され、プロセスの作成時にオペレーティング・システムにより付与されるユーザ名（OS User Name）を表示し、[モードカウント]（ロック・モードおよびロック・インクリメント・カウント）を有します。ロックおよびロック要求の最新リストを表示させるためには、[再表示] アイコンを使用する必要があります。このインタフェースの詳細は、“[ロックの監視](#)”を参照してください。

[ModeCount] は、特定の [リファレンス] についての特定の [所有者] プロセスによって、保持されたロックを指定することができます。以下は、保持されたロックに対する [ModeCount] 値の例です。

ModeCount	説明
Exclusive	排他ロック、非エスカレート (LOCK +^a(1))
Shared	共有ロック、非エスカレート (LOCK +^a(1)#"S")
Exclusive_e	排他ロック、 エスカレート (LOCK +^a(1)#"E")
Shared_e	共有ロック、 エスカレート (LOCK +^a(1)#"SE")
Exclusive->Delock	ロック解除状態にある排他ロック。ロックはアンロックされていますが、現在のトランザクションの終了までロックの解除が延期されます。これは、標準のアンロック (LOCK -^a(1)) または 遅延アンロック LOCK -^a(1)#"D") により発生できます。
Exclusive,Shared	共有ロックと排他ロックの両方 (いずれの順序でも適用)。エスカレート・ロックも指定可能です。例 : Exclusive_e,Shared_e
Exclusive/n	増分排他ロック (LOCK +^a(1) が n 回発行される)。ロック・カウントが 1 の場合、カウントは表示されません (ただし、以下参照のこと)。増分共有ロックも指定可能です。例 : Shared/2
Exclusive/n->Delock	ロック解除状態にあるインクリメンタル排他ロック。ロックのインクリメントはすべてアンロックされていますが、現在のトランザクションの終了までロックの解除が延期されます。トランザクション内で、個々のインクリメントをアンロックすることで、これらのインクリメントはすぐに解除されます。ロックは、ロック・カウントが 1 のときにアンロックが発行されるまで、ロック解除状態になりません。この ModeCount 値 (ロック解除状態のインクリメンタル・ロック) は、それより前のすべてのロックが単独の操作 (引数なしの LOCK コマンドまたはロック・オペレーション・インジケータ (LOCK ^xyz(1)) のないロックのどちらか) によってアンロックされたときに発生します。
Exclusive/1+1e	2 つの排他ロック。一方は非エスカレートで、もう一方はエスカレートです。インクリメント・カウントはこれら 2 種類の排他ロックに対して別々に維持します。共有ロックも指定可能です。例 : Shared/1+1e
Exclusive/n,Shared/m	共有ロックと排他ロックの両方。ともに整数インクリメントとなります。

保持されたロック [ModeCount] は、インクリメントの有無にかかわらず、共有または排他、エスカレートまたは非エスカレートのロックのあらゆる組み合わせを表すことができます。排他ロックまたは共有ロック (エスカレートまたは非エスカレート) はロック解除状態にすることができます。

[ModeCount] によって、WaitExclusiveExact などのロック待機プロセスを指定することが可能です。以下は、待機ロック要求に対する [ModeCount] の値です。

ModeCount	説明
WaitSharedExact	保持または事前に要求されたまったくの同一ロックに対する共有ロックの待機 : LOCK + ^a(1,2)#"S" はロック ^a(1,2) に対する待機となります。
WaitExclusiveExact	保持または事前に要求されたまったくの同一ロックに対する排他ロックの待機 : LOCK + ^a(1,2) はロック ^a(1,2) に対する待機となります。
WaitSharedParent	保持または事前に要求されたロックの親に対する共有ロックの待機 : LOCK + ^a(1)#"S" はロック ^a(1,2) に対する待機となります。
WaitExclusiveParent	保持または事前に要求されたロックの親に対する排他ロックの待機 : LOCK + ^a(1) はロック ^a(1,2) に対する待機となります。
WaitSharedChild	保持または事前に要求されたロックの子に対する共有ロックの待機 : LOCK + ^a(1,2)#"S" はロック ^a(1) に対する待機となります。
WaitExclusiveChild	保持または事前に要求されたロックの子に対する排他ロックの待機 : LOCK + ^a(1,2) はロック ^a(1) に対する待機となります。

[ModeCount] によって、ロック要求をブロックしているロック (またはロック要求) を指定します。これは必ずしも [リファレンス] と同じではなく、ロック・キュー先頭で現在保持されているロックを指定します。このロック・キューには当ロック要求が待機しています。[リファレンス] は必ずしも、このロック要求を即座にブロックする被要求ロックを指定するものではありません。

[ModeCount] は、特定の [リファレンス] についての特定の [所有者] プロセスに対するその他のロック・ステータス値を指定することができます。以下は、それらのその他の [ModeCount] ステータス値です。

ModeCount	説明
LockPending	排他ロックの保留中です。サーバが排他ロックの付与プロセス中にある場合に、このステータスが発生することがあります。ロック・ペンディングの状態のロックは削除できません。
SharePending	共有ロックの保留中です。サーバが共有ロックの付与プロセス中にある場合に、このステータスが発生することがあります。ロック・ペンディングの状態のロックは削除できません。
DelockPending	アンロックの保留中です。サーバが保持ロックのアンロック・プロセス中にある場合に、このステータスが発生することがあります。ロック・ペンディングの状態のロックは削除できません。
Lost	ネットワークがリセットされたため、ロックが失われました。

[所有者のルーチン情報を表示] を選択して [ルーチン] 列を有効にします。これにより、所有者のプロセスが実行しているルーチンの名前が、そのルーチンで実行されている現在の行の番号が先頭に追加されて表示されます。

[SQL オプションを表示] を選択し、[ネームスペースのSQLテーブル名を表示] リストからネームスペースを選択して [SQL テーブル名] 列を有効にします。この列には、選択したネームスペースにある各プロセスに関連付けられた SQL テーブルの名前が表示されます。プロセスが SQL テーブルと関連付けられていない場合、この列の値は空になります。

[ロック表示] ウィンドウを使用して、ロックを削除することはできません。

14.3 管理ポータルでのロックの削除

重要 ロックを削除するのではなく、ベスト・プラクティスとして、ロックを作成したプロセスを特定し、それを終了します。ロックを削除すると、そのロックの目的によっては、システムに重大な影響を与える可能性があります。

システム上で現在保持されているロックを削除するには、管理ポータルに移動して、[システムオペレーション]→[ロック]→[ロックを管理] の順に選択します。任意のプロセス ([所有者]) について、[削除] または [プロセスが保持するすべてのロックを削除] をクリックします。

ロックの削除により、全形式の該当ロック (すべてのインクリメント・レベルのロック、すべての排他、排他エスカレート、および共有バージョンのロック) が解放となります。1 つのロックを削除すると、そのロック・キューで待機している次のロックが即座に適用となります。

また、SYS.Lock.DeleteOneLock() および SYS.Lock.DeleteAllLocks() メソッドを使用して、ロックを削除することもできます。

ロックを削除するには、WRITE 権限が必要になります。ロックの削除は、監査データベースに記録されます (有効になっている場合)。messages.log には記録されません。

14.4 ^LOCKTAB ユーティリティ

また、%SYS ネームスペースから ^LOCKTAB ユーティリティを使用して、ロックを表示および削除することもできます。

重要 ロックを削除するのではなく、ベスト・プラクティスとして、ロックを作成したプロセスを特定し、それを終了します。ロックを削除すると、そのロックの目的によっては、システムに重大な影響を与える可能性があります。

^LOCKTAB は、以下の形式のいずれかで実行できます。

- DO ^LOCKTAB : ロックを表示および削除できます。個々のロックの削除、指定のプロセスが所有しているすべてのロックの削除、またはシステム上のすべてのロックの削除を行う文字コード・コマンドが用意されています。
- DO View^LOCKTAB : ロックを表示できます。ロックを削除するオプションは用意されていません。

これらのユーティリティ名は、大文字と小文字が区別されることに注意してください。

以下のターミナル・セッション例は、^LOCKTAB が現在のロックを表示する方法を示しています。

```
%SYS>DO ^LOCKTAB

                                Node Name: MYCOMPUTER
                                LOCK table entries at 07:22AM 01/13/2018
                                16767056 bytes usable, 16774512 bytes available.

Entry Process      X#   S# Flg   W# Item Locked
1) 4900             1           ^["^c:\intersystems\iris\mgr\";%SYS("CSP","Daemon")
2) 4856             1           ^["^c:\intersystems\iris\mgr\";ISC.LMFMON("License Monitor")
3) 5016             1           ^["^c:\intersystems\iris\mgr\";ISC.Monitor.System
4) 5024             1           ^["^c:\intersystems\iris\mgr\";TASKMGR
5) 6796             1           ^["^c:\intersystems\iris\mgr\user\";ja(1)
6) 6796            1e           ^["^c:\intersystems\iris\mgr\user\";ja(1,1)
7) 6796             2           1 ^["^c:\intersystems\iris\mgr\user\";jb(1)Waiters: 3120(XC)
8) 3120             2           ^["^c:\intersystems\iris\mgr\user\";jc(1)
9) 2024             1           ^["^c:\intersystems\iris\mgr\user\";jd(1)

Command=>
```

LOCKTAB の表示では、保持された排他ロックを X# 列がリストして、保持された共有ロックを S# 列がリストします。X# または S# の数字は、ロック・インクリメント・カウントを示します。“e” 接尾語は、ロックがエスカレートとして定義されていることを示します。“D” 接尾語は、ロックがロック解除状態であることを示します。ロックはアンロックされていますが、現在のトランザクションの終了まで別のプロセスが使用することはできません。W# 列では待機ロック要求の数をリストします。上記の表示で示すように、process 6796 は増分共有ロック `b(1) を保持しています。Process 3120 はこのロックを待つロック要求を 1 つ有しています。このロック要求は、`b(1) の子 (C) に対する排他 (X) ロックのためにあります。

このユーティリティのヘルプを表示するには、Command=> プロンプトで疑問符 (?) を入力します。これには、この表示の解説方法およびロックを削除する文字コード・コマンド (ある場合) についての詳細が含まれています。

注釈 Flg 列の値で示されるように、ロック保留状態にあるロックは削除できません。

LOCKTAB ユーティリティを終了するには Q を入力します。

14.5 関連項目

- ・ [ロックと並行処理の制御](#)
- ・ [ロック要求とデッドロックの詳細](#)

15

トランザクション処理

ここでは、InterSystems IRIS® データ・プラットフォームでのトランザクション処理について説明します。トランザクションとは、特定のシーケンスで複数のアトミック処理を組み合わせる、動作の論理ユニットです。アトミック処理とは、エラーの状況を含め、どのような状況においても常に完全に実行される処理です。そのような処理は通常 1 つのコマンドに対応します。

15.1 InterSystems IRIS のトランザクションについて

InterSystems IRIS では、アトミック処理には 1 つの SQL `INSERT` 文、`UPDATE` 文、または `DELETE` 文、あるいは 1 つのグローバル `SET` または `KILL` が含まれます。

ただし、タスクを実行するには、アプリケーションは通常複数のアトミック処理を組み合わせる必要があります。例えば銀行業務で、ある口座から別の口座に送金する場合、送金元のテーブル内のフィールドから差し引いた額を、送金先のテーブルのフィールドに追加する必要があります。両方の更新で 1 つのトランザクションとなることを指定すると、両方の処理が実行されるか、両方とも実行されないかのいずれかになります。つまり、一方の処理のみを実行できません。

このような場合、トランザクション処理コマンドを使用して、完全なトランザクションを構成する一連の操作を定義します。1 つのコマンドでトランザクションの開始を示し、一連の多数のコマンドを置いた後に、他のコマンドでトランザクションの終了を示します。

通常、トランザクションは全体的に実行されます。プログラム・エラーやシステムの動作不良が原因で不完全なトランザクションになった場合、部分的に完了したトランザクションはロールバックされます。

アプリケーション開発者は、アプリケーション内でトランザクション・ロールバックを処理する必要があります。InterSystems IRIS は、リカバリ、`HALT` や `ResJob` の実行中など、システム障害時やさまざまな状況でトランザクション・ロールバックを自動的に処理します。

LogRollback 構成オプションが設定されている場合、InterSystems IRIS は、`messages.log` ファイルにロールバックを記録します。管理ポータルの [システムオペレーション]→[システムログ]→[メッセージログ] オプションを使用して、`messages.log` を表示できます。

15.2 アプリケーションでのトランザクション管理

InterSystems IRIS では、ユース・ケースに応じて、SQL または ObjectScript 内でトランザクションを定義します。

重要

これらの手法は完全に置き換え可能ではありません。特定のトランザクションは、すべて ObjectScript か、すべて SQL で管理する必要があります。例えば、ObjectScript であるトランザクションを開始した場合、TSTART には TCOMMIT など、ObjectScript コマンドを使用して残りのトランザクションを処理します。SQL コマンドを使用した場合は、START TRANSACTION には COMMIT など、SQL コマンドを使用する必要があります。ただし、これらのコードは互いに呼び出すようにすることができ、入れ子にすることも可能です。

15.2.1 トランザクション・コマンド

InterSystems IRIS は、ANSI SQL 操作 COMMIT WORK と ROLLBACK WORK をサポートしています (InterSystems SQL では、キーワード WORK はオプションとなります)。また、InterSystems SQL 拡張機能 SET TRANSACTION、START TRANSACTION、SAVEPOINT、および %INTRANS もサポートしています。また、InterSystems IRIS には、M 言語 Type A 標準のトランザクション・コマンドのうち、実装されているものもあります。

以下のテーブルは、これらの SQL と ObjectScript コマンドの概要です。

テーブル 15-1: トランザクション・コマンド

SQL コマンド	ObjectScript コマンド	定義
SET TRANSACTION		トランザクションを開始せずに、トランザクション・パラメータを設定します。
START TRANSACTION	TSTART	トランザクションの開始を表します。
%INTRANS	\$TLEVEL	トランザクションが現在進行中かどうかを検出します。 <ul style="list-style-type: none"> ・ %INTRANS によって使用される <0 は、トランザクション処理中を意味しますが、ジャーナリングは無効です。\$TLEVEL では使用されません。 ・ 0 = トランザクション処理中ではありません。 ・ >0 は、トランザクション処理中を意味します。
SAVEPOINT		トランザクション内のポイントを指定します。セーブポイントまでの部分的なロール・バックに使用できます。
COMMIT	TCOMMIT	トランザクションの正常終了を示します。
ROLLBACK	TROLLBACK	トランザクションの失敗を示します。トランザクションの開始後に実行されたすべてのデータベース更新は、ロールバックされるか、元の状態に戻ります。

これらの ObjectScript と SQL のコマンドは完全に互換性があり、置き換え可能ですが、以下の例外があります。

ObjectScript TSTART と SQL START TRANSACTION はどちらも、トランザクションが進行中でない場合にトランザクションを開始します。ただし、START TRANSACTION では、入れ子になったトランザクションはサポートされません。そのため、入れ子になったトランザクションが必要な場合 (または必要になる可能性がある場合) には、トランザクションを TSTART で始めることをお勧めします。SQL 標準との互換性が必要な場合は、START TRANSACTION を使用してください。

15.2.2 トランザクションでの LOCK の使用

複数のプロセスがアクセスする可能性のあるグローバルにアクセスする際、常にそのグローバルで **LOCK** コマンドを使用して、データベースの整合性を保護する必要があります。グローバル変数に対応したロックを発行し、グローバルの値を変更し、次にロックをアンロックします。LOCK コマンドは、指定したロックのロックとアンロックの両方に使用されます。

グローバルの値を変更しようとするその他のプロセスはロックを要求し、そのロックは最初のプロセスがロックを解除するまで待機します。

トランザクションでロックを使用する際、以下の 3 つの考慮事項があります。

- ・ ロック/アンロック操作は、ロールバックを実行しません。
- ・ トランザクション内で、プロセスによって保持されたロックをアンロックすると、以下の 2 つのうちどちらかが発生する場合があります。
 - ロックがすぐにアンロックされる。ロックは直後に別のプロセスによって取得可能になります。
 - ロックがロック解除状態に置かれる。ロックはアンロックされていますが、現在のトランザクションの終了まで別のプロセスによって取得可能な状態になりません。

ロックがロック解除状態にある場合は、InterSystems IRIS はトランザクションがコミットされるかロールバックされるまで、アンロックを延期します。トランザクション内では、ロックはアンロック状態が表示され、同じ値の後続のロックが許可されます。しかし、トランザクションの範囲外では、ロックはロックされたままになります。詳細は、“[ロック管理](#)”を参照してください。

- ・ タイムアウトになるロック操作は、[\\$TEST](#) を設定します。トランザクション時に [\\$TEST](#) で設定される値はロールバックしません。

15.2.3 トランザクションでの [\\$INCREMENT](#) と [\\$SEQUENCE](#) の使用法

[\\$INCREMENT](#) または [\\$SEQUENCE](#) 関数呼び出しは、トランザクションの一部ではありません。したがって、トランザクション・ロールバックの一部としてロールバックされません。これらの関数を使用すると、LOCK コマンドを使用せずにインデックス値を取得できます。この機能は、トランザクションの継続中にカウンタ・グローバルをロックしたくない場合に役立ちます。

[\\$INCREMENT](#) は、1 つ以上のプロセスからインクリメント要求を受け取った順に個々の整数値を割り当てます。[\\$SEQUENCE](#) は複数プロセスに対して高速手段を提供するものであり、整数値のシーケンス (範囲) を各増分プロセスに割り当てることにより、同一のグローバル変数に対して一意 (重複なし) の整数を取得します。

注釈 [\\$INCREMENT](#) は、トランザクション内の 1 プロセスによってインクリメントされます。また、トランザクションの処理中、並行するトランザクションの別のプロセスによってもインクリメントされる場合があります。最初のトランザクションをロールバックすると、インクリメントが“スキップされ”、“使用されない”番号が発生する可能性があります。

15.2.4 アプリケーション内でのトランザクション・ロールバック

トランザクション中にエラーに遭遇した場合、以下の 3 つの方法でロールバックできます。

- ・ SQL ロールバック・コマンド ROLLBACK WORK の発行
- ・ ObjectScript ロールバック・コマンド TROLLBACK の発行
- ・ %ETN の呼び出し

注釈 トランザクションをロールバックする場合、既定クラスの IDKey はデクリメントされません。IDKey の値は、[\\$INCREMENT](#) 関数によって自動的に変更されます。

15.2.4.1 SQL あるいは ObjectScript ロールバック・コマンドの発行

アプリケーション開発者は、2 種類のロールバック・コマンドを使用して、トランザクションが失敗した場所を指定し、不完全なトランザクションを自動的にロールバックします。

- ・ マクロ・ソース・ルーチンで、`##sql(ROLLBACK WORK)` を使用
- ・ マクロあるいは中間ソース・コードで、ObjectScript `TROLLBACK` コマンドを使用

以下の例のように、ロールバック・コマンドは、エラー・トラップと併用する必要があります。

ObjectScript

```

ROU      ##sql(START TRANSACTION) set $ZT="ERROR"
        SET ^ZGLO(1)=100
        SET ^ZGLO=error
        SET ^ZGLO(1,1)=200
        ##sql(COMMIT WORK) Write !,"Transaction Committed" Quit
ERROR    ##sql(ROLLBACK WORK)
        Write !,"Transaction failed." Quit

```

この例のコードでは、トランザクションがコミットされる前にプログラム・エラーが発生した場合、\$ZT が ERROR サブルーチンを実行するように設定されています。ROU 行は、トランザクションを開始し、エラー・トラップを設定します。ROU+1 行と ROU+3 行では、グローバル ^ZGLO のノードを設定します。しかし、変数 error が未定義の場合、ROU+2 でプログラム・エラーが発生し、ROU+3 行は実行されません。プログラムは、ERROR サブルーチンを実行し、^ZGLO(1) の設定は取り消されます。ROU+2行が削除されていたら、^ZGLO は 2 回とも値を設定し、トランザクションはコミットされて、“Transaction committed” というメッセージが表示されます。

15.2.4.2 %ETN への呼び出し

ロールバック・コマンドでトランザクション・ロールバックを処理しない場合、エラー・トラップ・ユーティリティ %ETN が不完全なトランザクションを検出し、トランザクションのコミットまたはロールバックのいずれかをユーザに指示します。不完全なトランザクションをコミットすると、通常、論理データベースの整合性の低下を招くため、アプリケーション内でロールバックを処理する必要があります。

トランザクションの進行中に発生したエラーの後に %ETN を実行する場合、以下のロールバック・プロンプトが表示されます。

```

You have an open transaction.
Do you want to perform a Commit or Rollback?
Rollback =>

```

10 秒のタイムアウト時間内に何も応答がない場合、既定でロールバックされます。ジョブ起動ジョブあるいはアプリケーション・モード・ジョブでは、メッセージなしでトランザクションがロールバックされます。

%ETN 自体は、トランザクションのロールバックを引き起こしませんが、通常、%ETN は InterSystems IRIS の停止によって終了します。ObjectScript の停止時にトランザクション・ロールバックが発生し、システムが %HALT を実行して InterSystems IRIS プロセスのクリーンアップを実行します。%ETN には、BACK^%ETN というエントリ・ポイントがあり、halt ではなく quit で終了します。ルーチンが BACK^%ETN ではなく %ETN や FORE^%ETN を呼び出す場合、エラー処理プロセスの一部としてトランザクション・ロールバックを実行しません。

15.2.5 アプリケーション内のトランザクション処理の例

以下の例は、マクロ・ソース・ルーチンでトランザクションを処理する方法です。SQL コードでデータベース更新を実行します。以下の SQL 文は、ある口座から別の口座に資金を送金します。

ObjectScript

```
Transfer(from,to,amount)    // Transfer funds from one account to another
{
    TSTART
    &SQL(UPDATE A.Account
        SET A.Account.Balance = A.Account.Balance - :amount
        WHERE A.Account.AccountNum = :from)
    If SQLCODE TROLLBACK Quit "Cannot withdraw, SQLCODE = "_SQLCODE
    &SQL(UPDATE A.Account
        SET A.Account.Balance = A.Account.Balance + :amount
        WHERE A.Account.AccountNum = :to)
    If SQLCODE TROLLBACK QUIT "Cannot deposit, SQLCODE = "_SQLCODE
    TCOMMIT
    QUIT "Transfer succeeded"
}
```

15.3 自動トランザクション・ロールバック

以下の場合、トランザクション・ロールバックが自動的に発生します。

- ・ InterSystems IRIS の起動 (リカバリが必要な場合)。起動されると、InterSystems IRIS はリカバリが必要なことを判定し、コンピュータ上の不完全なトランザクションをすべてロールバックします。
- ・ **HALT** コマンド (現在のプロセスの場合) または **^RESJOB** ユーティリティ (その他のプロセスの場合) を使用したプロセスの終了。バックグラウンド・ジョブ (非インタラクティブ・プロセス) を停止すると、進行中の現在のトランザクションに対する変更が自動的にロールバックされます。インタラクティブ・プロセスを停止すると、進行中の現在のトランザクションに対する変更をコミットするかロールバックするかを確認するプロンプトが表示されます。**^RESJOB** をプログラマ・モード・ユーザ・プロセスで発行すると、ユーザにメッセージが表示され、現在のトランザクションをコミットするかロールバックするか尋ねられます。

また、システム管理者は、**^JOURNAL** ユーティリティを実行して、クラスタ固有のデータベースで不完全なトランザクションをロールバックできます。**^JOURNAL** ユーティリティのメイン・メニューから **Restore Globals From Journal** オプションを選択すると、ジャーナル・ファイルがリストアップされ、すべての不完全なトランザクションがロールバックされます。

15.4 トランザクション処理に関するシステム全体の問題

このセクションでは、トランザクション処理に関するさまざまなシステム全体の問題について説明します。バックアップに関連した問題の詳細は、["バックアップとリストア"](#)を参照してください。ECP に関連した問題の詳細は、["ECPリカバリ・プロセス、保証、および制限"](#)を参照してください。

15.4.1 トランザクション処理でのバックアップとジャーナリング

トランザクション処理を実装する際、以下のバックアップおよびジャーナリングの手順を検討してください。

InterSystems IRIS の各インスタンスにはジャーナルがあります。ジャーナルとは、最後のバックアップ以後、データベースに対して行われた変更を、時間順で記録している一連のファイルからなるログです。InterSystems IRIS のトランザクション処理は、データの論理的整合性を維持するために、ジャーナリングと共に機能します。

ジャーナルには、トランザクションにあるグローバルに対する SET 処理と KILL 処理が記録されます。これは、影響を受けるグローバルが保存されているデータベースのジャーナル設定には関係なく記録されます。さらに、**[グローバル・ジャーナル状態]**を“**Yes**”に設定したデータベースのグローバルに対するすべての SET 処理と KILL 処理も記録されます。

バックアップは、トランザクション処理中に実行できます。ただし、その結果として生ずるバックアップ・ファイルには、トランザクションの一部が含まれる場合があります。災害が発生し、バックアップからリストアする必要が生じた場合は、まず

バックアップ・ファイルをリストアしてから、リストアしたデータベースのコピーにジャーナル・ファイルを適用します。ジャーナル・ファイルを適用すると、バックアップの時点から災害発生時までの間にジャーナルに記述された更新がすべてリストアされます。トランザクションの一部を完了し、コミットされていないトランザクションをロールバックすることによってデータベースのトランザクションの整合性をリストアするには、ジャーナルの適用が必要です。これは、バックアップを実行した時点では、データベースにすべてのトランザクションが記録されていない可能性があるためです。詳細は、以下を参照してください。

- ・ [ジャーナリング](#)
- ・ [ジャーナルの重要性](#)

15.4.2 非同期エラーの通知

`%SYSTEM.Process` クラスの `AsynchError()` メソッドを使用して、ジョブが非同期エラーで割り込まれるかどうかを指定できます。

- ・ `%SYSTEM.Process.AsynchError(1)` は、非同期エラーの受け入れを可能にします。
- ・ `%SYSTEM.Process.AsynchError(0)` は、非同期エラーの受け入れを不可能にします。

`Config.Miscellaneous` クラスの `AsynchError` プロパティは、新規プロセスに対するシステム全体の既定として、プロセスが非同期エラーで割り込まれるかどうかを設定します。既定値は 1 で、“割り込まれる”を意味します。

特定のジョブで複数の非同期エラーが検出される場合、システムは少なくとも 1 つ非同期エラーを引き起こします。しかし、どのエラーが起こるかは判別できません。

非同期エラーには、以下が実装されています。

- ・ `<LOCKLOST>` – このジョブで一度ロックされたもののうち、いくつかがリセットされました。
- ・ `<DATALOST>` – このジョブで実行されたデータ修正のうち、いくつかはエラーとしてサーバから返されました。
- ・ `<TRANLOST>` – このジョブで開かれた分散型トランザクションが、サーバによって非同期的にロール・バックされました。

非同期エラーを受信しているジョブを無効にした場合でも、次に `ZSync` コマンドを実行すると、非同期エラーが引き起こされます。

`TStart`、`TCommit`、または `LOCK` 操作、およびネットワーク・グローバル参照が行われるたびに、InterSystems IRIS は保留になっている非同期エラーをチェックします。ネットワーク上の `SET` 操作と `KILL` 操作は非同期であるため、`SET` の生成時と非同期エラーの報告時の間に、任意の数の他の指示が挿入される場合があります。

15.5 現在のすべてのトランザクションの一時停止

`%SYSTEM.Process` クラスの `TransactionsSuspended()` メソッドを使用して、現在のプロセスの現在のトランザクションすべてを一時停止できます。これはブーリアン・メソッドです (1 = 現在のすべてのトランザクションを一時停止する、0 = 現在のすべてのトランザクションを再開する)。デフォルトは 0 です。

トランザクションが一時停止されている間、変更はトランザクション・ログに記録されないため、ロールバックすることはできません。トランザクションが一時停止された期間の前後に現在のトランザクションで行われた変更は、ロールバックすることができます。

トランザクションでグローバル変数を変更し、そのトランザクションが一時停止されている間にその変数を再度変更すると、ロールバックが試行されたときにエラーが発生する可能性があります。

ブーリアン・パラメータを指定せずに `TransactionsSuspended()` を呼び出すと、その設定を変更せずに現在のブーリアン設定が返されます。

16

%Status 値の操作

%Status 値 (status) を返す API を扱う場合、続行する前にステータスを確認し、成功の場合にのみ通常の処理を続行するのがベスト・プラクティスです。独自のコードでは、ステータス値を返す(および必要に応じて別の場所でその値を確認する) こともできます。

このページでは、ステータス値とその操作方法について説明します。

注釈 ステータスの確認は、本質的にはエラー・チェックではありません。予期できないエラーをトラップするには、コードで **TRY-CATCH** 処理も使用する必要があります。

16.1 ステータス値の操作の基本

InterSystems IRIS® データ・プラットフォームのクラスの多くに含まれるメソッドでは、**%Status** (**%Library.Status**) 値を返すことで、成功またはエラーを示します。ステータスがエラーを表す場合、エラーに関する情報もステータスに含まれます。例えば、**%Library.Persistent** の **%Save()** メソッドはステータスを返します。このようなメソッドでは、返されるステータスを必ず取得してください。その後、ステータスを確認して適切に処理を続行します。これには、以下の 2 つのシナリオが考えられます。

- ・ 成功の場合、ステータスは 1 になります。
- ・ 失敗の場合、ステータスは、エラー・ステータスおよび 1 つ以上のエラー・コードとテキスト・メッセージを含むエンコードされた文字列です。ステータス・テキスト・メッセージはユーザのロケールの言語でローカライズされています。InterSystems IRIS では、値を処理するためのメソッドとマクロが用意されているため、失敗の性質を理解することができます。

基本的なツールは以下のとおりです。

- ・ ステータスが成功またはエラーのどちらを表すかを確認するには、以下のいずれかを使用します。
 - **\$\$\$ISOK** マクロと **\$\$\$ISERR** マクロ。これらはインクルード・ファイル **%occStatus.inc** で定義されています。このインクルード・ファイルは、すべての**オブジェクト・クラス**で自動的に利用可能です。
 - **\$SYSTEM.Status.IsOK()** および **\$SYSTEM.Status.IsError()** メソッド。
- ・ エラーの詳細を表示するには、**\$SYSTEM.Status.DisplayError()** または **\$SYSTEM.OBJ.DisplayError()** を使用します。これらのメソッドは、互いに同等です。出力は現在のデバイスに書き込まれます。
- ・ エラーの詳細を含む文字列を取得するには、**\$SYSTEM.Status.GetErrorText()** を使用します。

\$SYSTEM 特殊変数は %SYSTEM パッケージに結合されます。これは、以前のリストのメソッドが %SYSTEM.Status および %SYSTEM.OBJ クラスにあることを意味します。詳細は、クラス・リファレンスを参照してください。

16.2 例

以下に例を示します。

```
Set object=##class(Sample.Person).%New()
Set object.Name="Smith,Janie"
Set tSC=object.%Save()
If $$$ISERR(tSC) {
    Do $SYSTEM.Status.DisplayError(tSC)
    Quit
}
```

以下に、\$SYSTEM.Status.GetErrorText() の使用法を示す部分的な例を示します。

```
If $$$ISERR(tSC) {
    // if error, log error message so users can see them
    Do ..LogMsg($System.Status.GetErrorText(tSC))
}
```

注釈 一部の ObjectScript プログラマは、文字 `t` を接頭語として使用して一時変数であることを示すため、コード・サンプルで “一時ステータス・コード” を意味する `tSC` が変数名として使用されることがあります。この規則を使用するのは自由ですが、この変数名に特別なことは何もありません。

16.3 バリエーション (%objlasterror)

%New() のような一部のメソッドは、%Status を返しません、ステータスを含む %objlasterror 変数を更新します。%New() は、成功時にはクラスのインスタンスに OREF を返し、失敗時には NULL 文字列を返します。以下の例で示すように、%objlasterror 変数にアクセスして、この種のメソッドのステータス値を取得することができます。

ObjectScript

```
Set session = ##class(%CSP.Session).%New()
If session="" {
    Write "session oref not created",!
    Write "%New error is ",!,$System.Status.GetErrorText(%objlasterror),!
} Else {
    Write "session oref is ",session,!
}
```

詳細は、%SYSTEM.Status のクラスを参照してください。

16.4 1 つのステータス値で報告される複数のエラー

1 つのステータス値が複数のエラーを表す場合、前述の手法で得られるのは最後のエラーの情報のみです。%SYSTEM.Status は、個々のエラーを取得するのに使用できるメソッド、GetOneErrorText() および GetOneStatusText() を提供します。以下に例を示します。

ObjectScript

```
CreateCustomErrors
    SET st1 = $System.Status.Error(83,"my unique error")
    SET st2 = $System.Status.Error(5001,"my unique error")
    SET allstatus = $System.Status.AppendStatus(st1,st2)
DisplayErrors
    WRITE "All together:",!
    WRITE $System.Status.GetErrorText(allstatus),!!
    WRITE "One by one",!
    WRITE "First error format:",!
    WRITE $System.Status.GetOneStatusText(allstatus,1),!
    WRITE "Second error format:",!
    WRITE $System.Status.GetOneStatusText(allstatus,2),!
```

もう 1 つのオプションは \$SYSTEM.Status.DecomposeStatus() です。このメソッドは、エラーの詳細の配列を (参照によって、2 つ目の引数として) 返します。以下に例を示します。

```
Do $SYSTEM.Status.DecomposeStatus(tSC,.errorlist)
//then examine the errorlist variable
```

変数 errorlist は、エラー情報が含まれる [多次元配列](#) です。以下に、部分的な例を示します (読みやすいように人為的に改行が追加されています)。

```
ZWRITE errorlist
errorlist=2
errorlist(1)="ERROR #5659: Property 'Sample.Person::SSN(1@Sample.Person,ID=)' required"
errorlist(1,"caller")="%ValidateObject+9^Sample.Person.1"
errorlist(1,"code")=5659
errorlist(1,"dcode")=5659
errorlist(1,"domain")="%ObjectErrors"
errorlist(1,"namespace")="SAMPLES"
errorlist(1,"param")=1
errorlist(1,"param",1)="Sample.Person::SSN(1@Sample.Person,ID=)"
...
errorlist(2)="ERROR #7209: Datatype value '' does not match
PATTERN '3N1'"-"2N1'"-"4N'"_"$c(13,10)"_ " >
ERROR #5802: Datatype validation failed on property 'Sample.Person:SSN',
with value equal to """"
errorlist(2,"caller")="zSSNIsValid+1^Sample.Person.1"
errorlist(2,"code")=7209
...
```

各エラー・メッセージをログに記録したい場合は、前述のロギングの例を以下のように変更できます。

```
If $$$ISERR(tSC) {
    // if error, log error message so users can see them
    Do $SYSTEM.Status.DecomposeStatus(tSC,.errorlist)
    For i=1:1:errorlist {
        Do ..LogMsg(errorlist(i))
    }
}
```

注釈 DecomposeStatus() を再び呼び出して同じエラー配列を渡すと、新しいエラーが配列に付加されます。

16.5 %Status を返す

ユーザがユーザ独自のカスタム・ステータス値を返すことができます。%Status を作成するには、以下の構文を使用します。

```
$$$ERROR($$GeneralError,"your error text here","parm","anotherparm")
```

または、以下も同じ意味です。

```
$SYSTEM.Status.Error($$GeneralError,"your error text here",
"parm","anotherparm")
```

ここで、"parm" および "anotherparm" は、処理が成功しなかったレコードのファイル名や識別子など、オプションの追加のエラー引数を表します。

以下に例を示します。

```
quit $$$ERROR($$$GeneralError,"Not enough information for request")
```

他のエラーに関する情報を含めるには、\$SYSTEM.Status.AppendStatus() を使用してステータス値を変更します。以下に例を示します。

```
set tSC=$SYSTEM.Status.AppendStatus(tSCfirst,tSCsecond)
quit tSC
```

16.6 %SYSTEM.Error

%SYSTEM.Error クラスは一般的なエラー・オブジェクトです。これは %Status エラー、[例外オブジェクト](#)、\$ZERROR エラー、または SQLCODE エラーから作成することができます。

%SYSTEM.Error クラス・メソッドを使用すれば、%Status を例外に、または例外を %Status に変換することができます。

16.7 関連項目

詳細は、"%SYSTEM.Status" クラスおよび "%Status (%Library.Status)" クラスのクラス・リファレンスを参照してください。

17

TRY-CATCH の使用法

エラー (特に予期しないエラー) 発生時のコードの動作を管理することをエラー処理といいます。エラー処理には、以下のオペレーションが含まれます。

- ・ エラーの原因となる条件の修正
- ・ エラー発生後に動作を再開させるための処理の実行
- ・ 実行フローの変更
- ・ エラーに関する[情報のログ記録](#)

InterSystems IRIS® データ・プラットフォームは、エラー処理のために TRY-CATCH メカニズムをサポートします。古いアプリケーションから移行されたコードでは、[従来のエラー処理](#)が行われる場合があります。これは引き続き全面的にサポートされていますが、新しいアプリケーションで使用することを意図したものではありません。

[%Status の処理](#)についても参照してください。こちらは厳密にはエラー処理ではありません。通常、ステータスの処理は TRY ブロック内で完結します。

17.1 概要

TRY-CATCH を使用すると、それぞれが TRY ブロックと呼ばれる区切られたコードのブロックを作成できます。TRY ブロック中にエラーが発生した場合、TRY ブロックに関連付けられた、例外処理のコードを含む CATCH ブロックに制御が渡されます。TRY ブロックには THROW コマンドを含めることもできます。これらのコマンドはそれぞれ、TRY ブロック内から明示的に例外を発行し、実行を CATCH ブロックに移します。

最も基本的な形式でこのメカニズムを使用するには、ObjectScript コード内に TRY ブロックを含めます。このブロック内で例外が発生する場合、関連付けられた CATCH ブロック内のコードが実行されます。TRY-CATCH ブロックの形式は以下のとおりです。

```
TRY {  
    protected statements  
} CATCH [ErrorHandler] {  
    error statements  
}  
further statements
```

以下はその説明です。

- ・ TRY コマンドは、中括弧で囲まれた ObjectScript のコード文を識別します。TRY は、引数を取りません。このコード・ブロックは、構造化された例外処理のための保護されたコードです。TRY ブロック内で例外が発生した場合、InterSystems IRIS は例外プロパティ (oref.Name、oref.Code、oref.Data、oref.Location)、\$ZERROR、および \$ECODE

を設定し、CATCH コマンドによって識別される、例外ハンドラに実行を移します。これは、例外のスローとして知られています。

- ・ `protected statements` は、通常の実行の一部である、ObjectScript 文です。(これらの文には、THROW コマンドへの呼び出しを含めることができます。このシナリオについては、以下のセクションで説明します。)
- ・ CATCH コマンドは、例外ハンドラを定義します。これは、TRY ブロックで例外が発生したときに実行されるコードのブロックです。
- ・ `ErrorHandle` 変数は、例外オブジェクトへのハンドルです。これは、InterSystems IRIS が実行時エラーに対して生成した例外オブジェクト、または THROW コマンド (以下のセクションで説明します) の呼び出しにより明示的に発行された例外オブジェクトのいずれかとなります。
- ・ `error statements` は、例外が発生したときに呼び出される ObjectScript 文です。
- ・ `further statements` は、例外がない場合に `protected statements` の実行の後に続く、または例外が発生して CATCH ブロックから制御が渡される場合に `error statements` の実行の後に続く、ObjectScript 文です。

`protected statements` の実行中のイベントに応じて、以下のイベントのいずれかが発生します。

- ・ エラーが発生しない場合、CATCH ブロックの外側に表示される `further statements` で実行が続行します。
- ・ エラーが発生する場合、制御が CATCH ブロックに渡され、`error statements` が実行されます。実行は、CATCH ブロックの内容によって異なります。
 - CATCH ブロックに THROW コマンドまたは GOTO コマンドが含まれている場合、指定された場所に制御が直接移動します。
 - CATCH ブロックに THROW コマンドまたは GOTO コマンドが含まれていない場合、CATCH ブロックから制御が渡され、`further statements` で実行が続行します。

17.2 TRY-CATCH と一緒に THROW を使用する

InterSystems IRIS は、実行時エラーの発生時に暗黙的な例外を発行します。明示的な例外を発行するには、[THROW](#) コマンドを使用できます。THROW コマンドは、TRY ブロックから CATCH 例外ハンドラへ実行を移します。THROW コマンドの構文は以下のとおりです。

THROW expression

expression は、例外処理のために InterSystems IRIS に用意されている `%Exception.AbstractException` クラスから継承するクラスのインスタンスです。`%Exception.AbstractException` の詳細は、以下のセクションを参照してください。

TRY/CATCH ブロックと THROW を一緒に発行する場合の形式は以下のとおりです。

```
TRY {
    protected statements
    THROW expression
    protected statements
}
CATCH exception {
    error statements
}
further statements
```

ここで、THROW コマンドは明示的に例外を発行します。TRY-CATCH ブロックの他の要素については、前のセクションで説明しています。

THROW の動作は、スローが発生する場所と THROW の引数によって異なります。

- ・ TRY ブロック内の THROW は、CATCH ブロックに制御を渡します。
- ・ CATCH ブロック内の THROW は、実行スタックの制御を次のエラー・ハンドラに渡します。例外が %Exception.SystemException オブジェクトの場合、次のエラー・ハンドラは、どのタイプ (CATCH または [従来のエラー処理](#)) でもかまいません。それ以外の場合は、CATCH による例外処理が必要となります。そうしないと、<NOCATCH> エラーがスローされます。

引数付きの THROW のために、制御が CATCH ブロックに渡される場合、ErrorHandle には引数からの値が含まれています。システム・エラーのために制御が CATCH ブロックに渡される場合、ErrorHandle は %Exception.SystemException オブジェクトです。ErrorHandle が指定されていない場合、制御が CATCH ブロックに渡された理由が示されません。

例えば、以下のように 2 つの数字を除算するコードがあるとします。

```
div(num,div) public {
  TRY {
    SET ans=num/div
  } CATCH errobj {
    IF errobj.Name="<DIVIDE>" { SET ans=0 }
    ELSE { THROW errobj }
  }
  QUIT ans
}
```

0 による除算エラーが発生すると、このコードは結果として 0 を返すように特別に設計されています。この他のエラーの場合、THROW はスタック上のエラーを次のエラー・ハンドラに送信します。

17.3 \$\$\$ThrowOnError および \$\$\$ThrowStatus マクロの使用法

InterSystems IRIS は、例外処理で使用するために、マクロを提供します。これらのマクロは、呼び出されると CATCH ブロックに例外オブジェクトをスローします。

以下の例は、%Prepare() メソッドによりエラー・ステータスが返されると、\$\$\$ThrowOnError マクロを呼び出します。

ObjectScript

```
#include %occStatus
TRY {
  SET myquery = "SELECT TOP 5 Name,Hipness,DOB FROM Sample.Person"
  SET tStatement = ##class(%SQL.Statement).%New()
  SET status = tStatement.%Prepare(myquery)
  $$$ThrowOnError(status)
  WRITE "%Prepare succeeded",!
  RETURN
}
CATCH sc {
  WRITE "In Catch block",!
  WRITE "error code: ",sc.Code,!
  WRITE "error location: ",sc.Location,!
  WRITE "error data: ", $LISTGET(sc.Data,2),!
  RETURN
}
```

以下の例は、%Prepare() メソッドにより返されたエラー・ステータスの値をテストした後に \$\$\$ThrowStatus を呼び出します。

ObjectScript

```
#include %occStatus
TRY {
    SET myquery = "SELECT TOP 5 Name,Hipness,DOB FROM Sample.Person"
    SET tStatement = ##class(%SQL.Statement).%New()
    SET status = tStatement.%Prepare(myquery)
    IF ($System.Status.IsError(status)) {
        WRITE "%Prepare failed",!
        $$$ThrowStatus(status)
    }
    ELSE {WRITE "%Prepare succeeded",!
        RETURN }
    }
CATCH sc {
    WRITE "In Catch block",!
    WRITE "error code: ",sc.Code,!
    WRITE "error location: ",sc.Location,!
    WRITE "error data: ",$LISTGET(sc.Data,2),!
    RETURN
}
```

詳細は、“システム・マクロ”を参照してください。

17.4 %Exception.SystemException と %Exception.AbstractException クラスの使用

InterSystems IRIS は、例外処理で使用するために、%Exception.SystemException クラスと %Exception.AbstractException クラスを提供します。%Exception.SystemException は %Exception.AbstractException クラスから継承しており、システム・エラーに使用されます。カスタム・エラーの場合、%Exception.AbstractException から継承するクラスを作成します。%Exception.AbstractException には、エラーの名前や発生した場所などのプロパティが含まれます。

システム・エラーが TRY ブロック内でキャッチされると、システムは %Exception.SystemException クラスの新しいインスタンスを作成し、そのインスタンス内にエラー情報を配置します。アプリケーション・プログラマは、カスタム例外をスローするとき、オブジェクトにエラー情報を入れる責任があります。

例外オブジェクトには、以下のプロパティがあります。

- ・ Name — <UNDEFINED> などのエラー名。
- ・ Code — エラー番号
- ・ Location — エラーの label+offset^routine の場所
- ・ Data — エラーによって報告される任意の追加データ (エラーの原因となる項目の名前など)

17.5 TRY-CATCH を使用する際の他の考慮事項

TRY-CATCH ブロックを使用する際に発生する可能性がある状況を、以下に説明します。

17.5.1 TRY-CATCH ブロック内での QUIT

TRY または CATCH ブロック内の QUIT コマンドは、TRY-CATCH ブロック全体の後で、ブロックから次の文に制御を渡します。

17.5.2 TRY-CATCH と実行スタック

TRY ブロックは、実行スタックで新しいレベルを導入しません。つまり、NEW コマンドのアクセス範囲境界ではありません。error statements は、そのエラーと同じレベルで実行されます。この結果として、protected statements 内に DO コマンドがあり、DO のターゲットも protected statements 内にある場合には、予期しない結果が発生する可能性があります。このような場合には、\$ESTACK 特殊変数が、相対的な実行レベルに関する情報を提供できます。

17.5.3 従来のエラー処理で TRY-CATCH を使用する

TRY-CATCH エラー処理は、実行スタックの異なるレベルで使用する \$ZTRAP エラー・トラップと互換性があります。例外は、\$ZTRAP を TRY 節の protected statements 内で使用できないことです。THROW コマンドでは、ユーザ定義のエラーは TRY-CATCH のみに制限されます。ZTRAP コマンドを使用したユーザ定義のエラーは、任意のタイプのエラー処理で使用できます。

18

エラー・ログ

各ネームスペースは、アプリケーション・エラー・ログを保持できます。ログには、そのネームスペースでコードを実行したときに発生したエラーが記録されます。一部のシステム・コードではこのログを自動的に書き込みます。独自のコードでもそのようにすることができます。

18.1 アプリケーション・エラーのログ作成

アプリケーション・エラー・ログに例外を記録するには、`%Exception.AbstractException.Log()` メソッドを使用します。通常、これは **TRY-CATCH** の **CATCH** ブロック内で行います。

18.2 管理ポータルを使用したアプリケーション・エラー・ログの表示

管理ポータルで、**[システムオペレーション]**→**[システムログ]**→**[アプリケーションエラーログ]** の順に選択します。これにより、アプリケーション・エラー・ログが発生したネームスペースの **[ネームスペース]** リストが表示されます。ヘッダを使用して、リストをソートできます。

ネームスペースの **[日付]** を選択すると、アプリケーション・エラー・ログが発生した日付と、その日付で記録されているエラーの数が表示されます。ヘッダを使用して、リストをソートできます。**[フィルタ]** を使用して、**[日付]** と **[数量]** の値に文字列をマッチングできます。

日付の **[エラー]** を選択すると、その日付のエラーが表示されます。**[エラー#]** の整数は、時系列順にエラーに割り当てられます。**[エラー#]** の ***COM** は、対象の日付のすべてのエラーに適用されるユーザ・コメントです。ヘッダを使用して、リストをソートできます。**[フィルタ]** を使用して、文字列をマッチングできます。

エラーの **[詳細]** を選択すると、**[エラー詳細]** ウィンドウが表示され、**特殊変数** の値や **[スタック]** の詳細など、エラー発生時の状態に関する情報が示されます。エラーに対応するスタック・トレースを表示するには、**[スタック]** をクリックするか、ページの下部までスクロールします。その後、このテーブルのその行の **+** アイコンをクリックして、`%objlasterror` 変数を探します。この変数には（存在する場合）エラーについての情報が含まれています。

個々のエラーにユーザ・コメントを指定できます。

[ネームスペース]、**[日付]**、および **[エラー]** のリストにはチェック・ボックスが含まれており、これを使用して対応する 1 つ以上のエラーのエラー・ログを削除できます。削除するものにチェックを付けて、**[削除]** ボタンを選択します。

18.3 ^%ERN を使用したアプリケーション・エラー・ログの表示

^%ERN ユーティリティはアプリケーション・エラーを調べ、現在のネームスペースについて記録されたすべてのエラーを確認できるようにします。これは、[管理ポータル](#)を使用する方法に代わるものとなります。

^%ERN ユーティリティを使用するには、以下の手順を実行します。

1. **ObjectScript シェル**で、「DO ^%ERN」と入力します。このユーティリティの名前では大文字と小文字が区別されます。ユーティリティ内でのプロンプトへの応答では、大文字と小文字は区別されません。
 いずれのプロンプトでも、「?」を入力すると、そのプロンプトの構文オプションが一覧表示され、「?L」を入力すると、定義されている値がすべて一覧表示されます。**Enter** キーを押すと、前のレベルに戻ることができます。
2. **For Date:** プロンプトで「?L」と入力し、エラーが発生したすべての日付のリストを表示します。
3. 次に同じプロンプトで、それらの日付のうちの 1 つを (mm/dd/yyyy の形式で) 入力します。年を省略した場合、現在の年と見なされます。これで、ルーチンにより、日付とその日付で記録されているエラーの数が返されます。または、以下の構文を使用して、このプロンプトからエラーのリストを取得できます。
 - ・ 「?L」を入力すると、エラーが生じた日付すべてのリストが、日付の新しい順に、記録済みエラーの数と共に表示されます。(T) 列は、何日前かを表します。(T) = 今日で、(T-7) = 7 日前です。対象の日のすべてのエラーにユーザ・コメントが定義されている場合、そのコメントが角かっこに囲まれて表示されます。リストの後に、**For Date:** プロンプトが再表示されます。日付または「T-n」を入力できます。
 - ・ 「[text]」を入力すると、部分文字列 text を含むすべてのエラーが一覧表示されます。「<text」を入力すると、エラー名コンポーネントに部分文字列 text を含むすべてのエラーが一覧表示されます。「^text」を入力すると、エラー位置コンポーネントに部分文字列 text を含むすべてのエラーが一覧表示されます。リストの後に、**For Date:** プロンプトが再表示されます。日付を入力します。
4. **Error:** プロンプトには、確認したいエラーの整数番号を入力します。対象の日の最初のエラーの場合は「1」、2 つ目のエラーの場合は「2」というように入力します。または、疑問符 (?) を入力すると、取得可能な応答のリストが表示されます。このユーティリティは、エラーに関する情報として、エラー名、エラー位置、時間、システム変数値、およびエラー発生時に実行されていたコード行を表示します。
Error: プロンプトに「*」を入力すると、コメントを表示できます。「*」を入力すると、対象の日のエラーすべてに適用されている現在のユーザ定義のコメントが表示されます。次に、そのすべてのエラーに適用されている既存のコメントに置き換わる新しいコメントを入力するよう求められます。
5. **Variable:** プロンプトには、変数に関する情報を取得するためのさまざまなオプションを指定できます。ローカル変数の名前 (添え字なしまたは添え字付き) を指定した場合、^%ERN は、スタック・レベルとその変数の値 (定義されている場合)、およびそのすべての下位ノードを返します。グローバル変数、プロセス・プライベート変数、または特殊システム変数を指定することはできません。
 「?」を入力すると、**Variable:** プロンプトで指定できるその他の構文オプションが一覧表示されます。
 - ・ ***A : Variable:** プロンプトに指定した場合、**Device:** プロンプトが表示されます。**Return** キーを押すと結果が表示されます。
 - ・ ***V : Variable:** プロンプトに指定した場合、**Variable(s):** プロンプトが表示されます。このプロンプトには、添え字なしのローカル変数を 1 つ、または添え字なしのローカル変数のコンマ区切りのリストを指定します。添え字付きの変数は拒否されます。すると、^%ERN は **Device:** プロンプトを表示します。**Return** キーを押すと結果が表示されます。^%ERN は、指定された各変数の値 (定義されている場合) とそのすべての下位ノードを返します。

- ・ `*L:Variable:` プロンプトに指定した場合、変数を現在のパーティションにロードします。すべてのプライベート変数を（パブリックとして）ロードしてから、ロードされたプライベート変数と競合しないすべてのパブリック変数をロードします。

18.4 関連項目

- ・ `%SYS.ProcessQuery.ExamStackByPid()` メソッド（`%ERN` により使用される `^mtemp` グローバルの詳細を提供します）

19

コマンド行ルーチンのデバッグ

ここでは、ObjectScript コードのテストとデバッグに使用する手法について説明します。InterSystems IRIS® データ・プラットフォームには、コードをデバッグする 2 つの方法があります。

- ・ ルーチン・コードで **BREAK** コマンドを使用して実行を中断し、実行内容を検証します。
- ・ **ZBREAK** コマンドで ObjectScript デバッガを呼び出して実行を中断し、コードと変数の両方を検証できます。

InterSystems IRIS には、ここで説明するように、ルーチンを中断し、完全なデバッグ機能をサポートするシェルに入る機能が含まれています。InterSystems IRIS には**保護されたデバッグ・シェル**も含まれています。これには、割り当てられた特権の超越や迂回をユーザができないという利点があります。

19.1 保護されたデバッグ・シェル

保護されたデバッグ・シェルは、機密データへのアクセスの制御に関して優れています。ユーザは、環境によって、変数のステップングや表示などの基本的なデバッグを実行できますが、ルーチンの実行パスや結果を変更するいかなる操作も許可されません。これによって、操作、悪意のあるロール・エスカレーション、より高い権限で実行するコードの挿入などの問題を引き起こす可能性のあるアクセスを防止できます。

既定では、デバッグ・プロンプトのユーザは、現在のレベルの特権を保持します。デバッグ・プロンプトの保護されたシェルを有効にし、このことによってユーザが発行できるコマンドを制限するには、そのユーザに対して**保護されたデバッグ・シェルを有効にする**必要があります。

現在のユーザに対して有効にした場合、保護されたデバッグ・シェルは、BREAK コマンドが実行された時点、**ブレークポイントまたはウォッチポイント**が検出された時点、または捕えられていないエラーが発生した時点で開始されます。

保護されたデバッグ・シェル内では、ユーザは以下のものを実行できません。

- ・ 変数を変更する可能性があるコマンド。
- ・ 変数を変更する可能性がある関数。
- ・ 他のルーチンを呼び出す可能性があるコマンド。
- ・ ルーチンのフローまたは環境に影響を及ぼすコマンド。

保護されたデバッグ・シェル内で、制限されたコマンドまたは関数をユーザが実行しようすると、InterSystems IRIS はそれぞれ <COMMAND> エラーまたは <FUNCTION> エラーをスローします。

19.1.1 制限されるコマンドと関数

この節では、保護されたデバッグ・シェル内で制限されるアクティビティのリストを示します。

- ・ [制限される ObjectScript コマンド](#)
- ・ [制限される ObjectScript 関数](#)
- ・ [制限されるオブジェクト構成](#)

19.1.1.1 制限される ObjectScript コマンド

保護されたデバッグ・シェルで制限される ObjectScript コマンドは以下のとおりです。

- ・ CLOSE
- ・ DO
- ・ FOR
- ・ 引数付きの GOTO
- ・ KILL
- ・ LOCK
- ・ MERGE
- ・ OPEN
- ・ QUIT
- ・ READ
- ・ RETURN
- ・ SET
- ・ TCOMMIT
- ・ TROLLBACK
- ・ TSTART
- ・ VIEW
- ・ XECUTE
- ・ ZINSERT
- ・ ZKILL
- ・ ZREMOVE
- ・ ZSAVE
- ・ ZW と ZZDUMP 以外のユーザ・コマンド

19.1.1.2 制限される ObjectScript 関数

保護されたデバッグ・シェルで制限される ObjectScript 関数は以下のとおりです。

- ・ \$CLASSMETHOD
- ・ \$COMPILE

- ・ \$DATA(,var) – 引数が 2 つのもののみ
- ・ \$INCREMENT
- ・ \$METHOD
- ・ \$ORDER(,var) – 引数が 3 つのもののみ
- ・ \$PROPERTY
- ・ \$QUERY(,var) – 引数が 3 つのもののみ
- ・ \$EXECUTE
- ・ \$ZF
- ・ \$ZSEEK
- ・ 任意の外部関数

19.1.1.3 制限されるオブジェクト構成

いかなるメソッドまたはプロパティの参照も許可されません。プロパティ参照は、propertyGet メソッドを起動する可能性があるため制限されます。制限されるオブジェクト・メソッドおよびプロパティ構文の構成の例の一部を以下に示します。

- ・ #class(classname).ClassMethod()
- ・ oref.Method()
- ・ **oref.Property**
- ・ \$SYSTEM.Class.Method()
- ・ ..Method()
- ・ ..Property

注釈 参照によって変数の受け渡しを行わない場合でも、メソッドによってパブリック変数を変更される可能性があります。プロパティ参照は propGet メソッドを起動する可能性があるため、いかなるプロパティ・アクセスも許可されません。

19.2 ObjectScript デバッガによるデバッグ

ObjectScript デバッガを使用すると、ルーチンのコードにデバッグ・コマンドを直接挿入し、ルーチンをテストできます。その後、コードの実行時にコマンドを発行し、アプリケーションの実行状況と処理フローをテストできます。以下は主な機能です。

- ・ ZBREAK コマンドを使用してコードにブレークポイントを設定し、その場所に達したときに特定の動作を実行します。
- ・ ローカル変数にウォッチポイントを設定し、その変数の値が変更されたときに特定の動作を実行します。
- ・ ブレークポイントとウォッチポイントの実行中は、別々のウィンドウで InterSystems IRIS とやり取りします。
- ・ 実行パスが変更されたときは常に実行をトレースし、その記録を（ターミナルあるいは別のデバイスに）出力します。
- ・ 実行スタックを表示します。
- ・ デバイスでアプリケーションを実行中、デバッグの入出力はセカンド・デバイスで行われます。これにより、InterSystems IRIS アプリケーションを全画面で実行しながら、アプリケーションのターミナル入出力を中断せずにデバッグできます。

19.2.1 ブレークポイントとウォッチポイントの使用法

ObjectScript デバッガには、プログラムの実行を中断する 2 つの方法があります。

- ・ ブレークポイント
- ・ ウォッチポイント

ブレークポイントは、ZBREAK コマンドで指定する InterSystems IRIS ルーチン内の位置です。実行中のルーチンが指定行に達すると、InterSystems IRIS はルーチンの実行を中断します。オプションとして、定義したデバッグを実行する場合もあります。最大 20 個のルーチンまでブレークポイントを設定できます。また、1 つのルーチン内に最大 20 個までのブレークポイントを設定できます。

ウォッチポイントは、ZBREAK コマンドで識別する変数です。SET コマンドあるいは KILL コマンドで値を変更すると、ルーチンの実行または ZBREAK コマンド内で定義したデバッグの実行、あるいはその両方を中断できます。システム変数にウォッチポイントを指定することはできません。

定義したブレークポイントとウォッチポイントは、1 つのセッション内でのみ維持されます。したがって、ブレークポイントとウォッチポイントの定義を 1 つのルーチンあるいは XECUTE コマンド文字列に格納すると、セッション間でそれらを簡単に元に戻せるため便利です。

19.2.2 ブレークポイントとウォッチポイントの設定

ZBREAK コマンドを使用して、ブレークポイントとウォッチポイントを設定します。

19.2.2.1 構文

```
ZBREAK location[:action:condition:execute_code]
```

以下を示します。

引数	説明
location	必須項目。(ブレークポイントを設定する)コード位置、(ウォッチポイントを設定する)ローカル変数あるいはシステム変数を指定します。指定した位置にブレークポイント/ウォッチポイントが既に定義されている場合、新規の設定により古い設定が置き換えられます。システム変数にウォッチポイントを指定することはできません。
action	オプション - ブレークポイント/ウォッチポイントがトリガされた(引き起こされた)ときの動作を指定します。ブレークポイントの場合、動作はコード行が実行される前に発生します。またウォッチポイントの場合、動作はローカル変数を変更するコマンドの実行後に発生します。動作は、大文字あるいは小文字で指定できますが、引用符で囲む必要があります。
condition	オプション - 中括弧または引用符で囲んだブーリアン式。ブレークポイントまたはウォッチポイントのトリガ時に評価されます。 <ul style="list-style-type: none"> condition が True (1) の場合、action が実行されます。 condition が False の場合、action は実行されず、execute_code 内のコードも実行されません。 condition が未指定の場合、既定では True です。
execute_code	オプション - condition が True の場合に実行される ObjectScript コードを指定します。コードがリテラルである場合、中括弧または引用符でそのコードを囲む必要があります。このコードは、動作が実行される前に実行されます。特殊なシステム変数 \$TEST の値は、コードを実行する前に保存されます。コードの実行後、デバッグされたプログラムに存在する \$TEST の値は復元されます。

注釈 疑問符 (?) を付けて ZBREAK を使用すると、ヘルプが表示されます。

19.2.2.2 コード位置でのブレークポイントの設定

\$TEXT 関数への呼び出しで使用するルーチン行参照として、コード位置を指定します。実行がコードの指定した位置に到達した場合は常に、コード行が実行される前にブレークポイントが発生します。ルーチン名を指定しない場合、InterSystems IRIS は、現在のルーチンを参照しているものと見なします。

19.2.2.3 ブレークポイント実行コード内の引数なしの GOTO

引数なしの GOTO は、ブレークポイントの実行コードで使用できます。これは、デバッガの Break プロンプトで引数なしの GOTO を実行するのと同じ効果があるため、次のブレークポイントまで引き続き実行されます。

例えば、テスト中のルーチンが現在のネームスペースにある場合、位置の値は以下のようになります。

値	ブレーク位置
label^rou	ルーチン rou にある行ラベル label の行の前で中断
label+3^rou	ルーチン rou にある行ラベル label から 3行目の前で中断
+3^rou	ルーチン rou の 3 行目で中断

テスト中のルーチンが現在メモリにロードされている場合 (つまり、暗黙あるいは明示的に ZLOAD が実行された場合)、以下のような位置の値を使用できます。

値	ブレーク位置
label	label の行ラベルの前で中断
label+3	label から 3 行目の前で中断
+3	3 行目の前で中断

19.2.2.4 ローカル変数名とシステム変数名でのウォッチポイントの設定

以下の状況で、ローカル変数名を使用してウォッチポイントが発生させることができます。

- ・ ローカル変数を生成したとき
- ・ SET コマンドがローカル変数値を変更したとき
- ・ KILL コマンドがローカル変数を削除したとき

変数名は、*a のようにアスタリスクが前に付きます。

配列変数名を指定すると、ObjectScript デバッガはすべての下位ノードを監視します。例えば、配列 a にウォッチポイントを設定し、これを a(5) あるいは a(5,1) に変更すると、ウォッチポイントがトリガされます。

ウォッチポイントの設定時、既存の変数は必要ありません。

以下の特殊なシステム変数も使用できます。

システム変数	トリガ・イベント
\$ZERROR	エラー発生時は常に、エラー・トラップの呼び出し前にトリガされます。
\$ZTRAP	エラー・トラップが設定、あるいはクリアされる際は常にトリガされます。
\$IO	明示的に SET コマンドを実行するとトリガされます。

19.2.2.5 Action 引数の値

以下のテーブルは、ZBREAK の action 引数に使用できる値の説明です。

引数	説明
"B"	既定値です。"T" 動作を含む場合を除き、ZBREAK *a:"TB" のように明示的に "B" 動作を組み込み、ブレークを起こす必要があります。実行を中断し、行の位置を示すキャレット(^)と共に、ブレークが起こった行を表示します。その後、ターミナル・プロンプトが表示され、やり取りが可能となります。引数なしの GOTO コマンドで実行を再開します。
"L"	GOTO コマンドがシングル・ステップの実行を開始し、各行の最初で停止する以外は、"B" と同じです。DO コマンド、ユーザ定義関数、XECUTE コマンドに遭遇した場合、シングル・ステップ・モードは、そのコマンドあるいは関数が完了するまで一時中断します。
"L+"	GOTO コマンドがシングル・ステップの実行を開始し、各行の最初で停止する以外は、"B" と同じです。DO コマンド、ユーザ定義関数、XECUTE コマンドは、シングル・ステップ・モードを中断しません。
"S"	GOTO コマンドがシングル・ステップの実行を開始し、各コマンドの最初で停止する以外は、"B" と同じです。DO コマンド、ユーザ定義関数、FOR コマンド、XECUTE コマンドに遭遇した場合、シングル・ステップ・モードは、そのコマンドあるいは関数が完了するまで一時中断します。

引数	説明
"S+"	GOTO コマンドがシングル・ステップの実行を開始し、各コマンドの最初で停止する以外は、"B"と同じです。DO コマンド、ユーザ定義関数、FOR コマンド、XECUTE コマンドは、シングル・ステップ・モードを中断しません。
"T"	他の引数と共に使用できます。トレース・デバイスにトレース・メッセージを出力します。この引数は、後述の ZBREAK /TRACE:ON コマンドでトレースを ON に設定した後にのみ機能します。トレース・デバイスは、ZBREAK /TRACE コマンドで定義しない限り、主デバイスになります。この引数をブレークポイントで使用すると、TRACE: ZBREAK at label2^rou2 というメッセージが表示されます。この引数をウォッチポイントで使用すると、監視中の変数名と実行中のコマンド名を指定するトレース・メッセージが表示されます。例えば次のコードの場合、変数 a は監視され、ルーチン test の行 test+1 で変更されました。TRACE: ZBREAK SET a=2 at test+1^test。"T" 動作を組み込む場合、ZBREAK *a:"TB" に "B" 動作も明示的に組み込む必要があります。これにより、実際のブレークが発生します。
"N"	このブレークポイント/ウォッチポイントでは、何も動作しません。condition 式は常に評価され、execute_code が実行されているかどうかを判定します。

19.2.2.6 ZBREAK の例

以下の例では、ローカル変数 a が削除されると常に実行を中断するウォッチポイントを設定します。動作は何も指定されていないため、"B" が指定されているものとします。

```
ZBREAK *a::"$DATA(a)"
```

以下の例は、(ルーチン内から発行されるコマンドではなく) ダイレクト・モードの ObjectScript コマンドに対して動作する、上記のウォッチポイントを示しています。キャレット (^) は、実行を中断させるコマンドを示します。

Terminal

```
USER>KILL a
KILL a
^
<BREAK>
USER ls0>
```

以下の例は、実行を中断し、label2^rou 行の先頭にシングル・ステップ・モードを設定するブレークポイントです。

```
ZBREAK label2^rou:"L"
```

以下の例は、ルーチン実行時にどのようにブレークが実行されるかを示しています。キャレット (^) は、実行が中断した位置を示します。

Terminal

```
USER>DO ^rou
label2 SET x=1
^
<BREAK>label2^rou
USER 2d0>
```

以下の例では、"N" 動作のため、行 label3^rou のブレークポイントは実行を中断しません。しかし、行 label3^rou に達した際に x<1 の場合、flag を x に設定します。

```
ZBREAK label3^rou:"N":"x<1":"SET flag=x"
```

以下の例は、a の値が変更されるたびに、^GLO のコードを実行するウォッチポイントを設定します。二重コロンは、condition 引数がないことを示しています。

```
ZBREAK *a:"N"::"XECUTE ^GLO"
```

以下の例は、b の値が変更されるたびに、トレース・メッセージを表示するウォッチポイントを設定します。トレース・メッセージは、トレース・モードが ZBREAK /TRACE:ON コマンドで ON になっている場合にのみ表示されます。

```
ZBREAK *b:"T"
```

以下の例は、変数 a が 5 に設定されると、シングル・ステップ・モードで実行を中断するウォッチポイントを設定します。

```
ZBREAK *a:"S":"a=5"
```

以下の例は、ブレークの発生時、キャレット記号 (^) は、変数 a を 5 に設定するコマンドを指定します。

Terminal

```
USER>DO ^test
FOR i=1:1:6 SET a=a+1
^
<BREAK>
test+3^test
USER 3f0>WRITE a
5
```

19.2.3 ブレークポイントとウォッチポイントを無効にする

以下のいずれかを無効にできます。

- ・ 特定のブレークポイントとウォッチポイント
- ・ すべてのブレークポイントあるいはウォッチポイント

19.2.3.1 特定のブレークポイントとウォッチポイントを無効にする

マイナス記号をブレークポイントあるいはウォッチポイントの位置指定の前に置くと、ブレークポイントあるいはウォッチポイントを無効にできます。以下のコマンドは、位置 label2^rou で事前に指定されたブレークポイントを無効にします。

```
ZBREAK -label2^rou
```

無効になったブレークポイントは“消失”しますが、その定義は維持されます。したがって、プラス記号をその前に置くと、無効にされたブレークポイントを有効にできます。以下のコマンドは、以前無効にされたブレークポイントを有効にします。

```
ZBREAK +label2^rou
```

19.2.3.2 すべてのブレークポイントとウォッチポイントを無効にする

位置を指定せずにプラス記号あるいはマイナス記号を使用して、すべてのブレークポイントまたはウォッチポイントを無効にできます。

記号	説明
ZBREAK -	定義済みのすべてのブレークポイントとウォッチポイントを無効にします。
ZBREAK +	定義済みのすべてのブレークポイントとウォッチポイントを有効にします。

19.2.4 ブレークポイントとウォッチポイントの実行を遅らせる

指定した数の反復の間、ブレークポイント/ウォッチポイントの実行を遅らせることもできます。実行する度ではなく定期的にブレークさせるように、ループ内にコード行を記述している場合もあります。このためには、通常どおりにブレークポイントを設定した後、位置の引数に続くカウントを無効にします。

以下の ZBREAK コマンドは、100 回反復される間、label2^rou のブレークポイントを無効にします。101 回目にこの行が実行されると、指定したブレークポイントの動作が発生します。

```
ZBREAK label2^rou      ; establish the breakpoint
ZBREAK -label2^rou#100 ; disable it for 100 iterations
```

重要 行が繰り返し実行されても、この行には FOR コマンドが含まれるため、遅らせたブレークポイントはデクリメントされません。

19.2.5 ブレークポイントとウォッチポイントの削除

次のように、ブレークポイント/ウォッチポイントの位置の前にマイナス記号を 2 つ付けて、ブレークポイントまたはウォッチポイントを個別に削除できます。

```
ZBREAK --label2^rou
```

ブレークポイント/ウォッチポイントを削除した後は、それを再度定義することにより削除したポイントをリセットできます。

すべてのブレークポイントを削除するには、以下のコマンドを発行します。

```
ZBREAK /CLEAR
```

このコマンドは、InterSystems IRIS プロセスが停止すると、自動的に実行されます。

19.2.6 シングル・ステップ・ブレークポイントの動作

シングル・ステップ実行を使用して、コードの各行あるいは各コマンドの最初で実行を停止できます。シングル・ステップ・ブレークポイントを設定し、各ステップで実行される動作と実行コードを指定できます。以下の構文を使用して、シングル・ステップ・ブレークポイントを定義します。

```
ZBREAK $:action[:condition:execute_code]
```

他のブレークポイントと異なり、ZBREAK \$ はブレークを引き起こしません。ブレークは、シングル・ステップとして自動的に発生するからです。ZBREAK \$ を使用すると、ルーチンのステップ実行時にデバッガがブレークする各ポイントで、動作を指定しコードを実行できます。これは、行あるいはコマンドの実行をトレースするのに特に役立ちます。例えば以下のコードは、アプリケーション ^TEST で実行された行をトレースします。

Terminal

```
USER>ZBREAK /TRACE:ON
USER>BREAK "L+"
USER>ZBREAK $:"T"
```

"T" 単体で（つまり、その他のアクション・コードなしで）指定すると、通常自動的に発生するシングル・ステップ・ブレークが抑制されます（また、他のアクション・コードを使用するかどうかにかかわらず、"N" アクション・コードを指定すれば、シングル・ステップ・ブレークが抑制されます）。

トレースとブレークの両方を発生させるには、以下の方法でシングル・ステップ・ブレークポイントの定義を設定します。

Terminal

USER>ZBREAK \$: "TB"

19.2.7 実行のトレース

以下の ZBREAK 形式を使用して、ZBREAK コマンドの "T" 動作を有効にするかどうかを制御できます。

ZBREAK /TRACE:state[:device]

state には、以下を指定できます。

状態	説明
ON	トレース可能
OFF	トレース不可能
ALL	次のコードと同等なコードを実行する場合、アプリケーションのトレース可能: ZBREAK /TRACE:ON[:device] BREAK "L+" ZBREAK \$:"T"

device で ALL あるいは ON 状態キーワードを使用する場合、トレース・メッセージは、主デバイスではなく、指定されたデバイスに転送されます。デバイスを開いていない場合、InterSystems IRIS は、WRITE オプションと APPEND オプションが付いたシーケンシャル・ファイルとしてそのデバイスを開こうとします。

OFF 状態キーワードを使用してデバイスを指定すると、現在ファイルが開いている場合に InterSystems IRIS はそのファイルを閉じます。

注釈 ZBREAK /TRACE:OFF により、ZBREAK /TRACE:ALL で設定したシングル・ステップ・ブレークポイントの定義が削除、または無効にされることはありません。また、ZBREAK /TRACE:ALL で設定した L+ シングル・ステップが消去されることもありません。また、ZBREAK --\$ と BREAK "C" コマンドを発行し、シングル・ステップを削除する必要があります。あるいはその代わりに、BREAK "OFF" シングル・コマンドを使用して、プロセスのデバッグをすべて無効にできます。

トレース・メッセージは、T 動作に対応するブレークポイントで生成されます。例外として、トレース・メッセージの形式は、すべてのブレークポイントで以下ようになります。

Trace: line_reference での ZBREAK

line_reference は、参照するブレークポイントの行です。

コマンドでステップを実行する場合、トレース・メッセージの形式は、シングル・ステップのブレークポイントと若干異なります。

Trace: line_reference source_offset での ZBREAK

line_reference は参照するブレークポイントの行で、source_offset は、ブレークが発生したソース行位置への 0 ベースのオフセットです。

オペレーティング・システムの注釈

- Windows – 別のデバイスへのトレース・メッセージは、COM1: など COM ポートに接続されているターミナル・デバイスに対して、Windows プラットフォームでサポートされています。コンソールあるいはターミナル・ウィンドウは使用できません。トレース・デバイスへのシーケンシャル・ファイルを指定できます。
- UNIX® – 以下の方法で、UNIX® プラットフォームで別のデバイスにトレース・メッセージを送信します。
 1. /dev/tty01 にログインします。

2. 次の tty コマンドを入力し、デバイス名を検証します。

```
$ tty
/dev/tty01
```

3. 以下のコマンドを発行し、デバイス競合を避けます。

```
$ exec sleep 50000
```

4. 作業ウィンドウに戻ります。
5. InterSystems IRIS を開始します。
6. 以下のトレース・コマンドを発行します。

```
ZBREAK /T:ON: "/dev/tty01"
```

7. プログラムを実行します。

T 動作を実行するブレイクポイントあるいはウォッチポイントを設定すると、/dev/tty01 に接続されたウィンドウに、トレース・メッセージが表示されます。

19.2.7.1 トレース・メッセージ形式

ブレイクポイントをコードに設定すると、以下のメッセージが表示されます。

```
Trace: ZBREAK at label2^rou2
```

ウォッチポイント変数を設定すると、以下のメッセージが表示されます。

```
Trace: ZBREAK SET var=val at label2^rou2
Trace: ZBREAK SET var=Array Val at label2^rou2
Trace: ZBREAK KILL var at label2^rou2
```

- ・ var は、監視される変数です。
- ・ val は、その変数に設定された新しい値です。

NEW コマンドを発行すると、トレース・メッセージは受け取りません。しかし、NEW レベルの変数で次に SET あるいは KILL を発行すると、変数のトレースが実行されます。変数が[ルーチンに参照渡しされる](#)と、その名前が変更された場合も、変数は引き続き追跡されます。

19.2.8 割り込みキーと Break コマンド

通常、割り込みキー（一般的に **CTRL-C**）を続けて押すと、トラップ可能な (<INTERRUPT>) エラーが生成されます。<INTERRUPT> エラーではなく、ブレイクを発生させる割り込み処理を設定するには、ZBREAK コマンド **ZBREAK** /**INTERRUPT:Break** を使用します。

これにより、デバイスのアプリケーション・レベルで割り込みが無効の場合も、割り込みキーを押したときにブレイクが発生します。

ターミナルから読み取り中に割り込みキーを押すと、ブレイク・モード・プロンプトを表示するため、**Enter** キーを押す必要があります。ブレイクではなくエラーを生成するよう割り込み処理をリセットするには、コマンド **ZBREAK** /**INTERRUPT:NORMAL** を発行します。

19.2.9 現在のデバッグ環境情報の表示

定義されたすべてのブレークポイントあるいはウォッチポイントを含め、現在のデバッグ環境の情報を表示するには、引数なしの ZBREAK コマンドを発行します。

引数なしの ZBREAK コマンドは、以下のデバッグ環境の状態を示します。

- ・ CTRL-C がブレークを生じさせるかどうか
- ・ “T” 動作で指定されたトレースの出力が、ZBREAK コマンドで表示されるかどうか
- ・ 有効/無効状態、動作、条件、実行可能コードを指定するフラグを持つ、すべての定義済みブレークポイントの位置
- ・ 有効/無効状態、動作、条件、実行可能コードを指定するフラグを持つウォッチポイントのすべての変数

このコマンドからの出力は、デバッグ・デバイスとして定義されたデバイスに表示されます。ZBREAK /DEBUG コマンドで異なるデバッグ・デバイスを定義しない限り、ユーザの主デバイスに出力されます。デバイスの定義は、“[デバッグ・デバイスの使用法](#)” のセクションで説明しています。

以下のテーブルは、各ブレークポイントとウォッチポイントに提供されるフラグの説明です。

表示セクション	意味
ブレークポイント/ウォッチポイントの識別	ブレークポイントのルーチンの行ウォッチポイントのローカルの変数
F:	ZBREAK コマンドで定義された動作タイプの情報を提供するフラグです。
S:	ZBREAK コマンドで定義されたブレークポイント/ウォッチポイントの実行を遅らせる回数です。
C:	ZBREAK コマンドで設定された Condition 引数です。
E:	ZBREAK コマンドで設定された Execute_code 引数です。

以下のテーブルは、ブレークポイント/ウォッチポイントの表示で F: 値をどのように解釈するかを示しています。F: 値は、最初の列にある適切な値のリストです。

値	意味
E	有効なブレークポイント/ウォッチポイント
D	無効なブレークポイント/ウォッチポイント
B	ブレークの実行
L	“L” の実行
L+	“L+” の実行
S	“S” の実行
S+	“S+” の実行
T	トレース・メッセージの出力

19.2.9.1 既定表示

最初に InterSystems IRIS を開いて ZB を使用するとき、出力は以下のようになります。

Terminal

```
USER>ZBREAK
BREAK:
No breakpoints
No watchpoints
```

これは、以下を示します。

- ・ トレースの実行は OFF です。
- ・ **CTRL-C** が押されると、ブレークは無効になります。
- ・ ブレークポイント/ウォッチポイントは未定義です。

19.2.9.2 既存のブレークポイントとウォッチポイントの表示

この例では、2 つのブレークポイントと 1 つのウォッチポイントの定義を示しています。

Terminal

```
USER>ZBREAK +3^test:::{WRITE "IN test"}
USER>ZBREAK -+3^test#5
USER>ZBREAK +5^test:"L"
USER>ZBREAK -+5^test
USER>ZBREAK *a:"T": "a=5"
USER>ZBREAK /TRACE:ON
USER>ZBREAK
BREAK: TRACE ON
+3^test F:EB S:5 C: E:"WRITE "IN test""
+5^test F:DL S:0 C: E:
a F:ET S:0 C:"a=5" E:
```

最初の 2 つの ZBREAK コマンドは、ブレークポイントの遅延を定義しています。その次の 2 つの ZBREAK は無効なブレークポイントを、5 番目の ZBREAK はウォッチポイントを定義しています。6 番目の ZBREAK コマンドはトレースの実行を有効にし、引数のない最後の ZBREAK コマンドは、現在のデバッグ設定情報を表示します。

例えば、ZBREAK は以下を表示します。

- ・ トレースは ON です。
- ・ **CTRL-C** が押されると、ブレークは無効になります。

その後出力は、2 つのブレークポイントと 1 つのウォッチポイントを示します。

- ・ 最初のブレークポイントの F フラグは EB、S フラグは 5 です。つまり、ブレークポイントは、行の 5 回目の実行で発生します。E フラグは、ブレーク用のターミナル・プロンプトが表示される前に実行される実行可能コードを表示します。
- ・ 2 番目のブレークポイントの F フラグは DL です。これは、ブレークポイントの無効を意味しますが、有効の場合ブレークが発生し、ブレークポイント位置の後ろの各コード行をシングル・ステップで実行します。
- ・ ウォッチポイントの F フラグは ET です。これは、ウォッチポイントが有効であることを意味します。トレース実行が ON の場合、トレース・メッセージがトレース・デバイスに表示されます。トレース・デバイスを定義していない場合、トレース・デバイスは主デバイスになります。
- ・ C フラグは、condition が True の場合にのみトレースが表示されることを意味します。

19.2.10 デバッグ・デバイスの使用法

デバッグ・デバイスは、以下のようなデバイスです。

- ・ ZBREAK コマンドは、デバッグ環境情報を表示します。

- ・ ターミナル・プロンプトは、ブレーク発生時に表示されます。

注釈 Windows プラットフォームでは、別のデバイスへのトレース・メッセージは、COM1: など COM ポートに接続されているターミナル・デバイスでのみサポートされています。

InterSystems IRIS を実行すると、デバッグ・デバイスが主デバイスとして自動的に設定されます。いつでも、デバッグ入出力は、コマンド ZBREAK /DEBUG:"device" で代替りのデバイスに送信できます。

注釈 また、オペレーティング・システム固有の動作も実行可能です。

UNIX® システムで tty01 デバイスをブレーク状態にするには、以下のコマンドを発行します。

```
ZBREAK /D: "/dev/tty01/"
```

CTRL-C あるいはブレークポイント、ウォッチポイントがトリガされてブレークが発生すると、デバイスに接続されたウィンドウに表示されます。そのウィンドウが、アクティブ・ウィンドウになります。

デバイスが開いていない場合、自動的に OPEN コマンドが実行されます。デバイスが指定した位置で既に関いている場合、あらゆる既存の OPEN パラメータが優先されます。

重要 指定したデバイスがインタラクティブでない（例えばターミナル）場合、ブレークから戻ることができません。しかし、システムでこの制約が強制されているわけではありません。

19.2.11 ObjectScript デバッガの例

最初に、以下に示す test という簡単なプログラムをデバッグします。目標は、変数 a に 1 を、変数 b に 2 を、変数 c に 3 を格納することです。

```
test; Assign the values 1, 2, and 3 to the variables a, b, and c
SET a=1
SET b=2
SET c=3 KILL a WRITE "in test, at end"
QUIT
```

しかし、test を実行すると、b と c のみが正しい値を取得します。

Terminal

```
USER>DO ^test
in test, at end
USER>WRITE
b=2
c=3
USER>
```

明らかにこのプログラムの問題は、変数 a が行 4 で KILL コマンドによって削除されることですが、この問題を特定するためにデバッガを使用する必要があると想定します。

ZBREAK コマンドを使用して、ルーチン test の各コード行 ("L" 動作) をシングル・ステップに設定できます。ステップを実行して a の値を書き込むことにより、4 行目に問題があることを判定します。

Terminal

```
USER>NEW
USER 1S1>ZBREAK
BREAK
No breakpoints
No watchpoints
USER 1S1>ZBREAK ^test:"L"
USER 1S1>DO ^test
SET a=1
^
<BREAK>test+1^test
```

```

USER 3d3>WRITE a
<UNDEFINED>^test
USER 3d3>GOTO
SET b=2
^
<BREAK>test+2^test
USER 3d3>WRITE a
1
USER 3d3>GOTO
SET c=3 KILL a WRITE "in test, at end"
^
<BREAK>test+3^test
USER 3d3>WRITE a
1
USER 3d3>GOTO
in test, at end
QUIT
^
<BREAK>test+4^test
USER 3d3>WRITE a
WRITE a
^
<UNDEFINED>^test
USER 3d3>GOTO
USER 1S1>

```

これでこの行を検証し、KILL a コマンドを発見できます。さらに複雑なコードでは、その行から (“S” 動作) コマンドでシングル・ステップを実行したい場合もあります。

問題が、DO、FOR、XECUTE コマンド、ユーザ定義関数で発生した場合、“L+” あるいは “S+” 動作を使用して、コードの下位レベルにある行あるいはコマンドからシングル・ステップを実行します。

19.2.12 ObjectScript デバッガ・エラーの理解

ObjectScript デバッガは、適切な InterSystems IRIS エラー・メッセージを持つ条件引数あるいは実行引数でエラーが発生すると、フラグを立てます。

エラーが `execute_code` 引数内に存在する場合、実行コードがエラー・メッセージの前に表示されると、実行コードは条件コードに囲まれます。条件特殊変数 (**\$TEST**) は、常に実行コードの最後で 1 に設定されるため、残りのデバッガ処理コードは適切に動作します。制御がルーチンに戻ると、ルーチン内の **\$TEST** 値はリストアされます。

サンプル・プログラム `test` に対し、以下の ZBREAK コマンドを発行するとします。

Terminal

```
USER>ZBREAK test+1^test:"B":"a=5":"WRITE b"
```

このプログラム `test` で変数 `b` は、行 `test+1` で定義されていないため、エラーが発生します。以下のようなエラーが表示されます。

```

IF a=5 XECUTE "WRITE b" IF 1
^
<UNDEFINED>test+1^test

```

`condition` を定義しなかった場合、人工的に `True` の条件を実行コードの前後に定義できます。以下はその例です。

Terminal

```
USER>IF 1 WRITE b IF 1
```

19.3 BREAK コマンドによるデバッグ

InterSystems IRIS の **BREAK** コマンドには、3 つの形式があります。

- ・ ルーチン・コードに挿入された引数なしの BREAK により、その場所においてブレークポイントが設定されます。コード実行中にこのブレークポイントと遭遇すると、実行が中断されて、ターミナル・プロンプトに戻ります。
- ・ 文字列引数付きの BREAK では、行ごとまたはコマンドごとでのコードのステップ実行を可能にするブレークポイントを設定/削除します。
- ・ 整数の引数付きの BREAK コマンドでは、CTRL-C によるユーザ割り込みを有効/無効化します。(詳細は“BREAK”コマンドを参照してください。)

19.3.1 ルーチンの実行を中断する引数なし BREAK の使用法

実行中のルーチンを中断し、プロセスをターミナル・プロンプトに戻すには、ルーチンで実行を一時的に中断させたい位置に、引数なしの BREAK を入力します。

InterSystems IRIS が BREAK に遭遇すると、以下のステップが実行されます。

1. 実行中のルーチンを中断します。
2. プロセスをターミナル・プロンプトに戻します。主デバイスの入出力リダイレクトを使用するアプリケーションをデバッグすると、リダイレクトはデバッグ・プロンプトでオフになり、デバッグ・コマンドからの出力がターミナルに表示されます。

これにより、ObjectScript コマンドを発行し、データを修正できます。また、エラーまたは追加の BREAK があっても、他のルーチンやサブルーチンを実行できます。デバッグ・ターミナル・プロンプトから ObjectScript コマンドを発行すると、このコマンドは直ちに実行されます。実行中のルーチンには挿入されません。このコマンド実行は、通常のターミナル・プロンプトの動作と同じです。1 つ違う点は、タブ文字によって続行されたコマンドがデバッグ・ターミナル・プロンプトから実行されることです。タブ文字によって続行されたコマンドは、通常のターミナル・プロンプトからは実行されません。

ルーチンが中断された位置で実行を再開するには、引数なしの GOTO コマンドを発行します。

コード内の引数なしの BREAK コマンドに後置条件を指定すると、ルーチンを変更しなくても、単に後置条件変数を設定するだけで同じコードを再実行できるので便利です。例えば、ルーチンに以下の行があるとします。

```
CHECK BREAK:$DATA(debug)
```

変数 debug を設定すると、ルーチンを中断し、ジョブをターミナル・プロンプトに戻すことができます。また、変数 debug をクリアすると、ルーチンの実行を継続できます。

詳細は、“[コマンド後置条件式](#)”を参照してください。

19.3.2 ルーチンの実行を中断する引数付き BREAK の使用法

中断したいルーチンのすべての位置に、引数なし BREAK コマンドを置く必要はありません。InterSystems IRIS では、コードのステップ実行を可能にする引数オプションがいくつか用意されています。コードのステップ実行は、シングル・ステップ (BREAK “S”) またはコマンド行 (BREAK “L”) を使用して行えます。これらの文字コード引数のすべてのリストについては、[BREAK](#) コマンドを参照してください。

BREAK “S” と BREAK “L” の一つの違いは、多数のコマンド行が複数ステップで構成されることにあります。このことは必ずしも明白になっているわけではありません。例えば、以下はすべて単一行 (および単一 ObjectScript コマンド) ですが、それぞれが 2 ステップで解析されます :SET x=1,y=2, KILL x,y, WRITE “hello”,!, IF x=1,y=2。

BREAK “S” と BREAK “L” は、ラベル行、コメント、および TRY 文を無視します (ただし、両方ともに TRY ブロックの閉じ中括弧でブレークします)。BREAK “S” は CATCH 文でブレークしますが (CATCH ブロックが入力された場合)、BREAK “L” はブレークしません。

BREAK でプロセスをターミナル・プロンプトに戻した場合、ブレークの状態はスタックされません。したがって、ブレークの状態を変更した場合、引数なしの GOTO を発行して実行中のルーチンに戻ったときに、その新規の状態が有効なままになります。

DO、XECUTE、FOR、またはユーザ定義関数が実行されるたびに、InterSystems IRIS はブレークの状態をスタックします。BREAK "C" を選択してブレークを OFF にすると、システムは DO、XECUTE、FOR、またはユーザ定義関数の最後にブレークの状態をリストアします。そうしない場合、InterSystems IRIS はスタックされた状態を無視します。

このように、下位のサブルーチン・レベルでブレークを有効にすると、ルーチンが上位のサブルーチン・レベルに戻った後、ブレークが継続されます。一方、上位レベルで有効だったブレークを下位サブルーチン・レベルで無効にした場合、上位レベルに戻ると、ブレークが再開されます。BREAK "C-" を使用して、すべてのレベルでブレークを無効にすることができます。

BREAK "L+" または BREAK "S+" を使用して、DO、XECUTE、FOR、またはユーザ定義関数内でブレークを有効にすることができます。

BREAK "L-" を使用すると、現在のレベルでのブレークが無効になりますが、前のレベルでの改行が有効になります。BREAK "S-" を使用すると、現在のレベルでのブレークが無効になりますが、前のレベルでのシングル・ステップ・ブレークが有効になります。

19.3.2.1 デバッグの停止

プロセスで作成されたすべてのデバッグを削除するには、BREAK "OFF" コマンドを使用します。このコマンドは、すべてのブレークポイントとウォッチポイントを削除し、すべてのプログラム・スタック・レベルでステップ実行を無効にします。また、デバッグとトレース・デバイスに関連するものも削除しますが、それらを閉じません。

BREAK "OFF" コマンドを実行することは、以下の一連のコマンドを発行することと同じです。

ObjectScript

```
ZBREAK /CLEAR
ZBREAK /TRACE:OFF
ZBREAK /DEBUG: ""
ZBREAK /ERRORTRAP:ON
BREAK "C-"
```

19.3.3 ターミナル・プロンプトのプログラム・スタック情報表示

BREAK コマンドがルーチンの実行を中断したとき、あるいはエラーが発生したとき、プログラム・スタックは、スタックされた情報の一部を維持します。このような状態になった場合、この情報の概要がターミナル・プロンプトの一部として表示されます (namespace>)。例えば、この情報は、USER 5d3> の形式で表示される場合があります。各要素は以下のとおりです。

文字	説明
5	5つのスタック・レベルがあることを示します。スタック・レベルは DO、FOR、XECUTE、NEW、ユーザ定義関数の呼び出し、エラー状態、またはブレーク状態が原因となる可能性があります。
d	最後にスタックされた項目が DO であることを示します。
3	3つの NEW 状態、パラメータ渡し、またはユーザ定義関数がスタックにあることを示します。NEW コマンド、パラメータ渡し、ユーザ定義関数がスタックされていない場合、この値は 0 です。

ターミナル・プロンプトの文字コードを以下のテーブルに示します。

テーブル 19-1: ターミナル・プロンプトのスタック・エラー・コード

プロンプト	定義
d	DO
e	ユーザ定義関数
f	FOR ループ
x	XECUTE
B	BREAK 状態
E	エラー状態
N	NEW 状態
S	サインオン状態

以下の例では、スタック・フレーム追加時において、コマンド行の文が結果のターミナル・プロンプトを伴って表示されています。

Terminal

```
USER>NEW
USER 1S1>NEW
USER 2N1>XECUTE "NEW WRITE 123 BREAK"
<BREAK>
USER 4x1>NEW
USER 5B1>BREAK
<BREAK>
USER 6N2>
```

QUIT 1 を使用して、プログラム・スタックを戻すことができます。以下は、スタックの巻き戻し時におけるターミナル・プロンプトの例です。

Terminal

```
USER 6f0>QUIT 1 /* an error occurred in a FOR loop. */
USER 5x0>QUIT 1 /* the FOR loop was in code invoked by XECUTE. */
USER 4f0>QUIT 1 /* the XECUTE was in a FOR loop. */
USER 3f0>QUIT 1 /* that FOR loop was nested inside another FOR loop. */
USER 2d0>QUIT 1 /* the DO command was used to execute the program. */
USER 1S0>QUIT 1 /* sign on state. */
USER>
```

19.3.4 FOR ループおよび WHILE ループ

FOR または WHILE のいずれかを使用して、同じ操作を実行できます。つまり、イベント (通常はカウンタのインクリメント) によって実行がループを抜けるまでループすることができます。ただし、どちらのループ構造を使用するかにより、コード・モジュールに対するシングル・ステップ (BREAK "S+" または BREAK "L+") デバッグの実行に影響があります。

FOR ループでは、スタックに新しいレベルがプッシュされます。WHILE ループでは、スタック・レベルは変更されません。FOR ループをデバッグする場合、FOR ループ内からスタックをポップすると (BREAK "C" GOTO または QUIT 1 を使用)、FOR コマンド構文の終了直後から、このコマンドでのシングル・ステップ・デバッグを続行できます。WHILE ループをデバッグする場合は、BREAK "C" GOTO または QUIT 1 を使用して発行してもスタックはポップされません。したがって、WHILE コマンドの終了後にシングル・ステップ・デバッグは続行されません。残りのコードはブレークなしで実行されます。

19.3.5 BREAK あるいはエラー後の実行の再開

BREAK あるいはエラーの後にターミナル・プロンプトに戻ると、InterSystems IRIS は、BREAK あるいはエラーの原因となったコマンドの位置を記録します。その後、ターミナル・プロンプトで引数なしの GOTO を入力するだけで、次のコマンドで実行を再開できます。

Terminal

```
USER 4f0>GOTO
```

引数付きの GOTO を入力すると、以下のように、ブレークあるいはエラーが発生する同じルーチン内の別の行頭から実行を再開できます。

Terminal

```
USER 4f0>GOTO label3
```

また、異なるルーチンの行頭から実行を再開できます。

Terminal

```
USER 4f0>GOTO label3^rou
```

または、引数なしの QUIT コマンドで、プログラム・スタックをクリアすることもできます。

Terminal

```
USER 4f0>QUIT
USER>
```

19.3.5.1 サンプル・ダイアログ

以下はサンプルで使用されるルーチンです。

```
MAIN ; 03 Jan 2019 11:40 AM
SET x=1,y=6,z=8
DO ^SUB1 WRITE !,"sum=",sum
QUIT

SUB1 ; 03 Jan 2019 11:42 AM
SET sum=x+y+z
QUIT
```

BREAK "L" により、ルーチン SUB1 でブレークは発生しません。

Terminal

```
USER>BREAK "L"
USER>DO ^MAIN
SET x=1,y=6,z=8
^
<BREAK>MAIN+1^MAIN
USER 2d0>GOTO
DO ^SUB1 WRITE !,"sum=",sum
^
<BREAK>MAIN+2^MAIN
USER 2d0>GOTO
sum=15
QUIT
^
<BREAK>MAIN+3^MAIN
USER 2d0>GOTO
USER>
```

BREAK "L+" により、ルーチン SUB1 でブレークも発生します。

Terminal

```

USER>BREAK "L+"
USER>DO ^MAIN
SET x=1,y=6,z=8
^
<BREAK>MAIN+1^MAIN
USER 2d0>GOTO
DO ^SUB1 WRITE !,"sum=",sum
^
<BREAK>MAIN+2^MAIN
USER 2d0>GOTO
SET sum=x+y+z
^
<BREAK>SUB1+1^SUB1
USER 3d0>GOTO
QUIT
^
<BREAK>SUB1+2^SUB1
USER 3d0>GOTO
sum=15
QUIT
^
<BREAK>MAIN+3^MAIN
USER 2d0>GOTO
USER>

```

19.3.6 ターミナル・プロンプトで使用する NEW コマンド

引数なしの **NEW** コマンドは、事実上すべての記号をシンボル・テーブルに格納します。したがって、空のシンボル・テーブルを実行できます。エラーあるいは **BREAK** の発生後に、このコマンドが特に役立ちます。

シンボル・テーブルの妨げにならずに他のルーチンを実行するには、ターミナル・プロンプトで引数なしの **NEW** コマンドを発行します。その後、システムは以下を実行します。

- ・ プログラム・スタックに現在のフレームをスタックします。
- ・ 新規スタック・フレームのターミナル・プロンプトを返します。

例えば以下のようになります。

Terminal

```

USER 4d0>NEW
USER 5B1>DO ^%T
3:49 PM
USER 5B1>QUIT 1
USER 4d0>GOTO

```

5B1> プロンプトは、システムが **BREAK** から入った現在のフレームをスタックしたことを示します。1 は、**NEW** コマンドが変数情報をスタックしたことを示し、この変数情報は、**QUIT 1** を発行することで削除できます。実行を再開する場合は、**QUIT 1** を発行して古いシンボル・テーブルをリストアし、**GOTO** を発行して実行を再開します。

NEW コマンド、パラメータ渡し、ユーザ定義関数を使用するたびに、システムは情報をスタックに格納し、後で現在のサブルーチンまたは **XECUTE** レベルの明示的または暗黙的 **QUIT** により特定の編集を削除して他の値をリストアする必要があります。

NEW コマンド、パラメータ渡し、ユーザ定義関数が実行されたかどうか (その結果一部の変数がスタックされます)、また実行された場合、既存情報をどのスタックまで戻すかがわかると役立ちます。

19.3.7 ターミナル・プロンプトで使用する QUIT コマンド

ターミナル・プロンプトから引数なしの **QUIT** コマンドを入力すると、すべての項目をプログラム・スタックから削除できます。

Terminal

```
USER 4f0>QUIT
USER>
```

プログラム・スタックからいくつかの項目のみを除外するには（例えば、現在実行中のサブルーチンから出て、前の DO レベルに戻る場合）、整数の引数で QUIT を使用します。例えば、QUIT 1 はプログラム・スタックの最後の項目を削除し、QUIT 3 は最後の 3 つの項目を削除します。以下はそのコード例です。

Terminal

```
9f0>QUIT 3
6d0>
```

19.3.8 InterSystems IRIS エラー・メッセージ

InterSystems IRIS は、<ERROR> のように、<> 山括弧内にエラー・メッセージを表示します。その後には、エラー発生時にルーチンが実行していた行の参照が続きます。キャレット記号 (^) は、行参照とルーチンを区切ります。また、エラー発生時に実行するコマンドの最初の文字の下に、キャレット記号と共に中間コード行が表示されます。例えば以下のようになります。

```
SET x=y+3 DO ^ABC
^
<UNDEFINED>label+3^rou
```

このエラー・メッセージは、rou ルーチンの label+3 行で、<UNDEFINED> エラー（変数 y への参照）が発生したことを示しています。この時点で、このメッセージも特殊変数 \$ZERROR の値となります。

19.4 スタックを表示する %STACK の使用法

%STACK ユーティリティを使用して、以下を実行できます。

- ・ プロセス実行スタックのコンテンツの表示
- ・ NEW コマンドあるいはパラメータ渡して“隠された”値を含むローカル変数値の表示
- ・ \$IO や \$JOB などのプロセス状態変数値の表示

19.4.1 %STACK の実行

以下のコマンドを入力して、%STACK を実行します。

Terminal

```
USER>DO ^%STACK
```


この例で示されているように、%STACK ユーティリティは、変数を持たない現在のプロセス・スタックを表示します。

Level	Type	Line	Source
1	SIGN ON		
2	DO		~DO ^StackTest
3	NEW ALL/EXCL		NEW (E)
4	DO	TEST1+1^StackTest	SET A=1 ~DO TEST1 QUIT ;level=2
5	NEW		NEW A
6	DO	TEST1+1^StackTest	~DO TEST2 ;level = 3
7	ERROR TRAP		SET \$ZTRAP="TrapLabel^StackTest"
8	XECUTE	TEST2+2^StackTest	~XECUTE "SET A=\$\$TEST3()"
9	\$\$EXTFUNC		^StackTest ~SET A=\$\$TEST3()
10	PARAMETER		AA
11	DIRECT BREAK	TEST3+1^StackTest	~BREAK
12	DO		^StackTest ~DO ^%STACK

現在の実行スタック表示の下で、%STACK により、スタックの表示動作を求める **Stack Display Action** プロンプトが表示されます。このプロンプトで疑問符 (?) を入力すると、ヘルプを表示できます。このプロンプトで **Return** キーを押すと、%STACK を終了できます。

19.4.2 プロセス実行スタックの表示

Stack Display Action で入力した内容によって、以下の 4 種類の形式で現在のプロセス実行スタックを表示できます。

- ・ *F を入力すると、変数なしのスタックを表示
- ・ *V を入力すると、特定のローカル変数付きスタックを表示
- ・ *P を入力すると、すべてのローカル変数付きスタックを表示
- ・ *A を入力すると、プロセス状態変数リストに先行するすべてのローカル変数付きスタックを表示

次に、%STACK により **Display on Device** プロンプトが表示されます。ここでは、この情報の表示先を指定できます。**Return** キーを押すと、この情報を現在のデバイスに表示できます。

19.4.2.1 変数なしスタックの表示

変数なしのプロセス実行スタックは、初めて %STACK ユーティリティを起動したとき、または **Stack Display Action** プロンプトで *F を入力したときに表示されます。

19.4.2.2 特定の変数付きスタックの表示

Stack Display Action プロンプトで、*V を入力します。これにより、このスタックで記録したいローカル変数の名前の入力を求められます。単一の変数、またはコンマで区切られた変数のリストを指定します。すべてのローカル変数の名前と値が返されます。以下の例では、変数 e が記録され、**Return** キーを押すと情報がターミナルに表示されます。

```
Stack Display Action: *V
Now loading variable information ... 2 done.
Variable(s): e
Display on
Device: <RETURN>
```

19.4.2.3 すべての定義済み変数付きスタックの表示

Stack Display Action プロンプトで *P を入力すると、すべての定義済みローカル変数の現在値と共に、プロセス実行スタックを表示できます。

19.4.2.4 状態変数を含むすべての変数付きスタックの表示

Stack Display Action プロンプトで *A を入力すると、すべての利用可能なレポートを表示できます。レポートは、以下の順序で発行されます。

- ・ プロセス状態内部変数

- すべてのローカル変数の名前と値を伴ったプロセス実行スタック

19.4.3 スタックの表示の理解

スタックの各項目は、フレームと呼ばれます。以下のテーブルは、各フレームに提供される情報を示します。

テーブル 19-2: %STACK ユーティリティ情報

見出し	説明
レベル	スタック内のレベルを識別します。スタック内の一番古い項目が 1 番です。関連付けられたレベル番号のないフレームは、そのすぐ上に表示されるレベルと同じになります。
タイプ	スタックのフレーム・タイプを識別します。タイプは次のようになります。DIRECT BREAK : BREAK コマンドが発生し、ダイレクト・モードに戻りました。DIRECT CALLIN : InterSystems IRIS コールイン・インタフェースを使用して、InterSystems IRIS プロセスが InterSystems IRIS 外部のアプリケーションから開始されました。DIRECT ERROR : エラーが発生し、ダイレクト・モードに戻りました。DO : DO コマンドが実行されました。ERROR TRAP : ルーチンが \$ZTRAP を設定した場合、このフレームは、エラーによって実行が継続する位置を識別します。FOR : FOR コマンドが実行されました。NEW : NEW コマンドが実行されました。NEW コマンドが引数付きの場合、引数は表示されます。SIGN ON : InterSystems IRIS プロセスの実行が開始されました。XECUTE : XECUTE コマンドが実行されました。\$XECUTE 関数が実行されました。\$\$EXTFUNC : ユーザ定義関数が実行されました。
行	使用可能な場合、label+offset^routine の形式で、フレームに関連付けられた ObjectScript ソース行を識別します。
ソース	使用可能な場合、ソース・コードを表示します。ソースが長すぎて表示領域に表示できない場合、水平スクロールを使用できます。デバイスが行表示対応の場合、ソースは折り返され、その後の行の先頭に ... が付きます。

以下のテーブルは、各フレーム・タイプごとにレベル、行、ソース値が使用できるかどうかを示します。レベルの下の "No" は、レベル番号がインクリメントされず、表示されないことを示します。

テーブル 19-3: フレーム・タイプと使用可能な値

フレーム・タイプ	レベル	行	ソース
DIRECT BREAK	あり	あり	あり
DIRECT CALL IN	あり	なし	なし
DIRECT ERROR	あり	あり	あり
DO	あり	あり *	あり
ERROR TRAP	なし	なし	なし (しかし、新規 \$ZTRAP 値は表示されます)
FOR	なし	あり	あり
NEW	なし	なし	NEW の形式 (包括的または排他的) と影響を受けた変数を示します。

フレーム・タイプ	レベル	行	ソース
PARAMETER	なし	なし	仮パラメータ・リストを示します。パラメータが 参照渡しされる 場合は、同じメモリ位置を指すその他の変数を示します。
SIGN ON	あり	なし	いいえ
XECUTE	あり	あり *	あり
\$\$EXTFUNC	あり	あり *	あり
* 行の値は、ターミナル・プロンプトから呼び出された場合、空白になります。			

19.4.3.1 %STACK 表示からの移動

%STACK 表示が複数画面にわたる場合、画面の左下にプロンプト `-- more --` が表示されます。最後のページには、プロンプト `-- fini --` が表示されます。? を入力すると、%STACK 表示内での移動に使用するキー操作が表示されます。

```
-- Filter Help --
<space> Display next page.
<return> Display one more line.
T Return to the beginning of the output.
B Back up one page (or many if arg>1).
R Redraw the current page.
/text Search for \qtext\q after the current page.
A View all the remaining text.
Q Quit.
? Display this screen
# specify an argument for B, L, or W actions.
L set the page length to the current argument.
W set the page width to the current argument.
```

`-- more --` あるいは `-- fini --` プロンプトが表示される場合は常に、上記に表示されたいずれかのコマンドを入力します。

B、L、W コマンドを入力する場合、コマンド文字の前に数値引数を入力します。例えば、2B と入力すると 2 ページ前に戻ります。また、20L を入力すると、20 行の長さにページが設定されます。

ページは、実際に表示される行数に合わせて設定してください。そうしない場合、ページを上下すると一部の行が表示されない可能性があります。既定のページの長さは 23 です。

19.4.3.2 特定のスタック・レベルでの変数の表示

指定されたスタック・フレーム・レベルに存在する変数を表示するには、Stack Display Action プロンプトで ?# と入力します。# には、スタック・フレーム・レベルを入力します。以下の例は、レベル 1 の変数を要求した場合の表示を示します。

```
Stack Display Action: ?1
The following Variables are defined for Stack Level: 1
E
Stack Display Action:
```

この情報を、%SYS.ProcessQuery VariableList クラス・クエリを使用して表示することもできます。

19.4.3.3 変数付きのスタック・レベルの表示

Stack Display Action プロンプトに ?? と入力すると、すべてのスタック・レベルで定義された変数を表示できます。?? と入力した場合に表示される例は、以下のとおりです。

```
Stack Display Action: ??
Now loading variable information ... 19
Base Stack Level: 5
A
Base Stack Level: 3
A B C D
Base Stack Level: 1
E
Stack Display Action:
```

19.4.3.4 プロセス状態変数の表示

\$IO のようなプロセス状態変数を表示するには、“Stack Display Action” プロンプトに *S と入力します。次の表に示すように、これらの定義された変数（プロセス状態内部）が表示されます。

プロセス状態内部	ドキュメント
\$D =	\$DEVICE 特殊変数
\$EC = ,M9,	\$ECODE 特殊変数
\$ES = 4	\$ESTACK 特殊変数
\$ET =	
\$H = 64700,50668	\$HOROLOG 特殊変数
\$I = TRM : 5008	\$IO 特殊変数
\$J = 5008	\$JOB 特殊変数
\$K = \$c(13)	\$KEY 特殊変数
\$P = TRM : 5008	\$PRINCIPAL 特殊変数
\$Roles = %All	\$ROLES 特殊変数
\$S = 268315992	\$STORAGE 特殊変数
\$T = 0	\$TEST 特殊変数
\$TL = 0	\$TLEVEL 特殊変数
\$USERNAME = glenn	\$USERNAME 特殊変数
\$X = 0	\$X 特殊変数
\$Y = 17	\$Y 特殊変数
\$ZA = 0	\$ZA 特殊変数
\$ZB = \$c(13)	\$ZB 特殊変数
\$ZC = 0	\$ZCHILD 特殊変数
\$ZE = <DIVIDE>	\$ZERROR 特殊変数
\$ZJ = 5	\$ZJOB 特殊変数
\$ZM = RY¥Latin1¥K¥UTF8¥	\$ZMODE 特殊変数
\$ZP = 0	\$ZPARENT 特殊変数

プロセス状態内部	ドキュメント
\$ZR = ^ a	\$ZREFERENCE 特殊変数
\$ZS = 262144	\$ZSTORAGE 特殊変数
\$ZT =	\$ZTRAP 特殊変数
\$ZTS = 64700,68668.58	\$ZTIMESTAMP 特殊変数
\$ZU(5) = USER	\$NAMESPACE
\$ZU(12) = c:¥intersystems¥iris¥mgr¥	NormalizeDirectory()
\$ZU(18) = 0	Undefined()
\$ZU(20) = USER	UserRoutinePath()
\$ZU(23,1) = 5	
\$ZU(34) = 0	
\$ZU(39) = USER	SysRoutinePath()
\$ZU(55) = 0	LanguageMode()
\$ZU(56,0) = \$Id: //iris/2018.1.1/kernel/common/src/emath.c#1 \$ 0	
\$ZU(56,1) = 1349	
\$ZU(61) = 16	
\$ZU(61,30,n) = 262160	
\$ZU(67,10,\$J) = 1	JobType
\$ZU(67,11,\$J) = glenn	UserName
\$ZU(67,12,\$J) = TRM:	ClientNodeName
\$ZU(67,13,\$J) =	ClientExecutableName
\$ZU(67,14,\$J) =	CSPSessionID
\$ZU(67,15,\$J) = 127.0.0.1	ClientIPAddress
\$ZU(67,4,\$J) = 0^0^0	State
\$ZU(67,5,\$J) = %STACK	Routine
\$ZU(67,6,\$J) = USER	Namespace
\$ZU(67,7,\$J) = TRM : 5008	CurrentDevice
\$ZU(67,8,\$J) = 923	LinesExecuted
\$ZU(67,9,\$J) = 46	GlobalReferences
\$ZU(68,1) = 0	NullSubscripts()
\$ZU(68,21) = 0	SynchCommit()
\$ZU(68,25) = 0	
\$ZU(68,27) = 1	
\$ZU(68,32) = 0	ZDateNull()

プロセス状態内部	ドキュメント
\$ZU(68,34) = 1	AsynchError()
\$ZU(68,36) = 0	
\$ZU(68,40) = 0	SetZEOF()
\$ZU(68,41) = 1	
\$ZU(68,43) = 0	OldZU5()
\$ZU(68,5) = 1	BreakMode()
\$ZU(68,6) = 0	
\$ZU(68,7) = 0	RefInKind()
\$ZU(131,0) = MYCOMPUTER	
\$ZU(131,1) = MYCOMPUTER:IRIS	
\$ZV = IRIS for Windows (x86-64) 2018.1.0 (Build 527U) Tue Feb 20 2018 22:47:10 EST	\$ZVERSION 特殊変数

19.4.3.5 スタックと変数の両方あるいはいずれかの出力

以下の動作を選択すると、出力デバイスを選択できます。

- ・ *P
- ・ *A
- ・ 表示したい変数を選択した後に、*V

19.5 その他のデバッグ・ツール

デバッグ・プロセスで役立つツールが他にもあります。これには、以下のものがあります。

- ・ [\\$SYSTEM.OBJ.ShowReferences](#) によるオブジェクトへの参照の表示
- ・ [エラー・トラップ・ユーティリティ](#) - %ETN および %ERN

19.5.1 \$SYSTEM.OBJ.ShowReferences によるオブジェクトへの参照の表示

指定されたオブジェクトへの参照を格納したすべての変数をプロセスのシンボル・テーブルに表示するには、%SYSTEM.OBJ クラスの ShowReferences(oref) メソッドを使用します。oref は指定されたオブジェクトの OREF (オブジェクト参照) です。OREF の詳細は、“[OREF の基本](#)”を参照してください。

19.5.2 エラー・トラップ・ユーティリティ

エラー・トラップ・ユーティリティ、%ETN と %ERN は、変数を格納し、エラーに関する他の適切な情報を記録することで、エラーの解析に役立ちます。

19.5.2.1 %ETN アプリケーション・エラー・トラップ

エラー・トラップを設定すると、アプリケーション・エラーの発生時、%ETN ユーティリティを実行できるため便利です。ユーティリティは、エラー時のジョブに関する重要な情報（実行スタックや各変数の値など）を保存します。この情報はアプリケーションのエラー・ログに保存されており、%ERN ユーティリティを使用して表示するか、または管理ポータルの **[アプリケーションエラーログの表示]** ページ（[システムオペレーション]→[システムログ]→[アプリケーションエラーログ]）を使用して確認できます。

以下のコードを使用して、このユーティリティにエラー・トラップを設定します。

```
SET $ZTRAP="%^%ETN"
```

注釈 プロシージャでは、外部ルーチンに \$ZTRAP を設定できません。この制限のため、プロシージャでは ^%ETN を使用できません（プロシージャであるクラス・メソッドを含む）。ただし、%ETN を呼び出すローカル・ラベルには \$ZTRAP を設定できます。

エラーが発生し、%ETN ユーティリティを呼び出すと、以下のようなメッセージが表示されます。

```
Error has occurred: <SYNTAX> at 10:30 AM
```

%ETN は HALT コマンド（プロセスを終了）で終了するため、アプリケーション・モードでルーチンを使用している場合にのみ、%ETN エラー・トラップを設定できます。ターミナル・プロンプトでは、エラーが発生したときに、ターミナルにエラーが表示され、デバッガ・プロンプトで直ちにエラーの分析を行えるので、便利です。以下のコードは、InterSystems IRIS がアプリケーション・モードの場合にのみ、エラー・トラップを設定します。

```
SET $ZTRAP=$SELECT($ZJ#2:" ",1:"%^%ETN")
```

19.5.2.2 %ERN アプリケーション・エラー・レポート

%ERN ユーティリティは、%ETN エラー・トラップ・ユーティリティによって記録されたアプリケーション・エラーを検証します。["%ERN を使用したアプリケーション・エラー・ログの表示"](#) を参照してください。

以下のコードでは、すべての変数を *LOAD でロードしてからルーチンをロードすることを示すために、ルーチン REPORT の ZLOAD が発行されています。エラー発生時、DO に関する情報などを記録するプログラム・スタックが空でない場合、ジョブの状態を再現できます。

Terminal

```
USER>DO ^%ERN
```

```
For Date: 4/30/2018    3 Errors
```

```
Error: ?L
```

```
1) "<DIVIDE>zMyTest+2^Sample.MyStuff.1" at 10:27 am. $I=|TRM|:|10044 ($X=0 $Y=17)
   $J=10044 $ZA=0 $ZB=$c(13) $ZS=262144 ($S=268242904)
   WRITE 5/0
```

```
2) <SUBSCRIPT>REPORT+4^REPORT at 03:16 pm. $I=|TRM|:|10044 ($X=0 $Y=57)
   $J=10044 $ZA=0 $ZB=$c(13) $ZS=2147483647 ($S=2199023047592)
   SET ^REPORT(%DAT,TYPE)=I
```

```
3) <UNDEFINED>zMyTest+2^Sample.MyStuff.1 *undef" at 10:13 pm. $I=|TRM|:|12416 ($X=0 $Y=7)
   $J=12416 $ZA=0 $ZB=$c(13) $ZS=262144 ($S=268279776)
   WRITE undef
```

```
Error: 2
```

```
2) <SUBSCRIPT>REPORT+4^REPORT at 03:16 pm. $I=|TRM|:|10044 ($X=0 $Y=57)
   $J=10044 $ZA=0 $ZB=$c(13) $ZS=2147483647 ($S=2199023047592)
   SET ^REPORT(%DAT,TYPE)=I
```

```
Variable: %DAT
%DAT="Apr 30 2018"
```

```
Variable: TYPE
TYPE=""
```

```
Variable: *LOAD
USER>ZLOAD REPORT

USER>WRITE

%DAT="Apr 30 2018"
%DS=" "
%TG="REPORT+1 "
I=88
TYPE=" "
XY="SET $X=250 WRITE *27,*91,DY+1,*59,DX+1,*72 SET $X=DX,$Y=DY"
USER>
```


A

(従来) ^%ETN を使用したエラー・ログ

古い形式の**エラー・ログ**では、^%ETN ユーティリティを使用します。これについて、ここでは参考用に説明します。

^%ETN ユーティリティは、例外をアプリケーション・エラー・ログに記録した後、終了します。^%ETN (またはそのエントリ・ポイントのいずれか) をユーティリティとして呼び出すことができます。

ObjectScript

```
DO ^%ETN
```

または、\$ZTRAP 特殊変数を ^%ETN (またはそのエントリ・ポイントのいずれか) に等しく設定することができます。

ObjectScript

```
SET $ZTRAP=" ^%ETN"
```

^%ETN またはそのエントリポイントのいずれかを指定できます。

- ・ FORE^%ETN (フォアグラウンド) は、例外を標準アプリケーション・エラー・ログに記録した後、HALT によって終了します。これは、ロールバック処理を起動します。これは、^%ETN と同じ処理です。
- ・ BACK^%ETN (バックグラウンド) は、例外を標準アプリケーション・エラー・ログに記録した後、QUIT によって終了します。これは、ロールバック処理を起動しません。
- ・ LOG^%ETN は、例外を標準アプリケーション・エラー・ログに記録した後、QUIT によって終了します。これは、ロールバック処理を起動しません。この例外は、標準の %Exception.SystemException またはユーザ定義の例外です。

例外を定義するには、LOG^%ETN を呼び出す前に \$ZERROR を意味のある値に設定します。この値は、ログ・エントリのエラー・メッセージ・フィールドとして使用されます。ユーザ定義の例外を直接 LOG^%ETN("This is my custom exception") に指定することもできます。この値は、ログ・エントリのエラー・メッセージ・フィールドとして使用されます。\$ZERROR を NULL 文字列に設定すると (SET \$ZERROR=""), LOG^%ETN は <LOG ENTRY> エラーを記録します。\$ZERROR を <INTERRUPT> に設定すると (SET \$ZERROR="<INTERRUPT>"), LOG^%ETN は <INTERRUPT LOG> エラーを記録します。

LOG^%ETN は、\$HOROLOG 日付とエラー番号の 2 つの要素を持つ %List 構造を返します。

以下の例では、\$ZERROR を変数に即座にコピーするお勧めのコーディング方法を使用しています。LOG^%ETN は %List 値を返します。

ObjectScript

```
SET err=$ZERROR
/* error handling code */
SET rtn = $$LOG^%ETN(err)
WRITE "logged error date: ", $LIST(rtn,1), !
WRITE "logged error number: ", $LIST(rtn,2)
```

LOG^%ETN または BACK^%ETN を呼び出すと、使用可能なプロセス・メモリが自動的に増加し、処理が実行されて、\$ZSTORAGE の元の値がリストアされます。ただし、〈STORE〉エラーの発生後に LOG^%ETN または BACK^%ETN を呼び出した場合、\$ZSTORAGE の元の値をリストアすることによって別の 〈STORE〉エラーがトリガされる可能性があります。このため、〈STORE〉エラーに対してこれらの ^%ETN エントリ・ポイントが呼び出された場合は、増加した使用可能メモリが維持されます。

B

(従来) 従来のエラー処理

ここでは、\$ZTRAP を使用したエラー処理について説明します。これは、従来のアプリケーションで見られるエラー処理の形式です。新しいアプリケーションでは、代わりに [TRY-CATCH](#) を使用します。

B.1 従来のエラー処理の概要

InterSystems IRIS® データ・プラットフォームには、アプリケーションでエラー・ハンドラを使用して従来のエラー処理を実行できるようにするための機能が備わっています。エラー・ハンドラは、アプリケーションの実行中に発生するあらゆるエラーを処理します。エラー発生時に実行する ObjectScript コマンドは、特殊変数で指定します。これらのコマンドは、エラーを直接処理するか、エラーを処理するルーチンを呼び出します。

エラー・ハンドラの作成には、以下の基本的な処理を実行します。

1. エラー処理を実行する 1 つ以上のルーチンを生成します。エラー処理を実行するコードを作成します。このコードは、アプリケーション全体に対する一般的なコードであっても、特定の処理における特定のエラー状態に対するものであってもかまいません。これにより、アプリケーションの特定個所で、カスタマイズされたエラー処理を実行できます。
2. アプリケーション内に 1 つ以上のエラー・ハンドラを設定し、それぞれのエラー・ハンドラで固有の適切なエラー処理を使用します。

エラーが発生してもエラー・ハンドラが設定されていなかった場合の振る舞いは、その InterSystems IRIS セッションの開始方法によって異なります。

1. ターミナル・プロンプトで InterSystems IRIS にサインオンし、エラー・トラップは設定していない場合、InterSystems IRIS は主デバイスにエラー・メッセージを表示し、プログラム・スタックの設定を変更せずにターミナル・プロンプトに戻ります。プログラマは、プログラムの実行を後で再開できます。
2. アプリケーション・モードで InterSystems IRIS を実行し、エラー・トラップを設定していない場合、InterSystems IRIS は主デバイスにエラー・メッセージを表示し、HALT コマンドを実行します。

B.1.1 内部的なエラー・トラップの振る舞い

特殊変数 \$ZTRAP (および \$ETRAP) による InterSystems IRIS のエラー処理とその有効範囲を最大限に活用するには、InterSystems IRIS が制御をあるルーチンから別のルーチンに移す方法を理解すると役立ちます。

InterSystems IRIS は、以下のいずれかが発生するたびに、“コンテキスト・フレーム” というデータ構造を構築します。


- ・ ルーチンが DO コマンドで別のルーチンを呼び出す場合 (この種類のフレームは“DO フレーム”とも呼ばれます)。
- ・ XECUTE コマンド引数により、ObjectScript コードが実行される場合 (この種類のフレームは“XECUTE フレーム”とも呼ばれます)。
- ・ ユーザ定義関数を実行した場合

プロセスのアドレス部にあるプライベート・データ構造の 1 つ、コール・スタックにフレームが構築されます。InterSystems IRIS はルーチン用として、フレームに以下の要素を格納します。

- ・ \$ZTRAP 特殊変数の値 (存在する場合)
- ・ \$ETRAP 特殊変数の値 (存在する場合)
- ・ サブルーチンから返される位置情報

例えば、ルーチン A がルーチン B を DO ^B で呼び出すと、InterSystems IRIS は、コール・スタックに A のコンテキストを保持する DO フレームを構築します。また、ルーチン B がルーチン C を呼び出すと、コール・スタックに B のコンテキストを保持する DO フレームを追加します。

図 II-1: コール・スタックのフレーム



上図のルーチン A がターミナル・プロンプトで DO コマンドを使用して呼び出される場合、図にはありませんが、追加の DO フレームがコール・スタックの一番下に存在しています。

B.1.2 現在のコンテキスト・レベル

現在のコンテキスト・レベルに関する情報を返すために以下を使用できます。

- ・ \$STACK 特殊変数は、現在の相対スタック・レベルを含みます。
- ・ \$ESTACK 特殊変数は、現在のスタック・レベルを含みます。任意のユーザ定義のポイントで、0 (レベル・ゼロ) に初期化できます。

- ・ \$STACK 関数は、現在のコンテキストと、コール・スタックに格納されているコンテキストに関する情報を返します。

B.1.2.1 \$STACK 特殊変数

\$STACK 特殊変数は、プロセスのコール・スタックに現在保存されているフレームの数を含みます。\$STACK 値は、基本的に、現在実行しているコンテキストの (ゼロ・ベースの) レベル番号です。したがって、イメージが開始される時、コマンドが処理されるまで、\$STACK の値は 0 です。

詳細は、“ObjectScript リファレンス” の “\$STACK” 特殊変数を参照してください。

B.1.2.2 \$ESTACK 特殊変数

\$ESTACK 特殊変数は、\$STACK 特殊変数に類似していますが、NEW コマンドで変数を 0 にリセットできるため (また以前の値を保存できます)、エラー処理には \$ESTACK の方が便利です。したがって、プロセスは、特定のコンテキストで \$ESTACK をリセットし、\$ESTACK レベル 0 のコンテキストとしてマークすることができます。その後エラーが発生した場合、エラー・ハンドラは \$ESTACK 値をテストし、コール・スタックをそのコンテキストまで戻すことができます。

詳細は、“ObjectScript リファレンス” の “\$ESTACK” 特殊変数を参照してください。

B.1.2.3 \$STACK 関数

\$STACK 関数は、現在のコンテキストと、コール・スタックに格納されているコンテキストに関する情報を返します。各コンテキストで、\$STACK 関数は以下の情報を提供します。

- ・ コンテキスト・タイプ (DO、XECUTE、ユーザ定義関数)
- ・ コンテキストで処理された最後のコマンドのエントリ参照とコマンド番号
- ・ コンテキストで処理された最後のコマンドを含むソース・ルーチン行あるいは XECUTE 文字列
- ・ コンテキストで発生した任意のエラーの \$ECODE 値 (\$ECODE が NULL でない場合、エラー処理中にのみ利用可能)

エラー発生時、すべてのコンテキスト情報はプロセス・エラー・スタックに即座に格納されます。その後、コンテキスト情報は、エラー・ハンドラが \$ECODE 値をクリアするまで \$STACK 関数によりアクセスできます。つまり、\$ECODE 値が NULL でない場合、\$STACK 関数は、エラー・スタックに保存されたコンテキストと同じ指定されたコンテキスト・レベルのアクティブなコンテキストではなく、エラー・スタックに保存されたコンテキストの情報を返します。

詳細は、“ObjectScript リファレンス” の “\$STACK” 関数を参照してください。

エラーが発生し、エラー・スタックが既に存在している場合、InterSystems IRIS は、エラー・スタックの同じコンテキスト・レベルに別のエラーに関する情報が存在しない限り、コンテキスト・レベルで新しいエラーに関する情報を記録します。この場合、エラー情報は、(エラー情報が既に記録されているかどうかにかかわらず) エラー・スタックの次のレベルに配置されます。

したがって、新規エラーのコンテキスト・レベルによっては、エラー・スタックは拡張される (1 つ以上のコンテキスト・レベルを追加される) か、あるいは、既存のエラー・スタックのコンテキスト・レベルに存在する情報が新規エラー情報を格納するために上書きされることがあります。

\$ECODE 特殊変数をクリアすると、プロセス・エラー・スタックが消去されることに注意してください。

B.1.3 エラー・コード

エラー発生時、InterSystems IRIS は \$ZERROR 特殊変数と \$ECODE 特殊変数に、エラーを示す値を設定します。\$ZERROR および \$ECODE の値は、エラーの直後に使用することを目的としています。これらの値は複数のルーチン呼び出しにわたっては保持されない可能性があるため、後で使用するために値を保持したい場合は、この値を変数にコピーする必要があります。

B.1.3.1 \$ZERROR 値

InterSystems IRIS は、\$ZERROR に以下を含む文字列を設定します。

- ・ 山括弧で囲まれた InterSystems IRIS のエラー・コード
- ・ エラー発生場所のラベル、オフセットおよびルーチン名
- ・ 一部のエラーでは、エラーの原因となった項目の名前などの追加情報

%Exception.SystemException クラスの AsSystemError() メソッドは、\$ZERROR と同じ形式で同じ値を返します。

以下の例では、InterSystems IRIS がエラーに遭遇した際に、\$ZERROR に設定されるメッセージ・タイプを示します。以下の例では、未定義のローカル変数 abc が、ルーチン MyTest のラベル PrintResult より行オフセット 2 の位置で呼び出されます。\$ZERROR は以下を含みます。

```
<UNDEFINED>PrintResult+2^MyTest *abc
```

存在しないクラスが行オフセット 3 の位置で実行されると、以下のエラーが発生します。

```
<CLASS DOES NOT EXIST>PrintResult+3^MyTest *%SYSTEM.XXQL
```

既存クラスの存在しないメソッドが行オフセット 4 の位置で実行されると、以下のエラーが発生します。

```
<METHOD DOES NOT EXIST>PrintResult+4^MyTest *BadMethod,%SYSTEM.SQL
```

128 文字までの任意の文字列として、特殊変数 \$ZERROR を明示的に設定することもできます。以下はその例です。

ObjectScript

```
SET $ZERROR="Any String"
```

\$ZERROR 値は、エラーの直後に使用することを目的としています。\$ZERROR 値は複数のルーチン呼び出しにわたっては保持されない可能性があるため、後で使用するために \$ZERROR 値を保持したいユーザはこの値を変数にコピーする必要があります。ユーザは \$ZERROR を使用したらすぐに NULL 文字列 ("") に設定することを強くお勧めします。詳細は、“ObjectScript リファレンス”の“\$ZERROR”特殊変数を参照してください。\$ZERROR エラーの取り扱いについての詳細は、“インターシステムズ・クラス・リファレンス”の %SYSTEM.Error クラス・メソッドを参照してください。

B.1.3.2 \$ECODE 値

エラー発生時、コンマに囲まれた文字列値を InterSystems IRIS は \$ECODE に設定します。この値は、エラーに対応する ANSI 標準エラー・コードを含んでいます。例えば、未定義のグローバル変数を参照する場合、InterSystems IRIS は \$ECODE セットに以下の文字列を設定します。

```
,M7,
```

エラーに対応 ANSI 標準エラー・コードを持たない場合、コンマに囲まれた文字列値 (文字 Z が先頭についた InterSystems IRIS エラー・コードが含まれる) に \$ECODE を InterSystems IRIS は設定します。例えば、プロセスがシンボル・テーブルの領域をすべて使用した場合、InterSystems IRIS は \$ZERROR 特殊変数にエラー・コード <STORE> を格納し、\$ECODE に以下の文字列を設定します。

```
,ZSTORE,
```

エラー発生後、エラー・ハンドラは \$ZERROR 特殊変数あるいは \$ECODE 特殊変数を検証して、特定のエラー・コードをテストできます。

注釈 エラー・ハンドラは、特定のエラーに関し、\$ECODE 特殊変数ではなく \$ZERROR 特殊変数を検証する必要があります。

詳細は、“ObjectScript リファレンス”の“\$ECODE”特殊変数を参照してください。

B.2 \$ZTRAP でのエラー処理

\$ZTRAP でエラーを処理するには、\$ZTRAP 特殊変数を location に設定し、引用符で囲まれた文字列として指定します。\$ZTRAP 特殊変数はエントリ参照に対して設定します。このエントリ参照によって、エラー発生時における制御の移管先となる location を指定します。次にその場所に \$ZTRAP コードを記述します。

\$ZTRAP を空以外の値に設定すると、これは既存のどの \$ETRAP エラー・ハンドラよりも優先されます。InterSystems IRIS は暗黙的に NEW \$ETRAP コマンドを実行し、\$ETRAP を "" に設定します。

B.2.1 プロシージャでの \$ZTRAP の設定

プロシージャ内では、そのプロシージャ内の行ラベル (プライベート・ラベル) にのみ \$ZTRAP 特殊変数を設定できます。プロシージャ・ブロック内から \$ZTRAP を外部ルーチンに設定することはできません。

\$ZTRAP 値を表示するとき、InterSystems IRIS はプライベート・ラベルの名前を返しません。代わりに、プロシージャの先頭からのそのプライベート・ラベルがある場所までのオフセットを返します。

詳細は、“ObjectScript リファレンス”の“\$ZTRAP”特殊変数を参照してください。

B.2.2 ルーチンでの \$ZTRAP の設定

ルーチン内では、\$ZTRAP 特殊変数を現在のルーチンのラベル、外部ルーチン、または外部ルーチン内のラベルに設定できます。外部ルーチンを参照できるのは、そのルーチンがプロシージャ・ブロック・コードではない場合だけです。以下の例では、LogErr^ErrRou をエラー・ハンドラとして構築します。エラー発生時には、InterSystems IRIS が ^ErrRou ルーチン内の LogErr ラベルにあるコードを実行します。

ObjectScript

```
SET $ZTRAP="LogErr^ErrRou"
```

\$ZTRAP 値を表示するとき、InterSystems IRIS はラベル名と (該当する場合には) ルーチン名を表示します。

ラベル名は最初の 31 文字内で一意である必要があります。ラベル名およびルーチン名では、大文字と小文字が区別されます。

ルーチン内では、\$ZTRAP には 3 つの形式があります。

- ・ SET \$ZTRAP="location"
- ・ SET \$ZTRAP="*location"。この形式では、この関数を呼び出したエラーの発生元であるコンテキストで実行されます。
- ・ SET \$ZTRAP="~%ETN" は、システムで提供されるエラー・ルーチン ~%ETN を実行します。このルーチンはルーチンを呼び出すエラーが発生したコンテキストにおいて実行されます。プロシージャ・ブロック内から ~%ETN (またはい

ずれかの外部ルーチン) を実行することはできません。コードが [Not ProcedureBlock] であることを指定するか、`%ETN エントリ・ポイント BACK`%ETN を呼び出す以下のようなルーチンを使用してください。

```
ClassMethod MyTest() as %Status
{
  SET $ZTRAP="Error"
  SET ans = 5/0 /* divide-by-zero error */
  WRITE "Exiting ##class(User.A).MyTest()",!
  QUIT ans
Error
  SET err=$ZERROR
  SET $ZTRAP=""
  DO BACK^%ETN
  QUIT $$$ERROR($$CacheError,err)
}
```

`%ETN とそのエントリ・ポイントの詳細は、“(従来) %ETN を使用したエラー・ログ” を参照してください。\$ZTRAP と共に使用する際の処理の詳細は、“SET \$ZTRAP=%ETN” を参照してください。

詳細は、“ObjectScript リファレンス” の “\$ZTRAP” 特殊変数を参照してください。

B.2.3 \$ZTRAP コードの記述

\$ZTRAP が指し示す location はさまざまな操作を実行して、エラーを表示、ログ記録、または修正することができます。実行したいエラー処理操作の如何にかかわらず、\$ZTRAP コードは以下の 2 つのタスクの実行により開始する必要があります。

- ・ \$ZTRAP を他の値、すなわちエラー・ハンドラの location または空文字列 (“”) のいずれかに設定します。(KILL \$ZTRAP. とすることはできないので、SET を使用する必要があります。)これを行う理由は、エラー処理中に別のエラーが発生した場合において、そのエラーが現在の \$ZTRAP エラー・ハンドラを呼び出してしまうためです。現在のエラー・ハンドラがユーザの関わるエラー・ハンドラである場合、無限ループとなってしまいます。
- ・ ある変数を \$ZERROR に設定します。後でコード内で \$ZERROR 値を参照する場合は、\$ZERROR 自体ではなく、この変数を参照します。これを行う理由は、\$ZERROR には最新のエラーが含まれているうえ、\$ZERROR 値が、内部ルーチン呼び出しなどの複数のルーチン呼び出しにわたって保持されない可能性があるからです。エラー処理中に別のエラーが発生した場合、\$ZERROR の値がその新しいエラーによって上書きされることになります。

ユーザは \$ZERROR を使用したらすぐに NULL 文字列 (“”) に設定することを強くお勧めします。

以下の例では、これらの不可欠な \$ZTRAP コード文を示します。

ObjectScript

```
MyErrorHandler
SET $ZTRAP=""
SET err=$ZERROR
/* error handling code
using err as the error
to be handled */
```

B.2.4 \$ZTRAP の使用法

アプリケーションの各ルーチンは、\$ZTRAP を設定して、独自の \$ZTRAP エラー・ハンドラを作成できます。エラー・トラップの発生時、InterSystems IRIS は以下の手順を実行します。

1. 特殊変数 \$ZERROR にエラー・メッセージを設定します。
2. プログラム・スタックを、エラー・トラップが設定された時点 (SET \$ZTRAP= の実行時) の状態にリセットします。つまり、システムは、エラー・トラップが設定された時点まで、スタックに存在するすべてのエントリを削除します。(\$ZTRAP がアスタリスク (*) で始まる文字列に設定されている場合、プログラム・スタックはリセットされません。)
3. \$ZTRAP の値により指定された位置でプログラムを再開します。\$ZTRAP の値は保持されます。

注釈 \$ZERROR 変数を、128 文字までの文字列として明示的に設定できます。通常、\$ZERROR は NULL 文字列にのみ設定しますが、\$ZERROR を値に設定してもかまいません。

B.2.5 エラー・トラップによる New コマンドの削除

エラー・トラップが発生しプログラム・スタック・エントリが削除されると、InterSystems IRIS は、スタックされたすべての NEW コマンドも削除し、SET \$ZTRAP= を含むサブルーチン・レベルに戻します。しかし、NEW コマンドが \$ZTRAP の設定前あるいは設定後にスタックに追加されたかどうかにかかわらず、そのサブルーチン・レベルで実行されたすべての NEW コマンドは維持されます。

以下に例を示します。

ObjectScript

```
Main
    SET A=1,B=2,C=3,D=4,E=5,F=6
    NEW A,B
    SET $ZTRAP="ErrSub"
    NEW C,D
    DO Sub1
    RETURN
Sub1()
    NEW E,F
    WRITE 6/0      // Error: division by zero
    RETURN
ErrSub()
    WRITE !,"Error is: ", $ZERROR
    WRITE
    RETURN
```

Sub1 のエラーがエラー・トラップを起動した場合、Sub1 にスタックされた E と F における以前の値は削除されます。しかし、A、B、C、D はスタックされたままになります。

B.2.6 \$ZTRAP フロー制御オプション

\$ZTRAP エラー・ハンドラがエラー処理のために呼び出され、クリーンアップとエラー記録処理を実行した後、このエラー・ハンドラには 3 つのフロー制御オプションがあります。

- ・ エラーを処理し、アプリケーションを継続します。
- ・ 制御を別のエラー・ハンドラに渡します。
- ・ アプリケーションを終了します。

B.2.6.1 アプリケーションの継続

\$ZTRAP がエラーを処理した後には、GOTO の発行によりアプリケーションを続行することができます。通常のアプリケーション処理を継続するために、\$ZERROR または \$ECODE 特殊変数値をクリアする必要はありません。しかし、\$ZTRAP をクリアし (空の文字列を設定)、別のエラーが発生した際に無限のエラー処理ループに入らないようにする必要があります。詳細は、“[エラー・ハンドラによるエラー処理](#)”を参照してください。

エラー処理を実行した後、\$ZTRAP エラー・ハンドラは GOTO コマンドを使用して、アプリケーション内で事前設定されたリスタートまたは継続ポイントに制御を移し、通常のアプリケーション処理を再開できます。


エラー・ハンドラがエラーを処理すると、\$ZERROR 特殊変数が値に設定されます。エラー・ハンドラが処理を終了すると、この値は必ずしもクリアされるわけではありません。ルーチンによっては、\$ZERROR を NULL 文字列に設定し直すものがあります。エラー・ハンドラを呼び出すエラーが次に発生したときに、\$ZERROR は上書きされます。この理由から、\$ZERROR 値はエラー・ハンドラのコンテキスト内でのみ使用する必要があります。この値を保持する場合は、これを変数にコピーして、\$ZERROR 自体ではなくその変数を参照します。その他のコンテキストで \$ZERROR を使用すると、信頼できる結果が生成されません。

B.2.6.2 別のエラー・ハンドラへの制御の移動

エラー条件を \$ZTRAP エラー・ハンドラで修正できない場合、特殊な形式の ZTRAP コマンドを使用して、別のエラー・ハンドラに制御を移すことができます。コマンド ZTRAP \$ZERROR は、エラー条件を再度示し、エラー・ハンドラを使って、InterSystems IRIS で次のコール・スタック・レベルまでコール・スタックが戻るようにします。InterSystems IRIS で次のエラー・ハンドラのレベルにコール・スタックを戻したら、そのエラー・ハンドラで処理が続行します。次のエラー・ハンドラは、\$ZTRAP または \$ETRAP で設定された可能性があります。

下図は、\$ZTRAP エラー処理ルーチンのフロー制御を示しています。

図 II-2: \$ZTRAP エラー・ハンドラ



B.3 \$ETRAP でのエラー処理

エラー・トラップの発生時に \$ETRAP が設定されている場合、InterSystems IRIS は以下の手順を実行します。

1. \$ECODE および \$ZERROR の値を設定します。

2. \$ETRAP の値であるコマンドを処理します。

既定で、DO、XECUTE、またはユーザ定義の各関数コンテキストは、それを呼び出したフレームの \$ETRAP エラー・ハンドラを継承します。つまり、どのコンテキスト・レベルであっても、指定された \$ETRAP エラー・ハンドラが最終的に \$ETRAP として定義されます。その定義が現在のレベルよりもいくつか低いスタック・レベルで行われていた場合でも同様です。

B.3.1 \$ETRAP エラー・ハンドラ

\$ETRAP 特殊変数には、エラーの発生時に実行される 1 つ以上の ObjectScript コマンドを含めることができます。SET コマンドを使用して、\$ETRAP を、制御をエラー処理ルーチンに移す 1 つ以上の InterSystems IRIS コマンドを含む文字列に設定します。以下の例では、制御が LogError コード・ラベル (ルーチン ErrRoutine の一部) に移されます。

ObjectScript

```
SET $ETRAP="DO LogError^ErrRoutine"
```

\$ETRAP 特殊変数内のコマンドの後ろには常に、暗黙の QUIT コマンドが続きます。引数付きの QUIT コマンドを必要とするユーザ定義の関数コンテキストで \$ETRAP エラー・ハンドラが呼び出されると、暗黙の QUIT コマンドは NULL 文字列の引数で終了します。

\$ETRAP はグローバル・スコープを持ちます。これは、通常、\$ETRAP の設定の前には NEW \$ETRAP が必要であることを意味します。そうしなければ、現在のコンテキストで \$ETRAP の値が設定されている場合、その値がそのコンテキストの範囲を超えて渡された後、制御は上位レベルのコンテキストにあるのに、\$ETRAP に格納された値は存在し続けることになります。このため、NEW \$ETRAP を指定しないと、\$ETRAP が設定されたときのコンテキストがもう存在しない場合、\$ETRAP が予期しないタイミングで実行される可能性があります。

詳細は、“ObjectScript リファレンス” の “\$ETRAP” 特殊変数を参照してください。

B.3.2 コンテキスト固有の \$ETRAP エラー・ハンドラ

以下の手順を実行することで、あらゆるコンテキストでコンテキスト固有の \$ETRAP エラー・ハンドラを作成できます。

1. NEW コマンドを使用して、\$ETRAP の新しいコピーを作成します。
2. \$ETRAP を新しい値に設定します。

ルーチンで最初に \$ETRAP の新しいコピーを作成せずに \$ETRAP を設定すると、現在のコンテキスト、それを呼び出したコンテキスト、およびコール・スタックに保存されている可能性のあるその他のコンテキストに、新しい \$ETRAP エラー・ハンドラが作成されます。このため、\$ETRAP を設定する前に、その新しいコピーを作成することをお勧めします。


\$ETRAP の新しいコピーを作成することで \$ETRAP はクリアされないことに注意してください。\$ETRAP の値は、NEW コマンドによって変更されないまま保持されます。

以下の図は、\$ETRAP エラー・ハンドラのスタックを作成する \$ETRAP 割り当てのシーケンスを示しています。この図が示している内容は次のとおりです。

- ・ ルーチン A は、\$ETRAP の新しいコピーを作成して、それを “GOTO `ERR” に設定し、ルーチン B を呼び出すための DO コマンドを含んでいます。
- ・ ルーチン B は、\$ETRAP に対して何も行わず (その結果、ルーチン A の \$ETRAP エラー・ハンドラを継承し)、ルーチン C を呼び出すための DO コマンドを含んでいます。
- ・ ルーチン C は \$ETRAP の新しいコピーを作成して、それを “GOTO `CERR” に設定し、ルーチン D を呼び出すための DO コマンドを含んでいます。
- ・ ルーチン D は、\$ETRAP の新しいコピーを作成した後、それをクリアして、コンテキストに \$ETRAP エラー・ハンドラを残しません。

ルーチン D でエラーが発生した場合 (\$ETRAP エラー・ハンドラが定義されていないコンテキスト)、InterSystems IRIS はルーチン D の DO フレームをコール・スタックから削除して、制御をルーチン C の \$ETRAP エラー・ハンドラに移します。ルーチン C の \$ETRAP エラー・ハンドラはさらに ^CERR にディスパッチしてエラーが処理されます。ルーチン C でエラーが発生した場合、InterSystems IRIS は制御をルーチン C の \$ETRAP エラー・ハンドラに移しますが、エラーが発生しているのは \$ETRAP エラー・ハンドラが定義されているコンテキストであるため、スタックを戻すことはしません。

図 II-3: \$ETRAP エラー・ハンドラ



B.3.3 \$ETRAP フロー制御オプション

\$ETRAP エラー・ハンドラが、エラーを処理し、クリーンアップまたはエラー・ログ処理を実行するために呼び出された場合、次のフロー制御オプションがあります。

- ・ エラーを処理し、アプリケーションを継続します。
- ・ 制御を別のエラー・ハンドラに渡します。
- ・ アプリケーションを終了します。

B.3.3.1 エラーの処理とアプリケーションの継続

エラーを処理するために \$ETRAP エラー・ハンドラが呼び出されると、InterSystems IRIS は、エラー状態が解除されるまでそのエラー状態がアクティブであると見なします。エラー状態を解除するには、\$ECODE 特殊変数を NULL 文字列に設定します。

ObjectScript

```
SET $ECODE=""
```

\$ECODE のクリアによって、プロセスのエラー・スタックもクリアされます。

エラー状態が解除されたら、通常は、GOTO コマンドを使用して、アプリケーションで事前設定されている再起動または継続のポイントに制御を移します。場合によっては、エラー状態の解除後に、終了して前のコンテキスト・レベルに戻る方が簡単なこともあります。

B.3.3.2 別のエラー・ハンドラへの制御の移動

エラー状態が解除されていない場合、\$ETRAP エラー・ハンドラが呼び出されたコンテキストを QUIT コマンドが終了すると、InterSystems IRIS はコール・スタック上の別のエラー・ハンドラに制御を渡します。このため、\$ECODE をクリアせずに、\$ETRAP コンテキストから QUIT を実行することで、前のレベルのエラー・ハンドラに制御を渡すことができます。

ルーチン C から呼び出されたルーチン D に、制御を ^CERR に移すエラーが含まれる場合、前もって \$ECODE が "" (空の文字列) に設定されていない ^CERR の QUIT コマンドは、前のコンテキスト・レベルの \$ETRAP エラー・ハンドラに制御を移します。一方、\$ECODE のクリアによってエラー状態が解除された場合、^CERR の QUIT は、ルーチン B の DO ^C コマンドに続く文に制御を移します。

B.3.3.3 アプリケーションの終了

コール・スタックに前のレベルのエラー・ハンドラが存在しておらず、\$ETRAP エラー・ハンドラがエラー状態を解除せずに QUIT を実行した場合、アプリケーションが終了します。アプリケーション・モードではその後 InterSystems IRIS は停止し、制御がオペレーティング・システムに渡されます。それに続いて、ターミナル・プロンプトが表示されます。

エラー状態が解除されているかどうかにかかわらず、\$ETRAP エラー・ハンドラのコンテキストを終了するには、QUIT コマンドを使用することに留意してください。引数なしの QUIT が必要なコンテキスト・レベルと引数ありの QUIT が必要なコンテキスト・レベル (ユーザ定義関数のコンテキスト) で、同じ \$ETRAP エラー・ハンドラを呼び出すことができるため、特定のコンテキスト・レベルで必要とされる QUIT コマンド形式を示すために \$QUIT 特殊変数が指定されます。

\$QUIT 特殊変数は、引数ありの QUIT が必要なコンテキストの場合は 1 を返し、引数なしの QUIT が必要なコンテキストの場合は 0 を返します。

\$ETRAP エラー・ハンドラは、\$QUIT を使用することで、以下のようにどちらの状況にも対処できます。

ObjectScript

```
Quit:$QUIT "" Quit
```

該当する場合、\$ETRAP エラー・ハンドラは、HALT コマンドを使用してアプリケーションを終了することができます。

B.4 エラー・ハンドラによるエラー処理

エラー・ハンドラでエラーが発生するとき、実行フローは、現在実行中のエラー・ハンドラ・タイプにより異なります。

B.4.1 \$ZTRAP エラー・ハンドラでのエラー

\$ZTRAP エラー・ハンドラで新規エラーが発生すると、InterSystems IRIS は、最初に検出したエラー・ハンドラに制御を渡し、必要に応じてコール・スタックを戻します。したがって、\$ZTRAP エラーが現在のスタック・レベルで \$ZTRAP をクリアせず、別のエラーが同じエラー・ハンドラで続けて発生した場合、\$ZTRAP ハンドラが同じコンテキスト・レベルで再度呼び出されるため、無限ループの原因となります。これを回避するには、エラー・ハンドラの開始時点において \$ZTRAP に他の値を設定します。

B.4.2 \$ETRAP エラー・ハンドラでのエラー

\$ETRAP エラー・ハンドラで新しいエラーが発生すると、InterSystems IRIS は、\$ETRAP エラー・ハンドラが呼び出されたコンテキスト・レベルが削除されるまでコール・スタックを戻します。その後、InterSystems IRIS はコール・スタック上の次のエラー・ハンドラ (存在する場合) に制御を渡します。

B.4.3 \$ZERROR 特殊変数と \$ECODE 特殊変数のエラー情報

元のエラーの処理中に別のエラーが発生した場合、2 番目のエラー情報により、\$ZERROR 特殊変数の最初のエラー情報が置き換えられます。しかし、InterSystems IRIS は、\$ECODE 特殊変数に新規情報を追加します。2 番目のエラーのコンテキスト・レベルによっては、InterSystems IRIS は新しい情報をプロセス・エラー・スタックに追加する場合もあります。

\$ECODE 特殊変数の既存の値が NULL 以外の場合、InterSystems IRIS は、新しいコンマ区切りの部分として、新規エラー・コードを現在の \$ECODE 値に追加します。以下のいずれかの状況が発生するまで、\$ECODE 特殊変数にエラー・コードが追加されます。

- ・ 以下のようにして、\$ECODE を明示的にクリアします。

ObjectScript

```
SET $ECODE = ""
```

- ・ \$ECODE の長さは、最大文字列長を超えています。

その後、次の新規エラー・コードにより、\$ECODE にある現在のエラー・コード・リストが置き換えられます。

エラーが発生し、エラー・スタックが既に存在している場合、InterSystems IRIS は、エラー・スタックの同じコンテキスト・レベルに別のエラーに関する情報が存在しない限り、コンテキスト・レベルで新しいエラーに関する情報を記録します。この場合、エラー情報は、(エラー情報が既に記録されているかどうかにかかわらず) エラー・スタックの次のレベルに配置されます。

したがって、新規エラーのコンテキスト・レベルによっては、エラー・スタックは拡張される (1 つ以上のコンテキスト・レベルを追加される) か、あるいは、既存のエラー・スタックのコンテキスト・レベルに存在する情報が新規エラー情報を格納するために上書きされることがあります。

\$ECODE 特殊変数をクリアすると、プロセス・エラー・スタックが消去されることに注意してください。

詳細は、“ObjectScript リファレンス” の “[\\$ECODE](#)” 特殊変数と “[\\$ZERROR](#)” 特殊変数を参照してください。\$ZERROR エラーの取り扱いについての詳細は、“インターシステムズ・クラス・リファレンス” の `%SYSTEM.Error` クラス・メソッドを参照してください。

B.5 強制エラー

\$ECODE 特殊変数を設定、あるいは ZTRAP コマンドを使用して、制御環境でエラーを発生させます。

B.5.1 \$ECODE の設定

\$ECODE 特殊変数に NULL 以外の文字列を設定し、エラーを発生させることができます。ルーチンで \$ECODE に NULL 以外の文字列を設定すると、InterSystems IRIS は \$ECODE に特定の文字列を設定し、その後、エラー状態を生成します。この状況で \$ZERROR 特殊変数は、以下のエラー・テキストと共に設定されます。

```
<ECODETRAP>
```

その後、通常のアプリケーション・エラーのように、エラー・ハンドラに制御を渡します。

エラー・ハンドラにロジックを追加し、\$ECODE の設定により発生させるエラーをチェックできます。エラー・ハンドラは、<ECODETRAP> エラー（例えば、“\$ZE[“ECODETRAP””] に対して \$ZERROR をチェックするか、または選択する特定の文字列値に対して \$ECODE をチェックできます。

B.5.2 アプリケーション固有エラーの生成

\$ECODE の ANSI 標準フォーマットは、1 つ以上のコンマに囲まれたエラー・コードのリストです。

- ・ 接頭語 “Z” 付きのエラーは、実装固有のエラーです。
- ・ 接頭語 “U” 付きのエラーは、アプリケーション固有のエラーです。

エラー・ハンドラで \$ECODE に接頭語 “U” を持つ適切なエラー・メッセージを設定し、ANSI 標準に合う独自のエラー・コードを生成できます。

ObjectScript

```
SET $ECODE=" ,Upassword expired ,"
```

B.6 ターミナル・プロンプトでのエラー処理

エラー・ハンドラ・セットを持たないターミナル・プロンプトで InterSystems IRIS にサインオンした後エラーを生成する場合、InterSystems IRIS は、入力したコード行でエラーが発生すると、以下の手順を実行します。

1. InterSystems IRIS は、プロセスの主デバイスで、エラー・メッセージを表示します。
2. エラーが発生したコール・スタック・レベルでプロセスを中断します。
3. プロセスをターミナル・プロンプトに戻します。

B.6.1 エラー・メッセージ形式の理解

エラー・メッセージとして、InterSystems IRIS は以下の 3 行を表示します。

1. エラーが発生したすべてのソース・コード行
2. ソース・コードの行の下で、エラーが発生したコマンドを指定するキャレット (^)
3. \$ZERROR のコンテンツを含む行

以下のターミナル・プロンプトの例では、2 番目の SET コマンドに未定義のローカル変数エラーがあります。

Terminal

```
USER>WRITE "hello",! SET x="world" SET y=zzz WRITE x,!
hello

WRITE "hello",! SET x="world" SET y=zzz WRITE x,!
      ^
<UNDEFINED> *zzz
USER>
```

以下の例では、ターミナル・プロンプトから実行される mytest という名前のプログラムにコードと同じ行があります。

Terminal

```
USER>DO ^mytest
hello

WRITE "hello",! SET x="world" SET y=zzz WRITE x,!
      ^
<UNDEFINED>WriteOut+2^mytest *zzz
USER 2d0>
```

この場合、\$ZERROR は、mytest の中で WriteOut という名前のラベルから 2 行のオフセット位置でエラーが発生したことを示しています。プロンプトが変更され、新規のプログラム・スタック・レベルが開始されたことを示している点に注意してください。

B.6.2 ターミナル・プロンプトの理解

既定で、ターミナル・プロンプトは現在のネームスペース名を示します。トランザクションが 1 つ以上開いている場合、[\\$TLEVEL](#) トランザクション・レベル・カウントも含まれます。[ZNSPACE](#) コマンドの説明に従って、既定のプロンプトを異なる内容で構成することができます。以下の例は、それらの既定の設定を示します。

Terminal

```
USER>
```

Terminal

```
TL1:USER>
```

ルーチンの実行中にエラーが発生すると、システムは、現在のプログラム・スタックを保存して、新規スタック・フレームを開始します。以下のような拡張プロンプトが表示されます。

Terminal

```
USER 2d0>
```

この拡張プロンプトは、プログラム・スタックに 2 つのエントリがあり、その最後の方が DO (d で示されています) の呼び出しであることを示しています。このエラーにより、プログラム・スタックに 2 つのエントリが置かれることに注意してください。次の DO 実行エラーにより、以下のプロンプト結果となります。

Terminal

```
USER 4d0>
```

説明の詳細は、“コマンド行ルーチンのデバッグ” の章の“[ターミナル・プロンプトのプログラム・スタック情報表示](#)”を参照してください。

B.6.3 エラーのリカバリ

以下のいずれかの手順を実行できます。

- ・ ターミナル・プロンプトからコマンドを発行します。
- ・ 変数とグローバル・データを表示、修正します。
- ・ エラーと他のルーチンを含むルーチンを編集します。
- ・ 他のルーチンを実行します。

これらの手順により、別のエラーを発生させる可能性もあります。

これらの手順の実行後、実行を再開する、あるいはプログラム・スタックの一部/すべてを削除することをお勧めします。

B.6.3.1 次に続くコマンドでの実行の再開

ターミナル・プロンプトから引数なしの GOTO を入力すれば、エラーを引き起こしたコマンドの次のコマンドで実行を再開できます。

Terminal

```
USER>DO ^mytest
hello

WRITE "hello",! SET x="world" SET y=zzz WRITE x,!
^
<UNDEFINED>WriteOut+2^mytest *zzz
USER 2d0>GOTO
world

USER>
```

B.6.3.2 別の行での実行の再開

ターミナル・プロンプトからラベル引数付きの GOTO を発行すると、別の行で実行を再開できます。

Terminal

```
USER 2d0>GOTO ErrSect
```

B.6.3.3 プログラム・スタックの削除

ターミナル・プロンプトから引数なしの QUIT コマンドを発行すると、プログラム・スタック全体を削除できます。

Terminal

```
USER 4d0>QUIT
USER>
```

B.6.3.4 プログラム・スタックの部分削除

ターミナル・プロンプトから整数の引数付きの QUIT n コマンドを発行すると、最後の（または最後にある複数の）プログラム・スタック・エントリを削除できます。

Terminal

```
USER 8d0>QUIT 1
USER 7E0>QUIT 3
USER 4d0>QUIT 1
USER 3E0>QUIT 1
USER 2d0>QUIT 1
USER 1S0>QUIT 1
USER>
```

この例では、プログラム・エラーによって 2 つのプログラム・スタック・エントリが作成されているので、GOTO を発行して実行を再開するには、“d” スタック・エントリに移動している必要があります。他に発生する内容に応じて、“d” スタック・エントリは偶数 (USER 2d0>) または奇数 (USER 3d0>) となります。

NEW \$ESTACK を使用すれば、指定のプログラム・スタック・レベルで終了できます。

Terminal

```
USER 4d0>NEW $ESTACK

USER 5E1>
/* more errors create more stack frames */

USER 11d7>QUIT $ESTACK

USER 4d0>
```

NEW \$ESTACK コマンドにて、エントリが 1 つプログラム・スタックに追加されます。